RAPID Reference Manual

Overview

BaseWare OS 3.2

Continue

ABB Flexible Automation

ABB

3HAC 5774-1
For BaseWare OS 3.2

RAPID Overview

Article number: 3HAC 5774-1
Issue: For BaseWare OS 3.2

ABB Robotics AB
S-721 68 Västerås
Sweden

CONTENTS

Page

1 Introduction

This is a reference manual containing a detailed explanation of the programming language as well as all *data types*, *instructions* and *functions*. If you are programming off-line, this manual will be particularly useful in this respect.

When you start to program the robot it is normally better to start with the User's Guide until you are familiar with the system.

1.1 Other Manuals

Before using the robot for the first time, you should read *Basic Operation*. This will provide you with the basics of operating the robot.

The User's Guide provides step-by-step instructions on how to perform various tasks, such as how to move the robot manually, how to program, or how to start a program when running production.

The Product Manual describes how to install the robot, as well as maintenance procedures and troubleshooting. This manual also contains a *Product Specification* which provides an overview of the characteristics and performance of the robot.

1.2 How to Read this Manual

To answer the questions *Which instruction should I use?* or *What does this instruction mean?*, see *RAPID Overview Chapter 3: RAPID Summary.* This chapter briefly describes all instructions, functions and data types grouped in accordance with the instruction pick-lists you use when programming. It also includes a summary of the syntax, which is particularly useful when programming off-line.

RAPID Overview Chapter 4: Basic Characteristics explains the inner details of the language. You would not normally read this chapter unless you are an experienced programmer.

RAPID Overview Chapter 5: Motion and I/O Principles describes the various coordinate systems of the robot, its velocity and other motion characteristics during different types of execution.

System DataTypes and Routines Chapters 1-3 describe all *data types*, *instructions* and *functions*. They are described in alphabetical order for your convenience.

This manual describes all the data and programs provided with the robot on delivery. In addition to these, there are a number of predefined data and programs supplied with the robot, either on diskette or, or sometimes already loaded.
RAPID Overview Chapter 7: Predefined Data and Programs describes what happens when these are loaded into the robot.

If you program off-line, you will find some tips in *RAPID Overview Chapter 6: Programming off-line*.

To make things easier to locate and understand, *RAPID Overview chapter 8* contains an *Index, Glossary* and *System DataTypes and Routines Chapter 4* contains an *index*.

Typographic conventions

The commands located under any of the five menu keys at the top of the teach pendant display are written in the form of **Menu: Command**. For example, to activate the Print command in the File menu, you choose **File: Print**.

The names on the function keys and in the entry fields are specified in bold italic typeface, e.g. ***Modpos***.

Words belonging to the actual programming language, such as instruction names, are written in italics, e.g. *MoveL*.

Examples of programs are always displayed in the same way as they are output to a diskette or printer. This differs from what is displayed on the teach pendant in the following ways:

- Certain control words that are masked in the teach pendant display are printed, e.g. words indicating the start and end of a routine.
- Data and routine declarations are printed in the formal form, e.g. *VAR num reg1;*.

Syntax rules

Instructions and functions are described using both simplified syntax and formal syntax. If you use the teach pendant to program, you generally only need to know the simplified syntax, since the robot automatically makes sure that the correct syntax is used.

Simplified syntax

Example:



- Optional arguments are enclosed in square brackets [ ]. These arguments can be omitted.
- Arguments that are mutually exclusive, i.e. cannot exist in the instruction at the same time, are separated by a vertical bar |.
- Arguments that can be repeated an arbitrary number of times are enclosed in braces { }.

Formal syntax

Example:
```
TPWrite
    [String':='] <expression (IN) of string>
    ['\'Num':=' <expression (IN) of num> ] |
    ['\'Bool':=' <expression (IN) of bool> ] |
    ['\'Pos':=' <expression (IN) of pos> ] |
    ['\'Orient':=' <expression (IN) of orient> ]';'
```

- The text within the square brackets [ ] may be omitted.
- Arguments that are mutually exclusive, i.e. cannot exist in the instruction at the same time, are separated by a vertical bar |.
- Arguments that can be repeated an arbitrary number of times are enclosed in braces { }.
- Symbols that are written in order to obtain the correct syntax are enclosed in single quotation marks (apostrophes) ' '.
- The data type of the argument (italics) and other characteristics are enclosed in angle brackets < >. See the description of the parameters of a routine for more detailed information.

The basic elements of the language and certain instructions are written using a special syntax, EBNF. This is based on the same rules, but with some additions.

Example:
```
GOTO <identifier>';'
    <identifier> ::= <ident>
        | <ID>
    <ident> ::= <letter> {<letter> | <digit> | '_'}
```

- The symbol ::= means *is defined as*.
- Text enclosed in angle brackets < > is defined in a separate line.

1 The Structure of the Language

The program consists of a number of instructions which describe the work of the robot. Thus, there are specific instructions for the various commands, such as one to move the robot, one to set an output, etc.

The instructions generally have a number of associated arguments which define what is to take place in a specific instruction. For example, the instruction for resetting an output contains an argument which defines which output is to be reset; e.g. *Reset do5*. These arguments can be specified in one of the following ways:

- as a numeric value, e.g. *5* or *4.6*
- as a reference to data, e.g. *reg1*
- as an expression, e.g. *5+reg1*2*
- as a function call, e.g. *Abs(reg1)*
- as a string value, e.g. *"Producing part A"*

There are three types of routines – *procedures*, *functions* and *trap routines*.

- A procedure is used as a subprogram.
- A function returns a value of a specific type and is used as an argument of an instruction.
- Trap routines provide a means of responding to interrupts. A trap routine can be associated with a specific interrupt; e.g. when an input is set, it is automatically executed if that particular interrupt occurs.

Information can also be stored in data, e.g. tool data (which contains all information on a tool, such as its TCP and weight) and numerical data (which can be used, for example, to count the number of parts to be processed). Data is grouped into different data types which describe different types of information, such as tools, positions and loads. As this data can be created and assigned arbitrary names, there is no limit (except that imposed by memory) on the number of data. These data can exist either globally in the program or locally within a routine.

There are three kinds of data – *constants*, *variables* and *persistents*.

- A constant represents a static value and can only be assigned a new value manually.
- A variable can also be assigned a new value during program execution.
- A persistent can be described as a “persistent” variable. When a program is saved the initialization value reflects the current value of the persistent.

Other features in the language are:

- Routine parameters
- Arithmetic and logical expressions
- Automatic error handling
- Modular programs
- Multi tasking

2 Controlling the Program Flow

The program is executed sequentially as a rule, i.e. instruction by instruction. Sometimes, instructions which interrupt this sequential execution and call another instruction are required to handle different situations that may arise during execution.

2.1 Programming principles

The program flow can be controlled according to five different principles:

- By calling another routine (procedure) and, when that routine has been executed, continuing execution with the instruction following the routine call.
- By executing different instructions depending on whether or not a given condition is satisfied.
- By repeating a sequence of instructions a number of times or until a given condition is satisfied.
- By going to a label within the same routine.
- By stopping program execution.

2.2 Calling another routine

Instruction	Used to:
ProcCall	Call (jump to) another routine
CallByVar	Call procedures with specific names
RETURN	Return to the original routine

2.3 Program control within the routine

Instruction	Used to:
Compact IF	Execute one instruction only if a condition is satisfied
IF	Execute a sequence of different instructions depending on whether or not a condition is satisfied
FOR	Repeat a section of the program a number of times
WHILE	Repeat a sequence of different instructions as long as a given condition is satisfied
TEST	Execute different instructions depending on the value of an expression
GOTO	Jump to a label
label	Specify a label (line name)

2.4 Stopping program execution

Instruction	Used to:
Stop	Stop program execution
EXIT	Stop program execution when a program restart is not allowed
Break	Stop program execution temporarily for debugging purposes

2.5 Stop current cycle

Instruction	Used to:
Exit cycle	Stop the current cycle and move the program pointer to the first instruction in the main routine. When the execution mode *CONT* is selected, execution will continue with the next program cycle.

3 Various Instructions

Various instructions are used to

- assign values to data
- wait a given amount of time or wait until a condition is satisfied
- insert a comment into the program
- load program modules.

3.1 Assigning a value to data

Data can be assigned an arbitrary value. It can, for example, be initialised with a constant value, e.g. 5, or updated with an arithmetic expression, e.g. *reg1+5*reg3*.

Instruction	Used to:
:=	Assign a value to data

3.2 Wait

The robot can be programmed to wait a given amount of time, or to wait until an arbitrary condition is satisfied; for example, to wait until an input is set.

Instruction	Used to:
WaitTime	Wait a given amount of time or to wait until the robot stops moving
WaitUntil	Wait until a condition is satisfied
WaitDI	Wait until a digital input is set
WaitDO	Wait until a digital output is set

3.3 Comments

Comments are only inserted into the program to increase its readability. Program execution is not affected by a comment.

Instruction	Used to:
comment	Comment on the program

3.4 Loading program modules

Program modules can be loaded from mass memory or erased from the program memory. In this way large programs can be handled with only a small memory.

Instruction	Used to:
Load	Load a program module into the program memory
UnLoad	Unload a program module from the program memory
Start Load	Load a program module into the program memory during execution
Wait Load	Connect the module, if loaded with *StartLoad*, to the program task
Save	Save a program module

Data type	Used to:
loadsession	Program a load session

3.5 Various functions

Function	Used to:
OpMode	Read the current operating mode of the robot
RunMode	Read the current program execution mode of the robot
Dim	Obtain the dimensions of an array
Present	Find out whether an optional parameter was present when a routine call was made
IsPers	Check whether a parameter is a persistent
IsVar	Check whether a parameter is a variable

3.6 Basic data

Data type	Used to define:
bool	Logical data (with the values true or false)
num	Numeric values (decimal or integer)
symnum	Numeric data with symbolic value
string	Character strings
switch	Routine parameters without value

3.7 Conversion function

Function	Used to:
StrToByte	Convert a byte to a string data with a defined byte data format.
ByteToStr	Convert a string with a defined byte data format to a byte data.

4 Motion Settings

Some of the motion characteristics of the robot are determined using logical instructions that apply to all movements:

- Maximum velocity and velocity override
- Acceleration
- Management of different robot configurations
- Payload
- Behaviour close to singular points
- Program displacement
- Soft servo
- Tuning values

4.1 Programming principles

The basic characteristics of the robot motion are determined by data specified for each positioning instruction. Some data, however, is specified in separate instructions which apply to all movements until that data changes.

The general motion settings are specified using a number of instructions, but can also be read using the system variable *C_MOTSET* or *C_PROGDISP.*

Default values are automatically set (by executing the routine *SYS_RESET* in system module BASE)

- at a cold start-up,
- when a new program is loaded,
- when the program is started from the beginning.

4.2 Defining velocity

The absolute velocity is programmed as an argument in the positioning instruction. In addition to this, the maximum velocity and velocity override (a percentage of the programmed velocity) can be defined.

Instruction	Used to define:
VelSet	The maximum velocity and velocity override

4.3 Defining acceleration

When fragile parts, for example, are handled, the acceleration can be reduced for part of the program.

Instruction	Used to define:
AccSet	The maximum acceleration

4.4 Defining configuration management

The robot's configuration is normally checked during motion. If joint (axis-by-axis) motion is used, the correct configuration will be achieved. If linear or circular motion are used, the robot will always move towards the closest configuration, but a check is performed to see if it is the same as the programmed one. It is possible to change this, however.

Instruction	Used to define:
ConfJ	Configuration control on/off during joint motion
ConfL	Configuration check on/off during linear motion

4.5 Defining the payload

To achieve the best robot performance, the correct payload must be defined.

Instruction	Used to define:
GripLoad	The payload of the gripper

4.6 Defining the behaviour near singular points

The robot can be programmed to avoid singular points by changing the tool orientation automatically.

Instruction	Used to define:
SingArea	The interpolation method through singular points

4.7 Displacing a program

When part of the program must be displaced, e.g. following a search, a program displacement can be added.

Instruction	Used to:
PDispOn	Activate program displacement
PDispSet	Activate program displacement by specifying a value
PDispOff	Deactivate program displacement
EOffsOn	Activate an external axis offset
EOffsSet	Activate an external axis offset by specifying a value
EOffsOff	Deactivate an external axis offset

Function	Used to:
DefDFrame	Calculate a program displacement from three positions
DefFrame	Calculate a program displacement from six positions
ORobT	Remove program displacement from a position

4.8 Soft servo

One or more of the robot axes can be made “soft”. When using this function, the robot will be compliant and can replace, for example, a spring tool.

Instruction	Used to:
SoftAct	Activate the soft servo for one or more axes
SoftDeact	Deactivate the soft servo

4.9 Adjust the robot tuning values

In general, the performance of the robot is self-optimising; however, in certain extreme cases, overrunning, for example, can occur. You can adjust the robot tuning values to obtain the required performance.

Instruction	Used to:
TuneServo[1]	Adjust the robot tuning values
TuneReset	Reset tuning to normal
PathResol	Adjust the geometric path resolution

Data type	Used to:
tunetype	Represent the tuning type as a symbolic constant

1. Only when the robot is equipped with the option “Advanced Motion”

4.10 World Zones

Up to 10 different volumes can be defined within the working area of the robot. These can be used for:

- Indicating that the robot's TCP is a definite part of the working area.
- Delimiting the working area for the robot and preventing a collision with the tool.
- Creating a working area common to two robots. The working area is then available only to one robot at a time.

Instruction	Used to:
WZBoxDef[1]	Define a box-shaped global zone
WZCylDef[1]	Define a cylindrical global zone
WZSphDef	Define a spherical global zone
WZLimSup[1]	Activate limit supervision for a global zone
WZDOSet[1]	Activate global zone to set digital outputs
WZDisable[1]	Deactivate supervision of a temporary global zone
WZEnable[1]	Activate supervision of a temporary global zone
WZFree[1]	Erase supervision of a temporary global zone

Data type	Used to:
wztemporary	Identify a temporary global zone
wzstationary	Identify a stationary global zone
shapedata	Describe the geometry of a global zone

4.11 Data for motion settings

Data type	Used to define:
motsetdata	Motion settings except program displacement
progdisp	Program displacement

1. Only when the robot is equipped with the option "Advanced functions"

5 Motion

The robot movements are programmed as pose-to-pose movements, i.e. “move from the current position to a new position”. The path between these two positions is then automatically calculated by the robot.

5.1 Programming principles

The basic motion characteristics, such as the type of path, are specified by choosing the appropriate positioning instruction.

The remaining motion characteristics are specified by defining data which are arguments of the instruction:

- Position data (end position for robot and external axes)
- Speed data (desired speed)
- Zone data (position accuracy)
- Tool data (e.g. the position of the TCP)
- Work-object data (e.g. the current coordinate system)

Some of the motion characteristics of the robot are determined using logical instructions which apply to all movements (See *Motion Settings* on page 9):

- Maximum velocity and velocity override
- Acceleration
- Management of different robot configurations
- Payload
- Behaviour close to singular points
- Program displacement
- Soft servo
- Tuning values

Both the robot and the external axes are positioned using the same instructions. The external axes are moved at a constant velocity, arriving at the end position at the same time as the robot.

5.2 Positioning instructions

Instruction	Type of movement:
MoveC	TCP moves along a circular path
MoveJ	Joint movement
MoveL	TCP moves along a linear path
MoveAbsJ	Absolute joint movement
MoveCDO	Moves the robot circularly and sets a digital output in the middle of the corner path.
MoveJDO	Moves the robot by joint movement and sets a digital output in the middle of the corner path.
MoveLDO	Moves the robot linearly and sets a digital output in the middle of the corner path.
MoveCSync[1]	Moves the robot circularly and executes a RAPID procedure
MoveJSync[1]	Moves the robot by joint movement and executes a RAPID procedure
MoveLSync[1]	Moves the robot linearly and executes a RAPID procedure

1. Only if the robot is equipped with the option “Advanced Functions”.

5.3 Searching

During the movement, the robot can search for the position of a work object, for example. The searched position (indicated by a sensor signal) is stored and can be used later to position the robot or to calculate a program displacement.

Instruction	Type of movement:
SearchC	TCP along a circular path
SearchL	TCP along a linear path

5.4 Activating outputs or interrupts at specific positions

Normally, logical instructions are executed in the transition from one positioning instruction to another. If, however, special motion instructions are used, these can be executed instead when the robot is at a specific position.

Instruction	Used to:
TriggIO[1]	Define a trigg condition to set an output at a given position
TriggInt[1]	Define a trigg condition to execute a trap routine at a given position

1. Only if the robot is equipped with the option “Advanced functions”

TriggEquip[1]	Define a trigg condition to set an output at a given position with the possibility to include time compensation for the lag in the external equipment
TriggC[1]	Run the robot (TCP) circularly with an activated trigg condition
TriggJ[1]	Run the robot axis-by-axis with an activated trigg condition
TriggL[1]	Run the robot (TCP) linearly with an activated trigg condition

Data type	Used to define:
triggdata[1]	Trigg conditions
aoitrigg	Analogue I/O trigger condition

1.Only if the robot is equipped with the option “Advanced Functions”.

5.5 Motion control if an error/interrupt takes place

In order to rectify an error or an interrupt, motion can be stopped temporarily and then restarted again.

Instruction	Used to:
StopMove	Stop the robot movements
StartMove	Restart the robot movements
StorePath[1]	Store the last path generated
RestoPath[1]	Regenerate a path stored earlier

1. Only if the robot is equipped with the option “Advanced Functions”.

5.6 Controlling external axes

The robot and external axes are usually positioned using the same instructions. Some instructions, however, only affect the external axis movements.

Instruction	Used to:
DeactUnit	Deactivate an external mechanical unit
ActUnit	Activate an external mechanical unit
MechUnit Load	Defines a payload for a mechanical unit

5.7 Independent axes

The robot axis 6 (and 4 on IRB 2400 /4400) or an external axis can be moved independently of other movements. The working area of an axis can also be reset, which will reduce the cycle times.

Function	Used to:
IndAMove[2]	Change an axis to independent mode and move the axis to an absolute position
IndCMove[2]	Change an axis to independent mode and start the axis moving continuously
IndDMove[2]	Change an axis to independent mode and move the axis a delta distance
IndRMove[2]	Change an axis to independent mode and move the axis to a relative position (within the axis revolution)
IndReset[2]	Change an axis to dependent mode or/and reset the working area
IndInpos[2]	Check whether an independent axis is in position
IndSpeed[2]	Check whether an independent axis has reached programmed *speed*
Instruction	Used to:
HollowWristReset[2]	Reset the position of the wrist joints on hollow wrist manipulators, such as IRB 5402 and IRB 5403.

2. Only if the robot is equipped with the option “Advanced Motion”.
(The Instruction HollowWristReset can only be used on Robot IRB 5402 AND IRB 5403.)

5.8 Path Correction

Instruction	Used to:
CorrCon[2]	Connect to a correction generator
CorrWrite[2]	Write offsets in the path coordinate system to a correction generator
CorrDiscon[2]	Disconnect from a previously connected correction generator
CorrClear[2]	Remove all connected correction generators
Function	Used to:
CorrRead[2]	Read the total corrections delivered by all connected correction generators
Data type	Used to:
Corrdescr[2]	Add geometric offsets in the path coordinate system

2. Only if the robot is equipped with the option “Advanced Motion”.
(The Instruction HollowWristReset can only be used on Robots IRB 5402 and IRB 5403.)

5.9 Conveyor tracking

Instruction	Used to:
WaitWObj[3]	Wait for work object on conveyor
DropWObj[3]	Drop work object on conveyor

3. Only if the robot is equipped with the option “Conveyor tracking”.

5.10 Load identification and Collision detection

Instruction	Used to:
MotionSup	Deactivates/activates motion supervision
ParIdPosValid	Valid robot position for parameter identification
ParIdRobValid	Valid robot type for parameter identification
LoadId	Load identification of tool or payload

5.11 Position functions

Function	Used to:
Offs	Add an offset to a robot position, expressed in relation to the work object
RelTool	Add an offset, expressed in the tool coordinate system
CalcRobT	Calculates *robtarget* from *jointtarget*
CPos	Read the current position (only *x*, *y*, *z* of the robot)
CRobT	Read the current position (the complete *robtarget*)
CJointT	Read the current joint angles
ReadMotor	Read the current motor angles
CTool	Read the current tooldata value
CWObj	Read the current wobjdata value
ORobT	Remove a program displacement from a position
MirPos	Mirror a position
CalcJointT	Calculates joint angles from *robtarget*
Distance	The distance between two positions

5.12 Motion data

Motion data is used as an argument in the positioning instructions.

Data type	Used to define:
robtarget	The end position
jointtarget	The end position for a *MoveAbsJ* instruction
speeddata	The speed
zonedata	The accuracy of the position (stop point or fly-by point)
tooldata	The tool coordinate system and the load of the tool
wobjdata	The work object coordinate system

5.13 Basic data for movements

Data type	Used to define:
pos	A position (x, y, z)
orient	An orientation
pose	A coordinate system (position + orientation)
confdata	The configuration of the robot axes
extjoint	The position of the external axes
robjoint	The position of the robot axes
o_robtarget	Original robot position when *Limit ModPos* is used
o_jointtarget	Original robot position when *Limit ModPos* is used for *MoveAbsJ*
loaddata	A load
mecunit	An external mechanical unit

6 Input and Output Signals

The robot can be equipped with a number of digital and analog user signals that can be read and changed from within the program.

6.1 Programming principles

The signal names are defined in the system parameters. These names are always available in the program for reading or setting I/O operations.

The value of an analog signal or a group of digital signals is specified as a numeric value.

6.2 Changing the value of a signal

Instruction	Used to:
InvertDO	Invert the value of a digital output signal
PulseDO	Generate a pulse on a digital output signal
Reset	Reset a digital output signal (to 0)
Set	Set a digital output signal (to 1)
SetAO	Change the value of an analog output signal
SetDO	Change the value of a digital output signal (symbolic value; e.g. *high/low*)
SetGO	Change the value of a group of digital output signals

6.3 Reading the value of an input signal

The value of an input signal can be read directly in the program, e.g. :

```
! Digital input
IF di1 = 1 THEN ...

! Digital group input
IF gi1 = 5 THEN ...

! Analog input
IF ai1 > 5.2 THEN ...
```

6.4 Testing input on output signals

Instruction	Used to:
WaitDI	Wait until a digital input is set or reset
WaitDO	Wait until a digital output is set on reset

Function	Used to:
TestDI	Test whether a digital input is set

6.5 Disabling and enabling I/O modules

I/O modules are automatically enabled at start-up, but they can be disabled during program execution and re-enabled later.

Instruction	Used to:
IODisable	Disable an I/O module
IOEnable	Enable an I/O module

6.6 Defining input and output signals

Data type	Used to define:
dionum	The symbolic value of a digital signal
signalai	The name of an analog input signal *
signalao	The name of an analog output signal *
signaldi	The name of a digital input signal *
signaldo	The name of a digital output signal *
signalgi	The name of a group of digital input signals *
signalgo	The name of a group of digital output signals *

Instruction	Used to:
AliasIO[1]	Define a signal with an alias name

* Only to be defined using system parameters.

1. Only if the robot is equipped with the option “Developer’s Functions”

7 Communication

There are four possible ways to communicate via serial channels:

- Messages can be output to the teach pendant display and the user can answer questions, such as about the number of parts to be processed.
- Character-based information can be written to or read from text files in mass memory. In this way, for example, production statistics can be stored and processed later in a PC. Information can also be printed directly on a printer connected to the robot.
- Binary information can be transferred between the robot and a sensor, for example.
- Binary information can be transferred between the robot and another computer, for example, with a link protocol.

7.1 Programming principles

The decision whether to use character-based or binary information depends on how the equipment with which the robot communicates handles that information. A file, for example, can have data that is stored in character-based or binary form.

If communication is required in both directions simultaneously, binary transmission is necessary.

Each serial channel or file used must first be opened. On doing this, the channel/file receives a descriptor that is then used as a reference when reading/writing. The teach pendant can be used at all times and does not need to be opened.

Both text and the value of certain types of data can be printed.

7.2 Communicating using the teach pendant

Instruction	Used to:
TPErase	Clear the teach pendant operator display
TPWrite	Write text on the teach pendant operator display
ErrWrite	Write text on the teach pendant display and simultaneously store that message in the progam's error log.
TPReadFK	Label the function keys and to read which key is pressed
TPReadNum	Read a numeric value from the teach pendant
TPShow	Choose a window on the teach pendant from RAPID

7.3 Reading from or writing to a character-based serial channel/file

Instruction	Used to:
Open[1]	Open a channel/file for reading or writing
Write[1]	Write text to the channel/file
Close[1]	Close the channel/file

Function	Used to:
ReadNum[1]	Read a numeric value
ReadStr[1]	Read a text string

7.4 Communicating using binary serial channels/files

Instruction	Used to:
Open[1]	Open a serial channel/file for binary transfer of data
WriteBin[1]	Write to a binary serial channel/file
WriteAnyBin[1]	Write to any binary serial channel/file
WriteStrBin[1]	Write a string to a binary serial channel/file
Rewind[1]	Set the file position to the beginning of the file
Close[1]	Close the channel/file
ClearIOBuff[1]	Clear input buffer of a serial channel

Function	Used to:
ReadBin[1]	Read from a binary serial channel
ReadAnyBin[1]	Read from any binary serial channel

7.5 Data for serial channels

Data type	Used to define:
iodev	A reference to a serial channel/file, which can then be used for reading and writing

1. Only if the robot is equipped with the option “Advanced functions”

8 Interrupts

Interrupts are used by the program to enable it to deal directly with an event, regardless of which instruction is being run at the time.

The program is interrupted, for example, when a specific input is set to one. When this occurs, the ordinary program is interrupted and a special trap routine is executed. When this has been fully executed, program execution resumes from where it was interrupted.

8.1 Programming principles

Each interrupt is assigned an interrupt identity. It obtains its identity by creating a variable (of data type *intnum*) and connecting this to a trap routine.

The interrupt identity (variable) is then used to order an interrupt, i.e. to specify the reason for the interrupt. This may be one of the following events:

- An input or output is set to one or to zero.
- A given amount of time elapses after an interrupt is ordered.
- A specific position is reached.

When an interrupt is ordered, it is also automatically enabled, but can be temporarily disabled. This can take place in two ways:

- All interrupts can be disabled. Any interrupts occurring during this time are placed in a queue and then automatically generated when interrupts are enabled again.
- Individual interrupts can be deactivated. Any interrupts occurring during this time are disregarded.

8.2 Connecting interrupts to trap routines

Instruction	Used to:
CONNECT	Connect a variable (interrupt identity) to a trap routine

8.3 Ordering interrupts

Instruction	Used to order:
ISignalDI	An interrupt from a digital input signal
SignalDO	An interrupt from a digital output signal
ITimer	A timed interrupt
TriggInt[1]	A position-fixed interrupt (from the Motion pick list)

8.4 Cancelling interrupts

Instruction	Used to:
IDelete	Cancel (delete) an interrupt

8.5 Enabling/disabling interrupts

Instruction	Used to:
ISleep	Deactivate an individual interrupt
IWatch	Activate an individual interrupt
IDisable	Disable all interrupts
IEnable	Enable all interrupts

8.6 Data type of interrupts

Data type	Used to define:
intnum	The identity of an interrupt

1. Only if the robot is equipped with the option “Advanced functions”

9 Error Recovery

Many of the errors that occur when a program is being executed can be handled in the program, which means that program execution does not have to be interrupted. These errors are either of a type detected by the robot, such as division by zero, or of a type that is detected by the program, such as errors that occur when an incorrect value is read by a bar code reader.

9.1 Programming principles

When an error occurs, the error handler of the routine is called (if there is one). It is also possible to create an error from within the program and then jump to the error handler.

If the routine does not have an error handler, a call will be made to the error handler in the routine that called the routine in question. If there is no error handler there either, a call will be made to the error handler in the routine that called that routine, and so on until the internal error handler of the robot takes over and outputs an error message and stops program execution.

In the error handler, errors can be handled using ordinary instructions. The system data *ERRNO* can be used to determine the type of error that has occurred. A return from the error handler can then take place in various ways.

In future releases, if the current routine does not have an error handler, the internal error handler of the robot takes over directly. The internal error handler outputs an error message and stops program execution with the program pointer at the faulty instruction.

So, a good rule already in this issue is as follows: if you want to call the error handler of the routine that called the current routine (propagate the error), then:

- Add an error handler in the current routine
- Add the instruction RAISE in this error handler.

9.2 Creating an error situation from within the program

Instruction	Used to:
RAISE	“Create” an error and call the error handler

9.3 Restarting/returning from the error handler

Instruction	Used to:
EXIT	Stop program execution in the event of a fatal error
RAISE	Call the error handler of the routine that called the current routine
RETRY	Re-execute the instruction that caused the error
TRYNEXT	Execute the instruction following the instruction that caused the error
RETURN	Return to the routine that called the current routine

9.4 Data for error handling

Data type	Used to define:
errnum	The reason for the error

10 System & Time

System and time instructions allow the user to measure, inspect and record time.

10.1 Programming principles

Clock instructions allow the user to use clocks that function as stopwatches. In this way the robot program can be used to time any desired event.

The current time or date can be retrieved in a string. This string can then be displayed to the operator on the teach pendant display or used to time and date-stamp log files.

It is also possible to retrieve components of the current system time as a numeric value. This allows the robot program to perform an action at a certain time or on a certain day of the week.

10.2 Using a clock to time an event

Instruction	Used to:
ClkReset	Reset a clock used for timing
ClkStart	Start a clock used for timing
ClkStop	Stop a clock used for timing

Function	Used to:
ClkRead	Read a clock used for timing

Data Type	Used for:
clock	Timing – stores a time measurement in seconds

10.3 Reading current time and date

Function	Used to:
CDate	Read the Current Date as a string
CTime	Read the Current Time as a string
GetTime	Read the Current Time as a numeric value

10.4 Retrieve time information from file

Function	Used to:
FileTime	Retrieve the last time for modification of a file.
ModTime	Retrieve the time of loading a specified module.

11 Mathematics

Mathematical instructions and functions are used to calculate and change the value of data.

11.1 Programming principles

Calculations are normally performed using the assignment instruction, e.g. *reg1:= reg2 + reg3 / 5*. There are also some instructions used for simple calculations, such as to clear a numeric variable.

11.2 Simple calculations on numeric data

Instruction	Used to:
Clear	Clear the value
Add	Add or subtract a value
Incr	Increment by 1
Decr	Decrement by 1

11.3 More advanced calculations

Instruction	Used to:
:=	Perform calculations on any type of data

11.4 Arithmetic functions

Function	Used to:
Abs	Calculate the absolute value
Round	Round a numeric value
Trunc	Truncate a numeric value
Sqrt	Calculate the square root
Exp	Calculate the exponential value with the base “e”
Pow	Calculate the exponential value with an arbitrary base
ACos	Calculate the arc cosine value
ASin	Calculate the arc sine value
ATan	Calculate the arc tangent value in the range [-90,90]
ATan2	Calculate the arc tangent value in the range [-180,180]
Cos	Calculate the cosine value
Sin	Calculate the sine value
Tan	Calculate the tangent value
EulerZYX	Calculate Euler angles from an orientation
OrientZYX	Calculate the orientation from Euler angles
PoseInv	Invert a pose
PoseMult	Multiply a pose
PoseVect	Multiply a pose and a vector
Vectmagn	Calculate the magnitude of a *pos* vector.
DotProd	Calculate the dot (or scalar) product of two pos vectors.
NOrient	Normalise unnormalised orientation (quarternion)

12 Spot Welding

The SpotWare package provides support for spot welding applications that are equipped with a weld timer and on/off weld gun.

The SpotWare application provides fast and accurate positioning combined with gun manipulation, process start and supervision of an external weld timer.

Communication with the welding equipment is carried out by means of digital inputs and outputs. Some serial weld timer interfaces are also supported: Bosch PSS5000, NADEX, ABB Timer. See separate documentation.

It should be noted that SpotWare is a package that can be extensively customised. The intention is that the user adapts some user data and routines to suit the environmental situation.

12.1 Spot welding features

The SpotWare package contains the following features:

- Fast and accurate positioning
- Handling of an on/off gun with two strokes
- Dual/single gun
- Gun pre-closing
- User defined supervision of the surrounding equipment before weld start
- User defined supervision of the surrounding equipment after the weld
- User defined open/close gun and supervision
- User defined pressure setting
- User defined preclose time calculation
- Monitoring of the external weld timer
- Weld error recovery with automatic rewelding
- Return to the spot weld position
- Spot counters
- Time or signal dependent motion release after a weld
- Quick start after a weld
- User-defined service routines
- Presetting and checking of gun pressure
- Simulated welding
- Reverse execution with gun control
- Parallel and serial weld timer interfaces

- Supports both program and start triggered weld timers
- SpotL/J current data information
- Spot identity info: the current spotdata parameter name (string format)
- Spot identity transfer to the serial weld timer BOSCH PSS 5000
- User defined autonomous supervision, such as state-controlled weld current signal and water cooling start. Note: This feature requires the MultiTasking option
- Manual weld, gun open and gun close initiated by digital input
- Weld process start disregarding the in position event, is possible
- Optional user defined error recovery

12.2 Principles of SpotWare

SpotWare is based on a separate handling of motion, spot welding and, if MultiTasking is installed, continuous supervision. On its way towards the programmed position, the motion task will trigger actions in the spot-welding tasks.

The triggers are activated by virtual digital signals.

The tasks work with their own internal encapsulated variables and with persistents which are fully transparent for all tasks.

For well defined entries, calls to user routines offer adaptations to the plant environment. A number of predefined parameters are also available to shape the behaviour of the SpotL/J instruction.

A program stop will only stop the motion task execution. The process and supervision carry on their tasks until they come to a well defined process stop. For example, this will finish the weld and open the gun, although the program has stopped.

The opening and closing of the gun are always executed by RAPID routines, even if activated manually from the I/O window on the teach-pendant. These gun routines may be changed from the simple on/off default functionality to a more complex like analog gun control and they may contain additional gun supervision.

Since the process and supervision tasks are acting on I/O triggers, they will be executed either by the trig that was sent by the motion (SpotL/J) or by manual activation (teach pendant or external). This offers the possibility of performing a stand-alone weld anywhere without programming a new position.

It is also possible to define new supervision events and to connect them to digital signal triggers. By default, a state dependent weld power and water cooling signal control are implemented.

Supported equipment:

- One weld timer monitoring with standard parallel (some serial) interface. The weld timer may be of the type, program schedule or start signal triggered.

- Any type of single/dual gun close and gun gap control.
- Any type of pressure preset.
- Event controlled SpotL/J-independent spot weld equipment such as contactors etc. (Note: MultiTasking option required).

12.3 Programming principles

Both the robot's linear movement and the spot weld process control are embedded in one instruction, *SpotL*.

Both the robot's joint movement and the spot weld process control are embedded in one instruction, *SpotJ*.

The spot welding process is specified by:

- Spotdata: spot weld process data
- Gundata: spot weld gun data
- The system modules SWUSRF and SWUSRC: RAPID routines and global data for customising purposes. See *Predefined Data and Programs ProcessWare*.
- System parameters: the I/O configuration. See User's Guide - System Parameters

12.4 Spot welding instructions

Instruction	Used to:
SpotL	Control the motion, gun closure/opening, and the welding process Move the TCP along a linear path and perform a spot weld at the end position
SpotJ	Control the motion, gun closure/opening, and the welding process Move the TCP along a non-linear path and perform a spot weld at the end position
SpotML	Spot welding with multiple guns

12.5 Spot welding data

Data type	Used to define:
spotdata	The spot weld process control
gundata	The spot weld gun
spotmdata	The spot weld process control for multiple guns
gunmdata	Spot weld gundata for multiple guns

13 Arc Welding

The ArcWare package supports most welding functions. Crater-filling and scraping starts can, for example, be programmed. Using ArcWare, the whole welding process can be controlled and monitored by the robot via a number of different digital and analog inputs and outputs.

13.1 Programming principles

The same instructions are used both to control the robot's movements and the actual welding process. The arc welding instructions indicate which weld data and seam data are used in the weld in question.

The weld settings for the actual weld phase are defined in weld data. The start and end phase are defined in seam data.

Any weaving is defined in weave data, which is also identified by the arc welding instruction.

Certain functions, such as a scraping start, are defined in the system parameters.

The welding process is divided into the following phases:



13.2 Arc welding instructions

Instruction	Type of movement:
ArcC	TCP along a circular path
ArcL	TCP along a linear path

13.3 Arc welding plus instructions

Instruction	Type of movement:
ArcRefresh	Refresh ArcWeld data
ArcKill	Kill the ArcWeld process
SpcWrite	Writes to a statistical process controller
SpcStat	Statistical process control status
SpcRead	Reads the current process status
SpcDump	Dumps statistical process control information
SpcDiscon	Disconnects from a statistical process controller
SpcCon	Connects to a statistical process controller

13.4 Arc welding data

Data type	Used to define:
welddata	The weld phase
seamdata	The start and end phase of a weld
weavedata	The weaving characteristics
arcdata	Arc process data

13.5 Arc welding plus data

Data type	Used to define:
spcdescr	Statistical process control descriptor
spcdata	Statistical process control data

14 GlueWare

The GlueWare package provides support for gluing applications that are equipped with one or two gluing guns.

The GlueWare application provides fast and accurate positioning combined with gun manipulation, process start and stop.

Communication with the glueing equipment is carried out by means of digital and analog outputs.

14.1 Glueing features

The GlueWare package contains the following features:

- Fast and accurate positioning
- Handling of on/off guns as well as proportional guns
- Two different guns can be handled in the same program, each gun controlled by one digital signal (on/off) and two analog signals (flows)
- Gun pre-opening and pre-closing respectively
- Simulated glueing

14.2 Programming principles

Both the robot's movement and the glue process control are embedded in one instruction, *GlueL* and *GlueC* respectively.

The glueing process is specified by:

- Gundata: glue gun data. See *Data types - ggundata*.
- The system module GLUSER: RAPID routines and global data for customizing purposes. See *Predefined Data and Programs - System Module GLUSER*.
- System parameters: the I/O configuration. See *System Parameters - Glueing*

14.3 Glue instructions

Instruction	Used to:
GlueL	Move the TCP along a linear path and perform gluing with the given data
GlueC	Move the TCP along a circular path and perform gluing with the given data

14.4 Glue data

Data type	Used to define:
ggundata	The glue gun used

14.5 DispenseWare data

Data type	Used to define:
beaddata	Dispensing Bead Data
equidata	Dispensing Equipment Data

15 External Computer Communication

The robot can be controlled from a superordinate computer. In this case, a special communications protocol is used to transfer information.

15.1 Programming principles

As a common communications protocol is used to transfer information from the robot to the computer and vice versa, the robot and computer can understand each other and no programming is required. The computer can, for example, change values in the program's data without any programming having to be carried out (except for defining this data). Programming is only necessary when program-controlled information has to be sent from the robot to the superordinate computer.

15.2 Sending a program-controlled message from the robot to a computer

Instruction	Used to:
SCWrite[1]	Send a message to the superordinate computer

1. Only if the robot is equipped with the option "RAP Serial Link".

16 RAPID Support Instructions

Various functions for supporting of the RAPID language:

- Get system data

16.1 Get system data

Instruction to fetch the value and (optional) the symbol name for the current system data of specified type.

Instruction	Used to:
GetSysData	Fetch data for and name of current active Tool or Work Object.

17 Service Instructions

A number of instructions are available to test the robot system. See the chapter on Troubleshooting Tools in the Product Manual for more information.

17.1 Directing a value to the robot's test signal

A reference signal, such as the speed of a motor, can be directed to an analog output signal located on the backplane of the robot.

Instruction	Used to:
TestSign	Define and activate a test signal

Data type	Used to define:
testsignal	The type of test signal

18 String Functions

String functions are used for operations with strings such as copying, concatenation, comparison, searching, conversion, etc.

18.1 Basic Operations

Data type	Used to define:
string	String. Predefined constants STR_DIGIT, STR_UPPER, STR_LOWER and STR_WHITE
Instruction/Operator	**Used to:**
:=	Assign a value (copy of string)
+	String concatenation
Function	**Used to:**
StrLen	Find string length
StrPart	Obtain part of a string

18.2 Comparison and Searching

Operator	Used to:
=	Test if equal to
<>	Test if not equal to
Function	**Used to:**
StrMemb	Check if character belongs to a set
StrFind	Search for character in a string
StrMatch	Search for pattern in a string
StrOrder	Check if strings are in order

18.3 Conversion

Function	Used to:
NumToStr	Convert a numeric value to a string
ValToStr	Convert a value to a string
StrToVal	Convert a string to a value
StrMap	Map a string
StrToByte	Convert a string to a byte
ByteToStr	Convert a byte to string data

19 Multitasking

Multitasking RAPID is a way to execute programs in (pseudo) parallel with the normal execution. One parallel program can be placed in the background or foreground of another program. It can also be on the same level as another program.
(See Basic Characteristics Multitasking.)

19.1 Basics

To use this function the robot must be configured with one extra TASK for each background program**.**

Up to 10 different tasks can be run in pseudo parallel. Each task consists of a set of modules, in the same way as the normal program. All the modules are local in each task.

Variables and constants are local in each task, but persistents are not. A persistent with the same name and type is reachable in all tasks. If two persistents have the same name, but their type or size (array dimension) differ, a runtime error will occur.

A task has its own trap handling and the event routines are triggered only on its own task system states (e.g. Start/Stop/Restart....).

19.2 Resource access Protection

Function	Used to
TestAndSet	Retrieve exclusive right to specific RAPID code areas or system resources.

20 Syntax Summary

20.1 Instructions

Data **:=** Value

AccSet Acc Ramp

ActUnit MecUnit

Add Name AddValue

ArcRefresh

ArcKill

Break

CallBy Var Name Number

Clear Name

ClkReset Clock

ClkStart Clock

ClkStop Clock

Close IODevice

! Comment

ConfJ [\On] | [\Off]

ConfL [\On] | [\Off]

CONNECT Interrupt **WITH** Trap routine

CorrClear

CorrCon Descr

CorrDiscon Descr

CorrWrite

CorrWrite Descr Data

CorrClear

DeactUnit MecUnit

Decr Name

DropWObj WObj

EOffsSet EAxOffs

ErrWrite [\W] Header Reason [\RL2] [\RL3] [\RL4]

Exit

ExitCycle

FOR Loop counter **FROM** Start value **TO** End value
[**STEP** Step value] **DO** ... **ENDFOR**

GOTO Label

GripLoad Load

GetSysData DestObject [\ ObjectName]

IDelete Interrupt

IF Condition **...**

IF Condition **THEN** ...
{**ELSEIF** Condition **THEN** ...}
[**ELSE** ...]

ENDIF

Incr Name

IndAMove MecUnit Axis [\ToAbsPos] | [\ToAbsNum] Speed
[\Ramp]

IndCMove MecUnit Axis Speed [\Ramp]

IndDMove MecUnit Axis Delta Speed [\Ramp]

IndReset MecUnit Axis [\RefPos] | [\RefNum] | [\Short] | [\Fwd] |
[\Bwd] | [\Old]

IndRMove MecUnit Axis [\ToRelPos] | [\ToRelNum] | [\Short] |
[\Fwd] | [\Bwd] Speed [\Ramp]

InvertDO Signal

IODisable UnitName MaxTime

IOEnable UnitName MaxTime

ISignalDI [\Single] Signal TriggValue Interrupt

ISignalDO [\Single] Signal TriggValue Interrupt

ISleep Interrupt

ITimer [\Single] Time Interrupt

IVarValue VarNo Value, Interrupt

IWatch Interrupt ParIdType LoadIdType Tool [\PayLoad] [\WObj] [\ConfAngle] [\SlowTest] [\Accuracy]

LoadId ParIdType LoadIdType Tool [\PayLoad] [\WObj] [\ConfAngle] [\SlowTest] [\Accuracy]

MechUnitLoad MechUnit AxisNo Load

MoveAbsJ [\Conc] ToJointPos Speed [\V] | [\T] Zone [\Z] Tool [\WObj]

MoveC [\Conc] CirPoint ToPoint Speed [\V] | [\T] Zone [\Z] Tool [\WObj]

MoveCDO CirPoint ToPoint Speed [\T] Zone Tool [\WObj] Signal Value

MoveCSync CirPoint ToPoint Speed [\T] Zone Tool [\WObj] ProcName

MoveJ [\Conc] ToPoint Speed [\V] | [\T] Zone [\Z] Tool [\WObj]

MoveJDO ToPoint Speed [\T] Zone Tool [\WObj] Signal Value

MoveJSync ToPoint Speed [\T] Zone Tool [\WObj] ProcName

MoveL [\Conc] ToPoint Speed [\V] | [\T] Zone [\Z] Tool [\WObj]

MoveLDO ToPoint Speed [\T] Zone Tool
[\WObj] Signal Value

MoveLSync ToPoint Speed [\T] Zone Tool
[\WObj] ProcName

MotionSup [\On] | [\Off] [\TuneValue]

Open Object [\File] IODevice [\Read] | [\Write] | [\Append] | [\Bin]

ParIdPosValid (ParIdType Pos AxValid [\ConfAngle])

ParIdRobValid (ParIdType)

PathResol Value

PDispOn [\Rot] [\ExeP] ProgPoint Tool [\WObj]

PDispSet DispFrame

Procedure { Argument }

PulseDO [\PLength] Signal

RAISE [Error no]

Reset Signal

RETURN [Return value]

Rewind IODevice

Save [\Task] ModuleName [\FilePath] [\File]

SearchC [\Stop] | [\PStop] | [\Sup] Signal SearchPoint CirPoint
ToPoint Speed [\V] | [\T] Tool [\WObj]

SearchL [\Stop] | [\PStop] | [\Sup] Signal SearchPoint ToPoint
Speed [\V] | [\T] Tool [\WObj]

Set Signal

SetAO Signal Value

SetDO [\SDelay] Signal Value

SetGO Signal Value

SingArea [\Wrist] | [\Arm] | [\Off]

SoftAct Axis Softness [\Ramp]

SpcCon Descr Status [\GrpSize] [\Teach] [\Strict] [\Header] [\BackupFile]

SpcDiscon Descr

SpcDump

SpcRead

SpcStat

SpotJ

SpotL ToPoint Speed Spot [\InPos] [\NoConc] [\Retract] Gun Tool [\WObj]

SpotML

Stop [\NoRegain]

TEST Test data **{CASE** Test value {**, Test value**} **:** ...} [**DEFAULT:** ...] **ENDTEST**

TPReadFK Answer String FK1 FK2 FK3 FK4 FK5 [\MaxTime] [\DIBreak] [\BreakFlag]

TPReadNum Answer String [\MaxTime] [\DIBreak] [\BreakFlag]

TPShow Window

TPWrite String [\Num] | [\Bool] | [\Pos] | [\Orient]

TriggC CirPoint ToPoint Speed [\T] Trigg_1 [\T2] [\T3] [\T4] Zone Tool [\WObj]

TriggEqip

TriggInt TriggData Distance [\Start] | [\Time] Interrupt

TriggIO TriggData Distance [\Start] | [\Time] [\DOp] | [\GOp] | [\AOp] SetValue [\DODelay] | [\AORamp]

TriggJ ToPoint Speed [\T] Trigg_1 [\T2] [\T3] [\T4] Zone Tool [\WObj]

TriggL ToPoint Speed [\T] Trigg_1 [\T2] [\T3] [\T4]
Zone Tool [\WObj]

TuneServo MecUnit Axis TuneValue

TuneServo MecUnit Axis TuneValue [\Type]

UnLoad FilePath [\File]

VelSet Override Max

WaitDI Signal Value [\MaxTime] [\TimeFlag]

WaitDO Signal Value [\MaxTime] [\TimeFlag]

WaitTime [\InPos] Time

WaitUntil [\InPos] Cond [\MaxTime] [\TimeFlag]

WaitWObj WObj [\RelDist]

WHILE Condition DO ... ENDWHILE

Write IODevice String [\Num] | [\Bool] | [\Pos] | [\Orient]
[\NoNewLine]

WriteBin IODevice Buffer NChar

WriteStrBin IODevice Str

WZBoxDef [\Inside] | [\Outside] Shape LowPoint HighPoint 1

WZCylDef [\Inside] | [\Outside] Shape CentrePoint Radius Height

WZDisable WorldZone

WZDOSet [\Temp] | [\Stat] WorldZone [\Inside] | [\Before] Shape
Signal SetValue

WZEnable WorldZone

WZFree WorldZone

WZLimSup [\Temp] | [\Stat] WorldZone Shape

WZSphDef [\Inside] | [\Outside] Shape CentrePoint Radius

20.2 Functions

Abs (Input)

ACos (Value)

AOutput (Signal)

ArgName (Parameter)

ASin (Value)

ATan (Value)

ATan2 (Y X)

ByteToStr (ByteData [\Hex] | [\Okt] | [\Bin] | [\Char])

ClkRead (Clock)

CorrRead

Cos (Angle)

CPos ([Tool] [\WObj])

CRobT ([Tool] [\WObj])

DefDFrame (OldP1 OldP2 OldP3 NewP1 NewP2 NewP3)

DefFrame (NewP1 NewP2 NewP3 [\Origin])

Dim (ArrPar DimNo)

DOutput (Signal)

DotProd (Vector1 Vector2)

EulerZYX ([\X] | [\Y] | [\Z] Rotation)

Exp (Exponent)

FileTime (Path [\ModifyTime] | [\AccessTime] | [\StatCTime])

GOutput (Signal)

GetTime ([\WDay] | [\Hour] | [\Min] | [\Sec])

IndInpos MecUnit Axis

IndSpeed MecUnit Axis [\InSpeed] | [\ZeroSpeed]

IsPers (DatObj)

IsVar (DatObj)

MirPos (Point MirPlane [\WObj] [\MirY])

ModTime (Object)

NOrient (Rotation)

NumToStr (Val Dec [\Exp])

Offs (Point XOffset YOffset ZOffset)

OrientZYX (ZAngle YAngle XAngle)

ORobT (OrgPoint [\InPDisp] | [\InEOffs])

PoseInv (Pose)

PoseMult (Pose1 Pose2)

PoseVect (Pose Pos)

Pow (Base Exponent)

Present (OptPar)

ReadBin (IODevice [\Time])

ReadMotor [\MecUnit] Axis

ReadNum (IODevice [\Time])

ReadStr (IODevice [\Time])

RelTool (Point Dx Dy Dz [\Rx] [\Ry] [\Rz])

Round (Val [\Dec])

Sin (Angle)

Sqrt (Value)

StrFind (Str ChPos Set [\NotInSet])

StrLen (Str)

StrMap (Str FromMap ToMap)

StrMatch (Str ChPos Pattern)

StrMemb (Str ChPos Set)

StrOrder (Str1 Str2 Order)

StrPart (Str ChPos Len)

StrToByte (ConStr [\Hex] | [\Okt] | [\Bin] | [\Char])

StrToVal (Str Val)

Tan (Angle)

TestDI (Signal)

TestAndSet Object

Trunc (Val [\Dec])

ValToStr (Val)

VectMagn (Vector)

1 Basic Elements

1.1 Identifiers

Identifiers are used to name modules, routines, data and labels;

e.g.
```
MODULE module_name
PROC routine_name()
VAR pos data_name;
label_name:
```

The first character in an identifier must be a letter. The other characters can be letters, digits or underscores "_".

The maximum length of any identifier is 16 characters, each of these characters being significant. Identifiers that are the same except that they are typed in the upper case, and vice versa, are considered the same.

Reserved words

The words listed below are reserved. They have a special meaning in the RAPID language and thus must not be used as identifiers.

There are also a number of predefined names for data types, system data, instructions, and functions, that must not be used as identifiers. See Chapters 7, 8, 9, 10 ,13, 14 and 15 in this manual.

ALIAS	AND	BACKWARD	CASE
CONNECT	CONST	DEFAULT	DIV
DO	ELSE	ELSEIF	ENDFOR
ENDFUNC	ENDIF	ENDMODULE	ENDPROC
ENDRECORD	ENDTEST	ENDTRAP	ENDWHILE
ERROR	EXIT	FALSE	FOR
FROM	FUNC	GOTO	IF
INOUT	LOCAL	MOD	MODULE
NOSTEPIN	NOT	NOVIEW	OR
PERS	PROC	RAISE	READONLY
RECORD	RETRY	RETURN	STEP
SYSMODULE	TEST	THEN	TO
TRAP	TRUE	TRYNEXT	VAR
VIEWONLY	WHILE	WITH	XOR

1.2 Spaces and new-line characters

The RAPID programming language is a free format language, meaning that spaces can be used anywhere except for in:

- identifiers
- reserved words
- numerical values
- placeholders.

New-line, tab and form-feed characters can be used wherever a space can be used, except for within comments.

Identifiers, reserved words and numeric values must be separated from one another by a space, a new-line, tab or form-feed character.

Unnecessary spaces and new-line characters will automatically be deleted from a program loaded into the program memory. Consequently, programs loaded from diskette and then stored again might not be identical.

1.3 Numeric values

A numeric value can be expressed as

- an integer, e.g. 3, -100, 3E2
- a decimal number, e.g. 3.5, -0.345, -245E-2

The value must be in the range specified by the ANSI IEEE 754-1985 standard (single precision) float format.

1.4 Logical values

A logical value can be expressed as TRUE or FALSE.

1.5 String values

A string value is a sequence of characters (ISO 8859-1) and control characters (non-ISO 8859-1 characters in the numeric code range 0-255). Character codes can be included, making it possible to include non-printable characters (binary data) in the string as well. String length max. 80 characters.

Example: "This is a string"
"This string ends with the BEL control character \07"

If a backslash (which indicates character code) or double quote character is included, it must be written twice.

Example: "This string contains a "" character"
"This string contains a \\ character"

1.6 Comments

Comments are used to make the program easier to understand. They do not affect the meaning of the program in any way.

A comment starts with an exclamation mark “!” and ends with a new-line character. It occupies an entire line and cannot occur between two modules;

e.g.
```
! comment
IF reg1 > 5 THEN
    ! comment
    reg2 := 0;
ENDIF
```

1.7 Placeholders

Placeholders can be used to temporarily represent parts of a program that are “not yet defined”. A program that contains placeholders is syntactically correct and may be loaded into the program memory.

Placeholder	Represents:
<TDN>	data type definition
<DDN>	data declaration
<RDN>	routine declaration
<PAR>	formal optional alternative parameter
<ALT>	optional formal parameter
<DIM>	formal (conformant) array dimension
<SMT>	instruction
<VAR>	data object (variable, persistent or parameter) reference
<EIT>	else if clause of if instruction
<CSE>	case clause of test instruction
<EXP>	expression
<ARG>	procedure call argument
<ID>	identifier

1.8 File header

A program file starts with the following file header:

```
%%%
 VERSION:1                    (Program version M94 or M94A)
 LANGUAGE:ENGLISH             (or some other language:
%%%                            GERMAN or FRENCH)
```

1.9 Syntax

Identifiers

<identifier> ::=
<ident>
| **<ID>**
<ident> ::= <letter> {<letter> | <digit> | '_'}

Numeric values

<num literal> ::=
<integer> [<exponent>]
| <integer> '.' [<integer>] [<exponent>]
| [<integer>] '.' <integer> [<exponent>]
<integer> ::= <digit> {<digit>}
<exponent> ::= ('E' | 'e') ['+' | '-'] <integer>

Logical values

<bool literal> ::= **TRUE** | **FALSE**

String values

<string literal> ::= '"' {<character> | <character code> } '"'
<character code> ::= '\' <hex digit> <hex digit>
<hex digit> ::= <digit> | A | B | C | D | E | F | a | b | c | d | e | f

Comments

<comment> ::=
'!' {<character> | <tab>} <newline>

Characters

<character> ::= -- *ISO 8859-1* --
<newline> ::= -- *newline control character* --
<digit> ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9
<letter> ::=
<upper case letter>
| <lower case letter>

<upper case letter> ::=

A | B | C | D | E | F | G | H | I | J
| K | L | M | N | O | P | Q | R | S | T
| U | V | W | X | Y | Z | À | Á | Â | Ã
| Ä | Å | Æ | Ç | È | É | Ê | Ë | Ì | Í
| Î | Ï | [1)] | Ñ | Ò | Ó | Ô | Õ | Ö | Ø
| Ù | Ú | Û | Ü | [2)] | [3)] | ß

<lower case letter> ::=

a | b | c | d | e | f | g | h | i | j
| k | l | m | n | o | p | q | r | s | t
| u | v | w | x | y | z | ß | à | á | â
| ã| ä | å | æ | ç | è | é | ê | ë | ì
| í | î | ï | [1)] | ñ | ò | ó | ô | õ | ö
| ø | ù | ú | û | ü | [2)] | [3)] | ÿ

1) Icelandic letter eth.
2) Letter Y with acute accent.
3) Icelandic letter thorn.

2 Modules

The program is divided into *program* and *system modules*. The program can also be divided into *modules* (see Figure 1).



Figure 1 The program can be divided into modules.

2.1 Program modules

A program module can consist of different data and routines. Each module, or the whole program, can be copied to diskette, RAM disk, etc., and vice versa.

One of the modules contains the entry procedure, a global procedure called *main*. Executing the program means, in actual fact, executing the main procedure. The program can include many modules, but only one of these will have a main procedure.

A module may, for example, define the interface with external equipment or contain geometrical data that is either generated from CAD systems or created on-line by digitizing (teach programming).

Whereas small installations are often contained in one module, larger installations may have a main module that references routines and/or data contained in one or several other modules.

2.2 System modules

System modules are used to define common, system-specific data and routines, such as tools. They are not included when a program is saved, meaning that any update made to a system module will affect all existing programs currently in, or loaded at a later stage into the program memory.

2.3 Module declarations

A module declaration specifies the name and attributes of that module. These attributes can only be added off-line, not using the teach pendant. The following are examples of the attributes of a module:

Attribute	If specified, the module:
SYSMODULE	is a system module, otherwise a program module
NOSTEPIN	cannot be entered during stepwise execution
VIEWONLY	cannot be modified
READONLY	cannot be modified, but the attribute can be removed
NOVIEW	cannot be viewed, only executed. Global routines can be reached from other modules and are always run as NOSTEPIN. The current values for global data can be reached from other modules or from the data window on the teach pendant. A module or a program containing a NOVIEW program module cannot be saved. Therefore, NOVIEW should primarily be used for system modules. NOVIEW can only be defined off-line from a PC.

e.g.
```
MODULE module_name (SYSMODULE, VIEWONLY)
   !data type definition
   !data declarations
   !routine declarations
ENDMODULE
```

A module may not have the same name as another module or a global routine or data.

2.4 Syntax

Module declaration

<module declaration> ::=
MODULE <module name> [<module attribute list>]
<type definition list>
<data declaration list>
<routine declaration list>
ENDMODULE

<module name> ::= <identifier>
<module attribute list> ::= ‘(‘ <module attribute> { ‘,’ <module attribute> } ‘)’
<module attribute> ::=
SYSMODULE
| **NOVIEW**
| **NOSTEPIN**
| **VIEWONLY**
| **READONLY**

(***Note.*** If two or more attributes are used they must be in the above order, the NOVIEW attribute can only be specified alone or together with the attribute SYSMODULE.)

<type definition list> ::= { <type definition> }
<data declaration list> ::= { <data declaration> }
<routine declaration list> ::= { <routine declaration> }

3 Routines

There are three types of routines (subprograms): *procedures*, *functions* and *traps*.

- Procedures do not return a value and are used in the context of instructions.
- Functions return a value of a specific type and are used in the context of expressions.
- Trap routines provide a means of dealing with interrupts. A trap routine can be associated with a specific interrupt and then, if that particular interrupt occurs at a later stage, will automatically be executed. A trap routine can never be explicitly called from the program.

3.1 Routine scope

The scope of a routine denotes the area in which the routine is visible. The optional local directive of a routine declaration classifies a routine as local (within the module), otherwise it is global.

Example:
```
LOCAL PROC local_routine (...
PROC global_routine (...
```

The following scope rules apply to routines (see the example in Figure 2):

- The scope of a global routine may include any module.
- The scope of a local routine comprises the module in which it is contained.
- Within its scope, a local routine hides any global routine or data with the same name.
- Within its scope, a routine hides instructions and predefined routines and data with the same name.



Figure 2 Example: The following routines can be called from Routine h:
Module1 - Routine c, d.
Module2 - All routines.

A routine may not have the same name as another routine or data in the same module. A global routine may not have the same name as a module or a global routine or global data in another module.

3.2 Parameters

The parameter list of a routine declaration specifies the arguments (actual parameters) that must/can be supplied when the routine is called.

There are four different types of parameters (in the access mode):

- Normally, a parameter is used only as an input and is treated as a routine variable. Changing this variable will not change the corresponding argument.
- An INOUT parameter specifies that a corresponding argument must be a variable (entire, element or component) or an entire persistent which can be changed by the routine.
- A VAR parameter specifies that a corresponding argument must be a variable (entire, element or component) which can be changed by the routine.
- A PERS parameter specifies that a corresponding argument must be an entire persistent which can be changed by the routine.

If an INOUT, VAR or PERS parameter is updated, this means, in actual fact, that the argument itself is updated, i.e. it makes it possible to use arguments to return values to the calling routine.

Example:
```
PROC routine1 (num in_par, INOUT num inout_par,
               VAR num var_par, PERS num pers_par)
```

A parameter can be optional and may be omitted from the argument list of a routine call. An optional parameter is denoted by a backslash “\” before the parameter.

Example:
```
PROC routine2 (num required_par \num optional_par)
```

The value of an optional parameter that is omitted in a routine call may not be referenced. This means that routine calls must be checked for optional parameters before an optional parameter is used.

Two or more optional parameters may be mutually exclusive (i.e. declared to exclude each other), which means that only one of them may be present in a routine call. This is indicated by a stroke “|” between the parameters in question.

Example:
```
PROC routine3 (\num exclude1 | num exclude2)
```

The special type, *switch*, may (only) be assigned to optional parameters and provides a means to use switch arguments, i.e. arguments that are only specified by names (not values). A value cannot be transferred to a switch parameter. The only way to use a switch parameter is to check for its presence using the predefined function, *Present*.

Example:
```
PROC routine4 (\switch on | switch off)
    ...
    IF Present (off ) THEN
    ...
ENDPROC
```

Arrays may be passed as arguments. The degree of an array argument must comply with the degree of the corresponding formal parameter. The dimension of an array parameter is “conformant” (marked with “*”). The actual dimension thus depends on the dimension of the corresponding argument in a routine call. A routine can determine the actual dimension of a parameter using the predefined function, *Dim*.

Example: PROC routine5 (VAR num pallet{*,*})

3.3 Routine termination

The execution of a procedure is either explicitly terminated by a RETURN instruction or implicitly terminated when the end (ENDPROC, BACKWARD or ERROR) of the procedure is reached.

The evaluation of a function must be terminated by a RETURN instruction.

The execution of a trap routine is explicitly terminated using the RETURN instruction or implicitly terminated when the end (ENDTRAP or ERROR) of that trap routine is reached. Execution continues from the point where the interrupt occurred.

3.4 Routine declarations

A routine can contain routine declarations (including parameters), data, a body, a backward handler (only procedures) and an error handler (see Figure 3). Routine declarations cannot be nested, i.e. it is not possible to declare a routine within a routine.



Figure 3 A routine can contain declarations, data, a body, a backward handler and an error handler.

Procedure declaration

Example: Multiply all elements in a num array by a factor;

```
PROC arrmul( VAR num array{*}, num factor)
   FOR index FROM 1 TO dim( array, 1 ) DO
      array{index} := array{index} * factor;
   ENDFOR
ENDPROC
```

Function declaration

A function can return any data type value, but not an array value.

Example: Return the length of a vector;

```
FUNC num veclen (pos vector)
 RETURN Sqrt(Pow(vector.x,2)+Pow(vector.y,2)+Pow(vector.z,2));
ENDFUNC
```

Trap declaration

Example: Respond to feeder empty interrupt;

```
TRAP feeder_empty
  wait_feeder;
  RETURN;
ENDTRAP
```

3.5 Procedure call

When a procedure is called, the arguments that correspond to the parameters of the procedure shall be used:

- Mandatory parameters must be specified. They must also be specified in the correct order.
- Optional arguments can be omitted.
- Conditional arguments can be used to transfer parameters from one routine call to another.

See the Chapter *Using function calls in expressions* on page 32 for more details.

The procedure name may either be statically specified by using an identifier (*early binding*) or evaluated during runtime from a string type expression (*late binding*). Even though early binding should be considered to be the “normal” procedure call form, late binding sometimes provides very efficient and compact code. Late binding is defined by putting percent signs before and after the string that denotes the name of the procedure.

Example:

```
! early binding
TEST products_id
CASE 1:
    proc1 x, y, z;
CASE 2:
    proc2 x, y, z;
CASE 3:
    ...
! same example using late binding
% “proc” + NumToStr(product_id, 0) % x, y, z;
    ...
! same example again using another variant of late binding
VAR string procname {3} :=[“proc1”, “proc2”, “proc3”];
    ...
% procname{product_id} % x, y, z;
    ...
```

Note that the late binding is available for procedure calls only, and not for function calls. If a reference is made to an unknown procedure using late binding, the system variable ERRNO is set to ERR_REFUNKPRC. If a reference is made to a procedure call error (syntax, not procedure) using late binding, the system variable ERRNO is set to ERR_CALLPROC.

3.6 Syntax

Routine declaration

```
<routine declaration> ::=
        [LOCAL] ( <procedure declaration>
                | <function declaration>
                | <trap declaration> )
        | <comment>
        | <RDN>
```

Parameters

```
<parameter list> ::=
        <first parameter declaration> { <next parameter declaration> }
<first parameter declaration> ::=
        <parameter declaration>
        | <optional parameter declaration>
        | <PAR>
<next parameter declaration> ::=
        ’,’ <parameter declaration>
         | <optional parameter declaration>
         | ’,’<optional parameter declaration>
         | ’,’ <PAR>
```

<optional parameter declaration> ::=
’\’ (<parameter declaration> | **<ALT>**)
{ ’|’ (<parameter declaration> | **<ALT>**) }
<parameter declaration> ::=
[**VAR** | **PERS** | **INOUT**] <data type>
<identifier> [’{’ (’*’ { ’,’ ’*’ }) | **<DIM>**] ’}’
| ’switch’ <identifier>

Procedure declaration

<procedure declaration> ::=
PROC <procedure name>
’(’ [<parameter list>] ’)’
<data declaration list>
<instruction list>
[**BACKWARD** <instruction list>]
[**ERROR** <instruction list>]
ENDPROC
<procedure name> ::= <identifier>
<data declaration list> ::= {<data declaration>}

Function declaration

<function declaration> ::=
FUNC <value data type>
<function name>
’(’ [<parameter list>] ’)’
<data declaration list>
<instruction list>
[**ERROR** <instruction list>]
ENDFUNC
<function name> ::= <identifier>

Trap routine declaration

<trap declaration> ::=
TRAP <trap name>
<data declaration list>
<instruction list>
[**ERROR** <instruction list>]
ENDTRAP
<trap name> ::= <identifier>

Procedure call

<procedure call> ::= <procedure> [<procedure argument list>] ';'
<procedure> ::=
<identifier>
| '%' <expression> '%'
<procedure argument list> ::= <first procedure argument> { <procedure argument> }
<first procedure argument> ::=
<required procedure argument>
| <optional procedure argument>
| <conditional procedure argument>
| **<ARG>**
<procedure argument> ::=
',' <required procedure argument>
| <optional procedure argument>
| ',' <optional procedure argument>
| <conditional procedure argument>
| ',' <conditional procedure argument>
| ',' **<ARG>**
<required procedure argument> ::= [<identifier> ':='] <expression>
<optional procedure argument> ::= '\' <identifier> [':=' <expression>]
<conditional procedure argument> ::= '\' <identifier> '?' (<parameter> | **<VAR>**)

4 Data Types

There are two different kinds of data types:

- An *atomic* type is atomic in the sense that it is not defined based on any other type and cannot be divided into parts or components, e.g. *num*.
- A *record* data type is a composite type with named, ordered components, e.g. *pos*. A component may be of an atomic or record type.

A record value can be expressed using an *aggregate* representation;

e.g. [300, 500, depth] pos record aggregate value.

A specific component of a record data can be accessed by using the name of that component;

e.g. pos1.x := 300; assignment of the x-component of pos1.

4.1 Non-value data types

Each available data type is either a *value* data type or a *non-value* data type. Simply speaking, a value data type represents some form of “value”. Non-value data cannot be used in value-oriented operations:

- Initialisation
- Assignment (:=)
- Equal to (=) and not equal to (<>) checks
- TEST instructions
- IN (access mode) parameters in routine calls
- Function (return) data types

The input data types (*signalai*, *signaldi*, *signalgi*) are of the data type *semi value*. These data can be used in value-oriented operations, except initialisation and assignment.

In the description of a data type it is only specified when it is a semi value or a non-value data type.

4.2 Equal (alias) data types

An *alias* data type is defined as being equal to another type. Data with the same data types can be substituted for one another.

Example:
```
VAR dionum high:=1;
VAR num level;          This is OK since dionum is an alias
level:= high;           data type for num
```

4.3 Syntax

<type definition>::=
[**LOCAL**] (<record definition>
| <alias definition>)
| <comment>
| <TDN>

<record definition>::=
RECORD <identifier>
<record component list> ';'
ENDRECORD

<record component list> ::=
<record component definition> |
<record component definition> <record component list>

<record component definition> ::=
<data type> <record component name>

<alias definition> ::=
ALIAS <data type> <identifier> ';'

<data type> ::= <identifier>

5 Data

There are three kinds of data: *variables*, *persistents* and *constants*.

- A variable can be assigned a new value during program execution.
- A persistent can be described as a “persistent” variable. This is accomplished by letting an update of the value of a persistent automatically cause the initialisation value of the persistent declaration to be updated. (When a program is saved the initialisation value of any persistent declaration reflects the current value of the persistent.)
- A constant represents a static value and cannot be assigned a new value.

A data declaration introduces data by associating a name (identifier) with a data type. Except for predefined data and loop variables, all data used must be declared.

5.1 Data scope

The scope of data denotes the area in which the data is visible. The optional local directive of a data declaration classifies data as local (within the module), otherwise it is global. Note that the local directive may only be used at module level, not inside a routine.

Example:
```
LOCAL VAR num local_variable;
VAR num global_variable;
```

Data declared outside a routine is called *program data*. The following scope rules apply to program data:

- The scope of predefined or global program data may include any module.
- The scope of local program data comprises the module in which it is contained.
- Within its scope, local program data hides any global data or routine with the same name (including instructions and predefined routines and data).

Program data may not have the same name as other data or a routine in the same module. Global program data may not have the same name as other global data or a routine in another module. A persistent may not have the same name as another persistent in the same program.

Data declared inside a routine is called *routine data*. Note that the parameters of a routine are also handled as routine data. The following scope rules apply to routine data:

- The scope of routine data comprises the routine in which it is contained.
- Within its scope, routine data hides any other routine or data with the same name.

See the example in Figure 4.



Figure 4 Example: The following data can be called from routine e:
Module1: Data c, d.
Module2: Data a, f, g, e1.
The following data can be called from routine h:
Module1: Data d.
Module2: Data a, f, g, h1, c.

Routine data may not have the same name as other data or a label in the same routine.

5.2 Variable declaration

A variable is introduced by a variable declaration.

Example: VAR num x;

Variables of any type can be given an array (of degree 1, 2 or 3) format by adding dimensional information to the declaration. A dimension is an integer value greater than 0.

Example: VAR pos pallet{14, 18};

Variables with value types may be initialised (given an initial value). The expression used to initialise a program variable must be constant. Note that the value of an uninitialized variable may be used, but it is undefined, i.e. set to zero.

Example: VAR string author_name := “John Smith”;
VAR pos start := [100, 100, 50];
VAR num maxno{10} := [1, 2, 3, 9, 8, 7, 6, 5, 4, 3];

The initialisation value is set when:

- the program is opened,
- the program is executed from the beginning of the program.

5.3 Persistent declaration

Persistents can only be declared at module level, not inside a routine, and **must always be given an initial value**. The initialisation value must be a single value (without data or operands), or a single aggregate with members which, in turn, are single values or single aggregates.

Example: PERS pos refpnt := [100.23, 778.55, 1183.98];

Persistents of any type can be given an array (of degree 1, 2 or 3) format by adding dimensional information to the declaration. A dimension is an integer value greater than 0.

Example: PERS pos pallet{14, 18} := [...];

Note that if the value of a persistent is updated, this automatically causes the initialisation value of the persistent declaration to be updated.

Example:
```
PERS num reg1 := 0;
  ...
reg1 := 5;
```
After execution, the program looks like this:
```
PERS num reg1 := 5;
  ...
reg1 := 5;
```

It is possible to declare two persistents with the same name in different modules, if they are local within the module (PERS LOCAL), without any error being generated by the system (different data scope). But **note the limitation** that these two persistents always have the same current value (use the same storage in the memory).

5.4 Constant declaration

A constant is introduced by a constant declaration. The value of a constant cannot be modified.

Example: CONST num pi := 3.141592654;

A constant of any type can be given an array (of degree 1, 2 or 3) format by adding dimensional information to the declaration. A dimension is an integer value greater than 0.

Example:
```
CONST pos seq{3} := [[614, 778, 1020],
                    [914, 998, 1021],
                    [814, 998, 1022]];
```

5.5 Initiating data

The initialisation value for a constant or variable can be a constant expression. **The initialisation value for a persistent can only be a literal expression**.

Example:
```
CONST num a := 2;
CONST num b := 3;
! Correct syntax
CONST num ab := a + b;
VAR num a_b := a + b;
PERS num a__b := 5;
! Faulty syntax
PERS num a__b := a + b;
```

In the table below, you can see what is happening in various activities such as warm start, new program, program start etc.

Table1

System event Affects	Power on (Warm start)	Open, Close or New program	Start program (Move PP to main)	Start program (Move PP to Routine)	Start program (Move PP to cursor)	Start program (Call Routine)	Start program (After cycle)	Start program (After stop)
Constant	Unchanged	Init	Init	Init	Unchanged	Unchanged	Unchanged	Unchanged
Variable	Unchanged	Init	Init	Init	Unchanged	Unchanged	Unchanged	Unchanged
Persistent	Unchanged	Init	Init	Init	Unchanged	Unchanged	Unchanged	Unchanged
Commanded interrupts	Re-ordered	Disappears	Disappears	Disappears	Unchanged	Unchanged	Unchanged	Unchanged
Start up routine SYS_RESET (with motion settings)	Not run	Run*	Run	Not run	Not run	Not run	Not run	Not run
Files	Closes	Closes	Closes	Closes	Unchanged	Unchanged	Unchanged	Unchanged
Path	Recreated at power on	Disappears	Disappears	Disappears	Disappears	Unchanged	Unchanged	Unchanged

* Generates an error when there is a semantic error in the actual task program.

5.6 Storage Class

The *storage class* of a data object determines when the system allocates and de-allocates memory for the data object. The storage class of a data object is determined by the kind of data object and the context of its declaration and can be either *static* or *volatile*.

Constants, persistents, and module variables are static, i.e. they have the same storage during the lifetime of a task. This means that any value assigned to an persistent or a module variable, always remains unchanged until the next assignment.

Routine variables are volatile. The memory needed to store the value of a volatile variable is allocated first upon the call of the routine in which the declaration of the variable is contained. The memory is later de-allocated at the point of the return to the caller of the routine. This means that the value of a routine variable is always undefined before the call of the routine and is always lost (becomes undefined) at the end of the execution of the routine.

In a chain of recursive routine calls (a routine calling itself directly or indirectly) each instance of the routine receives its own memory location for the “same” routine variable - a number of *instances* of the same variable are created.

5.7 Syntax

Data declaration

<data declaration> ::=
[**LOCAL**] (<variable declaration>
| <persistent declaration>
| <constant declaration>)
| <comment**>**
| **<DDN>**

Variable declaration

<variable declaration> ::=
VAR <data type> <variable definition> ’;’
<variable definition> ::=
<identifier> [’{’ <dim> { ’,’ <dim> } ’}’]
[’:=’ <constant expression>]
<dim> ::= <constant expression>

Persistent declaration

<persistent declaration> ::=
PERS <data type> <persistent definition> ’;’
<persistent definition> ::=
<identifier> [’{’ <dim> { ’,’ <dim> } ’}’]
’:=’ <literal expression>

Constant declaration

<constant declaration> ::=
CONST <data type> <constant definition> ’;’
<constant definition> ::=
<identifier> [’{’ <dim> { ’,’ <dim> } ’}’]
’:=’ <constant expression>
<dim> ::= <constant expression>

6 Instructions

Instructions are executed in succession unless a program flow instruction or an interrupt or error causes the execution to continue at some other place.

Most instructions are terminated by a semicolon ";". A label is terminated by a colon ":". Some instructions may contain other instructions and are terminated by specific keywords:

Instruction	Termination word
IF	ENDIF
FOR	ENDFOR
WHILE	ENDWHILE
TEST	ENDTEST

Example:
```
WHILE index < 100 DO
    .
    index := index + 1;
ENDWHILE
```

6.1 Syntax

<instruction list> ::= { <instruction> }
<instruction> ::=
[<instruction according to separate chapter in this manual>
| **<SMT>**

7 Expressions

An expression specifies the evaluation of a value. It can be used, for example:

- in an assignment instruction e.g. a:=*3*b/c*;
- as a condition in an IF instruction e.g. IF *a>=3* THEN ...
- as an argument in an instruction e.g. WaitTime *time*;
- as an argument in a function call e.g. a:=Abs*(3*b*);

7.1 Arithmetic expressions

An arithmetic expression is used to evaluate a numeric value.

Example: 2*pi*radius

Table 2 shows the different types of operations possible.

Table 2

Operator	Operation	Operand types	Result type
+	addition	num + num	num[3)]
+	unary plus; keep sign	+num or +pos	same[1)3)]
+	vector addition	pos + pos	pos
-	subtraction	num - num	num[3)]
-	unary minus; change sign	-num or -pos	same[1)3)]
-	vector subtraction	pos - pos	pos
*	multiplication	num * num	num[3)]
*	scalar vector multiplication	num * pos or pos * num	pos
*	vector product	pos * pos	pos
*	linking of rotations	orient * orient	orient
/	division	num / num	num
DIV [2)]	integer division	num DIV num	num
MOD [2)]	integer modulo; remainder	num MOD num	num

1. The result receives the same type as the operand. If the operand has an alias data type, the result receives the alias "base" type (num or pos).

2. Integer operations, e.g. 14 DIV 4=3, 14 MOD 4=2.
(Non-integer operands are illegal.)

3. Preserves integer (exact) representation as long as operands and result are kept within the integer subdomain of the num type.

7.2 Logical expressions

A logical expression is used to evaluate a logical value (TRUE/FALSE).

Example: a>5 AND b=3

Table 3 shows the different types of operations possible.

Table 3

Operator	Operation	Operand types	Result type
<	less than	num < num	bool
<=	less than or equal to	num <= num	bool
=	equal to	any 1)= any 1)	bool
>=	greater than or equal to	num >= num	bool
>	greater than	num > num	bool
<>	not equal to	any 1) <> any 1)	bool
AND	and	bool AND bool	bool
XOR	exclusive or	bool XOR bool	bool
OR	or	bool OR bool	bool
NOT	unary not; negation	NOT bool	bool

1) Only value data types. Operands must have equal types.

a AND b

b \ a	True	False
True	True	False
False	False	False

a XOR b

b \ a	True	False
True	False	True
False	True	False

a OR b

b \ a	True	False
True	True	True
False	True	False

NOT b

b	
True	False
False	True

7.3 String expressions

A string expression is used to carry out operations on strings.

Example: “IN” + “PUT” gives the result “INPUT”

Table 4 shows the one operation possible.

Table 4

Operator	Operation	Operand types	Result type
+	string concatenation	string + string	string

7.4 Using data in expressions

An entire variable, persistent or constant can be a part of an expression.

Example: 2*pi*radius

Arrays

A variable, persistent or constant declared as an array can be referenced to the whole array or a single element.

An array element is referenced using the index number of the element. The index is an integer value greater than 0 and may not violate the declared dimension. Index value 1 selects the first element. The number of elements in the index list must fit the declared degree (1, 2 or 3) of the array.

Example:
```
VAR num row{3};
VAR num column{3};
VAR num value;
   .
value := column{3};      only one element in the array
row := column;           all elements in the array
```

Records

A variable, persistent or constant declared as a record can be referenced to the whole record or a single component.

A record component is referenced using the component name.

Example:
```
VAR pos home;
VAR pos pos1;
VAR num yvalue;
..
yvalue := home.y;        the Y component only
pos1 := home;            the whole position
```

7.5 Using aggregates in expressions

An aggregate is used for record or array values.

Example:	pos := [x, y, 2*x];	pos record aggregate
	posarr := [[0, 0, 100], [0,0,z]];	pos array aggregate

It must be possible to determine the data type of an aggregate the context. The data type of each aggregate member must be equal to the type of the corresponding member of the determined type.

Example	**VAR** pos pl;	
	p1 :=[1, -100, 12];	aggregate type pos - determined by p1
	IF [1, -100, 12] = [a,b,b,] THEN	illegal since the data type of neither of the aggregates can be determined by the context.

7.6 Using function calls in expressions

A function call initiates the evaluation of a specific function and receives the value returned by the function.

Example: Sin(angle)

The arguments of a function call are used to transfer data to (and possibly from) the called function. The data type of an argument must be equal to the type of the corresponding parameter of the function. Optional arguments may be omitted but the order of the (present) arguments must be the same as the order of the formal parameters. In addition, two or more optional arguments may be declared to exclude each other, in which case, only one of them may be present in the argument list.

A required (compulsory) argument is separated from the preceding argument by a comma “,”. The formal parameter name may be included or omitted.

Example:	Polar(3.937, 0.785398)	two required arguments
	Polar(Dist:=3.937, Angle:=0.785398)...	using names

An optional argument must be preceded by a backslash “\” and the formal parameter name. A switch type argument is somewhat special; it may not include any argument expression. Instead, such an argument can only be either "present" or "not present".

Example:	Cosine(45)	one required argument
	Cosine(0.785398\Rad)	... and one switch
	Dist(p2)	one required argument
	Dist(\distance:=pos1, p2)	... and one optional

Conditional arguments are used to support smooth propagation of optional arguments through chains of routine calls. A conditional argument is considered to be “present” if the specified optional parameter (of the calling function) is present, otherwise it is simply considered to be omitted. Note that the specified parameter must be optional.

Example:
```
PROC Read_from_file (iodev File \num Maxtime)
    ..
    character:=ReadBin (File \Time?Maxtime);
! Max. time is only used if specified when calling the routine
! Read_from_file
    ..
ENDPROC
```

The parameter list of a function assigns an *access mode* to each parameter. The access mode can be either *in*, *inout*, *var* or *pers:*

- An IN parameter (default) allows the argument to be any expression. The called function views the parameter as a constant.
- An INOUT parameter requires the corresponding argument to be a variable (entire, array element or record component) or an entire persistent. The called function gains full (read/write) access to the argument.
- A VAR parameter requires the corresponding argument to be a variable (entire, array element or record component). The called function gains full (read/write) access to the argument.
- A PERS parameter requires the corresponding argument to be an entire persistent. The called function gains full (read/update) access to the argument.

7.7 Priority between operators

The relative priority of the operators determines the order in which they are evaluated. Parentheses provide a means to override operator priority. The rules below imply the following operator priority:

* / DIV MOD	- highest
+ -	
< > <> <= >= =	
AND	
XOR OR NOT	- lowest

An operator with high priority is evaluated prior to an operator with low priority. Operators of the same priority are evaluated from left to right.

Example

Expression	Evaluation order	Comment
a + b + c	(a + b) + c	left to right rule
a + b * c	a + (b * c)	* higher than +
a OR b OR c	(a OR b) OR c	Left to right rule
a AND b OR c AND d	(a AND b) OR (c AND d)	AND higher than OR
a < b AND c < d	(a < b) AND (c < d)	< higher than AND

7.8 Syntax

Expressions

<expression> ::=
<expr>
| **<EXP>**
<expr> ::= [**NOT**] <logical term> { (**OR** | **XOR**) <logical term> }
<logical term> ::= <relation> { **AND** <relation> }
<relation> ::= <simple expr> [<relop> <simple expr>]
<simple expr> ::= [<addop>] <term> { <addop> <term> }
<term> ::= <primary> { <mulop> <primary> }
<primary> ::=
<literal>
| <variable>
| <persistent>
| <constant>
| <parameter>
| <function call>
| <aggregate>
| ’(’ <expr> ’)’

Operators

<relop> ::= ’<’ | ’<=’ | ’=’ | ’>’ | ’>=’ | ’<>’
<addop> ::= ’+’ | ’-’
<mulop> ::= ’*’ | ’/’ | **DIV** | **MOD**

Constant values

<literal> ::= <num literal>
| <string literal>
| <bool literal>

Data

<variable> ::=
<entire variable>
| <variable element>
| <variable component>
<entire variable> ::= <ident>
<variable element> ::= <entire variable> ’{’ <index list> ’}’
<index list> ::= <expr> { ’,’ <expr> }
<variable component> ::= <variable> ’.’ <component name>
<component name> ::= <ident>
<persistent> ::=
<entire persistent>
| <persistent element>
| <persistent component>
<constant> ::=
<entire constant>
| <constant element>
| <constant component>

Aggregates

<aggregate> ::= ’[’ <expr> { ’,’ <expr> } ’]’

Function calls

<function call> ::= <function> ’(’ [<function argument list>] ’)’
<function> ::= <ident>
<function argument list> ::= <first function argument> { <function argument> }
<first function argument> ::=
<required function argument>
| <optional function argument>
| <conditional function argument>
<function argument> ::=
’,’ <required function argument>
| <optional function argument>
| ’,’ <optional function argument>
| <conditional function argument>
| ’,’ <conditional function argument>
<required function argument> ::= [<ident> ’:=’] <expr>
<optional function argument> ::= ’\’ <ident> [’:=’ <expr>]
<conditional function argument> ::= ’\’ <ident> ’?’ <parameter>

Special expressions

<constant expression> ::= <expression>
<literal expression> ::= <expression>

<conditional expression> ::= <expression>

Parameters

<parameter> ::=
<entire parameter>
| <parameter element>
| <parameter component>

8 Error Recovery

An execution error is an abnormal situation, related to the execution of a specific piece of a program. An error makes further execution impossible (or at least hazardous). “Overflow” and “division by zero” are examples of errors. Errors are identified by their unique *error number* and are always recognized by the robot. The occurrence of an error causes suspension of the normal program execution and the control is passed to an *error handler*. The concept of error handlers makes it possible to respond to and, possibly, recover from errors that arise during program execution. If further execution is not possible, the error handler can at least assure that the program is given a graceful abortion.

8.1 Error handlers

Any routine may include an error handler. The error handler is really a part of the routine, and the scope of any routine data also comprises the error handler of the routine. If an error occurs during the execution of the routine, control is transferred to its error handler.

Example:
```
FUNC num safediv( num x, num y)
    RETURN x / y;
ERROR
    IF ERRNO = ERR_DIVZERO THEN
        TPWrite "The number cannot be equal to 0";
        RETURN x;
    ENDIF
ENDFUNC
```

The system variable *ERRNO* contains the error number of the (most recent) error and can be used by the error handler to identify that error. After any necessary actions have been taken, the error handler can:

- Resume execution, starting with the instruction in which the error occurred. This is done using the RETRY instruction. If this instruction causes the same error again, up to four error recoveries will take place; after that execution will stop.
- Resume execution, starting with the instruction following the instruction in which the error occurred. This is done using the TRYNEXT instruction.
- Return control to the caller of the routine using the RETURN instruction. If the routine is a function, the RETURN instruction must specify an appropriate return value.
- Propagate the error to the caller of the routine using the RAISE instruction.

When an error occurs in a routine that does not contain an error handler or when the end of the error handler is reached (**ENDFUNC, ENDPROC** or **ENDTRAP**), the *system error handler* is called. The system error handler just reports the error and stops the execution.

In a chain of routine calls, each routine may have its own error handler. If an error occurs in a routine with an error handler, and the error is explicitly propagated using the RAISE instruction, the same error is raised again at the point of the call of the routine - the error is *propagated*. When the top of the call chain (the entry routine of the task) is reached without any error handler being found or when the end of any error handler is reached within the call chain, the system error handler is called. The system error handler just reports the error and stops the execution. Since a trap routine can only be called by the system (as a response to an interrupt), any propagation of an error from a trap routine is made to the system error handler.

Error recovery is not available for instructions in the backward handler. Such errors are always propagated to the system error handler.

In addition to errors detected and raised by the robot, a program can explicitly raise errors using the RAISE instruction. This facility can be used to recover from complex situations. It can, for example, be used to escape from deeply-nested code positions. Error numbers 1-90 may be used in the raise instruction. Explicitly-raised errors are treated exactly like errors raised by the system.

Note that it is not possible to recover from or respond to errors that occur within an error clause. Such errors are always propagated to the system error handler.

9 Interrupts

Interrupts are program-defined events, identified by *interrupt numbers*. An interrupt occurs when an *interrupt condition* is true. Unlike errors, the occurrence of an interrupt is not directly related to (synchronous with) a specific code position. The occurrence of an interrupt causes suspension of the normal program execution and control is passed to a *trap routine*.

Even though the robot immediately recognizes the occurrence of an interrupt (only delayed by the speed of the hardware), its response – calling the corresponding trap routine – can only take place at specific program positions, namely:

- when the next instruction is entered,
- any time during the execution of a waiting instruction, e.g. *WaitUntil*,
- any time during the execution of a movement instruction, e.g. *MoveL*.

This normally results in a delay of 5-120 ms between interrupt recognition and response, depending on what type of movement is being performed at the time of the interrupt.

The raising of interrupts may be *disabled* and *enabled*. If interrupts are disabled, any interrupt that occurs is queued and not raised until interrupts are enabled again. Note that the interrupt queue may contain more than one waiting interrupt. Queued interrupts are raised in *FIFO* order. Interrupts are always disabled during the execution of a trap routine.

When running stepwise and when the program has been stopped, no interrupts will be handled. Interrupts that are generated under these circumstances are not dealt with.

The maximum number of defined interrupts at any one time is limited to 70 **per program task.** The **total limitation** set by the I/O CPU is 100 interrupts.

9.1 Interrupt manipulation

Defining an interrupt makes it known to the robot. The definition specifies the interrupt condition and activates and enables the interrupt.

```
Example:    VAR intnum sig1int;
              .
            ISignalDI di1, high, sig1int;
```

An activated interrupt may in turn be deactivated (and vice versa).
During the deactivation time, any generated interrupts of the specified type are thrown away without any trap execution.

```
Example:    ISleep sig1int;                 deactivate
              .
            IWatch sig1int;                 activate
```

An enabled interrupt may in turn be disabled (and vice versa).
During the disable time, any generated interrupts of the specified type are queued and raised first when the interrupts are enabled again.

```
Example:    IDisable sig1int;                    disable
              .
            IEnable sig1int;                     enable
```

Deleting an interrupt removes its definition. It is not necessary to explicitly remove an interrupt definition, but a new interrupt cannot be defined to an interrupt variable until the previous definition has been deleted.

```
Example:    IDelete sig1int;
```

9.2 Trap routines

Trap routines provide a means of dealing with interrupts. A trap routine can be connected to a particular interrupt using the CONNECT instruction. When an interrupt occurs, control is immediately transferred to the associated trap routine (if any). If an interrupt occurs, that does not have any connected trap routine, this is treated as a fatal error, i.e. causes immediate termination of program execution.

```
Example:    VAR intnum empty;
            VAR intnum full;

              .PROC main()

              CONNECT empty WITH etrap;      connect trap routines
              CONNECT full WITH ftrap;
              ISignalDI di1, high, empty;    define feeder interrupts
              ISignalDI di3, high, full;
              .
              IDelete empty;
              IDelete full;
            ENDPROC

            TRAP etrap                       responds to “feeder
              open_valve;                    empty” interrupt
              RETURN;
            ENDTRAP

            TRAP ftrap                       responds to “feeder full”
              close_valve;                   interrupt
              RETURN;
            ENDTRAP
```

Several interrupts may be connected to the same trap routine. The system variable *INTNO* contains the interrupt number and can be used by a trap routine to identify an interrupt. After the necessary action has been taken, a trap routine can be terminated using the RETURN instruction or when the end (ENDTRAP or ERROR) of the trap routine is reached. Execution continues from the place where the interrupt occurred.

10 Backward execution

A program can be executed backwards one instruction at a time. The following general restrictions are valid for backward execution:

- The instructions IF, FOR, WHILE and TEST cannot be executed backwards.
- It is not possible to step backwards out of a routine when reaching the beginning of the routine.

10.1 Backward handlers

Procedures may contain a backward handler that defines the backward execution of a procedure call.

The backward handler is really a part of the procedure and the scope of any routine data also comprises the backward handler of the procedure.

Example:

```
PROC MoveTo ()
   MoveL p1,v500,z10,tool1;
   MoveC p2,p3,v500,z10,tool1;
   MoveL p4,v500,z10,tool1;
BACKWARD
   MoveL p4,v500,z10,tool1;
   MoveC p2,p3,v500,z10,tool1;
   MoveL p1,v500,z10,tool1;
ENDPROC
```

When the procedure is called during forward execution, the following occurs:



When the procedure is called during backwards execution, the following occurs:




```
PROC MoveTo ()
    MoveL p1,v500,z10,tool1;
    MoveC p2,p3,v500,z10,tool1;
    MoveL p4,v500,z10,tool1;
BACKWARD
    MoveL p4,v500,z10,tool1;
    MoveC p2,p3,v500,z10,tool1;
    MoveL p1,v500,z10,tool1;
ENDPROC
```

Instructions in the backward or error handler of a routine may not be executed backwards. Backward execution cannot be nested, i.e. two instructions in a call chain may not simultaneously be executed backwards.

A procedure with no backward handler cannot be executed backwards. A procedure with an empty backward handler is executed as “no operation”.

10.2 Limitation of move instructions in the backward handler

The move instruction type and sequence in the backward handler must be a mirror of the move instruction type and sequence for forward execution in the same routine:

```
PROC MoveTo ()
  MoveL p1,v500,z10,tool1;
  MoveC p2,p3,v500,z10,tool1;
  MoveL p4,v500,z10,tool1;
BACKWARD
  MoveL p4,v500,z10,tool1;
  MoveC p2,p3,v500,z10,tool1;
  MoveL p1,v500,z10,tool1;
ENDPROC
```



Note that the order of CirPoint *p2* and ToPoint *p3* in the MoveC should be the same.

By move instructions is meant all instructions that result in some movement of the robot or external axes such as MoveL, SearchC, TriggJ, ArcC, PaintL ...

Any departures from this programming limitation in the backward handler can result in faulty backward movement. Linear movement can result in circular movement and vice versa, for some part of the backward path.

11 Multitasking

The events in a robot cell are often in parallel, so why are the programs not in parallel?

Multitasking RAPID is a way to execute programs in (pseudo) parallel with the normal execution. The execution is started at power on and will continue for ever, unless an error occurs in that program. One parallel program can be placed in the background or foreground of another program. It can also be on the same level as another program.

To use this function the robot must be configured with one extra TASK for each background program**.**

Up to 10 different tasks can be run in pseudo parallel. Each task consists of a set of modules, in the same way as the normal program. All the modules are local in each task.

Variables and constants are local in each task, but persistents are not. A persistent with the same name and type is reachable in all tasks. If two persistents have the same name, but their type or size (array dimension) differ, a runtime error will occur.

A task has its own trap handling and the event routines are triggered only on its own task system states (e.g. Start/Stop/Restart....).

There are a few restrictions on the use of Multitasking RAPID.

- Do not mix up parallel programs with a PLC. The response time is the same as the interrupt response time for one task. This is true, of course, when the task is not in the background of another busy program
- There is only one physical Teach Pendent, so be careful that a TPWrite request is not mixed in the Operator Window for all tasks.
- When running a Wait instruction in manual mode, a simulation box will come up after 3 seconds. This will only occur in the main task.
- Move instructions can only be executed in the main task (the task bind to program instance 0, see User’s guide - System parameters).
- The execution of a task will halt during the time that some other tasks are accessing the file system, that is if the operator chooses to save or open a program, or if the program in a task uses the load/erase/read/write instructions.
- The Teach Pendent cannot access other tasks than the main task. So, the development of RAPID programs for other tasks can only be done if the code is loaded into the main task, or off-line.

For all settings, see User’s Guide - System parameters.

11.1 Synchronising the tasks

In many applications a parallel task only supervises some cell unit, quite independently of the other tasks being executed. In such cases, no synchronisation mechanism is necessary. But there are other applications which need to know what the main task is doing, for example.

Synchronising using polling

This is the easiest way to do it, but the performance will be the slowest.

Persistents are then used together with the instructions *WaitUntil*, *IF*, *WHILE* or *GOTO*.

If the instruction *WaitUntil* is used, it will poll internally every 100 ms. Do not poll more frequently in other implementations.

Example

TASK 0

```
MODULE module1
PERS bool startsync:=FALSE;
PROC main()

   startsync:= TRUE;
   .

ENDPROC
ENDMODULE
```

TASK 1

```
MODULE module2
PERS bool startsync:=FALSE;
PROC main()

   WaitUntil startsync;
   .

ENDPROC
ENDMODULE
```

Synchronising using an interrupt

The instruction *SetDO* and *ISignalDO* are used.

Example

TASK 0

```
MODULE module1
PROC main()

SetDO do1,1;
    .

ENDPROC
ENDMODULE
```

TASK 1

```
MODULE module2
VAR intnum isiint1;
PROC main()

CONNECT isiint1 WITH isi_trap;
ISignalDO do1, 1, isiint1;

WHILE TRUE DO
    WaitTime 200;
ENDWHILE

IDelete isiint1;

ENDPROC

TRAP isi_trap

    .

ENDTRAP
ENDMODULE
```

11.2 Intertask communication

All types of data can be sent between two (or more) tasks with persistent variables.

A persistent variable is global in all tasks. The persistent variable must be of the same type and size (array dimension) in all tasks that declared it. Otherwise a runtime error will occur.

All declarations must specify an init value to the persistent variable, but only the first module loaded with the declaration will use it.

Example

TASK 0

```
MODULE module1
PERS bool startsync:=FALSE;
PERS string stringtosend:=””;
PROC main()

    stringtosend:=”this is a test”;

    startsync:= TRUE

ENDPROC
ENDMODULE
```

TASK 1

```
MODULE module2
PERS bool startsync:=FALSE;
PERS string stringtosend:=””;
PROC main()

    WaitUntil startsync;

    !read string
    IF stringtosend = “this is a test” THEN

ENDPROC
ENDMODULE
```

11.3 Type of task

Each extra task (not 0) is started in the system start sequence. If the task is of type **STATIC**, it will be restarted at the current position (where PP was when the system was powered off), but if the type is set to **SEMISTATIC**, it will be restarted from the beginning each time the power is turned on. A **SEMISTATIC** task will also int the restart sequence reload modules specified in the system parameters if the module file is newer that the loaded module.

It is also possible to set the task to type **NORMAL**, then it will behave in the same was as task 0 (the main task, controlling the robot movement). The teach pendent can only be used to start task 0, so the only way to start other **NORMAL** tasks is to use CommunicationWare.

11.4 Priorities

The way to run the tasks as default is to run all tasks at the same level in a round robbin way (one basic step on each instance). But it is possible to change the priority of one task by putting the task in the background of another. Then the background will only execute when the foreground is waiting for some events, or has stopped the execution (idle). A robot program with move instructions will be in an idle state most of the time.

The example below describes some situations where the system has 10 tasks (see Figure 5)

Round robbin chain 1: tasks 0, 1, and 8 are busy

Round robbin chain 2: tasks 0, 3, 4, 5 and 8 are busy
tasks 1 and 2 are idle

Round robbin chain 3: tasks 2, 4 and 5 are busy
tasks 0, 1, 8 and 9 are idle.

Round robbin chain 4: tasks 6 and 7 are busy
tasks 0, 1, 2, 3, 4, 5, 8 and 9 are idle



Figure 5 The tasks can have different priorities.

11.5 Trust Level

TrustLevel handel the system behavior when a **SEMISTATIC** or **STATIC** task is stopped for some reason or not executable.

SysFail - This is the default behaivour, all other **NORMAL** tasks (eg the MAIN task) will also stop, and the system is set to state SYS_FAIL. All jogg and program start orders will be rejected. Only a new warm start reset the system. This should be used when the task has some security supervisions.

SysHalt - All **NORMAL** tasks will be stopped (normaly only the main task). The system is forced to “motors off”. When taking up the system to “motors on” it is possible to jogg the robot, but a new attempt to start the program will be rejected. A new warm start will reset the system.

SysStop - All **NORMAL** tasks will be stopped (eg the main task), but it is restartable. Jogging is also possible.

NoSafety - Only the actual task itself will stop.

See System parameters - Controller/Task

11.6 Task sizes

The system will supply a memory area with an installation depending on size. That area is shared by all tasks.

The value of a persistent variable will be stored in a separate part of the system, and not affect the memory area above. See System parameters - *AveragePers*.

11.7 Something to think about

When you specify task priorities, you must think about the following:

- Always use the interrupt mechanism or loops with delays in supervision tasks. Otherwise the teach pendent will never get any time to interact with the user. And if the supervision task is in foreground, it will never allow another task in background to execute.

11.8 Programming scheme

The first time

1. Define the new task under system parameters (controller/task)
   Set the ***type*** to **NORMAL** and the ***TrustLevel*** to **NoSafety**.

2. Specify all modules that should be preloaded to this new task, also under system parameters (controller/task-modules).

3. Create the modules that should be in the task from the TeachPendant (in the MAIN task) or off-line.

4. Test and debug the modules in the MAIN task, until the functionallity is satified. Note: this could only be done in **motors on** state.

5. Change the task ***type*** to **SEMISTATIC** (or **STATIC**).

6. Restart the system.

Iteration phase

In many cases an iteration with point 3 and 5 is enough. It's only when the program has to be tested in the MAIN task and execution of the RAPID code in two task at the same time could confuse the user, all point should be used. Note: if a **STATIC** task is used it has to be forced to reload the new changed module and restarted from the beginning. If all point below is used it will take care of that for you.

1. Change the task type to **NORMAL** to inhibit the task. A NORMAL task will not start when the system restart and if it's not the MAIN task not it will be affected by the start button at the teach pendant.

2. Restart the system.

3. Load the moulde(s) to the **MAIN** task, test, change and save the module(s). Note: Don't save the task, save each module according to the system parameters.

4. Change back the task ***type*** to **SEMISTATIC** (or **STATIC**).

5. Restart the system.

Finish phase

1. Set the ***TrustLevel*** to desired level eg **SysFail**

2. Restart the system

1 Coordinate Systems

1.1 The robot's tool centre point (TCP)

The position of the robot and its movements are always related to the tool centre point. This point is normally defined as being somewhere on the tool, e.g. in the muzzle of a glue gun, at the centre of a gripper or at the end of a grading tool.

Several TCPs (tools) may be defined, but only one may be active at any one time. When a position is recorded, it is the position of the TCP that is recorded. This is also the point that moves along a given path, at a given velocity.

If the robot is holding a work object and working on a stationary tool, a stationary TCP is used. If that tool is active, the programmed path and speed are related to the work object. See *Stationary TCPs* on page 8.

1.2 Coordinate systems used to determine the position of the TCP

The tool (TCP's) position can be specified in different coordinate systems to facilitate programming and readjustment of programs.

The coordinate system defined depends on what the robot has to do. When no coordinate system is defined, the robot's positions are defined in the base coordinate system.

1.2.1 Base coordinate system

In a simple application, programming can be done in the base coordinate system; here the z-axis is coincident with axis 1 of the robot (see Figure 1).



Figure 1 The base coordinate system.

The base coordinate system is located on the base of the robot:

- *The origin* is situated at the intersection of axis 1 and the base mounting surface.
- *The xy plane* is the same as the base mounting surface.
- *The x-axis* points forwards.
- *The y-axis* points to the left (from the perspective of the robot).
- *The z-axis* points upwards.

1.2.2 World coordinate system

If the robot is floor-mounted, programming in the base coordinate system is easy. If, however, the robot is mounted upside down (suspended), programming in the base coordinate system is more difficult because the directions of the axes are not the same as the principal directions in the working space. In such cases, it is useful to define a world coordinate system. The world coordinate system will be coincident with the base coordinate system, if it is not specifically defined.

Sometimes, several robots work within the same working space at a plant. A common world coordinate system is used in this case to enable the robot programs to communicate with one another. It can also be advantageous to use this type of system when the positions are to be related to a fixed point in the workshop. See the example in Figure 2.



Figure 2 Two robots (one of which is suspended) with a common world coordinate system.

1.2.3 User coordinate system

A robot can work with different fixtures or working surfaces having different positions and orientations. A user coordinate system can be defined for each fixture. If all positions are stored in object coordinates, you will not need to reprogram if a fixture must be moved or turned. By moving/turning the user coordinate system as much as the fixture has been moved/turned, all programmed positions will follow the fixture and no reprogramming will be required.

The user coordinate system is defined based on the world coordinate system (see Figure 3).



Figure 3 Two user coordinate systems describe the position of two different fixtures.

1.2.4 Object coordinate system

The user coordinate system is used to get different coordinate systems for different fixtures or working surfaces. A fixture, however, may include several work objects that are to be processed or handled by the robot. Thus, it often helps to define a coordinate system for each object in order to make it easier to adjust the program if the object is moved or if a new object, the same as the previous one, is to be programmed at a different location. A coordinate system referenced to an object is called an object coordinate system. This coordinate system is also very suited to off-line programming since the positions specified can usually be taken directly from a drawing of the work object. The object coordinate system can also be used when jogging the robot.

The object coordinate system is defined based on the user coordinate system (see Figure 4).



Figure 4 Two object coordinate systems describe the position of two different work objects located in the same fixture.

The programmed positions are always defined relative to an object coordinate system. If a fixture is moved/turned, this can be compensated for by moving/turning the user coordinate system. Neither the programmed positions nor the defined object coordinate systems need to be changed. If the work object is moved/turned, this can be compensated for by moving/turning the object coordinate system.

If the user coordinate system is movable, that is, coordinated external axes are used, then the object coordinate system moves with the user coordinate system. This makes it possible to move the robot in relation to the object even when the workbench is being manipulated.

1.2.5 Displacement coordinate system

Sometimes, the same path is to be performed at several places on the same object. To avoid having to re-program all positions each time, a coordinate system, known as the displacement coordinate system, is defined. This coordinate system can also be used in conjunction with searches, to compensate for differences in the positions of the individual parts.

The displacement coordinate system is defined based on the object coordinate system (see Figure 5).



Figure 5 If program displacement is active, all positions are displaced.

1.2.6 Coordinated external axes

Coordination of user coordinate system

If a work object is placed on an external mechanical unit, that is moved whilst the robot is executing a path defined in the object coordinate system, a movable user coordinate system can be defined. The position and orientation of the user coordinate system will, in this case, be dependent on the axes rotations of the external unit. The programmed path and speed will thus be related to the work object (see Figure 6) and there is no need to consider the fact that the object is moved by the external unit.



Figure 6 A user coordinate system, defined to follow the movements of a 3-axis external mechanical unit.

Coordination of base coordinate system

A movable coordinate system can also be defined for the base of the robot. This is of interest for the installation when the robot is mounted on a track or a gantry, for example. The position and orientation of the base coordinate system will, as for the moveable user coordinate system, be dependent on the movements of the external unit. The programmed path and speed will be related to the object coordinate system (Figure 7) and there is no need to think about the fact that the robot base is moved by an external unit. A coordinated user coordinate system and a coordinated base coordinate system can both be defined at the same time.



Figure 7 Coordinated interpolation with a track moving the base coordinate system of the robot.

To be able to calculate the user and the base coordinate systems when involved units are moved, the robot must be aware of:

- The calibration positions of the user and the base coordinate systems
- The relations between the angles of the external axes and the translation/rotation of the user and the base coordinate systems.

These relations are defined in the system parameters.

1.3 Coordinate systems used to determine the direction of the tool

The orientation of a tool at a programmed position is given by the orientation of the tool coordinate system. The tool coordinate system is referenced to the wrist coordinated system, defined at the mounting flange on the wrist of the robot.

1.3.1 Wrist coordinate system

In a simple application, the wrist coordinate system can be used to define the orientation of the tool; here the z-axis is coincident with axis 6 of the robot (see Figure 8).



Figure 8 The wrist coordinate system.

The wrist coordinate system cannot be changed and is always the same as the mounting flange of the robot in the following respects:

- *The origin* is situated at the centre of the mounting flange (on the mounting surface).
- *The x-axis* points in the opposite direction, towards the control hole of the mounting flange.
- *The z-axis* points outwards, at right angles to the mounting flange.

1.3.2 Tool coordinate system

The tool mounted on the mounting flange of the robot often requires its own coordinate system to enable definition of its TCP, which is the origin of the tool coordinate system. The tool coordinate system can also be used to get appropriate motion directions when jogging the robot.

If a tool is damaged or replaced, all you have to do is redefine the tool coordinate system. The program does not normally have to be changed.

The TCP (origin) is selected as the point on the tool that must be correctly positioned, e.g. the muzzle on a glue gun. The tool coordinate axes are defined as those natural for the tool in question.



Figure 9 Tool coordinate system, as usually defined for an arc-welding gun (left) and a spot welding gun (right).

The tool coordinate system is defined based on the wrist coordinate system (see Figure 10).



Figure 10 The tool coordinate system is defined relative to the wrist coordinate system, here for a gripper.

1.3.3 Stationary TCPs

If the robot is holding a work object and working on a stationary tool, a stationary TCP is used. If that tool is active, the programmed path and speed are related to the work object held by the robot.

This means that the coordinate systems will be reversed, as in Figure 11.



Figure 11 If a stationary TCP is used, the object coordinate system is usually based on the wrist coordinate system.

In the example in Figure 11, neither the user coordinate system nor program displacement is used. It is, however, possible to use them and, if they are used, they will be related to each other as shown in Figure 12.



Figure 12 Program displacement can also be used together with stationary TCPs.

1.4 Related information

	Described in:
Definition of the world coordinate system	User’s Guide - System Parameters
Definition of the user coordinate system	User’s Guide - Calibration Data Types - *wobjdata*
Definition of the object coordinate system	User’s Guide - Calibration Data Types - *wobjdata*
Definition of the tool coordinate system	User’s Guide - Calibration Data Types - *tooldata*
Definition of a tool centre point	User’s Guide - Calibration Data Types - *tooldata*
Definition of displacement frame	User’s Guide - Calibration RAPID Summary - *Motion Settings*
Jogging in different coordinate systems	User’s Guide - Jogging

2 Positioning during Program Execution

2.1 General

During program execution, positioning instructions in the robot program control all movements. The main task of the positioning instructions is to provide the following information on how to perform movements:

- The destination point of the movement (defined as the position of the tool centre point, the orientation of the tool, the configuration of the robot and the position of the external axes).
- The interpolation method used to reach the destination point, e.g. joint interpolation, linear interpolation or circle interpolation.
- The velocity of the robot and external axes.
- The zone data (defines how the robot and the external axes are to pass the destination point).
- The coordinate systems (tool, user and object) used for the movement.

As an alternative to defining the velocity of the robot and the external axes, the time for the movement can be programmed. This should, however, be avoided if the weaving function is used. Instead the velocities of the orientation and external axes should be used to limit the speed, when small or no TCP-movements are made.

In material handling and pallet applications with intensive and frequent movements, the drive system supervision may trip out and stop the robot in order to prevent overheating of drives or motors. If this occurs, the cycle time needs to be slightly increased by reducing programmed speed or acceleration.

2.2 Interpolation of the position and orientation of the tool

2.2.1 Joint interpolation

When path accuracy is not too important, this type of motion is used to move the tool quickly from one position to another. Joint interpolation also allows an axis to move from any location to another within its working space, in a single movement.

All axes move from the start point to the destination point at constant axis velocity (see Figure 13).



Figure 13 Joint interpolation is often the fastest way to move between two points as the robot axes follow the closest path between the start point and the destination point (from the perspective of the axis angles).

The velocity of the tool centre point is expressed in mm/s (in the object coordinate system). As interpolation takes place axis-by-axis, the velocity will not be exactly the programmed value.

During interpolation, the velocity of the limiting axis, i.e. the axis that travels fastest relative to its maximum velocity in order to carry out the movement, is determined. Then, the velocities of the remaining axes are calculated so that all axes reach the destination point at the same time.

All axes are coordinated in order to obtain a path that is independent of the velocity. Acceleration is automatically optimised to the max performance of the robot.

2.2.2 Linear interpolation

During linear interpolation, the TCP travels along a straight line between the start and destination points (see Figure 14).



Figure 14 Linear interpolation without reorientation of the tool.

To obtain a linear path in the object coordinate system, the robot axes must follow a non-linear path in the axis space. The more non-linear the configuration of the robot is, the more accelerations and decelerations are required to make the tool move in a straight line and to obtain the desired tool orientation. If the configuration is extremely non-linear (e.g. in the proximity of wrist and arm singularities), one or more of the axes will require more torque than the motors can give. In this case, the velocity of all axes will automatically be reduced.

The orientation of the tool remains constant during the entire movement unless a reorientation has been programmed. If the tool is reorientated, it is rotated at constant velocity.

A maximum rotational velocity (in degrees per second) can be specified when rotating the tool. If this is set to a low value, reorientation will be smooth, irrespective of the velocity defined for the tool centre point. If it is a high value, the reorientation velocity is only limited by the maximum motor speeds. As long as no motor exceeds the limit for the torque, the defined velocity will be maintained. If, on the other hand, one of the motors exceeds the current limit, the velocity of the entire movement (with respect to both the position and the orientation) will be reduced.

All axes are coordinated in order to obtain a path that is independent of the velocity. Acceleration is optimised automatically.

2.2.3 Circular interpolation

A circular path is defined using three programmed positions that define a circle segment. The first point to be programmed is the start of the circle segment. The next point is a support point (circle point) used to define the curvature of the circle, and the third point denotes the end of the circle (see Figure 15).

The three programmed points should be dispersed at regular intervals along the arc of the circle to make this as accurate as possible.

The orientation defined for the support point is used to select between the short and the long twist for the orientation from start to destination point.

If the programmed orientation is the same relative to the circle at the start and the destination points, and the orientation at the support is close to the same orientation relative to the circle, the orientation of the tool will remain constant relative to the path.



Figure 15 Circular interpolation with a short twist for part of a circle (circle segment) with a start point, circle point and destination point.

However, if the orientation at the support point is programmed closer to the orientation rotated 180°, the alternative twist is selected (see Figure 16).



Figure 16 Circular interpolation with a long twist for orientation is achieved by defining the orientation in the circle point in the opposite direction compared to the start point.

As long as all motor torques do not exceed the maximum permitted values, the tool will move at the programmed velocity along the arc of the circle. If the torque of any of the motors is insufficient, the velocity will automatically be reduced at those parts of the circular path where the motor performance is insufficient.

All axes are coordinated in order to obtain a path that is independent of the velocity. Acceleration is optimised automatically.

2.2.4 SingArea\Wrist

During execution in the proximity of a singular point, linear or circular interpolation may be problematic. In this case, it is best to use modified interpolation, which means that the wrist axes are interpolated axis-by-axis, with the TCP following a linear or circular path. The orientation of the tool, however, will differ somewhat from the programmed orientation.

In the *SingArea\Wrist* case the orientation in the circle support point will be the same as programmed. However, the tool will not have a constant direction relative to the circle plane as for normal circular interpolation. If the circle path passes a singularity, the orientation in the programmed positions sometimes must be modified to avoid big wrist movements, which can occur if a complete wrist reconfiguration is generated when the circle is executed (joints 4 and 6 moved 180 degrees each).

2.3 Interpolation of corner paths

The destination point is defined as a stop point in order to get point-to-point movement. This means that the robot and any external axes will stop and that it will not be possible to continue positioning until the velocities of all axes are zero and the axes are close to their destinations.

Fly-by points are used to get continuous movements past programmed positions. In this way, positions can be passed at high speed without having to reduce the speed unnecessarily. A fly-by point generates a corner path (parabola path) past the programmed

position, which generally means that the programmed position is never reached. The beginning and end of this corner path are defined by a zone around the programmed position (see Figure 17).



Figure 17 A fly-by point generates a corner path to pass the programmed position.

All axes are coordinated in order to obtain a path that is independent of the velocity. Acceleration is optimised automatically.

2.3.1 Joint interpolation in corner paths

The size of the corner paths (zones) for the TCP movement is expressed in mm (see Figure 18). Since the interpolation is performed axis-by-axis, the size of the zones (in mm) must be recalculated in axis angles (radians). This calculation has an error factor (normally max. 10%), which means that the true zone will deviate somewhat from the one programmed.

If different speeds have been programmed before or after the position, the transition from one speed to the other will be smooth and take place within the corner path without affecting the actual path.



Figure 18 During joint interpolation, a corner path is generated in order to pass a fly-by point.

2.3.2 Linear interpolation of a position in corner paths

The size of the corner paths (zones) for the TCP movement is expressed in mm (see Figure 19).



Figure 19 During linear interpolation, a corner path is generated in order to pass a fly-by point.

If different speeds have been programmed before or after the corner position, the transition will be smooth and take place within the corner path without affecting the actual path.

If the tool is to carry out a process (such as arc-welding, gluing or water cutting) along the corner path, the size of the zone can be adjusted to get the desired path. If the shape of the parabolic corner path does not match the object geometry, the programmed positions can be placed closer together, making it possible to approximate the desired path using two or more smaller parabolic paths.

2.3.3 Linear interpolation of the orientation in corner paths

Zones can be defined for tool orientations, just as zones can be defined for tool positions. The orientation zone is usually set larger than the position zone. In this case, the reorientation will start interpolating towards the orientation of the next position before the corner path starts. The reorientation will then be smoother and it will probably not be necessary to reduce the velocity to perform the reorientation.

The tool will be reorientated so that the orientation at the end of the zone will be the same as if a stop point had been programmed (see Figure 20a-c).



Figure 20a Three positions with different tool orientations are programmed as above.



Figure 20b If all positions were stop points, program execution would look like this.



Figure 20c If the middle position was a fly-by point, program execution would look like this.

The orientation zone for the tool movement is normally expressed in mm. In this way, you can determine directly where on the path the orientation zone begins and ends. If the tool is not moved, the size of the zone is expressed in angle of rotation degrees instead of TCP-mm.

If different reorientation velocities are programmed before and after the fly-by point, and if the reorientation velocities limit the movement, the transition from one velocity to the other will take place smoothly within the corner path.

2.3.4 Interpolation of external axes in corner paths

Zones can also be defined for external axes, in the same manner as for orientation. If the external axis zone is set to be larger than the TCP zone, the interpolation of the external axes towards the destination of the next programmed position, will be started before the TCP corner path starts. This can be used for smoothing external axes movements in the same way as the orientation zone is used för the smoothing of the wrist movements.

2.3.5 Corner paths when changing the interpolation method

Corner paths are also generated when one interpolation method is exchanged for another. The interpolation method used in the actual corner paths is chosen in such a way as to make the transition from one method to another as smooth as possible. If the corner path zones for orientation and position are not the same size, more than one interpolation method may be used in the corner path (see Figure 21).



Figure 21 Interpolation when changing from one interpolation method to another. Linear interpolation has been programmed between p1 and p2; joint interpolation between p2 and p3; and Sing Area\Wrist interpolation between p3 and p4.

If the interpolation is changed from a normal TCP-movement to a reorientation without a TCP-movement or vice versa, no corner zone will be generated. The same will be the case if the interpolation is changed to or from an external joint movement without TCP-movement.

2.3.6 Interpolation when changing coordinate system

When there is a change of coordinate system in a corner path, e.g. a new TCP or a new work object, joint interpolation of the corner path is used. This is also applicable when changing from coordinated operation to non-coordinated operation, or vice versa.

2.3.7 Corner paths with overlapping zones

If programmed positions are located close to each other, it is not unusual for the programmed zones to overlap. To get a well-defined path and to achieve optimum velocity at all times, the robot reduces the size of the zone to half the distance from one overlapping programmed position to the other (see Figure 22). The same zone radius is always used for inputs to or outputs from a programmed position, in order to obtain symmetrical corner paths.



Figure 22 Interpolation with overlapping position zones. The zones around p2 and p3 are larger than half the distance from p2 to p3. Thus, the robot reduces the size of the zones to make them equal to half the distance from p2 to p3, thereby generating symmetrical corner paths within the zones.

Both position and orientation corner path zones can overlap. As soon as one of these corner path zones overlap, that zone is reduced (see Figure 23).



Figure 23 Interpolation with overlapping orientation zones. The orientation zone at p2 is larger than half the distance from p2 to p3 and is thus reduced to half the distance from p2 to p3. The position zones do not overlap and are consequently not reduced; the orientation zone at p3 is not reduced either.

2.3.8 Planning time for fly-by points

Occasionally, if the next movement is not planned in time, programmed fly-by points can give rise to a stop point. This may happen when:

- A number of logical instructions with long program execution times are programmed between short movements.
- The points are very close together at high speeds.

If stop points are a problem then use concurrent program execution.

2.4 Independent axes

An independent axis is an axis moving independently of other axes in the robot system. It is possible to change an axis to independent mode and later back to normal mode again.

A special set of instructions handles the independent axes. Four different move instructions specify the movement of the axis. For instance, the *IndCMove* instruction starts the axis for continuous movement. The axis then keeps moving at a constant speed (regardless of what the robot does) until a new independent-instruction is executed.

To change back to normal mode a reset instruction, *IndReset*, is used. The reset instruction can also set a new reference for the measurement system - a type of new synchronization of the axis. Once the axis is changed back to normal mode it is possible to run it as a normal axis.

2.4.1 Program execution

An axis is immediately change to independent mode when an *Ind_Move* instruction is executed. This takes place even if the axis is being moved at the time, such as when a previous point has been programmed as a fly-by point, or when simultaneous program execution is used.

If a new *Ind_Move* instruction is executed before the last one is finished, the new instruction immediately overrides the old one.

If a program execution is stopped when an independent axis is moving, that axis will stop. When the program is restarted the independent axis starts automatically. No active coordination between independent and other axes in normal mode takes place.

If a loss of voltage occurs when an axis is in independent mode, the program cannot be restarted. An error message is then displayed, and the program must be started from the beginning.

Note that a mechanical unit may not be deactivated when one of its axes is in independent mode.

2.4.2 Stepwise execution

During stepwise execution, an independent axis is executed only when another instruction is being executed. The movement of the axis will also be stepwise in line with the execution of other instruments, see Figure 24.



Figure 24 Stepwise execution of independent axes.

2.4.3 Jogging

Axes in independent mode cannot be jogged. If an attempt is made to execute the axis manually, the axis does not move and an error message is displayed. Execute an *IndReset* instruction or move the program pointer to main, in order to leave the independent mode.

2.4.4 Working range

The physical working range is the total movement of the axis.

The logical working range is the range used by RAPID instructions and read in the jogging window.

After synchronization (updated revolution counter), the physical and logical working range coincide. By using the *IndReset* instruction the logical working area can be moved, see Figure 25.



Figure 25 The logical working range can be moved, using the instruction IndReset.

The resolution of positions is decreased when moving away from logical position 0. Low resolution together with stiff tuned controller can result in unacceptable torque, noise and controller instability. Check the controller tuning and axis performance close to the working range limit at installation. Also check if the position resolution and path performance are acceptable.

2.4.5 Speed and acceleration

In manual mode with reduced speed, the speed is reduced to the same level as if the axis was running as non-independent. Note that the *IndSpeed\InSpeed* function will not be TRUE if the axis speed is reduced.

The *VelSet* instruction and speed correction in percentage via the production window, are active for independent movement. Note that correction via the production window inhibits TRUE value from the *IndSpeed\InSpeed* function.

In independent mode, the lowest value of acceleration and deceleration, specified in the configuration file, is used both for acceleration and deceleration. This value can be reduced by the ramp value in the instruction (1 - 100%). The *AccSet* instruction does not affect axes in independent mode.

2.4.6 Robot axes

Only robot axis 6 can be used as an independent axis. Normally the *IndReset* instruction is used only for this axis. However, the *IndReset* instruction can also be used for axis 4 on IRB 2400 and 4400 models. If *IndReset* is used for robot axis 4, then axis 6 must not be in the independent mode.

If axis 6 is used as an independent axis, singularity problems may occur because the normal 6-axes coordinate transform function is still used. If a problem occurs, execute the same program with axis 6 in normal mode. Modify the points or use *SingArea\Wrist* or *MoveJ* instructions.

The axis 6 is also internally active in the path performance calculation. A result of this is that an internal movement of axis 6 can reduce the speed of the other axes in the system.

The independent working range for axis 6 is defined with axis 4 and 5 in home position. If axis 4 or 5 is out of home position the working range for axis 6 is moved due to the gear coupling. However, the position read from teach pendant for axis 6 is compensated with the positions of axis 4 and 5 via the gear coupling.

2.5 Soft Servo

In some applications there is a need for a servo, which acts like a mechanical spring. This means that the force from the robot on the work object will increase as a function of the distance between the programmed position (behind the work object) and the contact position (robot tool - work object).

The relationship between the position deviation and the force, is defined by a parameter called ***softness.*** The higher the softness parameter, the larger the position deviation required to obtain the same force.

The softness parameter is set in the program and it is possible to change the softness values anywhere in the program. Different softness values can be set for different joints and it is also possible to mix joints having normal servo with joints having soft servo.

Activation and deactivation of soft servo as well as changing of softness values can be made when the robot is moving. When this is done, a tuning will be made between the different servo modes and between different softness values to achieve smooth transitions. The tuning time can be set from the program with the parameter ramp. With *ramp = 1*, the transitions will take 0.5 seconds, and in the general case the transition time will be *ramp x 0.5* in seconds.

Note that deactivation of soft servo should not be done when there is a force between the robot and the work object.

With high softness values there is a risk that the servo position deviations may be so big that the axes will move outside the working range of the robot.

2.6 Stop and restart

A movement can be stopped in three different ways:

1. *For a normal stop* the robot will stop on the path, which makes a restart easy.

2. *For a stiff stop* the robot will stop in a shorter time than for the normal stop, but the deceleration path will not follow the programmed path. This stop method is, for example, used for search stop when it is important to stop the motion as soon as possible.

3. *For a quick-stop* the mechanical brakes are used to achieve a deceleration distance, which is as short as specified for safety reasons. The path deviation will usually be bigger for a quick-stop than for a stiff stop.

After a stop (any of the types above) a restart can always be made on the interrupted path. If the robot has stopped outside the programmed path, the restart will begin with a return to the position on the path, where the robot should have stopped.

A restart following a power failure is equivalent to a restart after a quick-stop. It should be noted that the robot will always return to the path before the interrupted program operation is restarted, even in cases when the power failure occurred while a logical instruction was running. When restarting, all times are counted from the beginning; for example, positioning on time or an interruption in the instruction *WaitTime*.

2.7 Related information

	Described in:
Definition of speed	Data Types - *speeddata*
Definition of zones (corner paths)	Data Types - *zonedata*
Instruction for joint interpolation	Instructions - *MoveJ*
Instruction for linear interpolation	Instructions - *MoveL*
Instruction for circular interpolation	Instructions - *MoveC*
Instruction for modified interpolation	Instructions - *SingArea*
Singularity	Motion and I/O Principles- *Singularity*
Concurrent program execution	Motion and I/O Principles- *Synchronisation with logical instructions*
CPU Optimization	User’s Guide - System parameters

3 Synchronisation with logical instructions

Instructions are normally executed sequentially in the program. Nevertheless, logical instructions can also be executed at specific positions or during an ongoing movement.

A logical instruction is any instruction that does not generate a robot movement or an external axis movement, e.g. an I/O instruction.

3.1 Sequential program execution at stop points

If a positioning instruction has been programmed as a stop point, the subsequent instruction is not executed until the robot and the external axes have come to a standstill, i.e. when the programmed position has been attained (see Figure 26).



Figure 26 A logical instruction after a stop point is not executed until the destination position has been reached.

3.2 Sequential program execution at fly-by points

If a positioning instruction has been programmed as a fly-by point, the subsequent logical instructions are executed some time before reaching the largest zone (for position, orientation or external axes). See Figure 27 and Figure 28. These instructions are then executed in order.



Figure 27 A logical instruction following a fly-by point is executed before reaching the largest zone.



Figure 28 A logical instruction following a fly-by point is executed before reaching the largest zone.

The time at which they are executed *(DT*) comprises the following time components:

- The time it takes for the robot to plan the next move: approx. 0.1 seconds.
- The robot delay (servo lag) in seconds: 0 - 1.0 seconds depending on the velocity and the actual deceleration performance of the robot.

3.3 Concurrent program execution

Concurrent program execution can be programmed using the argument *\Conc* in the positioning instruction. This argument is used to:

- Execute one or more logical instructions at the same time as the robot moves in order to reduce the cycle time (e.g. used when communicating via serial channels).

When a positioning instruction with the argument *\Conc* is executed, the following logical instructions are also executed (in sequence):

- If the robot is not moving, or if the previous positioning instruction ended with a stop point, the logical instructions are executed as soon as the current positioning instruction starts (at the same time as the movement). See Figure 29.
- If the previous positioning instruction ends at a fly-by point, the logical instructions are executed at a given time (*DT*) before reaching the largest zone (for position, orientation or external axes). See Figure 30.



Figure 29 In the case of concurrent program execution after a stop point, a positioning instruction and subsequent logical instructions are started at the same time.



Figure 30 In the case of concurrent program execution after a fly-by point, the logical instructions start executing before the positioning instructions with the argument \Conc are started.

Instructions which indirectly affect movements, such as *ConfL* and *SingArea*, are executed in the same way as other logical instructions. They do not, however, affect the movements ordered by previous positioning instructions.

If several positioning instructions with the argument *\Conc* and several logical instructions in a long sequence are mixed, the following applies:

- Logical instructions are executed directly, in the order they were programmed. This takes place at the same time as the movement (see Figure 31) which means that logical instructions are executed at an earlier stage on the path than they were programmed.



Figure 31 If several positioning instructions with the argument \Conc are programmed in sequence, all connected logical instructions are executed at the same time as the first position is executed.

During concurrent program execution, the following instructions are programmed to end the sequence and subsequently re-synchronise positioning instructions and logical instructions:

- a positioning instruction to a stop point without the argument *\Conc*,
- the instruction *WaitTime* or *WaitUntil* with the argument *\Inpos*.

3.4 Path synchronisation

In order to synchronise process equipment (for applications such as gluing, painting and arc welding) with the robot movements, different types of path synchronisation signals can be generated.

With a so-called positions event, a trig signal will be generated when the robot passes a predefined position on the path. With a time event, a signal will be generated in a predefined time before the robot stops at a stop position. Moreover, the control system also handles weave events, which generate pulses at predefined phase angles of a weave motion.

All the position synchronised signals can be achieved both before (look ahead time) and after (delay time) the time that the robot passes the predefined position. The position is defined by a programmed position and can be tuned as a path distance before the programmed position.

Typical repeat accuracy for a set of digital outputs on the path is +/- 2ms.

In the event of a power failure and restart in a Trigg instruction, all trigg events will be generated once again on the remaining movement path for the trigg instruction.

3.5 Related information

	Described in:
Positioning instructions	RAPID Summary - *Motion*
Definition of zone size	Data Types - *zonedata*

4 Robot Configuration

It is usually possible to attain the same robot tool position and orientation in several different ways, using different sets of axis angles. We call these different robot configurations. If, for example, a position is located approximately in the middle of a work cell, some robots can get to that position from above and from below (see Figure 32). This can also be achieved by turning the front part of the robot upper arm (axis 4) upside down while rotating axes 5 and 6 to the desired position and orientation (see Figure 33).



Figure 32 Two different arm configurations used to attain the same position and orientation. In one of the configurations, the arms point upwards and to attain the other configuration, axis 1 must be rotated 180 degrees.



Figure 33 Two different wrist configurations used to attain the same position and orientation. In the configuration in which the front part of the upper arm points upwards (lower), axis 4 has been rotated 180 degrees, axis 5 through 180 degrees and axis 6 through 180 degrees in order to attain the configuration in which the front part of the upper arm points downwards (upper).

Usually you want the robot to attain the same configuration during program execution as the one you programmed. To do this, you can make the robot check the configuration and, if the correct configuration is not attained, program execution will stop. If the configuration is not checked, the robot may unexpectedly start to move its arms and wrists which, in turn, may cause it to collide with peripheral equipment.

For a rotational robot axis the robot configuration is specified by defining the appropriate quarter revolutions of the axis.



Figure 34 Quarter revolution for a positive joint angle: $int(joint\tan gle \times 2/\pi)$.



Figure 35 Quarter revolution for a negative joint angle: $int(joint\tan gle \times 2/\pi - 1)$.

For a linear robot axis the value defines a meter interval for the robot axis. Value 0 means a position between 0 and 1 meters, 1 means a position between 1 and 2 meters. For negative values, -1 means a position between -1 and 0 meters, etc.



Figure 36 Configuration values for a linear axis

The configuration data set consists of four components cf1, cf4, cf6, cfx. cf1 specifies the value for axis 1, cf4, for axis 4, cf6 for axis 6. cfx is an additional component that is used for robot types with structures that need an extra component.

The configuration check involves comparing the configuration of the programmed position with that of the robot. The configuration of a specific axis is accepted if the

attained axis value is within +/- 45 degrees from the specified quadrant for a rotational joint or within +/- 0.5 m from the specified meter value of a linear axis.



Figure 37 Example: Accepted axis value 135 deg < joint 4 <-45 deg for rotational axis 4 when cf4 = 0



Figure 38 Example: Accepted axis value -1.5 m < joint 1 < 0.5 m for linear axis 1 when cf1 = -1

During linear movement, the robot always moves to the closest possible configuration. If, however, the configuration check is active, program execution stops as soon as:

- The configuration of the programmed position will not be attained.
- The reorientation needed by any one of the wrist axes to get to the programmed position exceeds a limit (140-180 degrees).

During axis-by-axis or modified linear movement using a configuration check, the robot always moves to the programmed axis configuration. If the programmed axes configurations will not be reached, program execution stops before starting the movement. If the configuration check is not active, the robot moves to the specified position and orientation with the closest configuration.

When the execution of a programmed position is stopped because of a configuration error, it may often be caused by one or more of the following reasons:

- The position is programmed off-line with a faulty configuration.
- The robot tool has been changed causing the robot to take another configuration than was programmed.
- The position is subject to an active frame operation (displacement, user, object, base).

The correct configuration in the destination position can be found by positioning the robot near it and reading the configuration on the teach pendant.

If the configuration parameters change because of active frame operation, the configuration check can be deactivated.

4.1 Related information

	Described in:
Definition of robot configuration	Data Types - *confdata*
Activating/deactivating the configuration check	RAPID Summary - *Motion Settings*

5 Robot kinematic models

5.1 Robot kinematics

The position and orientation of a robot is determined from the kinematic model of its mechanical structure. The specific mechanical unit models must be defined for each installation. For standard ABB master and external robots, these models are predefined in the controller.

5.1.1 Master robot

The kinematic model of the master robot models the position and orientation of the tool of the robot relative to its base as function of the robot joint angles.

The kinematic parameters specifying the arm-lengths, offsets and joint attitudes, are predefined in the configuration file for each robot type.



Figure 39 Kinematic structure of an IRB1400 robot

A calibration procedure supports the definition of the base frame of the master robot relative to the world frame.



Figure 40 Base frame of master robot

5.1.2 External robot

Coordination with an external robot also requires a kinematic model for the external robot. A number of predefined classes of 2 and 3 dimensional mechanical structures are supported.



Figure 41 Kinematic structure of an ORBIT 160B robot using predefined model

Calibration procedures to define the base frame relative to the world frame are supplied for each class of structures.



Figure 42 Base frame of an ORBIT_160B robot



Figure 43 Reference points on turntable for base frame calibration of an ORBIT_160B robot in the home position using predefined model

5.2 General kinematics

Mechanical structures not supported by the predefined structures may be modelled by using a general kinematic model. This is possible for external robots.

Modelling is based on the Denavit-Hartenberg convention according to Introduction to Robotics, Mechanics & Control, John J. Craigh (Addison-Wesley 1986)



Figure 44 Kinematic structure of an ORBIT 160B robot using general kinematics model

A calibration procedure supports the definition of the base frame of the external robot relative to the world frame.



Figure 45 Base frame of an ORBIT_160B robot using general kinematics model



Figure 46 Reference points on turntable for base frame calibration of an ORBIT_160B robot in the home position (joints = 0 degrees)

5.3 Related information

	Described in:
Definition of general kinematics of an external robot	User's Guide - System Parameters

6 Motion Supervision/Collision Detection

Motion supervision is the name of a collection of functions for high sensitivity, model-based supervision of the robot's movements. Motion supervision includes functionality for the detection of collision, jamming, and incorrect load definition. This functionality is called Collision Detection.

6.1 Introduction

The collision detection will trig if the load is incorrectly defined. If the load data is not known, the load identification functionality can be used to define it.

When the collision detection is triggered, the motor torques are reversed and the mechanical brakes applied in order to stop the robot. The robot then backs up a short distance along the path in order to remove any residual forces which may be present if a collision or jam occurred. After this, the robot stops again and remains in the motors on state. A typical collision is illustrated in the figure below.

The motion supervision is only active when at least one axis (including external axes) is in motion. When all axes are standing still, the function is deactivated. This is to avoid unnecessary trigging due to external process forces.

6.2 Tuning of Collision Detection levels

The collision detection uses a variable supervision level. At low speeds it is more sensitive than during high speeds. For this reason, no tuning of the function should be required by the user during normal operating conditions. However, it is possible to turn the function on and off and to tune the supervision levels. Separate tuning parameters are available for jogging and program execution. The different tuning parameters are described in more detail in the User's Guide under System Parameters: Manipulator.

There is a RAPID instruction called MotionSup which turns the function on and off and modifies the supervision level. This is useful in applications where external process forces act on the robot in certain parts of the cycle. The MotionSup instruction is described in more detail in the LoadId & CollDetect Manual.

6.3 Motion supervision dialogue box

Select motion supervision under the special menu in the jogging window. This displays a dialogue box which allows the motion supervision to be turned on and off. **This will only affect the robot during jogging**. If the motion supervision is turned off in the dialogue box and a program is executed, the collision detection can still be active during the running of the program. If the program is then stopped and the robot jogged, the status flag in the dialogue window is set to on again. This is a safety measure to avoid turning the function off by accident.

Figure: Typical collision

Phase 1 - The motor torque is reversed to stop the robot.

Phase 2 - The motor speed is reversed to remove residual forces on the tool and robot.





6.4 Digital outputs

The digital output MotSupOn is high when the collision detection function is active and low when it is not active. Note that a change in the state of the function takes effect when a motion starts. Thus, if the collision detection is active and the robot is moving, MotSupOn is high. If the robot is stopped and the function turned off, MotSupOn is still high. When the robot starts to move, MotSupOn switches to low.

The digital output MotSupTrigg goes high when the collision detection triggers. It stays high until the error code is acknowledged, either from the teach pendant or through the digital input AckErrDialog.

The digital outputs are described in more detail in the User's Guide under System Parameters: IO Signals.

6.5 Limitations

The motion supervision is only available for the robot axes. It is not available for track motions, orbit stations, or any other external manipulators.

In RobotWare 3.1, the motion supervision is only available for the IRB6400 robot family.

The collision detection is deactivated when at least one axis is run in independent joint mode. This is also the case even when it is an external axis which is run as an independent joint.

The collision detection may trigger when the robot is used in soft servo mode. Therefore, it is advisable to turn the collision detection off when the robot is in soft sevo mode.

If the RAPID instruction MotionSup is used to turn off the collision detection, this will only take effect once the robot starts to move. As a result, the digital output MotSupOn may temporarily be high at program start before the robot starts to move.

The distance the robot backs up after a collision is proportional to the speed of the motion before the collision. If repeated low speed collisions occur, the robot may not back up sufficiently to relieve the stress of the collision. As a result, it may not be possible to jog the robot without the supervision triggering. In this case use the jog menu to turn off the collision detection temporarily and jog the robot away from the obstacle.

In the event of a stiff collision during program execution, it may take a few seconds before the robot starts to back up.

6.6 Related information

	Described in:
RAPID instruction MotionSup	RAPID Summary - *Motion*
System parameters for tuning	System Parameters - *Manipulator*
Motion supervision IO Signals	System Parameters - *IO Signals*
Load Identification	Option Manual - *LoadId & CollDetect*

7 Singularities

Some positions in the robot working space can be attained using an infinite number of robot configurations to position and orient the tool. These positions, known as singular points (singularities), constitute a problem when calculating the robot arm angles based on the position and orientation of the tool.

Generally speaking, a robot has two types of singularities: arm singularities and wrist singularities. Arm singularities are all configurations where the wrist centre (the intersection of axes 4, 5 and 6) ends up directly above axis 1 (see Figure 47). Wrist singularities are configurations where axis 4 and axis 6 are on the same line, i.e. axis 5 has an angle equal to 0 (see Figure 48).



Figure 47 Arm singularity occurs where the wrist centre and axis 1 intersect.



Figure 48 Wrist singularity occurs when axis 5 is 0 degrees.

7.1 Singularity points/IRB 6400C

Between the robot's working space, with the arm directed forward and the arm directed backward, there is a singularity point above the robot. There is also a singularity point on the sides of the robot. These points contain a singularity and have kinematic limitations. A position in these points cannot be specified as forward/ backward and can only be reached with *MoveAbsJ*. When the robot is in a singular point:

- It is only possible to use *MoveAbsJ* or to jog the robot axis by axis.

7.2 Program execution through singularities

During joint interpolation, the robot never has any problem passing singular points.

When executing a linear or circular path close to a singularity, the velocities in some joints (1 and 6/4 and 6) may be very high. In order not to exceed the maximum joint velocities, the linear path velocity is reduced.

The high joint velocities may be reduced by using the mode (*Sing Area\Wrist*) when the wrist axes are interpolated in joint angles while still maintaining the linear path of the robot tool. An orientation error compared to the full linear interpolation is however introduced.

Note that the robot configuration changes dramatically when the robot passes close to a singularity with linear or circular interpolation. In order to avoid the reconfiguration, the first position on the other side of the singularity should be programmed with an orientation that makes the reconfiguration unnecessary.

Also note that the robot should not be in its singularity when external joints only are moved, as this may cause robot joints to make unnecessary movements.

7.3 Jogging through singularities

During joint interpolation, the robot never has any problem passing singular points.

During linear interpolation the robot can pass singular points but at a decreased speed.

7.4 Related information

	Described in:
Controlling how the robot is to act on execution near singular points	Instructions - *SingArea*

8 World Zones

8.1 Using global zones

When using this function, the robot stops or an output is automatically set if the robot is inside a special user-defined area. Here are some examples of applications:

- When two robots share a part of their respective work areas. The possibility of the two robots colliding can be safely eliminated by the supervision of these signals.
- When external equipment is located inside the robot's work area. A forbidden work area can be created to prevent the robot colliding with this equipment.
- Indication that the robot is at a position where it is permissible to start program execution from a PLC.

8.2 Using World Zones

To indicate that the tool centre point is in a specific part of the working area.
To limit the working area of the robot in order to avoid collision with the tool.
To make a common work area for two robots available to only one robot at a time.

8.3 Definition of World Zones in the world coordinate system

All World Zones are to be defined in the world coordinate system.
The sides of the Boxes are parallel to the coordinate axes and Cylinder axis is parallel to the Z axis of the world coordinate system.



A World Zone can be defined to be inside or outside the shape of the Box, Sphere or the Cylinder.

8.4 Supervision of the Robot TCP



The movement of the tool centre point is supervised and not any other points on the robot.

The TCP is always supervised irrespective of the mode of operation, for example, program execution and jogging.

8.4.1 Stationary TCPs

If the robot is holding a work object and working on a stationary tool, a stationary TCP is used. If that tool is active, the tool will not move and if it is inside a World Zone then it is always inside.

8.5 Actions

8.5.1 Set a digital output when the tcp is inside a World Zone.

This action sets a digital output when the tcp is inside a World Zone. It is useful to indicate that the robot has stopped in a specified area.



8.5.2 Set a digital output before the tcp reaches a World Zone.

This action sets a digital output before the tcp reaches a World Zone. It can be used to stop the robot just inside a World Zone



8.5.3 Stop the robot before the tcp reaches a World Zone.

A World Zone can be defined to be outside the work area. The robot will then stop with the Tool Centre Point just outside the World Zone, when heading towards the Zone



When the robot has been moved into a World Zone defined as an outside work area, for example, by releasing the brakes and manually pushing, then the only ways to get out of the Zone are by jogging or by manual pushing with the brakes released.

8.6 Minimum size of World Zones.

Supervision of the movement of the tool centre points is done at discrete points with a time interval between them that depends on the path resolution.
It is up to the user to make the zones large enough so the robot cannot move right through a zone without being checked inside the Zone.

Min. size of zone for used path_resolution and max. speed			
Speed / Resol.	**1000 mm/s**	**2000 mm/s**	**4000 mm/s**
1	**25 mm**	**50 mm**	**100 mm**
2	**50 mm**	**100 mm**	**200 mm**
3	**75 mm**	**150 mm**	**300 mm**

If the same digital output is used for more than one World Zone, the distance between the Zones must exceed the minimum size, as shown in the table above, to avoid an incorrect status for the output.



It is possible that the robot can pass right through a corner of a zone without it being noticed, if the time that the robot is inside the zone is too short. Therefore, make the size of the zone larger than the dangerous area.

8.7 Maximum number of World Zones

A maximum of **ten** World Zones can be defined at the same time.

8.8 Power failure, restart, and run on

Stationary World Zones will be deleted at power off and must be reinserted at power on by an event routine connected to the event POWER ON.

Temporary World Zones will survive a power failure but will be erased when a new program is loaded or when a program is started from the main program.

The digital outputs for the World Zones will be updated first at ***Motors on***.

8.9 Related information

RAPID Reference Manual	
Motion and I/O Principles:	Coordinate Systems
Data Types:	wztemporary
	wzstationary
	shapedata
Instructions:	WZBoxDef
	WZSphDef
	WZCylDef
	WZLimSup
	WZDOSet
	WZDisable
	WZEnable
	WZFree
	WZTempFree

9 I/O Principles

The robot generally has one or more I/O boards. Each of the boards has several digital and/or analog channels which must be connected to logical signals before they can be used. This is carried out in the system parameters and has usually already been done using standard names before the robot is delivered. Logical names must always be used during programming.

A physical channel can be connected to several logical names, but can also have no logical connections (see Figure 49).



Figure 49 To be able to use an I/O board, its channels must be given logical names. In the above example, the physical output 2 is connected to two different logical names. IN16, on the other hand, has no logical name and thus cannot be used.

9.1 Signal characteristics

The characteristics of a signal are depend on the physical channel used as well as how the channel is defined in the system parameters. The physical channel determines time delays and voltage levels (see the Product Specification). The characteristics, filter times and scaling between programmed and physical values, are defined in the system parameters.

When the power supply to the robot is switched on, all signals are set to zero. They are not, however, affected by emergency stops or similar events.

An output can be set to one or zero from within the program. This can also be done using a delay or in the form of a pulse. If a pulse or a delayed change is ordered for an output, program execution continues. The change is then carried out without affecting the rest of the program execution. If, on the other hand, a new change is ordered for the same output before the given time elapses, the first change is not carried out (see Figure 50).



Figure 50 The instruction SetDO is not carried out at all because a new command is given before the time delay has elapsed.

9.2 Signals connected to interrupt

RAPID interrupt functions can be connected to digital signal changes. The function can be called on a raising or falling edge of the signal. However, if the digital signal changes very quickly, the interrupt can be missed.

Ex:

If a function is connected to a signal called do1 and you make a program like:
SetDO do1,1;
SetDO do1,0;
The signal will first go to High and then Low in a few milliseconds. In this case you may loose the interrupt. To be sure that you will get the interrupt, make sure that the output is set before resetting it.

Ex:
SetDO do1,1;
WaitDO do1 ,1;
SetDO do1,0;
In this way you will never loose any interrupt.

9.3 System signals

Logical signals can be interconnected by means of special system functions. If, for example, an input is connected to the system function *Start*, a program start is automatically generated as soon as this input is enabled. These system functions are generally only enabled in automatic mode. For more information, see Chapter 9, System Parameters, or the chapter on *Installation and Commissioning - PLC Communication* in the Product Manual.

9.4 Cross connections

Digital signals can be interconnected in such a way that they automatically affect one another:

- An output signal can be connected to one or more input or output signals.
- An input signal can be connected to one or more input or output signals.
- If the same signal is used in several cross connections, the value of that signal is the same as the value that was last enabled (changed).
- Cross connections can be interlinked, in other words, one cross connection can affect another. They must not, however, be connected in such a way so as to form a ”vicious circle”, e.g. cross-connecting *di1* to *di2* whilst *di2* is cross-connected to *di1*.
- If there is a cross connection on an input signal, the corresponding physical connection is automatically disabled. Any changes to that physical channel will thus not be detected.
- Pulses or delays are not transmitted over cross connections.
- Logical conditions can be defined using NOT, AND and OR (Option: Advanced functions).

Examples:

- *di2=di1*
- *di3=di2*
- *do4=di2*

If *di1* changes, *di2*, *di3* and *do4* will be changed to the corresponding value.

- *do8=do7*
- *do8=di5*

If *do7* is set to 1, *do8* will also be set to 1. If *di5* is then set to 0, *do8* will also be changed (in spite of the fact that *do7* is still 1).

- *do5=di6* and *do1*

Do5 is set to 1 if both *di6* and *do1* is set to 1.

9.5 Limitations

A maximum of 10 signals can be pulsed at the same time and a maximum of 20 signals can be delayed at the same time.

9.6 Related information

	Described in:
Definition of I/O boards and signals	User’s Guide - System Parameters
Instructions for handling I/O	RAPID Summary - *Input and Output Signals*
Manual manipulation of I/O	User’s Guide - Inputs and Outputs

1 Programming Off-line

RAPID programs can easily be created, maintained and stored in an ordinary office computer. All information can be read and changed directly using a normal text editor. This chapter explains the working procedure for doing this. In addition to off-line programming, you can use the computer tool, QuickTeach.

1.1 File format

The robot stores and reads RAPID programs in TXT format (ASCII) and can handle both DOS and UNIX text formats. If you use a word-processor to edit programs, these must be saved in TXT format (ASCII) before they are used in the robot.

1.2 Editing

When a program is created or changed in a word-processor, all information will be handled in the form of text. This means that information about data and routines will differ somewhat from what is displayed on the teach pendant.

Note that the value of a stored position is only displayed as an *on the teach pendant, whereas the text file will contain the actual position value (x, y, z, etc.).

In order to minimise the risk of errors in the syntax (faulty programs), you should use a template. A template can take the form of a program that was created previously on the robot or using QuickTeach. These programs can be read directly to a word-processor without having to be converted.

1.3 Syntax check

Programs must be syntactically correct before they are loaded into the robot. By this is meant that the text must follow the fundamental rules of the RAPID language. One of the following methods should be used to detect errors in the text:

- Save the file on diskette and try to open it in the robot. If there are any syntactical errors, the program will not be accepted and an error message will be displayed. To obtain information about the type of error, the robot stores a log called PGMCPL1.LOG on the internal RAM disk. Copy this log to a diskette using the robot's File Manager. Open the log in a word-processor and you will be able to read which lines were incorrect and receive a description of the error.
- Open the file in QuickTeach or ProgramMaker.
- Use a RAPID syntax check program for the PC.

When the program is syntactically correct, it can be checked and edited in the robot. To make sure that all references to routines and data are correct, use the command **File: Check Program**. If the program has been changed in the robot, it can be stored on diskette again and processed or stored in a PC.

1.4 Examples

The following shows examples of what routines look like in text format.

```
%%%
    VERSION: 1
     LANGUAGE: ENGLISH
%%%
MODULE main
VAR intnum process_int ;
! Demo of RAPID program
PROC main()
    MoveL p1, v200, fine, gun1;
ENDPROC


TRAP InvertDo12
! Trap routine for TriggInt
    TEST INTNO
        CASE process_int:
        InvertDO do12;
        DEFAULT:
        TPWrite “Unknown trap , number=”\Num:=INTNO;
    ENDTEST
ENDTRAP

LOCAL FUNC num MaxNum(num t1, num t2)
    IF t1 > t2 THEN
        RETURN t1;
    ELSE
        RETURN t2;
    ENDIF
ENDFUNC
ENDMODULE
```

1.5 Making your own instructions

In order to make programming easier, you can customize your own instructions. These are created in the form of normal routines, but, when programming and test-running, function as instructions:

- They can be taken from the instruction pick list and programmed as normal instructions.
- The complete routine will be run during step-by-step execution.

• Create a new system module where you can place your routines that will function as instructions. Alternatively, you can place them in the USER system module.

- Create a routine in this system module with the name that you want your new instruction to be called. The arguments of the instruction are defined in the form of routine parameters. Note that the name of the parameters will be displayed in the window during programming and should therefore be given names that the user will understand.
- Place the routine in one of the Most Common pick lists.
- If the instruction is to behave in a certain way during backward program execution, this can be done in the form of a backward handler. If there is no such handler, it will not be possible to get past the instruction during backward program execution (see Chapter 13 in this manual - Basic Characteristics). A backward handler can be entered using the command **Routine: Add Backward Handler** from the *Program Routines* window.
- Test the routine thoroughly so that it works with different types of input data (arguments).
- Change the module attribute to NOSTEPIN. The complete routine will then be run during step-by-step execution. This attribute, however, must be entered off-line.

Example: To make the gripper easier to handle, two new instructions are made, *GripOpen* and *GripClose*. The output signal's name is given to the instruction's argument, e.g. *GripOpen gripper1*.

```
MODULE My_instr (SYSMODULE, NOSTEPIN)
PROC GripOpen (VAR signaldo Gripper)
    Set Gripper;
    WaitTime 0.2;
ENDPROC
PROC GripClose (VAR signaldo Gripper)
    Reset Gripper;
    WaitTime 0.2;
ENDPROC
ENDMODULE
```

1 System Module *User*

In order to facilitate programming, predefined data is supplied with the robot. This data does not have to be created and, consequently, can be used directly.

If this data is used, initial programming is made easier. It is, however, usually better to give your own names to the data you use, since this makes the program easier for you to read.

1.1 Contents

User comprises five numerical data (registers), one work object data, one clock and two symbolic values for digital signals.

Name	Data type	Declaration
reg1	num	VAR num reg1:=0
reg2	.	.
reg3	.	.
reg4	.	.
reg5	num	VAR num reg5:=0
wobj1	wobjdata	PERS wobjdata wobj1:=wobj0
clock1	clock	VAR clock clock1
high	dionum	CONST dionum high:=1
low	dionum	CONST dionum low:=0
edge	dionum	CONST dionum edge:=2

User is a system module, which means that it is always present in the memory of the robot regardless of which program is loaded.

1.2 Creating new data in this module

This module can be used to create such data and routines that must always be present in the program memory regardless of which program is loaded, e.g. tools and service routines.

- Choose **View: Modules** from the Program window.
- Select the system module *User* and press Enter [↵].
- Change, create data and routines in the normal way (see *Programming and Testing)*.

1.3 Deleting this data

Warning: If the Module is deleted, the CallByVar instruction will not work.

To delete all data (i.e. the entire module)

- Choose **View: Modules** from the Program window.
- Select the module *User.*
- Press Delete ⌫.

To change or delete individual data

- Choose **View: Data** from the Program window.
- Choose **Data: In All Modules**.
- Select the desired data. If this is not shown, press the ***Types*** function key to select the correct data type.
- Change or delete in the normal way (see *Programming and Testing)*.

INDEX

Glossary

Argument	The parts of an instruction that can be changed, i.e. everything except the name of the instruction.
Automatic mode	The applicable mode when the operating mode selector is set to  .
Component	One part of a record.
Configuration	The position of the robot axes at a particular location.
Constant	Data that can only be changed manually.
Corner path	The path generated when passing a fly-by point.
Declaration	The part of a routine or data that defines its properties.
Dialog/Dialog box	Any dialog boxes appearing on the display of the teach pendant must always be terminated (usually by pressing ***OK*** or ***Cancel***) before they can be exited.
Error handler	A separate part of a routine where an error can be taken care of. Normal execution can then be restarted automatically.
Expression	A sequence of data and associated operands; e.g. *reg1+5* or *reg1>5*.
Fly-by point	A point which the robot only passes in the vicinity of – without stopping. The distance to that point depends on the size of the programmed zone.
Function	A routine that returns a value.
Group of signals	A number of digital signals that are grouped together and handled as one signal.
Interrupt	An event that temporarily interrupts program execution and executes a trap routine.
I/O	Electrical inputs and outputs.
Main routine	The routine that usually starts when the ***Start*** key is pressed.
Manual mode	The applicable mode when the operating mode switch is set to  .
Mechanical unit	A group of external axes.
Module	A group of routines and data, i.e. a part of the program.
Motors On/Off	The state of the robot, i.e. whether or not the power supply to the motors is switched on.
Operator's panel	The panel located on the front of the control system.
Orientation	The direction of an end effector, for example.
Parameter	The input data of a routine, sent with the routine call. It corresponds to the argument of an instruction.
Persistent	A variable, the value of which is persistent.
Procedure	A routine which, when called, can independently form an instruction.

Program	The set of instructions and data which define the task of the robot. Programs do not, however, contain system modules.
Program data	Data that can be accessed in a complete module or in the complete program.
Program module	A module included in the robot's program and which is transferred when copying the program to a diskette, for example.
Record	A compound data type.
Routine	A subprogram.
Routine data	Local data that can only be used in a routine.
Start point	The instruction that will be executed first when starting program execution.
Stop point	A point at which the robot stops before it continues on to the next point.
System module	A module that is always present in the program memory. When a new program is read, the system modules remain in the program memory.
System parameters	The settings which define the robot equipment and properties; configuration data in other words.
Tool Centre Point (TCP)	The point, generally at the tip of a tool, that moves along the programmed path at the programmed velocity.
Trap routine	The routine that defines what is to be done when a specific interrupt occurs.
Variable	Data that can be changed from within a program, but which loses its value (returns to its initial value) when a program is started from the beginning.
Window	The robot is programmed and operated by means of a number of different windows, such as the Program window and the Service window. A window can always be exited by choosing another window.
Zone	The spherical space that surrounds a fly-by point. As soon as the robot enters this zone, it starts to move to the next position.

RAPID Reference Manual

SystemDataTypes and Routines

BaseWare OS 3.2

Continue

ABB Flexible Automation

ABB

ABB

3HAC 5761-1
For BaseWare OS 3.2

System Data Types and Routines

Data Types and System Data

Instructions

Functions

Index

Article number: 3HAC 5761-1
Issue: For BaseWare OS 3.2

ABB Robotics AB
S-721 68 Västerås
Sweden

CONTENTS

wobjdata	Work object data
wzstationary	Stationary world zone data
wztemporary	Temporary world zone data
zonedata	Zone data

aiotrigg Analog I/O trigger condition

aiotrigg (Analog I/O Trigger) is used to define the condition to generate an interrupt for an analog input or output signal.

Description

Data of the type *aiotrigg* defines the way a low and a high threshold will be used to determine whether the logical value of an analog signal satisfies a condition to generate an interrupt.

Example

```
VAR intnum sig1int;
CONNECT sig1int WITH iroutine1;
ISignalAI \Single, ai1, AIO_BETWEEN, 1.5, 0.5, 0, sig1int;
```

Orders an interrupt which is to occur the first time the logical value of the analog input signal *ai1* is between *0.5* and *1.5*. A call is then made to the *iroutine1* trap routine.

Predefined data

The following symbolic constants of the data type *aiotrigg* are predefined and can be used when specifying a condition for the instructions *ISignalAI* and *ISignalAO*.

Value	Symbolic constant	Comment
1	AIO_ABOVE_HIGH	Signal will generate interrupts if above specified high value
2	AIO_BELOW_HIGH	Signal will generate interrupts if below specified high value
3	AIO_ABOVE_LOW	Signal will generate interrupts if above specified low value
4	AIO_BELOW_LOW	Signal will generate interrupts if below specified low value
5	AIO_BETWEEN	Signal will generate interrupts if between specified low and high values
6	AIO_OUTSIDE	Signal will generate interrupts if below specified low value or above specified high value
7	AIO_ALWAYS	Signal will always generate interrupts

Characteristics

aiotrigg is an alias data type for *num* and consequently inherits its characteristics.

Related information

	Described in:
Interrupt from analog input signal	Instructions - *ISignalAI*
Interrupt from analog output signal	Instructions - *ISignalAO*
Data types in general, alias data types	Basic Characteristics - *Data Types*

bool Logical values

Bool is used for logical values (true/false).

Description

The value of data of the type *bool* can be either *TRUE* or *FALSE*.

Examples

```
flag1 := TRUE;
```

flag is assigned the value TRUE.

```
VAR bool highvalue;
VAR num reg1;
   .
highvalue := reg1 > 100;
```

highvalue is assigned the value *TRUE* if *reg1 is greater than 100*; otherwise, *FALSE* is assigned.

```
IF highvalue Set do1;
```

The *do1* signal is set if *highvalue* is *TRUE*.

```
highvalue := reg1 > 100;
mediumvalue := reg1 > 20 AND NOT highvalue;
```

mediumvalue is assigned the value *TRUE* if *reg1* is between *20* and *100*.

Related information

	Described in:
Logical expressions	Basic Characteristics - *Expressions*
Operations using logical values	Basic Characteristics - *Expressions*

byte Decimal values 0 - 255

Byte is used for decimal values (0 - 255) according to the range of a byte.

This data type is used in conjunction with instructions and functions that handle the bit manipulations and convert features.

Description

Data of the type *byte* represents a decimal byte value.

Examples

CONST num parity_bit := 8;

VAR byte data1 := 130;

Definition of a variable *data1* with a decimal value 130.

BitClear data1, parity_bit;

Bit number 8 (*parity_bit*) in the variable *data1* will be set to 0, e.g. the content of the variable *data1* will be changed from 130 to 2 (decimal representation).

Error handling

If an argument of the type *byte* has a value that is not in the range between 0 and 255, an error is returned on program execution.

Characteristics

Byte is an alias data type for *num* and consequently inherits its characteristics.

Related information

	Described in:
Alias data types	Basic Characteristics- *Data Types*
Bit functions	RAPID Summary - *Bit Functions*

clock Time measurement

Clock is used for time measurement. A *clock* functions like a stopwatch used for timing.

Description

Data of the type *clock* stores a time measurement in seconds and has a resolution of 0.01 seconds.

Example

```
VAR clock clock1;

    ClkReset clock1;
```

The clock, *clock1*, is declared and reset. Before using *ClkReset*, *ClkStart*, *ClkStop* and *ClkRead*, you must declare a variable of data type *clock* in your program.

Limitations

The maximum time that can be stored in a clock variable is approximately 49 days (4,294,967 seconds). The instructions *ClkStart*, *ClkStop* and *ClkRead* report clock overflows in the very unlikely event that one occurs.

A clock must be declared as a *VAR* variable type, not as a *persistent* variable type.

Characteristics

Clock is a non-value data type and cannot be used in value-oriented operations.

Related Information

	Described in:
Summary of Time and Date Instructions	RAPID Summary - *System & Time*
Non-value data type characteristics	Basic Characteristics - *Data Types*

confdata Robot configuration data

Confdata is used to define the axis configurations of the robot.

Description

All positions of the robot are defined and stored using rectangular coordinates. When calculating the corresponding axis positions, there will often be two or more possible solutions. This means that the robot is able to achieve the same position, i.e. the tool is in the same position and with the same orientation, with several different positions or configurations of the robots axes.

Some robot types use iterative numerical methods to determine the robot axes positions. In these cases the configuration parameters may be used to define good starting values for the joints to be used by the iterative procedure.

To unambiguously denote one of these possible configurations, the robot configuration is specified using four axis values. For a rotating axis, the value defines the current quadrant of the robot axis. The quadrants are numbered 0, 1, 2, etc. (they can also be negative). The quadrant number is connected to the current joint angle of the axis. For each axis, quadrant 0 is the first quarter revolution, 0 to 90°, in a positive direction from the zero position; quadrant 1 is the next revolution, 90 to 180°, etc. Quadrant -1 is the revolution 0° to (-90°), etc. (see Figure 1).



Figure 1 The configuration quadrants for axis 6.

For a linear axis, the value defines a meter interval for the robot axis. For each axis, value 0 means a position between 0 and 1 meters, 1 means a position between 1 and 2 meters. For negative values, -1 means a position between -1 and 0 meters, etc. (see Figure 2).



Figure 2 Configuration values for a linear axis.

Robot Configuration data for IRB540, 640

Only the configuration parameter cf6 is used.

Robot Configuration data for IRB1400, 2400, 3400, 4400, 6400

Only the three configuration parameters cf1, cf4 and cf6 are used.

Robot Configuration data for IRB5400

All four configuration parameters are used. cf1, cf4, cf6 for joints 1, 4, and 6 respectively and cfx for joint 5.

Robot configuration data for 6400C

The IRB 6400C requires a slightly different way of unambiguously denoting one robot configuration. The difference lies in the interpretation of the confdata *cf1*.

cf1 is used to select one of two possible main axes (axes 1, 2 and 3) configurations:

- cf1 = 0 is the forward configuration
- cf1 = 1 is the backward configuration

Figure 3 shows an example of a forward configuration and a backward configuration giving the same position and orientation.



Figure 3 Same position and orientation with two different main axes configurations.

The forward configuration is the front part of the robot's working area with the arm directed forward. The backward configuration is the service part of the working area with the arm directed backwards.

Robot configuration data for IRB5404, 5406

The robots have two rotation axes (arms 1 and 2) and one linear axis (arm 3).

cf1 is used for the rotating axis 1

cfx is used for the rotating axis 2

cf4 and cf6 are not used

Robot Configuration data for IRB5413, 5414, 5423

The robots have two linear axes (arms 1 and 2) and one or two rotating axes (arms 4 and 5) (Arm 3 locked)

cf1 is used for the linear axis 1

cfx is used for the linear axis 2

cf4 is used for the rotating axis 4

cf6 is not used

Robot configuration data for IRB840

The robot has three linear axes (arms 1, 2 and 3) and one rotating axis (arm 4).

cf1 is used for the linear axis 1

cfx is used for the linear axis 2

cf4 is used for the rotating axis 4

cf6 is not used

Because of the robot's mainly linear structure, the correct setting of the configuration parameters c1, cx is of less importance.

Components

cf1 Data type: *num*

Rotating axis:

The current quadrant of axis 1, expressed as a positive or negative integer.

Linear axis:

The current meter interval of axis 1, expressed as a positive or negative integer.

cf4 Data type: *num*

Rotating axis:

The current quadrant of axis 4, expressed as a positive or negative integer.

Linear axis:

The current meter interval of axis 4, expressed as a positive or negative integer.

cf6 Data type: *num*

Rotating axis:

The current quadrant of axis 6, expressed as a positive or negative integer.

Linear axis:

The current meter interval of axis 6, expressed as a positive or negative integer.

cfx Data type: *num*

Rotating axis:

For the IRB5400 robot, the current quadrant of axis 5, expressed as a positive or negative integer. For other robots, using the current quadrant of axis 2, expressed as a positive or negative integer.

Linear axis:

The current meter interval of axis 2, expressed as a positive or negative integer.

Example

VAR confdata conf15 := [1, -1, 0, 0]

A robot configuration *conf15* is defined as follows:

- The axis configuration of the robot axis 1 is quadrant *1*, i.e. 90-180^{o}.
- The axis configuration of the robot axis 4 is quadrant *-1*, i.e. 0-(-90^{o}).
- The axis configuration of the robot axis 6 is quadrant *0*, i.e. 0 - 90^{o}.
- The axis configuration of the robot axis 5 is quadrant *0*, i.e. 0 - 90^{o}.

Structure

< dataobject of *confdata* >
< *cf1* of *num* >
< *cf4* of *num* >
< *cf6* of *num* >
< *cf*x of *num* >

Related information

	Described in:
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Handling configuration data	Motion and I/O Principles - *Robot Configuration*

dionum Digital values 0 - 1

Dionum (digital input output numeric) is used for digital values (0 or 1).

This data type is used in conjunction with instructions and functions that handle digital input or output signals.

Description

Data of the type *dionum* represents a digital value 0 or 1.

Examples

CONST dionum close := 1;

Definition of a constant *close* with a value equal to *1*.

SetDO grip1, close;

The signal *grip1* is set to *close*, i.e. 1.

Error handling

If an argument of the type *dionum* has a value that is neither equal to 0 nor 1, an error is returned on program execution.

Characteristics

Dionum is an alias data type for *num* and consequently inherits its characteristics.

Related information

	Described in:
Summary input/output instructions	RAPID Summary - *Input and Output Signals*
Configuration of I/O	User's Guide - *System Parameters*
Alias data types	Basic Characteristics- *Data Types*

errnum Error number

Errnum is used to describe all recoverable (non fatal) errors that occur during program execution, such as division by zero.

Description

If the robot detects an error during program execution, this can be dealt with in the error handler of the routine. Examples of such errors are values that are too high and division by zero. The system variable *ERRNO*, of type *errnum*, is thus assigned different values depending on the nature of an error. The error handler may be able to correct an error by reading this variable and then program execution can continue in the correct way.

An error can also be created from within the program using the RAISE instruction. This particular type of error can be detected in the error handler by specifying an error number (within the range 1-90 or booked with instruction *BookErrNo*) as an argument to RAISE.

Examples

```
reg1 := reg2 / reg3;
.
ERROR
   IF ERRNO = ERR_DIVZERO THEN
      reg3 := 1;
      RETRY;
   ENDIF
```

If *reg3* = 0, the robot detects an error when division is taking place. This error, however, can be detected and corrected by assigning *reg3* the value *1*. Following this, the division can be performed again and program execution can continue.

```
CONST errnum machine_error := 1;
.
IF di1=0 RAISE machine_error;
.
ERROR
   IF ERRNO=machine_error RAISE;
```

An error occurs in a machine (detected by means of the input signal *di1*). A jump is made to the error handler in the routine which, in turn, calls the error handler of the calling routine where the error may possibly be corrected. The constant, *machine_error*, is used to let the error handler know exactly what type of error has occurred.

Predefined data

The system variable ERRNO can be used to read the latest error that occurred. A number of predefined constants can be used to determine the type of error that has occurred.

Name	Cause of error
ERR_ALRDYCNT	The interrupt variable is already connected to a TRAP routine
ERR_ARGDUPCND	More than one present conditional argument for the same parameter
ERR_ARGNAME	Argument is expression, not present or of type switch when executing ArgName
ERR_ARGNOTPER	Argument is not a persistent reference
ERR_ARGNOTVAR	Argument is not a variable reference
ERR_AXIS_ACT	Axis is not active
ERR_AXIS_IND	Axis is not independent
ERR_AXIS_MOVING	Axis is moving
ERR_AXIS_PAR	Parameter axis in instruction TestSign and SetCurrRef is wrong.
ERR_CALLIO_INTER	If an IOEnable or IODisable request is interrupted by another request to the same unit
ERR_CALLPROC	Procedure call error (not procedure) at runtime (late binding)
ERR_CNTNOTVAR	CONNECT target is not a variable reference
ERR_CNV_NOT_ACT	The conveyor is not activated.
ERR_CNV_CONNECT	The *WaitWobj* instruction is already active.
ERR_CNV_DROPPED	The object that the instruction *WaitWobj* was waiting for has been dropped.
ERR_DEV_MAXTIME	Timeout when executing a ReadBin, ReadNum or a ReadStr instruction
ERR_DIVZERO	Division by zero
ERR_EXCRTYMAX	Max. number of retries exceeded
ERR_EXECPHR	An attempt was made to execute an instruction using a place holder
ERR_FILEACC	A file is accessed incorrectly
ERR_FILEOPEN	A file cannot be opened
ERR_FILNOTFND	File not found
ERR_FNCNORET	No return value
ERR_FRAME	Unable to calculate new frame
ERR_ILLDIM	Incorrect array dimension
ERR_ILLQUAT	Attempt to use illegal orientation (quaternion) valve

ERR_ILLRAISE	Error number in RAISE out of range
ERR_INOMAX	No more interrupt numbers available
ERR_IODISABLE	Timeout when executing IODisable
ERR_IOENABLE	Timeout when executing IOEnable
ERR_IOERROR	I/O Error from instruction Save
ERR_LOADED	The program module is already loaded
ERR_LOADID_FATAL	Only internal use in LoadId
ERR_LOADID_RETRY	Only internal use in LoadId
ERR_MAXINTVAL	The integer value is too large
ERR_MODULE	Incorrect module name in instruction Save
ERR_NAME_INVALID	If the unit name does not exist or if the unit is not allowed to be disabled
ERR_NEGARG	Negative argument is not allowed
ERR_NOTARR	Data is not an array
ERR_NOTEQDIM	The array dimension used when calling the routine does not coincide with its parameters
ERR_NOTINTVAL	Not an integer value
ERR_NOTPRES	A parameter is used, despite the fact that the corresponding argument was not used at the routine call
ERR_OUTOFBND	The array index is outside the permitted limits
ERR_PATH	Missing destination path in instruction Save
ERR_PATHDIST	Too long regain distance for StartMove instruction
ERR_PID_MOVESTOP	Only internal use in LoadId
ERR_PID_RAISE_PP	Error from ParIdRobValid or ParIdPosValid
ERR_RANYBIN_EOF	End of file is detected before all bytes are read in instruction ReadAnyBin
ERR_RCVDATA	An attempt was made to read non numeric data with ReadNum
ERR_REFUNKDAT	Reference to unknown entire data object
ERR_REFUNKFUN	Reference to unknown function
ERR_REFUNKPRC	Reference to unknown procedure at linking time or at run time (late binding)
ERR_REFUNKTRP	Reference to unknown trap
ERR_SC_WRITE	Error when sending to external computer
ERR_SIGSUPSEARCH	The signal has already a positive value at the beginning of the search process
ERR_STEP_PAR	Parameter Step in SetCurrRef is wrong
ERR_STRTOOLNG	The string is too long
ERR_SYM_ACCESS	Symbol read/write access error
ERR_TP_DIBREAK	A TPRead instruction was interrupted by a digital input
ERR_TP_MAXTIME	Timeout when executing a TPRead instruction

ERR_UNIT_PAR	Parameter Mech_unit in TestSign and SetCurrRef is wrong
ERR_UNKINO	Unknown interrupt number
ERR_UNKPROC	Incorrect reference to the load session in instruction WaitLoad
ERR_UNLOAD	Unload error in instruction UnLoad or WaitLoad
ERR_WAIT_MAXTIME	Timeout when executing a WaitDI or WaitUntil instruction
ERR_WHLSEARCH	No search stop

Characteristics

Errnum is an alias data type for *num* and consequently inherits its characteristics.

Related information

	Described in:
Error recovery	RAPID Summary - *Error Recovery* Basic Characteristics - *Error Recovery*
Data types in general, alias data types	Basic Characteristics - *Data Types*

extjoint Position of external joints

Extjoint is used to define the axis positions of external axes, positioners or workpiece manipulators.

Description

The robot can control up to six external axes in addition to its six internal axes, i.e. a total of twelve axes. The six external axes are logically denoted: a, b, c, d, e, f. Each such logical axis can be connected to a physical axis and, in this case, the connection is defined in the system parameters.

Data of the type *extjoint* is used to hold position values for each of the logical axes a - f.

For each logical axis connected to a physical axis, the position is defined as follows:

- For rotating axes – the position is defined as the rotation in degrees from the calibration position.
- For linear axes – the position is defined as the distance in mm from the calibration position.

If a logical axis is not connected to a physical one, the value 9E9 is used as a position value, indicating that the axis is not connected. At the time of execution, the position data of each axis is checked and it is checked whether or not the corresponding axis is connected. If the stored position value does not comply with the actual axis connection, the following applies:

- If the position is not defined in the position data (value is 9E9), the value will be ignored if the axis is connected and not activated. But if the axis is activated, it will result in an error.
- If the position is defined in the position data, although the axis is not connected, the value will be ignored.

If an external axis offset is used (instruction *EOffsOn* or *EOffsSet*), the positions are specified in the ExtOffs coordinate system.

Components

eax_a *(external axis a)* Data type: *num*

The position of the external logical axis “a”, expressed in degrees or mm (depending on the type of axis).

eax_b *(external axis b)* Data type: *num*

The position of the external logical axis “b”, expressed in degrees or mm (depending on the type of axis).

...

eax_f *(external axis f)* Data type: *num*

The position of the external logical axis “f”, expressed in degrees or mm (depending on the type of axis).

Example

VAR extjoint axpos10 := [11, 12.3, 9E9, 9E9, 9E9, 9E9] ;

The position of an external positioner, *axpos10*, is defined as follows:

- The position of the external logical axis “a” is set to 11, expressed in degrees or mm (depending on the type of axis).
- The position of the external logical axis “b” is set to 12.3, expressed in degrees or mm (depending on the type of axis).
- Axes c to f are undefined.

Structure

< dataobject of *extjoint* >
< *eax_a* of *num* >
< *eax_b* of *num* >
< *eax_c* of *num* >
< *eax_d* of *num* >
< *eax_e* of *num* >
< *eax_f* of *num* >

Related information

	Described in:
Position data	Data Types - *robtarget*
ExtOffs coordinate system	Instructions - *EOffsOn*

intnum Interrupt identity

Intnum (interrupt numeric) is used to identify an interrupt.

Description

When a variable of type *intnum* is connected to a trap routine, it is given a specific value identifying the interrupt. This variable is then used in all dealings with the interrupt, such as when ordering or disabling an interrupt.

More than one interrupt identity can be connected to the same trap routine. The system variable *INTNO* can thus be used in a trap routine to determine the type of interrupt that occurs.

Examples

```
VAR intnum feeder_error;
.
CONNECT feeder_error WITH correct_feeder;
ISignalDI di1, 1, feeder_error;
```

An interrupt is generated when the input *di1* is set to *1*. When this happens, a call is made to the *correct_feeder* trap routine.

```
VAR intnum feeder1_error;
VAR intnum feeder2_error;
.
PROC init_interrupt();
.
   CONNECT feeder1_error WITH correct_feeder;
   ISignalDI di1, 1, feeder1_error;
   CONNECT feeder2_error WITH correct_feeder;
   ISignalDI di2, 1, feeder2_error;
.
ENDPROC
.
TRAP correct_feeder
   IF INTNO=feeder1_error THEN
   .
   ELSE
   .
   ENDIF
.
ENDTRAP
```

An interrupt is generated when either of the inputs *di1* or *di2* is set to *1*. A call is then made to the *correct_feeder* trap routine. The system variable INTNO is used in the trap routine to find out which type of interrupt has occurred.

Limitations

The maximum number of active variables of type *intnum* at any one time (between *CONNECT* and *IDelete*) is limited to 40.The maximum number of interrupts, in the queue for execution of *TRAP* routine at any one time, is limited to 30.

Characteristics

Intnum is an alias data type for *num* and thus inherits its properties.

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Alias data types	Basic Characteristics- *Data Types*

iodev Serial channels and files

Iodev (I/O device) is used for serial channels, such as printers and files.

Description

Data of the type *iodev* contains a reference to a file or serial channel. It can be linked to the physical unit by means of the instruction *Open* and then used for reading and writing.

Example

```
VAR iodev file;
.
Open "flp1:LOGDIR/INFILE.DOC", file\Read;
input := ReadNum(file);
```

The file *INFILE.DOC* is opened for reading. When reading from the file, *file* is used as a reference instead of the file name.

Characteristics

Iodev is a non-value data type.

Related information

	Described in:
Communication via serial channels	RAPID Summary - *Communication*
Configuration of serial channels	User’s Guide - *System Parameters*
Characteristics of non-value data types	Basic Characteristics - *Data Types*

jointtarget Joint position data

Jointtarget is used to define the position that the robot and the external axes will move to with the instruction *MoveAbsJ*.

Description

Jointtarget defines each individual axis position, for both the robot and the external axes.

Components

robax *(robot axes)* Data type: *robjoint*

Axis positions of the robot axes in degrees.

Axis position is defined as the rotation in degrees for the respective axis (arm) in a positive or negative direction from the axis calibration position.

extax *(external axes)* Data type: *extjoint*

The position of the external axes.

The position is defined as follows for each individual axis (*eax_a*, *eax_b* ... *eax_f*):

- For rotating axes, the position is defined as the rotation in degrees from the calibration position.
- For linear axes, the position is defined as the distance in mm from the calibration position.

External axes *eax_a* ... are logical axes. How the logical axis number and the physical axis number are related to each other is defined in the system parameters.

The value 9E9 is defined for axes which are not connected. If the axes defined in the position data differ from the axes that are actually connected on program execution, the following applies:

- If the position is not defined in the position data (value 9E9) the value will be ignored, if the axis is connected and not activated. But if the axis is activated it will result in error.
- If the position is defined in the position data yet the axis is not connected, the value is ignored.

Examples

```
CONST jointtarget calib_pos := [ [ 0, 0, 0, 0, 0, 0], [ 0, 9E9, 9E9, 9E9, 9E9, 9E9] ];
```

The normal calibration position for IRB2400 is defined in *calib_pos* by the data type *jointtarget.* The normal calibration position 0 (degrees or mm) is also defined for the external logical axis a. The external axes b to f are undefined.

Structure

```
< dataobject of jointtarget >
   < robax of robjoint >
      < rax_1 of num >
      < rax_2 of num >
      < rax_3 of num >
      < rax_4 of num >
      < rax_5 of num >
      < rax_6 of num >
   < extax of extjoint >
      < eax_a of num >
      < eax_b of num >
      < eax_c of num >
      < eax_d of num >
      < eax_e of num >
      < eax_f of num >
```

Related information

	Described in:
Move to joint position	Instructions - *MoveAbsJ*
Positioning instructions	RAPID Summary - *Motion*
Configuration of external axes	User’s Guide - *System Parameters*

loaddata Load data

Loaddata is used to describe loads attached to the mechanical interface of the robot (the robot's mounting flange).

Load data usually defines the payload (grip load is defined by the instruction *GripLoad*) of the robot, i.e. the load held in the robot gripper. The tool load is specified in the tool data (*tooldata*) which includes load data.

Description

Specified loads are used to set up a model of the dynamics of the robot so that the robot movements can be controlled in the best possible way.

It is important to always define the actual tool load and when used, the payload of the robot too. Incorrect definitions of load data can result in overloading of the robot mechanical structure.

When incorrect load data is specified, it can often lead to the following consequences:

- If the value in the specified load data is greater than that of the value of the true load;
  -> The robot will not be used to its maximum capacity
  -> Impaired path accuracy including a risk of overshooting
  -> Risk of overloading the mechanical structure
- If the value in the specified load data is less than the value of the true load;
  -> Impaired path accuracy including a risk of overshooting
  -> Risk of overloading the mechanical structure

The payload is connected/disconnected using the instruction *GripLoad*.

Components

mass Data type: *num*

The weight of the load in kg.

cog *(centre of gravity)* Data type: *pos*

The centre of gravity of a tool load expressed in the wrist coordinate system. If a stationary tool is used, it means the centre of gravity for the tool holding the work object.

The centre of gravity of a payload expressed in the tool coordinate system. The object coordinate system when a stationary tool is used.

aom *(axes of moment)* Data type: *orient*

Tool load (Ref. to Figure 4)

The orientation of the coordinate system defined by the inertial axes of the tool load. Expressed in the wrist coordinate system as a quaternion (q1, q2, q3, q4). If a stationary tool is used, it means the inertial axes for the tool holding the work object.

The orientation of the tool load coordinate system must coincide with the orientation of the wrist coordinate system. **It must always be set to 1, 0, 0, 0.**

Pay load (Ref. to figure 1 and 2)

The orientation of the coordinate system defined by the inertial axes of the payload. Expressed in the tool coordinate system as a quaternion (q1, q2, q3, q4). The object coordinate system if a stationary tool is used.

The orientation of the payload coordinate system must coincide with the orientation of the wrist coordinate system. **It must always be set to 1, 0, 0, 0**.

Because of this limitation, the best way is to define the orientation of the tool coordinate system (*tool frame*) to coincide with the orientation of the wrist coordinate system.



Figure 4 Restriction on the orientation of tool load and payload coordinate system.



Figure 5 The centre of gravity and inertial axes of the payload.

ix *(inertia x)* Data type: *num*

The moment of inertia of the load around the x-axis of the tool load or payload coordinate system in kgm^2.

Correct definition of the inertial moments will allow optimal utilisation of the path planner and axes control. This may be of special importance when handling large sheets of metal, etc. All inertial moments of inertia *ix, iy* and *iz* equal to 0 kgm^2 imply a point mass.

Normally, the inertial moments must only be defined when the distance from the mounting flange to the centre of gravity is less than the dimension of the load (see Figure 6).



Figure 6 The moment of inertia must normally be defined when the distance is less than the load dimension.

iy *(inertia y)* Data type: *num*

The inertial moment of the load around the y-axis, expressed in kgm^2.

For more information, see *ix*.

iz *(inertia z)* Data type: *num*

The inertial moment of the load around the z-axis, expressed in kgm^2.

For more information, see *ix*.

Examples

PERS loaddata piece1 := [5, [50, 0, 50], [1, 0, 0, 0], 0, 0, 0];

The payload in Figure 4 is described using the following values:

- Weight *5* kg.
- The centre of gravity is x = *50*, y = *0* and z = *50* mm in the tool coordinate system.
- The payload is a point mass.

```
Set gripper;
WaitTime 0.3;
GripLoad piece1;
```

Connection of the payload, *piece1*, specified at the same time as the robot grips the load *piece1*.

```
Reset gripper;
WaitTime 0.3;
GripLoad load0;
```

Disconnection of a payload, specified at the same time as the robot releases a payload.

Limitations

The payload should only be defined as a persistent variable (PERS) and not within a routine. Current values are then saved when storing the program on diskette and are retrieved on loading.

Arguments of the type load data in the *GripLoad* instruction should only be an entire persistent (not array element or record component).

Predefined data

The load *load0* defines a payload, the weight of which is equal to 0 kg, i.e. no load at all. This load is used as the argument in the instruction *GripLoad* to disconnect a payload.

The load *load0* can always be accessed from the program, but cannot be changed (it is stored in the system module *BASE*).

PERS loaddata load0 := [0.001, [0, 0, 0.001], [1, 0, 0, 0],0, 0 ,0];

Structure

< dataobject of *loaddata* >
< *mass* of *num* >
< *cog* of *pos* >
< *x* of *num* >
< *y* of *num* >
< *z* of *num* >
< *aom* of *orient* >
< *q1* of *num* >
< *q2* of *num* >
< *q3* of *num* >
< *q4* of *num* >
< *ix* of *num* >
< *iy* of *num* >
< *iz* of *num* >

Related information

	Described in:
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Definition of tool loads	Data Types - *tooldata*
Activation of payload	Instructions - *GripLoad*

loadsession Program load session

Loadsession is used to define different load sessions of RAPID program modules.

Description

Data of the type *loadsession* is used in the instructions *StartLoad* and *WaitLoad*, to identify the load session. *Loadsession* only contains a reference to the load session.

Characteristics

Loadsession is a *non-value* data type and cannot be used in value-oriented operations.

Related information

	Described in:
Loading program modules during execution	Instructions - *StartLoad, WaitLoad*
Characteristics of non-value data types	Basic Characteristics - *Data Types*

mecunit Mechanical unit

Mecunit is used to define the different mechanical units which can be controlled and accessed from the robot and the program.

The names of the mechanical units are defined in the system parameters and, consequently, must not be defined in the program.

Description

Data of the type *mecunit* only contains a reference to the mechanical unit.

Limitations

Data of the type *mecunit* must not be defined in the program. The data type can, on the other hand, be used as a parameter when declaring a routine.

Predefined data

The mechanical units defined in the system parameters can always be accessed from the program (installed data).

Characteristics

Mecunit is a *non-value* data type. This means that data of this type does not permit value-oriented operations.

Related information

	Described in:
Activating/Deactivating mechanical units	Instructions - *ActUnit, DeactUnit*
Configuration of mechanical units	User's Guide - *System Parameters*
Characteristics of non-value data types	Basic Characteristics - *Data Types*

motsetdata Motion settings data

Motsetdata is used to define a number of motion settings that affect all positioning instructions in the program:

- Max. velocity and velocity override
- Acceleration data
- Behavior around singular points
- Management of different robot configurations
- Override of path resolution
- Motion supervision

This data type does not normally have to be used since these settings can only be set using the instructions *VelSet*, *AccSet*, *SingArea*, *ConfJ*, *ConfL*, *PathResol* and *MotionSup*.

The current values of these motion settings can be accessed using the system variable *C_MOTSET.*

Description

The current motion settings (stored in the system variable *C_MOTSET*) affect all movements.

Components

vel.oride Data type: *veldata/num*

Velocity as a percentage of programmed velocity.

vel.max Data type: *veldata/num*

Maximum velocity in mm/s.

acc.acc Data type: *accdata/num*

Acceleration and deceleration as a percentage of the normal values.

acc.ramp Data type: *accdata/num*

The rate by which acceleration and deceleration increases as a percentage of the normal values.

sing.wrist Data type: *singdata/bool*

The orientation of the tool is allowed to deviate somewhat in order to prevent wrist singularity.

sing.arm Data type: *singdata/bool*

The orientation of the tool is allowed to deviate somewhat in order to prevent arm singularity (not implemented).

sing.base Data type: *singdata/bool*

The orientation of the tool is not allowed to deviate.

conf.jsup Data type: *confsupdata/bool*

Supervision of joint configuration is active during joint movement.

conf.lsup Data type: *confsupdata/bool*

Supervision of joint configuration is active during linear and circular movement.

conf.ax1 Data type: *confsupdata/num*

Maximum permitted deviation in degrees for axis 1 (not used in this version).

conf.ax4 Data type: *confsupdata/num*

Maximum permitted deviation in degrees for axis 4 (not used in this version).

conf.ax6 Data type: *confsupdata/num*

Maximum permitted deviation in degrees for axis 6 (not used in this version).

pathresol Data type: *num*

Current override in percentage of the configured path resolution.

motionsup Data type: *bool*

Mirror RAPID status (TRUE = On and FALSE = Off) of motion supervision function.

tunevalue Data type**:** *num*

Current RAPID override as a percentage of the configured tunevalue for the motion supervision function.

Limitations

One and only one of the components *sing.wrist*, *sing.arm* or *sing.base* may have a value equal to TRUE.

Example

```
IF C_MOTSET.vel.oride > 50 THEN
   ...
ELSE
   ...
ENDIF
```

Different parts of the program are executed depending on the current velocity override.

Predefined data

C_MOTSET describes the current motion settings of the robot and can always be accessed from the program (installed data). *C_MOTSET*, on the other hand, can only be changed using a number of instructions, not by assignment.

The following default values for motion parameters are set

- at a cold start-up
- when a new program is loaded
- when starting program execution from the beginning.

```
PERS motsetdata C_MOTSET := [
   [ 100, 500 ],-> veldata
   [ 100, 100 ],-> accdata
   [ FALSE, FALSE, TRUE ],-> singdata
   [ TRUE, TRUE, 30, 45, 90],-> confsupdata
   [100 ],-> path resolution
   [TRUE ],-> motionsup
   [100 ] ];-> tunevalue
```

Structure

```
<dataobject of motsetdata>
   <vel of veldata >          -> Affected by instruction VelSet
      < oride of num >
      < max of num >
   <acc of accdata >          -> Affected by instruction AccSet
      < acc of num >
      < ramp of num >
   <sing of singdata >        -> Affected by instruction SingArea
      < wrist of bool >
      < arm of bool >
      < base of bool >
   <conf of confsupdata >     -> Affected by instructions ConfJ and ConfL
      < jsup of bool >
      <lsup of bool >
      < ax1 of num >
      < ax4 of num >
      < ax6 of num >
   <pathresol of num>         -> Affected by instruction PathResol
   <motionsup of bool>        -> Affected by instruction MotionSup
   <tunevalue of num>         -> Affected by instruction MotionSup
```

Related information

	Described in:
Instructions for setting motion parameters	RAPID Summary - *Motion Settings*

num Numeric values (registers)

Num is used for numeric values; e.g. counters.

Description

The value of the *num* data type may be

- an integer; e.g. -5,
- a decimal number; e.g. 3.45.

It may also be written exponentially; e.g.2E3 (= $2*10^3$ = 2000), 2.5E-2 (= 0.025).

Integers between -8388607 and +8388608 are always stored as exact integers.

Decimal numbers are only approximate numbers and should not, therefore, be used in *is equal to* or *is not equal to* comparisons. In the case of divisions, and operations using decimal numbers, the result will also be a decimal number; i.e. not an exact integer.

E.g.

```
a := 10;
b := 5;
IF a/b=2 THEN
...
```

As the result of a/b is not an integer, this condition is not necessarily satisfied.

Example

```
VAR num reg1;
  .
reg1 := 3;
```

reg1 is assigned the value *3*.

```
a := 10 DIV 3;
b := 10 MOD 3;
```

Integer division where *a* is assigned an integer (=3) and *b* is assigned the remainder (=1).

Predefined data

The constant pi (π) is already defined in the system module *BASE*.

CONST num pi := 3.1415926;

The constants EOF_BIN and EOF_NUM are already defined in the system.

CONST num EOF_BIN := -1;

CONST num EOF_NUM := 9.998E36;

Related information

	Described in:
Numeric expressions	Basic Characteristics - *Expressions*
Operations using numeric values	Basic Characteristics - *Expressions*

o_jointtarget Original joint position data

o_jointtarget (original joint target) is used in combination with the function *Absolute Limit Modpos*. When this function is used to modify a position, the original position is stored as a data of the type *o_jointtarget*.

Description

If the function *Absolute Limit Modpos* is activated and a named position in a movement instruction is modified with the function *Modpos,* then the original programmed position is saved.

Example of a program before *Modpos*:

```
CONST jointtarget jpos40     := [[0, 0, 0, 0, 0, 0],
                                 [0, 9E9, 9E9, 9E9, 9E9, 9E9]];
...
MoveAbsJ jpos40, v1000, z50, tool1;
```

The same program after *ModPos* in which the point *jpos40 is* corrected to 2 degrees for robot axis 1:

```
CONST jointtarget jpos40        := [[2, 0, 0, 0, 0, 0],
                                    [0, 9E9, 9E9, 9E9, 9E9, 9E9]];
CONST o_jointtarget o_jpos40 := [[0, 0, 0, 0, 0, 0],
                                    [0, 9E9, 9E9, 9E9, 9E9, 9E9]];
...
MoveAbsJ jpos40, v1000, z50, tool1;
```

The original programmed point has now been saved in *o_jpos40* (by the data type *o_jointtarget*) and the modified point saved in *jpos40* (by the data type *jointtarget*).

By saving the original programmed point, the robot can monitor that further *Modpos* of the point in question are within the acceptable limits from the original programmed point.

The fixed name convention means that an original programmed point with the name *xxxxx* is saved with the name *o_xxxxx* by using *Absolute Limit Modpos*.

Components

robax *(robot axes)* Data type: *robjoint*

Axis positions of the robot axes in degrees.

extax *(external axes)* Data type: *extjoint*

The position of the external axes.

Structure

```
< dataobject of o_jointtarget >
   < robax of robjoint>
      < rax_1 of num >
      < rax_2 of num >
      < rax_3 of num >
      < rax_4 of num >
      < rax_5 of num >
      < rax_6 of num >
   < extax of extjoint >
      < eax_a of num >
      < eax_b of num >
      < eax_c of num >
      < eax_d of num >
      < eax_e of num >
      < eax_f of num >
```

Related information

	Described in:
Position data	Data Types - *Jointtarget*
Configuration of Limit Modpos	User’s Guide - *System Parameters*

orient Orientation

Orient is used for orientations (such as the orientation of a tool) and rotations (such as the rotation of a coordinate system).

Description

The orientation is described in the form of a quaternion which consists of four elements: *q1*, *q2*, *q3* and *q4*. For more information on how to calculate these, see below.

Components

q1 Data type: *num*

Quaternion 1.

q2 Data type: *num*

Quaternion 2.

q3 Data type: *num*

Quaternion 3.

q4 Data type: *num*

Quaternion 4.

Example

```
VAR orient orient1;
.
orient1 := [1, 0, 0, 0];
```

The *orient1* orientation is assigned the value q1=1, q2-q4=0; this corresponds to no rotation.

Limitations

The orientation must be normalised; i.e. the sum of the squares must equal 1:

$$q_1^2 + q_2^2 + q_3^2 + q_4^2 = 1$$

What is a Quaternion?

The orientation of a coordinate system (such as that of a tool) can be described by a rotational matrix that describes the direction of the axes of the coordinate system in relation to a reference system (see Figure 7).



Figure 7 The rotation of a coordinate system is described by a quaternion.

The rotated coordinate systems axes (**x**, **y**, **z**) are vectors which can be expressed in the reference coordinate system as follows:

$\mathbf{x} = (x_1, x_2, x_3)$

$\mathbf{y} = (y_1, y_2, y_3)$

$\mathbf{z} = (z_1, z_2, z_3)$

This means that the x-component of the x-vector in the reference coordinate system will be x_1, the y-component will be x_2, etc.

These three vectors can be put together in a matrix, a rotational matrix, where each of the vectors form one of the columns:

$$\begin{bmatrix} x_1 & y_1 & z_1 \\ x_2 & y_2 & z_2 \\ x_3 & y_3 & z_3 \end{bmatrix}$$

A quaternion is just a more concise way to describe this rotational matrix; the quaternions are calculated based on the elements of the rotational matrix:

$$q1 = \frac{\sqrt{x_1 + y_2 + z_3 + 1}}{2}$$

$$q2 = \frac{\sqrt{x_1 - y_2 - z_3 + 1}}{2} \qquad \text{sign } q2 = \text{sign } (y_3\text{-}z_2)$$

$$q3 = \frac{\sqrt{y_2 - x_1 - z_3 + 1}}{2} \qquad \text{sign } q3 = \text{sign } (z_1\text{-}x_3)$$

$$q4 = \frac{\sqrt{z_3 - x_1 - y_2 + 1}}{2} \qquad \text{sign } q4 = \text{sign } (x_2\text{-}y_1)$$

Example 1

A tool is orientated so that its Z'-axis points straight ahead (in the same direction as the X-axis of the base coordinate system). The Y'-axis of the tool corresponds to the Y-axis of the base coordinate system (see Figure 8). How is the orientation of the tool defined in the position data (robtarget)?

The orientation of the tool in a programmed position is normally related to the coordinate system of the work object used. In this example, no work object is used and the base coordinate system is equal to the world coordinate system. Thus, the orientation is related to the base coordinate system.



Figure 8 The direction of a tool in accordance with example 1.

The axes will then be related as follows:

$\mathbf{x}' = -\mathbf{z} = (0, 0, -1)$

$\mathbf{y}' = \mathbf{y} = (0, 1, 0)$

$\mathbf{z}' = \mathbf{x} = (1, 0, 0)$

Which corresponds to the following rotational matrix: $\begin{bmatrix} 0 & 0 & 1 \\ 0 & 1 & 0 \\ -1 & 0 & 0 \end{bmatrix}$

The rotational matrix provides a corresponding quaternion:

$$q1 = \frac{\sqrt{0+1+0+1}}{2} = \frac{\sqrt{2}}{2} = 0{,}707$$

$$q2 = \frac{\sqrt{0-1-0+1}}{2} = 0$$

$$q3 = \frac{\sqrt{1-0-0+1}}{2} = \frac{\sqrt{2}}{2} = 0{,}707 \qquad \text{sign } q3 = \text{sign } (1+1) = +$$

$$q4 = \frac{\sqrt{0-0-1+1}}{2} = 0$$

Example 2

The direction of the tool is rotated 30^{o} about the X'- and Z'-axes in relation to the wrist coordinate system (see Figure 8). How is the orientation of the tool defined in the tool data?



Figure 9 The direction of the tool in accordance with example 2.

The axes will then be related as follows:

x' = (cos30^o, 0, -sin30^o)

x' = (0, 1, 0)

x' = (sin30^o, 0, cos30^o)

Which corresponds to the following rotational matrix: $\begin{bmatrix} \cos 30° & 0 & \sin 30° \\ 0 & 1 & 0 \\ -\sin 30° & 0 & \cos 30° \end{bmatrix}$

The rotational matrix provides a corresponding quaternion:

$$q1 = \frac{\sqrt{\cos 30° + 1 + \cos 30° + 1}}{2} = 0{,}965926$$

$$q2 = \frac{\sqrt{\cos 30° - 1 - \cos 30° + 1}}{2} = 0$$

$$q3 = \frac{\sqrt{1 - \cos 30° - \cos 30° + 1}}{2} = 0{,}258819$$

sign q3 = sign (sin30^o+sin30^o) = +

$$q4 = \frac{\sqrt{\cos 30° - \cos 30° - 1 + 1}}{2} = 0$$

Structure

<dataobject of *orient*>
  <*q1* of *num*>
  <*q2* of *num*>
  <*q3* of *num*>
  <*q4* of *num*>

Related information

	Described in:
Operations on orientations	Basic Characteristics - *Expressions*

o_robtarget Original position data

o_robtarget (*original robot target*) is used in combination with the function *Absolute Limit Modpos*. When this function is used to modify a position, the original position is stored as a data of the type *o_robtarget*.

Description

If the function *Absolute Limit Modpos* is activated and a named position in a movement instruction is modified with the function *Modpos*, then the original programmed position is saved.

Example of a program before *Modpos*:

```
CONST robtarget p50       := [[500, 500, 500], [1, 0, 0, 0], [1, 1, 0, 0],
                              [500, 9E9, 9E9, 9E9, 9E9, 9E9] ];
...
MoveL p50, v1000, z50, tool1;
```

The same program after *ModPos* in which the point *p50 is* corrected to 502 in the x-direction:

```
CONST robtarget p50       := [[502, 500, 500], [1, 0, 0, 0], [1, 1, 0, 0],
                              [500, 9E9, 9E9, 9E9, 9E9, 9E9] ];
CONST o_robtarget o_p50   := [[500, 500, 500], [1, 0, 0, 0], [1, 1, 0, 0],
                              [ 500, 9E9, 9E9, 9E9, 9E9, 9E9] ];
...
MoveL p50, v1000, z50, tool1;
```

The original programmed point has now been saved in *o_p50* (by the data type *o_robtarget*) and the modified point saved in *p50* (by the data type *robtarget*).

By saving the original programmed point, the robot can monitor that further *Modpos* of the point in question are within the acceptable limits from the original programmed point.

The fixed name convention means that an original programmed point with the name *xxxxx* is saved with the name *o_xxxxx* by using *Absolute Limit Modpos*.

Components

trans *(translation)* Data type: *pos*

The position (x, y and z) of the tool centre point expressed in mm.

rot *(rotation)* Data type: *orient*

The orientation of the tool, expressed in the form of a quaternion (q1, q2, q3 and q4).

robconf *(robot configuration)* Data type: *confdata*

The axis configuration of the robot (cf1, cf4, cf6 and cfx).

extax *(external axes)* Data type: *extjoint*

The position of the external axes.

Structure

```
< dataobject of o_robtarget >
   < trans of pos >
      < x of num >
      < y of num >
      < z of num >
   < rot of orient >
      < q1 of num >
      < q2 of num >
      < q3 of num >
      < q4 of num >
   < robconf of confdata >
      < cf1 of num >
      < cf4 of num >
      < cf6 of num >
      < cfx of num >
   < extax of extjoint >
      < eax_a of num >
      < eax_b of num >
      < eax_c of num >
      < eax_d of num >
      < eax_e of num >
      < eax_f of num >
```

Related information

	Described in:
Position data	Data Types - *Robtarget*
Configuration of Limit Modpos	User's Guide - *System Parameters*

pos Positions (only X, Y and Z)

Pos is used for positions (only X, Y and Z).

The *robtarget* data type is used for the robot's position including the orientation of the tool and the configuration of the axes.

Description

Data of the type *pos* describes the coordinates of a position: X, Y and Z.

Components

x Data type: *num*

The X-value of the position.

y Data type: *num*

The Y-value of the position.

z Data type: *num*

The Z-value of the position.

Examples

```
VAR pos pos1;
.
pos1 := [500, 0, 940];
```

The *pos1* position is assigned the value: X=500 mm, Y=0 mm, Z=940 mm.

```
pos1.x := pos1.x + 50;
```

The *pos1* position is shifted *50* mm in the X-direction.

Structure

```
<dataobject of pos>
   <x of num>
   <y of num>
   <z of num>
```

Related information

	Described in:
Operations on positions	Basic Characteristics - *Expressions*
Robot position including orientation	Data Types- *robtarget*

pose Coordinate transformations

Pose is used to change from one coordinate system to another.

Description

Data of the type *pose* describes how a coordinate system is displaced and rotated around another coordinate system. The data can, for example, describe how the tool coordinate system is located and oriented in relation to the wrist coordinate system.

Components

trans *(translation)* Data type: *pos*

The displacement in position (x, y and z) of the coordinate system.

rot *(rotation)* Data type: *orient*

The rotation of the coordinate system.

Example

```
VAR pose frame1;
.
frame1.trans := [50, 0, 40];
frame1.rot := [1, 0, 0, 0];
```

The *frame1* coordinate transformation is assigned a value that corresponds to a displacement in position, where X=*50* mm, Y=*0* mm, Z=*40* mm; there is, however, no rotation.

Structure

```
<dataobject of pose>
    <trans of pos>
    <rot of orient>
```

Related information

	Described in:
What is a Quaternion?	Data Types - *orient*

progdisp Program displacement

Progdisp is used to store the current program displacement of the robot and the external axes.

This data type does not normally have to be used since the data is set using the instructions *PDispSet, PDispOn, PDispOff, EOffsSet, EOffsOn* and *EOffsOff*. It is only used to temporarily store the current value for later use.

Description

The current values for program displacement can be accessed using the system variable *C_PROGDISP.*

For more information, see the instructions *PDispSet, PDispOn, EOffsSet* and *EOffsOn*.

Components

pdisp *(program displacement)* Data type: *pose*

The program displacement for the robot, expressed using a translation and an orientation. The translation is expressed in mm.

eoffs *(external offset)* Data type: *extjoint*

The offset for each of the external axes. If the axis is linear, the value is expressed in mm; if it is rotating, the value is expressed in degrees.

Example

```
VAR progdisp progdisp1;
.
SearchL sen1, psearch, p10, v100, tool1;
PDispOn \ExeP:=psearch, *, tool1;
EOffsOn \ExeP:=psearch, *;
.
progdisp1:=C_PROGDISP;
PDispOff;
EOffsOff;
.
PDispSet progdisp1.pdisp;
EOffsSet progdisp1.eoffs;
```

First, a program displacement is activated from a searched position. Then, it is temporarily deactivated by storing the value in the variable *progdisp1* and, later on, re-activated using the instructions *PDispSet* and *EOffsSet*.

Predefined data

The system variable *C_PROGDISP* describes the current program displacement of the robot and external axes, and can always be accessed from the program (installed data). *C_PROGDISP*, on the other hand, can only be changed using a number of instructions, not by assignment.

Structure

< dataobject of *progdisp* >
<*pdisp* of *pose*>
< *trans* of *pos* >
< *x* of *num* >
< *y* of *num* >
< *z* of *num* >
< *rot* of *orient* >
< *q1* of *num* >
< *q2* of *num* >
< *q3* of *num* >
< *q4* of *num* >
< *eoffs* of *extjoint* >
< *eax_a* of *num* >
< *eax_b* of *num* >
< *eax_c* of *num* >
< *eax_d* of *num* >
< *eax_e* of *num* >
< *eax_f* of *num* >

Related information

	Described in:
Instructions for defining program displacement	RAPID Summary - *Motion Settings*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*

robjoint Joint position of robot axes

Robjoint is used to define the axis position in degrees of the robot axes.

Description

Data of the type *robjoint* is used to store axis positions in degrees of the robot axes 1 to 6. Axis position is defined as the rotation in degrees for the respective axis (arm) in a positive or negative direction from the axis calibration position.

Components

rax_1 *(robot axis 1)* Data type: *num*

The position of robot axis 1 in degrees from the calibration position.

...

rax_6 *(robot axis 6)* Data type: *num*

The position of robot axis 6 in degrees from the calibration position.

Structure

< dataobject of *robjoint* >
< *rax_1* of *num* >
< *rax_2* of *num* >
< *rax_3* of *num* >
< *rax_4* of *num* >
< *rax_5* of *num* >
< *rax_6* of *num* >

Related information

	Described in:
Joint position data	Data Types - *jointtarget*
Move to joint position	Instructions - *MoveAbsJ*

robtarget Position data

Robtarget (*robot target*) is used to define the position of the robot and external axes.

Description

Position data is used to define the position in the positioning instructions to which the robot and external axes are to move.

As the robot is able to achieve the same position in several different ways, the axis configuration is also specified. This defines the axis values if these are in any way ambiguous, for example:

- if the robot is in a forward or backward position,
- if axis 4 points downwards or upwards,
- if axis 6 has a negative or positive revolution.

The position is defined based on the coordinate system of the work object, including any program displacement. If the position is programmed with some other work object than the one used in the instruction, the robot will not move in the expected way. Make sure that you use the same work object as the one used when programming positioning instructions. Incorrect use can injure someone or damage the robot or other equipment.

Components

trans *(translation)* Data type: *pos*

The position (x, y and z) of the tool centre point expressed in mm.

The position is specified in relation to the current object coordinate system, including program displacement. If no work object is specified, this is the world coordinate system.

rot *(rotation)* Data type: *orient*

The orientation of the tool, expressed in the form of a quaternion (q1, q2, q3 and q4).

The orientation is specified in relation to the current object coordinate system, including program displacement. If no work object is specified, this is the world coordinate system.

robconf *(robot configuration)* Data type: *confdata*

The axis configuration of the robot (cf1, cf4, cf6 and cfx). This is defined in the form of the current quarter revolution of axis 1, axis 4 and axis 6. The first positive quarter revolution 0 to 90 ^{o} is defined as 0. The component cfx is only used for the robot model IRB5400.

For more information, see data type *confdata*.

extax *(external axes)* Data type: *extjoint*

The position of the external axes.

The position is defined as follows for each individual axis (*eax_a*, *eax_b* ... *eax_f*):

- For rotating axes, the position is defined as the rotation in degrees from the calibration position.
- For linear axes, the position is defined as the distance in mm from the calibration position.

External axes *eax_a* ... are logical axes. How the logical axis number and the physical axis number are related to each other is defined in the system parameters.

The value 9E9 is defined for axes which are not connected. If the axes defined in the position data differ from the axes that are actually connected on program execution, the following applies:

- If the position is not defined in the position data (value 9E9), the value will be ignored, if the axis is connected and not activated. But if the axis is activated, it will result in an error.
- If the position is defined in the position data although the axis is not connected, the value is ignored.

Examples

```
CONST robtarget p15 := [ [600, 500, 225.3], [1, 0, 0, 0], [1, 1, 0, 0],
[ 11, 12.3, 9E9, 9E9, 9E9, 9E9] ];
```

A position *p15* is defined as follows:

- The position of the robot: *x = 600*, *y = 500* and *z = 225.3* mm in the object coordinate system.
- The orientation of the tool in the same direction as the object coordinate system.
- The axis configuration of the robot: axes 1 and 4 in position 90-180^{o}, axis 6 in position 0-90^{o}.
- The position of the external logical axes, a and b, expressed in degrees or mm (depending on the type of axis). Axes c to f are undefined.

```
VAR robtarget p20;
. . .
p20 := CRobT();
p20 := Offs(p20,10,0,0);
```

The position *p20* is set to the same position as the current position of the robot by calling the function *CRobT*. The position is then moved *10* mm in the x-direction.

Limitations

When using the configurable edit function *Absolute Limit Modpos*, the number of characters in the name of the data of the type *robtarget*, is limited to 14 (in other cases 16).

Structure

```
< dataobject of robtarget >
    < trans of pos >
        < x of num >
        < y of num >
        < z of num >
    < rot of orient >
        < q1 of num >
        < q2 of num >
        < q3 of num >
        < q4 of num >
    < robconf of confdata >
        < cf1 of num >
        < cf4 of num >
        < cf6 of num >
        < cfx of num >
    < extax of extjoint >
        < eax_a of num >
        < eax_b of num >
        < eax_c of num >
        < eax_d of num >
        < eax_e of num >
        < eax_f of num >
```

Related information

	Described in:
Positioning instructions	RAPID Summary - *Motion*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Handling configuration data	Motion and I/O Principles - *Robot Configuration*
Configuration of external axes	User’s Guide - *System Parameters*
What is a quaternion?	Data Types - *Orient*

shapedata World zone shape data

shapedata is used to describe the geometry of a world zone.

Description

World zones can be defined in 3 different geometrical shapes:

- a straight box, with all sides parallel to the world coordinate system and defined by a *WZBoxDef* instruction
- a sphere, defined by a *WZSphDef* instruction
- a cylinder, parallel to the z axis of the world coordinate system and defined by a *WZCylDef* instruction

The geometry of a world zone is defined by one of the previous instructions and the action of a world zone is defined by the instruction *WZLimSup* or *WZDOSet*.

Example

```
VAR wzstationary pole;
VAR wzstationary conveyor;
...
PROC ...
   VAR shapedata volume;
   ...
   WZBoxDef \Inside, volume, p_corner1, p_corner2;
   WZLimSup \Stat, conveyor, volume;
   WZCylDef \Inside, volume, p_center, 200, 2500;
   WZLimSup \Stat, pole, volume;
ENDPROC
```

A *conveyor* is defined as a box and the supervision for this area is activated. A *pole* is defined as a cylinder and the supervision of this zone is also activated. If the robot reaches one of these areas, the motion is stopped.

Characteristics

shapedata is a non-value data type.

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Define box-shaped world zone	Instructions - *WZBoxDef*
Define sphere-shaped world zone	Instructions - *WZSphDef*
Define cylinder-shaped world zone	Instructions - *WZCylDef*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone digital output set	Instructions - *WZDOSet*

signal*xx* Digital and analog signals

Data types within *signalxx* are used for digital and analog input and output signals.

The names of the signals are defined in the system parameters and are consequently not to be defined in the program.

Description

Data type	Used for
signalai	analog input signals
signalao	analog output signals
signaldi	digital input signals
signaldo	digital output signals
signalgi	groups of digital input signals
signalgo	groups of digital output signals

Variables of the type *signalxo* only contain a reference to the signal. The value is set using an instruction, e.g. *DOutput*.

Variables of the type *signalxi* contain a reference to a signal as well as the possibility to retrieve the value directly in the program, if used in value context.

The value of an input signal can be read directly in the program, e.g. :

```
! Digital input
IF di1 = 1 THEN ...

! Digital group input
IF gi1 = 5 THEN ...

! Analog input
IF ai1 > 5.2 THEN ...
```

Limitations

Data of the data type *signalxx* must not be defined in the program. However, if this is in fact done, an error message will be displayed as soon as an instruction or function that refers to this signal is executed. The data type can, on the other hand, be used as a parameter when declaring a routine.

Predefined data

The signals defined in the system parameters can always be accessed from the program by using the predefined signal variables (installed data). It should however be noted that if other data with the same name is defined, these signals cannot be used.

Characteristics

Signalxo is a *non-value* data type. Thus, data of this type does not permit value-oriented operations.

Signalxi is a semi-value data type.

Related information

	Described in:
Summary input/output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User’s Guide - *System Parameters*
Characteristics of non-value data types	Basic Characteristics - *Data Types*

speeddata Speed data

Speeddata is used to specify the velocity at which both the robot and the external axes move.

Description

Speed data defines the velocity:

- at which the tool centre point moves,
- of the reorientation of the tool,
- at which linear or rotating external axes move.

When several different types of movement are combined, one of the velocities often limits all movements. The velocity of the other movements will be reduced in such a way that all movements will finish executing at the same time.

The velocity is also restricted by the performance of the robot. This differs, depending on the type of robot and the path of movement.

Components

v_tcp *(velocity tcp)* Data type: *num*

The velocity of the tool centre point (TCP) in mm/s.

If a stationary tool or coordinated external axes are used, the velocity is specified relative to the work object.

v_ori *(velocity orientation)* Data type: *num*

The velocity of reorientation about the TCP expressed in degrees/s.

If a stationary tool or coordinated external axes are used, the velocity is specified relative to the work object.

v_leax *(velocity linear external axes)* Data type: *num*

The velocity of linear external axes in mm/s.

v_reax *(velocity rotational external axes)* Data type: *num*

The velocity of rotating external axes in degrees/s.

Example

VAR speeddata vmedium := [1000, 30, 200, 15];

The speed data *vmedium* is defined with the following velocities:

- *1000* mm/s for the TCP.
- *30* degrees/s for reorientation of the tool.
- *200* mm/s for linear external axes.
- *15* degrees/s for rotating external axes.

vmedium.v_tcp := 900;

The velocity of the TCP is changed to *900* mm/s.

Predefined data

A number of speed data are already defined in the system module *BASE*.

Name	TCP speed	Orientation	Linear ext. axis	Rotating ext. axis
v5	5 mm/s	500°/s	5000 mm/s	1000°/s
v10	10 mm/s	500°/s	5000 mm/s	1000°/s
v20	20 mm/s	500°/s	5000 mm/s	1000°/s
v30	30 mm/s	500°/s	5000 mm/s	1000°/s
v40	40 mm/s	500°/s	5000 mm/s	1000°/s
v50	50 mm/s	500°/s	5000 mm/s	1000°/s
v60	60 mm/s	500°/s	5000 mm/s	1000°/s
v80	80 mm/s	500°/s	5000 mm/s	1000°/s
v100	100 mm/s	500°/s	5000 mm/s	1000°/s
v150	150 mm/s	500°/s	5000 mm/s	1000°/s
v200	200 mm/s	500°/s	5000 mm/s	1000°/s
v300	300 mm/s	500°/s	5000 mm/s	1000°/s
v400	400 mm/s	500°/s	5000 mm/s	1000°/s
v500	500 mm/s	500°/s	5000 mm/s	1000°/s
v600	600 mm/s	500°/s	5000 mm/s	1000°/s
v800	800 mm/s	500°/s	5000 mm/s	1000°/s
v1000	1000 mm/s	500°/s	5000 mm/s	1000°/s
v1500	1500 mm/s	500°/s	5000 mm/s	1000°/s
v2000	2000 mm/s	500°/s	5000 mm/s	1000°/s
v2500	2500 mm/s	500°/s	5000 mm/s	1000°/s
v3000	3000 mm/s	500°/s	5000 mm/s	1000°/s
v4000	4000 mm/s	500°/s	5000 mm/s	1000°/s
v5000	5000 mm/s	500°/s	5000 mm/s	1000°/s
vmax	5000 mm/s	500°/s	5000 mm/s	1000°/s
v6000	6000 mm/s	500°/s	5000 mm/s	1000°/s
v7000	7000 mm/s	500°/s	5000 mm/s	1000°/s

Structure

< dataobject of *speeddata* >
  < *v_tcp* of *num* >
  < *v_ori* of *num* >
  < *v_leax* of *num* >
  < *v_reax* of *num* >

Related information

	Described in:
Positioning instructions	RAPID Summary - *Motion*
Motion/Speed in general	Motion and I/O Principles - *Positioning during Program Execution*
Defining maximum velocity	Instructions - *VelSet*
Configuration of external axes	User’s Guide - *System Parameters*
Motion performance	Product Specification

string Strings

String is used for character strings.

Description

A character string consists of a number of characters (a maximum of 80) enclosed by quotation marks (“”),

e.g. “This is a character string”.

If the quotation marks are to be included in the string, they must be written twice,

e.g. “This string contains a ““character”.

Example

```
VAR string text;
.
text := “start welding pipe 1”;
TPWrite text;
```

The text *start welding pipe 1* is written on the teach pendant.

Limitations

A string may have from 0 to 80 characters; inclusive of extra quotation marks.

A string may contain any of the characters specified by ISO 8859-1 as well as control characters (non-ISO 8859-1 characters with a numeric code between 0-255).

Predefined data

A number of predefined string constants are available in the system and can be used together with string functions.

Name	Character set
STR_DIGIT	<digit> ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9
STR_UPPER	<upper case letter> ::= A | B | C | D | E | F | G | H | I | J | K | L | M | N | O | P | Q | R | S | T | U | V | W | X | Y | Z | À | Á | Â | Ã | Ä | Å | Æ | Ç | È | É | Ê | Ë | Ì | Í | Î | Ï | 1) | Ñ | Ò | Ó | Ô | Õ | Ö | Ø | Ù | Ú | Û | Ü | 2) | 3)
STR_LOWER	<lower case letter> ::= a | b | c | d | e | f | g | h | i | j | k | l | m | n | o | p | q | r | s | t | u | v | w | x | y | z | à | á | â | ã | ä | å | æ | ç | è | é | ê | ë | ì | í | î | ï | 1) | ñ | ò | ó | ô | õ | ö | ø | ù | ú | û | ü | 2) | 3) | ß | ÿ
STR_WHITE	<blank character> ::=

1) Icelandic letter eth.
2) Letter Y with acute accent.
3) Icelandic letter thorn.

The constants flp1, ram1disk and stEmpty are already defined in the system module *BASE*.

```
CONST string flp1 := “flp1:”;

CONST string ram1disk := “ram1disk:”;

CONST string stEmpty := “”;
```

Related information

	Described in:
Operations using strings	Basic Characteristics - *Expressions*
String values	Basic Characteristics - *Basic Elements*

symnum Symbolic number

Symnum is used to represent an integer with a symbolic constant.

Description

A *symnum* constant is intended to be used when checking the return value from the functions *OpMode* and *RunMode*. See example below.

Example

```
IF RunMode() = RUN_CONT_CYCLE THEN
.
.
ELSE
.
.
ENDIF
```

Predefined data

The following symbolic constants of the data type *symnum* are predefined and can be used when checking return values from the functions *OpMode* and *RunMode*.

Value	Symbolic constant	Comment
0	RUN_UNDEF	Undefined running mode
1	RUN_CONT_CYCLE	Continuous or cycle running mode
2	RUN_INSTR_FWD	Instruction forward running mode
3	RUN_INSTR_BWD	Instruction backward running mode
4	RUN_SIM	Simulated running mode

Value	Symbolic constant	Comment
0	OP_UNDEF	Undefined operating mode
1	OP_AUTO	Automatic operating mode
2	OP_MAN_PROG	Manual operating mode max. 250 mm/s
3	OP_MAN_TEST	Manual operating mode full speed, 100%

Characteristics

Symnum is an alias data type for *num* and consequently inherits its characteristics.

Related information

	Described in:
Data types in general, alias data types	Basic Characteristics - *Data Types*

System Data

System data is the internal data of the robot that can be accessed and read by the program. It can be used to read the current status, e.g. the current maximum velocity.

The following table contains a list of all system data.

Name	Description	Data Type	Changed by	See also
C_MOTSET	Current motion settings, i.e.: - max. velocity and velocity override - max. acceleration - movement about singular points - monitoring the axis configuration - path resolution - motion supervision with tunevalue	motsetdata	Instructions - VelSet - AccSet - SingArea - ConfL,ConfJ - PathResol - MotionSup	Data Types - *motsetdata* Instructions - *VelSet* Instructions - *AccSet* Instructions - *SingArea* Instructions - *ConfL, ConfJ* Instructions - *PathResol* Instructions - *MotionSup*
C_PROGDISP	Current program displacement for robot and external axes.	progdisp	Instructions - PDispSet - PDispOn - PDispOff - EOffsSet - EOffsOn - EOffsOff	Data Types - *progdisp* Instructions - *PDispSet* Instructions - *PDispOn* Instructions - *PDispOff* Instructions - *EOffsSet* Instructions - *EOffsOn* Instructions - *EOffsOff*
ERRNO	The latest error that occurred	errnum	The robot	Data Types - *errnum* RAPID Summary - *Error Recovery*
INTNO	The latest interrupt that occurred	intnum	The robot	Data Types - *intnum* RAPID Summary -*Interrupts*

taskid Task identification

Taskid is used to identify available program tasks in the system.

The names of the program tasks are defined in the system parameters and, consequently, must not be defined in the program.

Description

Data of the type *taskid* only contains a reference to the program task.

Limitations

Data of the type *taskid* must not be defined in the program. The data type can, on the other hand, be used as a parameter when declaring a routine.

Predefined data

The program tasks defined in the system parameters can always be accessed from the program (installed data).

For all program tasks in the system, predefined variables of the data type *taskid* will be available. The variable identity will be "taskname"+"Id", e.g. for MAIN task the variable identity will be MAINId, TSK1 - TSK1Id etc.

Characteristics

Taskid is a *non-value* data type. This means that data of this type does not permit value-oriented operations.

Related information

	Described in:
Saving program modules	Instruction - *Save*
Configuration of program tasks	User's Guide - *System Parameters*
Characteristics of non-value data types	Basic Characteristics - *Data Types*

testsignal Test signal

The data type *testsignal* is used when a test of the robot system is performed.

Description

A number of predefined test signals are available in the robot system. The *testsignal* data type is available in order to simplify programming of instructions regarding service and test.

Examples

TestSign 2, revolution_counter, Orbit, 2, 1, 0;

revolution_counter is a constant of the *testsignal* data type.

Predefined data

A number of predefined constants for the various test signals in the robot system are loaded into the system at start-up. The service manual describes the test signals more thoroughly.

Characteristics

Testsignal is an alias data type for *num* and consequently inherits its characteristics.

Related information

	Described in:
Data types in general, alias data types	Basic Characteristics - *Data Types*

tooldata Tool data

Tooldata is used to describe the characteristics of a tool, e.g. a welding gun or a gripper.

If the tool is fixed in space (a stationary tool), common tool data is defined for this tool and the gripper holding the work object.

Description

Tool data affects robot movements in the following ways:

- The tool centre point (TCP) refers to a point that will satisfy the specified path and velocity performance. If the tool is reorientated or if coordinated external axes are used, only this point will follow the desired path at the programmed velocity.
- If a stationary tool is used, the programmed speed and path will relate to the work object.
- Programmed positions refer to the position of the current TCP and the orientation in relation to the tool coordinate system. This means that if, for example, a tool is replaced because it is damaged, the old program can still be used if the tool coordinate system is redefined.

Tool data is also used when jogging the robot to:

- Define the TCP that is not to move when the tool is reorientated.
- Define the tool coordinate system in order to facilitate moving in or rotating about the tool directions.

It is important to always define the actual tool load and when used, the payload of the robot too. Incorrect definitions of load data can result in overloading of the robot mechanical structure.

When incorrect tool load data is specified, it can often lead to the following consequences:

- If the value in the specified load is greater than that of the value of the true load;
  -> The robot will not be used to its maximum capacity
  -> Impaired path accuracy including a risk of overshooting
- If the value in the specified load is less than the value of the true load;
  -> Impaired path accuracy including a risk of overshooting
  -> Risk of overloading the mechanical structure

Components

robhold *(robot hold)* Data type: *bool*

Defines whether or not the robot is holding the tool:

- *TRUE* -> The robot is holding the tool.
- *FALSE* -> The robot is not holding the tool, i.e. a stationary tool.

tframe *(tool frame)* Data type: *pose*

The tool coordinate system, i.e.:

- The position of the TCP (x, y and z) in mm, expressed in the wrist coordinate system (See figure 1).
- The orientation of the tool coordinate system, expressed in the wrist coordinate system as a quaternion (q1, q2, q3 and q4) (See figure 1).

If a stationary tool is used, the definition is defined in relation to the world coordinate system.

If the direction of the tool is not specified, the tool coordinate system and the wrist coordinate system will coincide.



Figure 10 Definition of the tool coordinate system.

tload *(tool load)* Data type: *loaddata*

The load of the tool, i.e.:

- The weight of the tool in kg.
- The centre of gravity of the tool (x, y and z) in mm, expressed in the wrist coordinate system
- The orientation of the tool load coordinate system expressed in the wrist coordinate system, defining the inertial axes of the tool.
  The orientation of the tool load coordinate system must coincide with the orientation of the wrist coordinate system. **This must always be set to 1, 0, 0, 0**.
- The moments of inertia of the tool relative to its centre of mass around the tool load coordinate axes in kgm^2.
  If all inertial components are defined as being 0 kgm^2, the tool is handled as a point mass.



Figure 11 Tool load parameter definitions

For more information (such as coordinate system for stationary tool or restrictions), see the data type *loaddata*.

If a stationary tool is used, the load of the gripper holding the work object must be defined.

Note that only the load of the tool is to be specified. The payload handled by a gripper is connected and disconnected by means of the instruction *GripLoad*.

Examples

```
PERS tooldata gripper := [ TRUE, [[97.4, 0, 223.1], [0.924, 0, 0.383 ,0]],
   [5, [23, 0, 75], [1, 0, 0, 0], 0, 0, 0]];
```

The tool in Figure 10 is described using the following values:

- The robot is holding the tool.
- The TCP is located at a point *223.1* mm straight out from axis 6 and *97.4* mm along the X-axis of the wrist coordinate system.
- The X and Z directions of the tool are rotated 45^{o} in relation to the wrist coordinate system.
- The tool weighs 5 kg.
- The centre of gravity is located at a point *75* mm straight out from axis 6 and *23* mm along the X-axis of the wrist coordinate system.
- The load can be considered a point mass, i.e. without any moment of inertia.

```
gripper.tframe.trans.z := 225.2;
```

The TCP of the tool, *gripper*, is adjusted to *225.2* in the z-direction.

Limitations

The tool data should be defined as a persistent variable *(PERS)* and should not be defined within a routine. The current values are then saved when the program is stored on diskette and are retrieved on loading.

Arguments of the type tool data in any motion instruction should only be an entire persistent (not array element or record component).

Predefined data

The tool *tool0* defines the wrist coordinate system, with the origin being the centre of the mounting flange. *Tool0* can always be accessed from the program, but can never be changed (it is stored in system module *BASE*).

```
PERS tooldata tool0 := [ TRUE, [ [0, 0, 0], [1, 0, 0 ,0] ],
                              [0.001, [0, 0, 0.001], [1, 0, 0, 0], 0, 0, 0] ];
```

Structure

< dataobject of *tooldata* >
  < *robhold* of *bool* >
  < *tframe* of *pose* >
    < *trans* of *pos* >
      < *x* of *num* >
      < *y* of *num* >
      < *z* of *num* >
    < *rot* of *orient* >
      < *q1* of *num* >
      < *q2* of *num* >
      < *q3* of *num* >
      < *q4* of *num* >
  < *tload* of *loaddata* >
    < *mass* of *num* >
    < *cog* of *pos* >
      < *x* of *num* >
      < *y* of *num* >
      < *z* of *num* >
    < *aom* of *orient* >
      < *q1* of *num* >
      < *q2* of *num* >
      < *q3* of *num* >
      < *q4* of *num* >
    < *ix* of *num* >
    < *iy* of *num* >
    < *iz* of *num* >

Related information

	Described in:
Positioning instructions	RAPID Summary - *Motion*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Definition of payload	Instructions - *Gripload*
Definition of load	Data types - *Load data*

tpnum Teach Pendant Window number

tpnum is used to represent the Teach Pendant Window number with a symbolic constant.

Description

A *tpnum* constant is intended to be used in instruction *TPShow.* See example below.

Example

TPShow TP_PROGRAM;

The *Production Window* will be active if the system is in *AUTO* mode and the *Program Window* will be active if the system is in *MAN* mode, after execution of this instruction.

Predefined data

The following symbolic constants of the data type *tpnum* are predefined and can be used in instruction *TPShow*:

Value	Symbolic constant	Comment
1	TP_PROGRAM	AUTO: Production Window MAN: Program Window
2	TP_LATEST	Latest used Teach Pendant Window
3	TP_SCREENVIEWER	Screen Viewer window, if this option is active

Characteristics

tpnum is an alias data type for *num* and consequently inherits its characteristics.

Related information

	Described in:
Data types in general, alias data types	Basic Characteristics - *Data Types*
Communicating using the teach pendant	RAPID Summary - *Communication*
Switch window on the teach pendant	Instructions - *TPShow*

triggdata Positioning events - trigg

Triggdata is used to store data about a positioning event during a robot movement.

A positioning event can take the form of setting an output signal or running an interrupt routine at a specific position along the movement path of the robot.

Description

To define the conditions for the respective measures at a positioning event, variables of the type *triggdata* are used. The data contents of the variable are formed in the program using one of the instructions *TriggIO* or *TriggInt*, and are used by one of the instructions *TriggL*, *TriggC* or *TriggJ*.

Example

```
VAR triggdata gunoff;

TriggIO gunoff, 5 \DOp:=gun, off;

TriggL p1, v500, gunoff, fine, gun1;
```

The digital output signal *gun* is set to the value *off* when the TCP is at a position *5* mm before the point *p1*.

Characteristics

Triggdata is a non-value data type.

Related information

	Described in:
Definition of triggs	Instructions - *TriggIO*, *TriggInt*
Use of triggs	Instructions - *TriggL*, *TriggC*, *TriggJ*
Characteristics of non-value data types	Basic Characteristics- *Data Types*

tunetype Servo tune type

Tunetype is used to represent an integer with a symbolic constant.

Description

A *tunetype* constant is intended to be used as an argument to the instruction *TuneServo*. See example below.

Example

```
TuneServo MHA160R1, 1, 110 \Type:= TUNE_KP;
```

Predefined data

The following symbolic constants of the data type *tunetype* are predefined and can be used as argument for the instruction *TuneServo.*

Value	Symbolic constant	Comment
0	TUNE_DF	Reduces overshoots
1	TUNE_KP	Affects position control gain
2	TUNE_KV	Affects speed control gain
3	TUNE_TI	Affects speed control integration time
4	TUNE_FRIC_LEV	Affects friction compensation level
5	TUNE_FRIC_RAMP	Affects friction compensation ramp
6	TUNE_DG	Reduces overshoots
7	TUNE_DH	Reduces vibrations with heavy loads
8	TUNE_DI	Reduces path errors
9	TUNE_DK	Reseved for future use

Characteristics

Tunetype is an alias data type for *num* and consequently inherits its characteristics.

Related information

	Described in:
Data types in general, alias data types	Basic Characteristics - *Data Types*
Use of data type tunetype	Instructions - *TuneServo*

wobjdata Work object data

Wobjdata is used to describe the work object that the robot welds, processes, moves within, etc.

Description

If work objects are defined in a positioning instruction, the position will be based on the coordinates of the work object. The advantages of this are as follows:

- If position data is entered manually, such as in off-line programming, the values can often be taken from a drawing.
- Programs can be reused quickly following changes in the robot installation. If, for example, the fixture is moved, only the user coordinate system has to be redefined.
- Variations in how the work object is attached can be compensated for. For this, however, some sort of sensor will be required to position the work object.

If a stationary tool or coordinated external axes are used the work object must be defined, since the path and velocity would then be related to the work object instead of the TCP.

Work object data can also be used for jogging:

- The robot can be jogged in the directions of the work object.
- The current position displayed is based on the coordinate system of the work object.

Components

robhold *(robot hold)* Data type: *bool*

Defines whether or not the robot is holding the work object:

- *TRUE* -> The robot is holding the work object, i.e. using a stationary tool.
- *FALSE* -> The robot is not holding the work object, i.e. the robot is holding the tool.

ufprog *(user frame programmed)* Data type: *bool*

Defines whether or not a fixed user coordinate system is used:

- *TRUE* -> Fixed user coordinate system.
- *FALSE* -> Movable user coordinate system, i.e. coordinated external axes are used.

ufmec *(user frame mechanical unit)* Data type: *string*

The mechanical unit with which the robot movements are coordinated. Only specified in the case of movable user coordinate systems (*ufprog* is *FALSE*).

Specified with the name that is defined in the system parameters, e.g. "orbit_a".

uframe *(user frame)* Data type: *pose*

The user coordinate system, i.e. the position of the current work surface or fixture (see Figure 12):

- The position of the origin of the coordinate system (x, y and z) in mm.
- The rotation of the coordinate system, expressed as a quaternion (q1, q2, q3, q4).

If the robot is holding the tool, the user coordinate system is defined in the world coordinate system (in the wrist coordinate system if a stationary tool is used).

When coordinated external axes are used (*ufprog* is *FALSE*), the user coordinate system is defined in the system parameters.

oframe *(object frame)* Data type: *pose*

The object coordinate system, i.e. the position of the current work object (see Figure 12):

- The position of the origin of the coordinate system (x, y and z) in mm.
- The rotation of the coordinate system, expressed as a quaternion (q1, q2, q3, q4).

The object coordinate system is defined in the user coordinate system.



Figure 12 The various coordinate systems of the robot (when the robot is holding the tool).

Example

```
PERS wobjdata wobj2 :=[ FALSE, TRUE, "", [ [300, 600, 200], [1, 0, 0 ,0] ],
   [ [0, 200, 30], [1, 0, 0 ,0] ] ];
```

The work object in Figure 12 is described using the following values:

- The robot is not holding the work object.
- The fixed user coordinate system is used.
- The user coordinate system is not rotated and the coordinates of its origin are x= *300*, y = *600* and z = *200* mm in the world coordinate system.
- The object coordinate system is not rotated and the coordinates of its origin are x= *0*, y= *200* and z= *30* mm in the user coordinate system.

```
wobj2.oframe.trans.z := 38.3;
```

- The position of the work object *wobj2* is adjusted to *38.3* mm in the z-direction.

Limitations

The work object data should be defined as a persistent variable (*PERS)* and should not be defined within a routine. The current values are then saved when the program is stored on diskette and are retrieved on loading.

Arguments of the type work object data in any motion instruction should only be an entire persistent (not array element or record component).

Predefined data

The work object data *wobj0* is defined in such a way that the object coordinate system coincides with the world coordinate system. The robot does not hold the work object.

Wobj0 can always be accessed from the program, but can never be changed (it is stored in system module *BASE*).

```
PERS wobjdata wobj0 :=    [ FALSE, TRUE, "", [ [0, 0, 0], [1, 0, 0 ,0] ],
                          [ [0, 0, 0], [1, 0, 0 ,0] ] ];
```

Structure

< dataobject of *wobjdata* >
< *robhold* of *bool* >
< *ufprog* of *bool*>
< *ufmec* of *string* >
< *uframe* of *pose* >
< *trans* of *pos* >
< *x* of *num* >
< *y* of *num* >
< *z* of *num* >
< *rot* of *orient* >
< *q1* of *num* >
< *q2* of *num* >
< *q3* of *num* >
< *q4* of *num* >
< *oframe* of *pose* >
< *trans* of *pos* >
< *x* of *num* >
< *y* of *num* >
< *z* of *num* >
< *rot* of *orient* >
< *q1* of *num* >
< *q2* of *num* >
< *q3* of *num* >
< *q4* of *num* >

Related information

	Described in:
Positioning instructions	RAPID Summary - *Motion*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Coordinated external axes	Motion and I/O Principles - *Coordinate Systems*
Calibration of coordinated external axes	User’s Guide - *System Parameters*

wzstationary Stationary world zone data

wzstationary (world zone stationary) is used to identify a stationary world zone and can only be used in an event routine connected to the event POWER ON.

A world zone is supervised during robot movements both during program execution and jogging. If the robot's TCP reaches this world zone, the movement is stopped or a digital output signal is set or reset.

Description

A *wzstationary* world zone is defined and activated by a *WZLimSup* or a *WZDOSet* instruction.

WZLimSup or *WZDOSet* gives the variable or the persistent of data type *stationary* a numeric value. The value identifies the world zone.

A stationary world zone is always active and is only erased by a warm start (switch power off then on, or change system parameters). It is not possible to deactivate, activate or erase a stationary world zone via RAPID instructions.

Stationary world zones should be active from power on and should be defined in a POWER ON event routine or a semistatic task.

Example

```
VAR wzstationary conveyor;
...
PROC ...
   VAR shapedata volume;
   ...
   WZBoxDef \Inside, volume, p_corner1, p_corner2;
   WZLimSup \Stat, conveyor, volume;
ENDPROC
```

A *conveyor* is defined as a straight box (the volume below the belt). If the robot reaches this volume, the movement is stopped.

Limitations

A *wzstationary* data can be defined as a variable (VAR) or as a persistent (PERS). It can be global in task or local within module, but not local within a routine.

Arguments of the type *wzstationary* should only be entire data (not array element or record component).

An init value for data of the type *wzstationary* is not used by the control system. When there is a need to use a persistent variable in a multi-tasking system, set the init value to 0 in both tasks,
e.g. PERS wzstationary share_workarea := [0];

Example

For a complete example see instruction *WZLimSup*.

Characteristics

wzstationary is an alias data type of *wztemporary* and inherits its characteristics.

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Temporary world zone	Data Types - *wztemporary*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone digital output set	Instructions - *WZDOSet*

wztemporary Temporary world zone data

wztemporary (world zone temporary) is used to identify a temporary world zone and can be used anywhere in the RAPID program for the MAIN task.

A world zone is supervised during robot movements both during program execution and jogging. If the robot's TCP reaches this world zone, the movement is stopped or a digital output signal is set or reset.

Description

A *wztemporary* world zone is defined and activated by a *WZLimSup* or a *WZDOSet* instruction.

WZLimSup or *WZDOSet* gives the variable or the persistent of data type *wztemporary* a numeric value. The value identifies the world zone.

Once defined and activated, a temporary world zone can be deactivated by *WZDisable*, activated again by *WZEnable*, and erased by *WZFree*.

All temporary world zones in the MAIN task are automatically erased and all data objects of type *wztemporary* in the MAIN task are set to 0:

- when a new program is loaded in the MAIN task
- when starting program execution from the beginning in the MAIN task

Example

```
VAR wztemporary roll;
...
PROC ...
   VAR shapedata volume;
   CONST pos t_center := [1000, 1000, 1000];
   ...
   WZCylDef \Inside, volume, t_center, 400, 1000;
   WZLimSup \Temp, roll, volume;
ENDPROC
```

A wztemporary variable, *roll*, is defined as a cylinder. If the robot reaches this volume, the movement is stopped.

Limitations

A *wztemporary* data can be defined as a variable (VAR) or as a persistent (PERS). It can be global in a task or local within a module, but not local within a routine.

Arguments of the type *wztemporary* must only be entire data, not an array element or record component.

A temporary world zone must only be defined (*WZLimSup* or *WZDOSet*) and free (*WZFree*) in the MAIN task. Definitions of temporary world zones in the background would affect the program execution in the MAIN task The instructions *WZDisable* and *WZEnable* can be used in the background task. When there is a need to use a persistent variable in a multi-tasking system, set the init value to 0 in both tasks,
e.g. PERS wztemporary share_workarea := [0];

Example

For a complete example see instruction *WZDOSet*.

Structure

<dataobject of *wztemporary*>
<*wz* of *num*>

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Stationary world zone	Data Types - *wzstationary*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone digital output set	Instructions - *WZDOSet*
Deactivate world zone	Instructions - *WZDisable*
Activate world zone	Instructions - *WZEnable*
Erase world zone	Instructions - *WZFree*

zonedata Zone data

Zonedata is used to specify how a position is to be terminated, i.e. how close to the programmed position the axes must be before moving towards the next position.

Description

A position can be terminated either in the form of a stop point or a fly-by point.

A stop point means that the robot and external axes must reach the specified position (stand still) before program execution continues with the next instruction.

A fly-by point means that the programmed position is never attained.
Instead, the direction of motion is changed before the position is reached.
Two different zones (ranges) can be defined for each position:

- The zone for the TCP path.
- The extended zone for reorientation of the tool and for external axes.



Figure 13 The zones for a fly-by point.

Zones function in the same way during joint movement, but the zone size may differ somewhat from the one programmed.

The zone size cannot be larger than half the distance to the closest position (forwards or backwards). If a larger zone is specified, the robot automatically reduces it.

The zone for the TCP path

A corner path (parabola) is generated as soon as the edge of the zone is reached (see Figure 13).

The zone for reorientation of the tool

Reorientation starts as soon as the TCP reaches the extended zone. The tool is reoriented in such a way that the orientation is the same leaving the zone as it would have been in the same position if stop points had been programmed. Reorientation will be smoother if the zone size is increased, and there is less of a risk of having to reduce the velocity to carry out the reorientation.



Figure 14a Three positions are programmed, the last with different tool orientation.



Figure 14b If all positions were stop points, program execution would look like this.



Figure 14c If the middle position was a fly-by point, program execution would look like this

The zone for external axes

External axes start to move towards the next position as soon as the TCP reaches the extended zone. In this way, a slow axis can start accelerating at an earlier stage and thus execute more evenly.

Reduced zone

With large reorientations of the tool or with large movements of the external axes, the extended zone and even the TCP zone can be reduced by the robot. The zone will be defined as the smallest relative size of the zone based upon the zone components (see next page) and the programmed motion.



Figure 15 Example of reduced zone for reorientation of the tool to 36% of the motion due to zone_ori.



Figure 16 Example of reduced zone for reorientation of the tool and TCP path to 15% of the motion due to zone_ori.

When external axes are active they affect the relative sizes of the zone according to these formulas:

$$\frac{\text{pzone\_eax}}{\text{length of movement P1 - P2}}$$

$$\frac{\text{zone\_leax}}{\text{length of max linear ext. axis movement P1 - P2}}$$

$$\frac{\text{zone\_reax}}{\text{angle of max reorientation of rotating ext. axis P1 - P2}}$$

NOTE: If the TCP zone is reduced because of zone_ori, zone_leax or zone_reax the path planner enters a mode that can handle the case of no TCP movement. If there is a TCP movement when in this mode the speed is not compensated for the curvature of the path in a corner zone. For instance, this will cause a 30% speed reduction in a 90 degree corner. If this is a problem, increase the limiting zone component.

Components

finep *(fine point)* Data type: *bool*

Defines whether the movement is to terminate as a stop point (fine point) or as a fly-by point.

- *TRUE* -> The movement terminates as a stop point.
  The remaining components in the zone data are not used.
- *FALSE* -> The movement terminates as a fly-by point.

pzone_tcp *(path zone TCP)* Data type: *num*

The size (the radius) of the TCP zone in mm.

The extended zone will be defined as the smallest relative size of the zone based upon the following components and the programmed motion.

pzone_ori *(path zone orientation)* Data type: *num*

The zone size (the radius) for the tool reorientation. The size is defined as the distance of the TCP from the programmed point in mm.

The size must be larger than the corresponding value for *pzone_tcp*.
If a lower value is specified, the size is automatically increased to make it the same as *pzone_tcp*.

pzone_eax *(path zone external axes)* Data type: *num*

The zone size (the radius) for external axes. The size is defined as the distance of the TCP from the programmed point in mm.

The size must be larger than the corresponding value for *pzone_tcp*.
If a lower value is specified, the size is automatically increased to make it the same as *pzone_tcp*.

zone_ori *(zone orientation)* Data type: *num*

The zone size for the tool reorientation in degrees. If the robot is holding the work object, this means an angle of rotation for the work object.

zone_leax *(zone linear external axes)* Data type: *num*

The zone size for linear external axes in mm.

zone_reax *(zone rotational external axes)* Data type: *num*

The zone size for rotating external axes in degrees.

Examples

VAR zonedata path := [FALSE, 25, 40, 40, 10, 35, 5];

The zone data *path* is defined by means of the following characteristics:

- The zone size for the TCP path is *25* mm.
- The zone size for the tool reorientation is *40* mm (TCP movement).
- The zone size for external axes is *40* mm (TCP movement).

If the TCP is standing still, or there is a large reorientation, or there is a large external axis movement, with respect to the zone, the following apply instead:

- The zone size for the tool reorientation is *10* degrees.
- The zone size for linear external axes is *35* mm.
- The zone size for rotating external axes is *5* degrees.

path.pzone_tcp := 40;

The zone size for the TCP path is adjusted to *40* mm.

Predefined data

A number of zone data are already defined in the system module *BASE*.

Stop points

Name	
fine	0 mm

Fly-by points

	TCP movement			Tool reorientation		
Name	TCP path	Orientation	Ext. axis	Orientation	Linear axis	Rotating axis
z1	1 mm	1 mm	1 mm	0.1 ^{o}	1 mm	0.1 ^{o}
z5	5 mm	8 mm	8 mm	0.8 ^{o}	8 mm	0.8 ^{o}
z10	10 mm	15 mm	15 mm	1.5 ^{o}	15 mm	1.5 ^{o}
z15	15 mm	23 mm	23 mm	2.3 ^{o}	23 mm	2.3 ^{o}
z20	20 mm	30 mm	30 mm	3.0 ^{o}	30 mm	3.0^{o}
z30	30 mm	45 mm	45 mm	4.5 ^{o}	45 mm	4.5 ^{o}
z40	40 mm	60 mm	60 mm	6.0 ^{o}	60 mm	6.0 ^{o}
z50	50 mm	75 mm	75 mm	7.5 ^{o}	75 mm	7.5 ^{o}
z60	60 mm	90 mm	90 mm	9.0 ^{o}	90 mm	9.0 ^{o}
z80	80 mm	120 mm	120 mm	12 ^{o}	120 mm	12 ^{o}
z100	100 mm	150 mm	150 mm	15 ^{o}	150 mm	15 ^{o}
z150	150 mm	225 mm	225 mm	23 ^{o}	225 mm	23 ^{o}
z200	200 mm	300 mm	300 mm	30 ^{o}	300 mm	30 ^{o}

Structure

< data object of *zonedata* >
  < *finep* of *bool* >
  < *pzone_tcp* of *num* >
  < *pzone_ori* of *num* >
  < *pzone_eax* of *num* >
  < *zone_ori* of *num* >
  < *zone_leax* of *num* >
  < *zone_reax* of *num* >

Related information

	Described in:
Positioning instructions	RAPID Summary - *Motion*
Movements/Paths in general	Motion and I/O Principles - *Positioning during Program Execution*
Configuration of external axes	User’s Guide - *System Parameters*

CONTENTS

InvertDO	Inverts the value of a digital output signal
IODisable	Disable I/O unit
IOEnable	Enable I/O unit
ISignalAI	Interrupts from analog input signal
ISignalAO	Interrupts from analog output signal
ISignalDI	Orders interrupts from a digital input signal
ISignalDO	Interrupts from a digital output signal
ISleep	Deactivates an interrupt
ITimer	Orders a timed interrupt
IVarValue	Orders a variable value interrupt
IWatch	Activates an interrupt
label	Line name
Load	Load a program module during execution
MechUnitLoad	Defines a payload for a mechanical unit
MoveAbsJ	Moves the robot to an absolute joint position
MoveC	Moves the robot circularly
MoveJ	Moves the robot by joint movement
MoveL	Moves the robot linearly
MoveCDO	Moves the robot circularly and sets digital output in the corner
MoveJDO	Moves the robot by joint movement and sets digital output in the corner
MoveLDO	Moves the robot linearly and sets digital output in the corner
MoveCSync	Moves the robot circularly and executes a RAPID procedure
MoveJSync	Moves the robot by joint movement and executes a RAPID procedure
MoveL Sync	Moves the robot linearly and executes a RAPID procedure
Open	Opens a file or serial channel
PathResol	Override path resolution
PDispOff	Deactivates program displacement
PDispOn	Activates program displacement
PDispSet	Activates program displacement using a value
PulseDO	Generates a pulse on a digital output signal
RAISE	Calls an error handler

ReadAnyBin	Read data from a binary serial channel or file
Reset	Resets a digital output signal
RestoPath	Restores the path after an interrupt
RETRY	Restarts following an error
RETURN	Finishes execution of a routine
Rewind	Rewind file position
Save	Save a program module
SearchC	Searches circularly using the robot
SearchL	Searches linearly using the robot
Set	Sets a digital output signal
SetAO	Changes the value of an analog output signal
SetDO	Changes the value of a digital output signal
SetGO	Changes the value of a group of digital output signals
SingArea	Defines interpolation around singular points
SpyStart	Start recording of execution time data
SpyStop	Stop recording of time execution data
SoftAct	Activating the soft servo
SoftDeact	Deactivating the soft servo
StartLoad	Load a program module during execution
StartMove	Restarts robot motion
Stop	Stops program execution
StopMove	Stops robot motion
StorePath	Stores the path when an interrupt occurs
TEST	Depending on the value of an expression ...
TestSign	Output of test signals
TPErase	Erases text printed on the teach pendant
TPReadFK	Reads function keys
TPReadNum	Reads a number from the teach pendant
TPShow	Switch window on the teach pendant
TPWrite	Writes on the teach pendant
TriggC	Circular robot movement with events
TriggEquip	Defines a fixed position-time I/O event
TriggInt	Defines a position related interrupt
TriggIO	Defines a fixed position I/O event
TriggJ	Axis-wise robot movements with events

TriggL	Linear robot movements with events
TRYNEXT	Jumps over an instruction which has caused an error
TuneReset	Resetting servo tuning
TuneServo	Tuning servos
UnLoad	UnLoad a program module during execution
WaitDI	Waits until a digital input signal is set
WaitDO	Waits until a digital output signal is set
WaitLoad	Connect the loaded module to the task
VelSet	Changes the programmed velocity
WHILE	Repeats as long as ...
Write	Writes to a character-based file or serial channel
WriteAnyBin	Writes data to a binary serial channel or file
WriteBin	Writes to a binary serial channel
WriteStrBin	Writes a string to a binary serial channel
WaitTime	Waits a given amount of time
WaitUntil	Waits until a condition is met
WZBoxDef	Define a box-shaped world zone
WZCylDef	Define a cylinder-shaped world zone
WZDisable	Deactivate temporary world zone supervision
WZDOSet	Activate world zone to set digital output
WZEnable	Activate temporary world zone supervision
WZFree	Erase temporary world zone supervision
WZLimSup	Activate world zone limit supervision
WZSphDef	Define a sphere-shaped world zone

":=" Assigns a value

The ":=" instruction is used to assign a new value to data. This value can be anything from a constant value to an arithmetic expression, e.g. *reg1+5*reg3*.

Examples

reg1 := 5;

reg1 is assigned the value *5*.

reg1 := reg2 - reg3;

reg1 is assigned the value that the *reg2-reg3* calculation returns.

counter := counter + 1;

counter is incremented by one.

Arguments

Data := Value

Data Data type: All

The data that is to be assigned a new value.

Value Data type: Same as Data

The desired value.

Examples

tool1.tframe.trans.x := tool1.tframe.trans.x + 20;

The TCP for *tool1* is shifted *20* mm in the X-direction.

pallet{5,8} := Abs(value);

An element in the *pallet* matrix is assigned a value equal to the absolute value of the *value* variable.

Limitations

The data (whose value is to be changed) must not be

- a constant
- a non-value data type.

The data and value must have similar (the same or alias) data types.

Syntax

(EBNF)
<assignment target> ’:=’ <expression> ’;’
<assignment target> ::=
<variable>
| <persistent>
| **<parameter>**
| **<VAR>**

Related information

	Described in:
Expressions	Basic Characteristics - *Expressions*
Non-value data types	Basic Characteristics - *Data Types*
Assigning an initial value to data	Basic Characteristics - *Data* Programming and Testing
Manually assigning a value to data	Programming and Testing

AccSet Reduces the acceleration

AccSet is used when handling fragile loads. It allows slower acceleration and deceleration, which results in smoother robot movements.

Examples

AccSet 50, 100;

The acceleration is limited to 50% of the normal value.

AccSet 100, 50;

The acceleration ramp is limited to 50% of the normal value.

Arguments

AccSet Acc Ramp

Acc Data type: *num*

Acceleration and deceleration as a percentage of the normal values.
100% corresponds to maximum acceleration. Maximum value: 100%.
Input value < 20% gives 20% of maximum acceleration.

Ramp Data type: *num*

The rate at which acceleration and deceleration increases as a percentage of the normal values (see Figure 17). Jerking can be restricted by reducing this value.
100% corresponds to maximum rate. Maximum value: 100%.
Input value < 10% gives 10% of maximum rate.



Figure 17 Reducing the acceleration results in smoother movements.

Program execution

The acceleration applies to both the robot and external axes until a new *AccSet* instruction is executed.

The default values (100%) are automatically set

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Syntax

AccSet
[Acc ':='] < expression (**IN**) of *num* > ','
[Ramp ':='] < expression (**IN**) of *num* > ';'

Related information

	Described in:
Positioning instructions	RAPID Summary - *Motion*

ActUnit Activates a mechanical unit

ActUnit is used to activate a mechanical unit.

It can be used to determine which unit is to be active when, for example, common drive units are used.

Example

ActUnit orbit_a;

Activation of the *orbit_a* mechanical unit.

Arguments

ActUnit MecUnit

MecUnit *(Mechanical Unit)* Data type: *mecunit*

The name of the mechanical unit that is to be activated.

Program execution

When the robot and external axes have come to a standstill, the specified mechanical unit is activated. This means that it is controlled and monitored by the robot.

If several mechanical units share a common drive unit, activation of one of these mechanical units will also connect that unit to the common drive unit.

Limitations

Instruction ActUnit cannot be used in

- program sequence StorePath ... RestoPath
- event routine RESTART

The movement instruction previous to this instruction, should be terminated with a stop point in order to make a restart in this instruction possible following a power failure.

Syntax

ActUnit
[MecUnit ':='] < variable (**VAR**) of *mecunit*> ';'

Related information

	Described in:
Deactivating mechanical units	Instructions - *DeactUnit*
Mechanical units	Data Types - *mecunit*
More examples	Instructions - *DeactUnit*

Add Adds a numeric value

Add is used to add or subtract a value to or from a numeric variable or persistent.

Examples

Add reg1, 3;

3 is added to *reg1*, i.e. reg1:=reg1+3.

Add reg1, -reg2;

The value of *reg2* is subtracted from *reg1*, i.e. reg1:=reg1-reg2.

Arguments

Add Name AddValue

Name Data type: *num*

The name of the variable or persistent to be changed.

AddValue Data type: *num*

The value to be added.

Syntax

Add
[Name ':='] < var or pers (**INOUT**) of *num* > ','
[AddValue ':='] < expression (**IN**) of *num* > ';'

Related information

	Described in:
Incrementing a variable by 1	Instructions - *Incr*
Decrementing a variable by 1	Instructions - *Decr*
Changing data using an arbitrary expression, e.g. multiplication	Instructions - *:=*

AliasIO Define I/O signal with alias name

AliasIO is used to define a signal of any type with an alias name or to use signals in built-in task modules.

Signals with alias names can be used for predefined generic programs, without any modification of the program before running in different robot installations.

The instruction *AliasIO* must be run before any use of the actual signal. See example 1 below for *loaded* modules and example 2 below for *builtin* modules.

Example 1

```
VAR signaldo alias_do;

PROC prog_start()
   AliasIO config_do, alias_do;
ENDPROC
```

The routine *prog_start* is connected to the START event in system parameters. The program defined digital output signal *alias_do* is connected to the configured digital output signal *config_do* at program start (start the program from beginning).

Arguments

AliasIO FromSignal ToSignal

FromSignal Data type: *signalxx* or *string*

Loaded modules:

The signal identifier named according to the configuration (data type *signalxx*) from which the signal descriptor is copied. The signal must be defined in the IO configuration.

Built-in modules:

A reference (CONST, VAR, PERS or parameter of these) containing the name of the signal (data type *string*) from which the signal descriptor after search in the system is copied. The signal must be defined in the IO configuration.

ToSignal Data type: *signalxx*

The signal identifier according to the program (data type *signalxx*) to which the signal descriptor is copied. The signal must be declared in the RAPID program.

The same data type must be used (or find) for the arguments *FromSignal* and *ToSignal* and must be one of type *signalxx* (*signalai*, *signalao*, *signaldi*, *signaldo*, *signalgi* or *signalgo*).

Program execution

The signal descriptor value is copied from the signal given in argument *FromSignal* to the signal given in argument *ToSignal*.

Example 2

```
VAR signaldi alias_di;

PROC prog_start()
   CONST string config_string := "config_di";
   AliasIO config_string, alias_di;
ENDPROC
```

The routine *prog_start* is connected to the START event in system parameters. The program defined digital output signal *alias_di* is connected to the configured digital output signal *config_di* (via constant *config_string*) at program start (start the program from the beginning).

Limitation

When starting the program, the alias signal cannot be used until the *AliasIO* instruction is executed.

Instruction *AliasIO* **must** be placed

- either in the event routine executed at program start (event START)
- or in the program part executed after every program start (before use of the signal)

Instruction *AliasIO* is not available for programming from the Teach Pendant (only from Program Maker). Option *Developer's Functions* is required.

Syntax

AliasIO
[FromSignal ':='] < reference (**REF**) of *anytype*> ','
[ToSignal ':='] < variable (**VAR**) of *anytype*> ';'

Related information

	Described in:
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User's Guide - System Parameters
Defining event routines	User's Guide - System Parameters
Loaded/Built-in task modules	User's Guide - System Parameters

Break Break program execution

Break is used to make an immediate break in program execution for RAPID program code debugging purposes.

Example

```
..
Break;
...
```

Program execution stops and it is possible to analyse variables, values etc. for debugging purposes.

Program execution

The instruction stops program execution at once, without waiting for the robot and external axes to reach their programmed destination points for the movement being performed at the time. Program execution can then be restarted from the next instruction.

If there is a *Break* instruction in some event routine, the routine will be executed from the beginning of the next event.

Syntax

Break’;’

Related information

	Described in:
Stopping for program actions	Instructions - *Stop*
Stopping after a fatal error	Instructions - *EXIT*
Terminating program execution	Instructions - *EXIT*
Only stopping robot movements	Instructions - *StopMove*

ProcCall Calls a new procedure

A procedure call is used to transfer program execution to another procedure. When the procedure has been fully executed, program execution continues with the instruction following the procedure call.

It is usually possible to send a number of arguments to the new procedure. These control the behaviour of the procedure and make it possible for the same procedure to be used for different things.

Examples

```
weldpipe1;
```

Calls the *weldpipe1* procedure.

```
errormessage;
Set do1;
   .

PROC errormessage()
   TPWrite "ERROR";
ENDPROC
```

The *errormessage* procedure is called. When this procedure is ready, program execution returns to the instruction following the procedure call, *Set do1*.

Arguments

Procedure { Argument }

Procedure Identifier

The name of the procedure to be called.

Argument Data type: In accordance with the procedure declaration

The procedure arguments (in accordance with the parameters of the procedure).

Example

```
weldpipe2 10, lowspeed;
```

Calls the *weldpipe2* procedure, including two arguments.

```
weldpipe3 10 \speed:=20;
```

Calls the *weldpipe3* procedure, including one mandatory and one optional argument.

Limitations

The procedure's arguments must agree with its parameters:

- All mandatory arguments must be included.
- They must be placed in the same order.
- They must be of the same data type.
- They must be of the correct type with respect to the access-mode (input, variable or persistent).

A routine can call a routine which, in turn, calls another routine, etc. A routine can also call itself, i.e. a recursive call. The number of routine levels permitted depends on the number of parameters, but more than 10 levels are usually permitted.

Syntax

(EBNF)
<procedure> [<argument list>] ';'

<procedure> ::= <identifier>

Related information

	Described in:
Arguments, parameters	Basic Characteristics - *Routines*
More examples	Program Examples

CallByVar Call a procedure by a variable

CallByVar (*Call By Variable*) can be used to call procedures with specific names, e.g. *proc_name1, proc_name2, proc_name3 ... proc_namex* via a variable.

Example

```
reg1 := 2;
CallByVar “proc”, reg1;
```

The procedure *proc2* is called.

Arguments

CallByVar Name Number

Name Data type: *string*

The first part of the procedure name, e.g. *proc_name*.

Number Data type: *num*

The numeric value for the number of the procedure. This value will be converted to a string and gives the 2:nd part of the procedure name e.g. *1*. The value must be a positive integer.

Example

Static selection of procedure call

```
TEST reg1
   CASE 1:
      lf_door door_loc;
   CASE 2:
      rf_door door_loc;
   CASE 3:
      lr_door door_loc;
   CASE 4:
      rr_door door_loc;
   DEFAULT:
      EXIT;
ENDTEST
```

Depending on whether the value of register *reg1* is 1, 2, 3 or 4, different procedures are called that perform the appropriate type of work for the selected door.

The door location in argument *door_loc*.

Dynamic selection of procedure call with RAPID syntax

```
reg1 := 2;
%”proc”+NumToStr(reg1,0)% door_loc;
```

The procedure *proc2* is called with argument *door_loc*.

Limitation: All procedures must have a specific name e.g. *proc1, proc2, proc3.*

Dynamic selection of procedure call with CallByVar

```
reg1 := 2;
CallByVar “proc”,reg1;
```

The procedure *proc2* is called.

Limitation: All procedures must have specific name, e.g. *proc1, proc2, proc3,* and no arguments can be used.

Limitations

Can only be used to call procedures without parameters.

Execution of CallByVar takes a little more time than execution of a normal procedure call.

Error handling

In the event of a reference to an unknown procedure, the system variable ERRNO is set to ERR_REFUNKPRC.

In the event of the procedure call error (not procedure), the system variable ERRNO is set to ERR_CALLPROC.

These errors can be handled in the error handler.

Syntax

CallByVar
[Name ‘:=’] <expression (**IN**) of *string*>’,’
[Number ‘:=‘] <expression (**IN**) of *num*>’;’

Related information

	Described in:
Calling procedures	Basic Characteristic - *Routines* User’s Guide - *The programming language RAPID*

Clear Clears the value

Clear is used to clear a numeric variable or persistent , i.e. it sets it to 0.

Example

Clear reg1;

Reg1 is cleared, i.e. reg1:=0.

Arguments

Clear Name

Name Data type: *num*

The name of the variable or persistent to be cleared.

Syntax

Clear
[Name ':='] < var or pers (**INOUT**) of *num* > ';'

Related information

	Described in:
Incrementing a variable by 1	Instructions - *Incr*
Decrementing a variable by 1	Instructions - *Decr*

ClkReset Resets a clock used for timing

ClkReset is used to reset a clock that functions as a stop-watch used for timing.

This instruction can be used before using a clock to make sure that it is set to 0.

Example

```
ClkReset clock1;
```

The clock *clock1* is reset.

Arguments

ClkReset Clock

Clock Data type: *clock*

The name of the clock to reset.

Program execution

When a clock is reset, it is set to 0.

If a clock is running, it will be stopped and then reset.

Syntax

```
ClkReset
   [ Clock ':=' ] < variable (VAR) of clock > ';'
```

Related Information

	Described in:
Other clock instructions	RAPID Summary - *System & Time*

ClkStart Starts a clock used for timing

ClkStart is used to start a clock that functions as a stop-watch used for timing.

Example

ClkStart clock1;

The clock *clock1* is started.

Arguments

ClkStart Clock

Clock Data type: *clock*

The name of the clock to start.

Program execution

When a clock is started, it will run and continue counting seconds until it is stopped.

A clock continues to run when the program that started it is stopped. However, the event that you intended to time may no longer be valid. For example, if the program was measuring the waiting time for an input, the input may have been received while the program was stopped. In this case, the program will not be able to “see” the event that occurred while the program was stopped.

A clock continues to run when the robot is powered down as long as the battery back-up retains the program that contains the clock variable.

If a clock is running it can be read, stopped or reset.

Example

```
VAR clock clock2;

ClkReset clock2;
ClkStart clock2;
WaitUntil DInput(di1) = 1;
ClkStop clock2;
time:=ClkRead(clock2);
```

The waiting time for *di1* to become 1 is measured.

Syntax

ClkStart
[Clock ':='] < variable (**VAR**) of *clock* > ';'

Related Information

	Described in:
Other clock instructions	RAPID Summary - *System & Time*

ClkStop Stops a clock used for timing

ClkStop is used to stop a clock that functions as a stop-watch used for timing.

Example

ClkStop clock1;

The clock *clock1* is stopped.

Arguments

ClkStop Clock

Clock Data type: *clock*

The name of the clock to stop.

Program execution

When a clock is stopped, it will stop running.

If a clock is stopped, it can be read, started again or reset.

Syntax

ClkStop
[Clock ':='] < variable (**VAR**) of *clock* > ';'

Related Information

	Described in:
Other clock instructions	RAPID Summary - *System & Time*
More examples	Instructions - *ClkStart*

Close Closes a file or serial channel

Close is used to close a file or serial channel.

Example

Close channel2;

The serial channel referred to by *channel2* is closed.

Arguments

Close IODevice

IODevice Data type: *iodev*

The name (reference) of the file or serial channel to be closed.

Program execution

The specified file or serial channel is closed and must be re-opened before reading or writing. If it is already closed, the instruction is ignored.

Syntax

Close
[IODevice ':='] <variable (**VAR**) of *iodev*>';'

Related information

	Described in:
Opening a file or serial channel	RAPID Summary - *Communication*

ClearIOBuff Clear input buffer of a serial channel

ClearIOBuff (*Clear I/O Buffer*) is used to clear the input buffer of a serial channel. All buffered characters from the input serial channel are discarded.

Example

```
VAR iodev channel2;
...
Open "sio1:", channel2 \Bin;
ClearIOBuff channel2;
```

The input buffer for the serial channel referred to by *channel2* is cleared.

Arguments

ClearIOBuff IODevice

IODevice Data type: *iodev*

The name (reference) of the serial channel whose input buffer is to be cleared.

Program execution

All buffered characters from the input serial channel are discarded. Next read instructions will wait for new input from the channel.

Limitations

This instruction can only be used for serial channels.

Syntax

ClearIOBuff
[IODevice ':='] <variable (**VAR**) of *iodev*>';'

Related information

	Described in:
Opening a serial channel	RAPID Summary - *Communication*

comment Comment

Comment is only used to make the program easier to understand. It has no effect on the execution of the program.

Example

```
! Goto the position above pallet
MoveL p100, v500, z20, tool1;
```

A comment is inserted into the program to make it easier to understand.

Arguments

! Comment

Comment Text string

Any text.

Program execution

Nothing happens when you execute this instruction.

Syntax

(EBNF)
'!' {<character>} <newline>

Related information

	Described in:
Characters permitted in a comment	Basic Characteristics- *Basic Elements*
Comments within data and routine declarations	Basic Characteristics- *Basic Elements*

ConfJ Controls the configuration during joint movement

ConfJ (Configuration Joint) is used to specify whether or not the robot's configuration is to be controlled during joint movement. If it is not controlled, the robot can sometimes use a different configuration than that which was programmed.

With ConfJ\Off, the robot cannot switch main axes configuration - it will search for a solution with the same main axes configuration as the current one. It moves to the closest wrist configuration for axes 4 and 6.

Examples

```
ConfJ \Off;
MoveJ *, v1000, fine, tool1;
```

The robot moves to the programmed position and orientation. If this position can be reached in several different ways, with different axis configurations, the closest possible position is chosen.

```
ConfJ \On;
MoveJ *, v1000, fine, tool1;
```

The robot moves to the programmed position, orientation and axis configuration. If this is not possible, program execution stops.

Arguments

ConfJ [\On] | [\Off]

\On Data type: *switch*

The robot always moves to the programmed axis configuration. If this is not possible using the programmed position and orientation, program execution stops.

The IRB5400 robot will move to the pogrammed axis configuration or to an axis configuration close the the programmed one. Program execution will not stop if it is impossible to reach the programmed axis configuration.

\Off Data type: *switch*

The robot always moves to the closest axis configuration.

Program execution

If the argument *\On* (or no argument) is chosen, the robot always moves to the programmed axis configuration. If this is not possible using the programmed position and orientation, program execution stops before the movement starts.

If the argument *\Off* is chosen, the robot always moves to the closest axis configuration. This may be different to the programmed one if the configuration has been incorrectly specified manually, or if a program displacement has been carried out.

The control is active by default. This is automatically set

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Syntax

```
ConfJ
   [ ’\’ On] | [ ’\’ Off] ’;’
```

Related information

	Described in:
Handling different configurations	Motion Principles - *Robot Configuration*
Robot configuration during linear movement	Instructions - *ConfL*

ConfL Monitors the configuration during linear movement

ConfL (Configuration Linear) is used to specify whether or not the robot's configuration is to be monitored during linear or circular movement. If it is not monitored, the configuration at execution time may differ from that at programmed time. It may also result in unexpected sweeping robot movements when the mode is changed to joint movement.

NOTE: For the IRB 6400 robot the monotoring is always off independant of the switch.

Examples

```
ConfL \On;
MoveL *, v1000, fine, tool1;
```

Program execution stops when the programmed configuration is not possible to reach from the current position.

```
SingArea \Wrist;
Confl \On;
MoveL *, v1000, fine, tool1;
```

The robot moves to the programmed position, orientation and wrist axis configuration. If this is not possible, program execution stops.

```
ConfL \Off;
MoveL *, v1000, fine, tool1;
```

No error message is displayed when the programmed configuration is not the same as the configuration achieved by program execution.

Arguments

ConfL [\On] | [\Off]

\On Data type: *switch*

The robot configuration is monitored.

\Off Data type: *switch*

The robot configuration is not monitored.

Program execution

During linear or circular movement, the robot always moves to the programmed position and orientation that has the closest possible axis configuration. If the argument *\On* (or no argument) is chosen, then the program execution stops as soon as:

- the configuration of the programmed position will not be attained from the current position.
- the needed reorientation of any one of the wrist axes to get to the programmed position from the current position exceeds a limit (140-180 degrees).

However, it is possible to restart the program again, although the wrist axes may continue to the wrong configuration. At a stop point, the robot will check that the configurations of all axes are achieved, not only the wrist axes.

If SingArea\Wrist is also used, the robot always moves to the programmed wrist axes configuration and at a stop point the remaining axes configurations will be checked.

If the argument *\Off* is chosen, there is no monitoring.

Monitoring is active by default. This is automatically set

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Syntax

```
ConfL
   [ '\' On] | [ '\' Off] ';'
```

Related information

	Described in:
Handling different configurations	Motion and I/O Principles- *Robot Configuration*
Robot configuration during joint movement	Instructions - *ConfJ*

CONNECT Connects an interrupt to a trap routine

CONNECT is used to find the identity of an interrupt and connect it to a trap routine.

The interrupt is defined by ordering an interrupt event and specifying its identity. Thus, when that event occurs, the trap routine is automatically executed.

Example

```
VAR intnum feeder_low;
CONNECT feeder_low WITH feeder_empty;
ISignalDI di1, 1 , feeder_low;
```

An interrupt identity *feeder_low* is created which is connected to the trap routine *feeder_empty*. The interrupt is defined as *input di1 is getting high*. In other words, when this signal becomes high, the *feeder_empty* trap routine is executed.

Arguments

CONNECT Interrupt WITH Trap routine

Interrupt Data type: *intnum*

The variable that is to be assigned the identity of the interrupt. This must not be declared within a routine (routine data).

Trap routine Identifier

The name of the trap routine.

Program execution

The variable is assigned an interrupt identity which can then be used when ordering or disabling interrupts. This identity is also connected to the specified trap routine.

Note that before an event can be handled, an interrupt must also be ordered, i.e. the event specified.

Limitations

An interrupt (interrupt identity) cannot be connected to more than one trap routine. Different interrupts, however, can be connected to the same trap routine.

When an interrupt has been connected to a trap routine, it cannot be reconnected or transferred to another routine; it must first be deleted using the instruction *IDelete*.

Error handling

If the interrupt variable is already connected to a TRAP routine, the system variable ERRNO is set to ERR_ALRDYCNT.

If the interrupt variable is not a variable reference, the system variable ERRNO is set to ERR_CNTNOTVAR.

If no more interrupt numbers are available, the system variable ERRNO is set to ERR_INOMAX.

These errors can be handled in the ERROR handler.

Syntax

(EBNF)
CONNECT <connect target> **WITH** <trap>';'

<connect target> ::= <variable>
| <parameter>
| <VAR>
<trap> ::= <identifier>

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
More information on interrupt management	Basic Characteristics- *Interrupts*

DeactUnit Deactivates a mechanical unit

DeactUnit is used to deactivate a mechanical unit.

It can be used to determine which unit is to be active when, for example, common drive units are used.

Examples

```
DeactUnit orbit_a;
```

Deactivation of the *orbit_a* mechanical unit.

```
MoveL p10, v100, fine, tool1;
DeactUnit track_motion;
MoveL p20, v100, z10, tool1;
MoveL p30, v100, fine, tool1;
ActUnit track_motion;
MoveL p40, v100, z10, tool1;
```

The unit *track_motion* will be stationary when the robot moves to *p20* and *p30*. After this, both the robot and *track_motion* will move to *p40*.

```
MoveL p10, v100, fine, tool1;
DeactUnit orbit1;
ActUnit orbit2;
MoveL p20, v100, z10, tool1;
```

The unit *orbit1* is deactivated and *orbit2* activated.

Arguments

DeactUnit MecUnit

MecUnit *(Mechanical Unit)* Data type: *mecunit*

The name of the mechanical unit that is to be deactivated.

Program execution

When the robot and external axes have come to a standstill, the specified mechanical unit is deactivated. This means that it will neither be controlled nor monitored until it is re-activated.

If several mechanical units share a common drive unit, deactivation of one of the mechanical units will also disconnect that unit from the common drive unit.

Limitations

Instruction DeactUnit cannot be used

- in program sequence StorePath ... RestoPath
- in event routine RESTART
- when one of the axes in the mechanical unit is in independent mode.

The movement instruction previous to this instruction, should be terminated with a stop point in order to make a restart in this instruction possible following a power failure.

Syntax

DeactUnit
[MecUnit ':='] < variable (**VAR**) of *mecunit*> ';'

Related information

	Described in:
Activating mechanical units	Instructions - *ActUnit*
Mechanical units	Data Types - *mecunit*

Decr Decrements by 1

Decr is used to subtract 1 from a numeric variable or persistent.

Example

Decr reg1;

1 is subtracted from *reg1*, i.e. reg1:=reg1-1.

Arguments

Decr Name

Name Data type: *num*

The name of the variable or persistent to be decremented.

Example

```
TPReadNum no_of_parts, "How many parts should be produced? ";
WHILE no_of_parts>0 DO
   produce_part;
   Decr no_of_parts;
ENDWHILE
```

The operator is asked to input the number of parts to be produced. The variable *no_of_parts* is used to count the number that still have to be produced.

Syntax

Decr
[Name ':='] < var or pers (**INOUT**) of *num* > ';'

Related information

	Described in:
Incrementing a variable by 1	Instructions - *Incr*
Subtracting any value from a variable	Instructions - *Add*
Changing data using an arbitrary expression, e.g. multiplication	Instructions - *:=*

EOffsOff Deactivates an offset for external axes

EOffsOff (External Offset Off) is used to deactivate an offset for external axes.

The offset for external axes is activated by the instruction *EOffsSet* or *EOffsOn* and applies to all movements until some other offset for external axes is activated or until the offset for external axes is deactivated.

Examples

EOffsOff;

Deactivation of the offset for external axes.

```
MoveL p10, v500, z10, tool1;
EOffsOn \ExeP:=p10, p11;
MoveL p20, v500, z10, tool1;
MoveL p30, v500, z10, tool1;
EOffsOff;
MoveL p40, v500, z10, tool1;
```

An offset is defined as the difference between the position of each axis at *p10* and *p11*. This displacement affects the movement to *p20* and *p30,* but not to *p40*.

Program execution

Active offsets for external axes are reset.

Syntax

EOffsOff ‘;’

Related information

	Described in:
Definition of offset using two positions	Instructions - *EOffsOn*
Definition of offset using values	Instructions - *EOffsSet*
Deactivation of the robot’s motion displacement	Instructions - *PDispOff*

EOffsOn Activates an offset for external axes

EOffsOn (External Offset On) is used to define and activate an offset for external axes using two positions.

Examples

```
MoveL p10, v500, z10, tool1;
EOffsOn \ExeP:=p10, p20;
```

Activation of an offset for external axes. This is calculated for each axis based on the difference between positions *p10* and *p20*.

```
MoveL p10, v500, fine, tool1;
EOffsOn *;
```

Activation of an offset for external axes. Since a stop point has been used in the previous instruction, the argument \ExeP does not have to be used. The displacement is calculated on the basis of the difference between the actual position of each axis and the programmed point (*) stored in the instruction.

Arguments

EOffsOn [\ExeP] ProgPoint

[\ExeP] *(Executed Point)* Data type: *robtarget*

The new position of the axes at the time of the program execution. If this argument is omitted, the current position of the axes at the time of the program execution is used.

ProgPoint *(Programmed Point)* Data type: *robtarget*

The original position of the axes at the time of programming.

Program execution

The offset is calculated as the difference between *ExeP* and *ProgPoint* for each separate external axis. If *ExeP* has not been specified, the current position of the axes at the time of the program execution is used instead. Since it is the actual position of the axes that is used, the axes should not move when *EOffsOn* is executed.

This offset is then used to displace the position of external axes in subsequent positioning instructions and remains active until some other offset is activated (the instruction *EOffsSet* or *EOffsOn*) or until the offset for external axes is deactivated (the instruction *EOffsOff*).

Only one offset for each individual external axis can be activated at any one time. Several *EOffsOn*, on the other hand, can be programmed one after the other and, if they are, the different offsets will be added.

The external axes' offset is automatically reset

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Example

```
SearchL sen1, psearch, p10, v100, tool1;
PDispOn \ExeP:=psearch, *, tool1;
EOffsOn \ExeP:=psearch, *;
```

A search is carried out in which the searched position of both the robot and the external axes is stored in the position *psearch*. Any movement carried out after this starts from this position using a program displacement of both the robot and the external axes. This is calculated based on the difference between the searched position and the programmed point (*) stored in the instruction.

Syntax

```
EOffsOn
   [ '\' ExeP ':=' < expression (IN) of robtarget > ',']
   [ ProgPoint ':=' ] < expression (IN) of robtarget > ';'
```

Related information

	Described in:
Deactivation of offset for external axes	Instructions - *EOffsOff*
Definition of offset using values	Instructions - *EOffsSet*
Displacement of the robot’s movements	Instructions - *PDispOn*
Coordinate Systems	Motion Principles- *Coordinate Systems*

EOffsSet Activates an offset for external axes using a value

EOffsSet (External Offset Set) is used to define and activate an offset for external axes using values.

Example

```
VAR extjoint eax_a_p100 := [100, 0, 0, 0, 0, 0];
.
EOffsSet eax_a_p100;
```

Activation of an offset *eax_a_p100* for external axes, meaning (provided that the external axis “a” is linear) that:

- The ExtOffs coordinate system is displaced 100 mm for the logical axis “a” (see Figure 18).
- As long as this offset is active, all positions will be displaced 100 mm in the direction of the x-axis.



Figure 18 Displacement of an external axis.

Arguments

EOffsSet EAxOffs

EAxOffs *(External Axes Offset)* Data type: *extjoint*

The offset for external axes is defined as data of the type *extjoint*, expressed in:

- mm for linear axes
- degrees for rotating axes

Program execution

The offset for external axes is activated when the *EOffsSet* instruction is activated and remains active until some other offset is activated (the instruction *EOffsSet* or *EOffsOn*) or until the offset for external axes is deactivated (the *EOffsOff*).

Only one offset for external axes can be activated at any one time. Offsets cannot be added to one another using *EOffsSet*.

The external axes' offset is automatically reset

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Syntax

EOffsSet
[EAxOffs ':='] < expression (**IN**) of *extjoint*> ';'

Related information

	Described in:
Deactivation of offset for external axes	Instructions - *EOffsOff*
Definition of offset using two positions	Instructions - *EOffsSet*
Displacement of the robot's movements	Instructions - *PDispOn*
Definition of data of the type *extjoint*	Data Types - *extjoint*
Coordinate Systems	Motion Principles- *Coordinate Systems*

ErrWrite Write an Error Message

ErrWrite (Error Write) is used to display an error message on the teach pendant and write it in the robot message log.

Example

```
ErrWrite “PLC error”, “Fatal error in PLC” \RL2:=”Call service”;
Stop;
```

A message is stored in the robot log. The message is also shown on the teach pendant display.

```
ErrWrite \ W, “ Search error”, “No hit for the first search”;
RAISE try_search_again;
```

A message is stored in the robot log only. Program execution then continues.

Arguments

ErrWrite [\W] Header Reason [\RL2] [\RL3] [\RL4]

[\W] *(Warning)* Data type: *switch*

Gives a warning that is stored in the robot error message log only (not shown directly on the teach pendant display).

Header Data type: *string*

Error message heading (max. 24 characters).

Reason Data type: string

Reason for error (line 1 of max. 40 characters).

[\RL2] *(Reason Line 2)* Data type: *string*

Reason for error (line 2 of max. 40 characters).

[\RL3] *(Reason Line 3)* Data type: *string*

Reason for error (line 3 of max. 40 characters).

[\RL4] *(Reason Line 4)* Data type: *string*

Reason for error (line 4 of max. 40 characters).

Program execution

An error message (max. 5 lines) is displayed on the teach pendant and written in the robot message log.

ErrWrite always generates the program error no. 80001 or in the event of a warning (argument \W) generates no. 80002.

Limitations

Total string length (Header+Reason+\RL2+\RL3+\RL4) is limited to 145 characters.

Syntax

ErrWrite
['\' W ',']
[Header ':='] < expression (**IN**) of *string*> ','
[Reason ':='] < expression (**IN**) of *string*>
['\' RL2 ':=' < expression (**IN**) of *string*>]
['\' RL3 ':=' < expression (**IN**) of *string*>]
['\' RL4 ':=' < expression (**IN**) of *string*>] ';'

Related information

	Described in:
Display a message on the teach pendant only	Instructions - *TPWrite*
Message logs	Service

EXIT Terminates program execution

EXIT is used to terminate program execution. Program restart will then be blocked, i.e. the program can only be restarted from the first instruction of the main routine (if the start point is not moved manually).

The *EXIT* instruction should be used when fatal errors occur or when program execution is to be stopped permanently. The *Stop* instruction is used to temporarily stop program execution.

Example

```
ErrWrite "Fatal error","Illegal state";
EXIT;
```

Program execution stops and cannot be restarted from that position in the program.

Syntax

EXIT ’;’

Related information

	Described in:
Stopping program execution temporarily	Instructions - *Stop*

ExitCycle Break current cycle and start next

ExitCycle is used to break the current cycle and move the PP back to the first instruction in the main routine.

If the program is executed in continuous mode, it will start to execute the next cycle. If the execution is in cycle mode, the execution will stop at the first instruction in the main routine.

Example

```
VAR num cyclecount:=0;
VAR intnum error_intno;

PROC main()
   IF cyclecount = 0 THEN
      CONNECT error_intno WITH error_trap;
      ISignalDI di_error,1,error_intno;
   ENDIF
   cyclecount:=cyclecount+1;
   ! start to do something intelligent
   ....

ENDPROC

TRAP error_trap
   TPWrite “ERROR, I will start on the next item”;
   ExitCycle;
ENDTRAP
```

This will start the next cycle if the signal *di_error* is set.

Program execution

Execution of *ExitCycle* in the MAIN program task, results in the following in the MAIN task:

- On-going robot movements stops
- All robot paths that are not performed at all path levels (both normal and *StorePath* level) are cleared
- All instructions that are started but not finished at all execution levels (both normal and TRAP level) are interrupted
- The program pointer is moved to the first instruction in the main routine
- The program execution continues to execute the next cycle

Execution of *ExitCycle* in some other program task (besides MAIN) results in the following in the actual task:

- All instructions that are started but not finished on all execution levels (both normal and TRAP level) are interrupted
- The program pointer is moved to the first instruction in the main routine
- The program execution continues to execute the next cycle

All other modal things in the program and system are **not** affected by *ExitCycle* such as:

- The actual value of variables or persistents
- Any motion settings such as *StorePath-RestoPath* sequence, world zones, etc.
- Open files, directories, etc.
- Defined interrupts, etc.

Syntax

ExitCycle';'

Related information

	Described in:
Stopping after a fatal error	Instructions - *EXIT*
Terminating program execution	Instructions - *EXIT*
Stopping for program actions	Instructions - *Stop*
Finishing execution of a routine	Instructions - *RETURN*

FOR Repeats a given number of times

FOR is used when one or several instructions are to be repeated a number of times.

If the instructions are to be repeated as long as a given condition is met, the *WHILE* instruction is used.

Example

```
FOR i FROM 1 TO 10 DO
   routine1;
ENDFOR
```

Repeats the *routine1* procedure *10* times.

Arguments

FOR Loop counter FROM Start value TO End value [STEP Step value] DO ... ENDFOR

Loop counter Identifier

The name of the data that will contain the value of the current loop counter. The data is declared automatically and its name should therefore not be the same as the name of any data that exists already.

Start value Data type: *Num*

The desired start value of the loop counter.
(usually integer values)

End value Data type: *Num*

The desired end value of the loop counter.
(usually integer values)

Step value Data type: *Num*

The value by which the loop counter is to be incremented (or decremented) each loop.
(usually integer values)

If this value is not specified, the step value will automatically be set to 1 (or -1 if the start value is greater than the end value).

Example

```
FOR i FROM 10 TO 2 STEP -1 DO
   a{i} := a{i-1};
ENDFOR
```

The values in an array are adjusted upwards so that a{10}:=a{9}, a{9}:=a{8} etc.

Program execution

1. The expressions for the start, end and step values are calculated.
2. The loop counter is assigned the start value.
3. The value of the loop counter is checked to see whether its value lies between the start and end value, or whether it is equal to the start or end value. If the value of the loop counter is outside of this range, the FOR loop stops and program execution continues with the instruction following ENDFOR.
4. The instructions in the FOR loop are executed.
5. The loop counter is incremented (or decremented) in accordance with the step value.
6. The FOR loop is repeated, starting from point 3.

Limitations

The loop counter (of data type *num*) can only be accessed from within the FOR loop and consequently hides other data and routines that have the same name. It can only be read (not updated) by the instructions in the FOR loop.

Decimal values for start, end or stop values, in combination with exact termination conditions for the FOR loop, cannot be used (undefined whether or not the last loop is running).

Syntax

(EBNF)
FOR <loop variable> **FROM** <expression> **TO** <expression>
   [**STEP** <expression>] **DO**
   <instruction list>
ENDFOR

<loop variable> ::= <identifier>

Related information

	Described in:
Expressions	Basic Characteristics - *Expressions*
Identifiers	Basic Characteristics - *Basic Elements*

GetSysData Get system data

GetSysData fetches the value and optional symbol name for the current system data of specified data type.

With this instruction it is possible to fetch data for and the name of the current active Tool or Work Object.

Example

```
PERS tooldata curtoolvalue := [TRUE, [[0, 0, 0], [1, 0, 0, 0]],
                                [0, [0, 0, 0], [1, 0, 0, 0], 0, 0, 0]];
VAR string curtoolname;

GetSysData curtoolvalue;
```

Copy current active tool data value to the persistent variable *curtoolvalue*.

```
GetSysData curtoolvalue \ObjectName := curtoolname;
```

Copy also current active tool name to the variable *curtoolname*.

Arguments

GetSysData DestObject [\ ObjectName]

DestObject Data type: *anytype*

Persistent for storage of current active system data value.

The data type of this argument also specifies the type of system data (Tool or Work Object) to fetch.

[\ObjectName] Data type: *string*

Option argument (variable or persistent) to also fetch the current active system data name.

Program execution

When running the instruction *GetSysData* the current data value is stored in the specified persistent in argument *DestObject*.

If argument *\ObjectName* is used, the name of the current data is stored in the specified variable or persistent in argument *ObjectName*.

Current system data for Tool or Work Object is activated by execution of any move instruction or can be manually set in the jogging window.

Syntax

GetSysData
[DestObject':='] < persistent(**PERS**) of *anytype*>
['\'ObjectName':=' < expression (**INOUT**) of *string*>] ';'

Related information

	Described in:
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*

GOTO Goes to a new instruction

GOTO is used to transfer program execution to another line (a label) within the same routine.

Examples

```
GOTO next;
   .
next:
```

Program execution continues with the instruction following *next*.

```
reg1 := 1;
next:
   .
reg1 := reg1 + 1;
IF reg1<=5 GOTO next;
```

The *next* program loop is executed five times.

```
IF reg1>100 GOTO highvalue;
lowvalue:
   .
GOTO ready;
highvalue:
   .
ready:
```

If *reg1 is greater than 100*, the *highvalue* program loop is executed; otherwise the *lowvalue* loop is executed.

Arguments

GOTO Label

Label Identifier

The label from where program execution is to continue.

Limitations

It is only possible to transfer program execution to a label within the same routine.

It is only possible to transfer program execution to a label within an IF or TEST instruction if the GOTO instruction is also located within the same branch of that instruction.

It is only possible to transfer program execution to a label within a FOR or WHILE instruction if the GOTO instruction is also located within that instruction.

Syntax

(EBNF)
GOTO <identifier>';'

Related information

	Described in:
Label	Instructions - *label*
Other instructions that change the program flow	RAPID Summary - *Controlling the Program Flow*

GripLoad Defines the payload of the robot

GripLoad is used to define the payload which the robot holds in its gripper.

Description

It is important to always define the actual tool load and when used, the payload of the robot too. Incorrect definitions of load data can result in overloading of the robot mechanical structure.

When incorrect load data is specified, it can often lead to the following consequences:

- If the value in the specified load data is greater than that of the value of the true load;
  -> The robot will not be used to its maximum capacity
  -> Impaired path accuracy including a risk of overshooting

If the value in the specified load data is less than the value of the true load;
-> Impaired path accuracy including a risk of overshooting
-> Risk of overloading the mechanical structure

Examples

GripLoad piece1;

The robot gripper holds a load called *piece1*.

GripLoad load0;

The robot gripper releases all loads.

Arguments

GripLoad Load

Load Data type: *loaddata*

The load data that describes the current payload.

Program execution

The specified load affects the performance of the robot.

The default load, 0 kg, is automatically set

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Syntax

GripLoad
[Load ':='] < persistent (**PERS**) of *loaddata* > ';'

Related information

	Described in:
Definition of load data	Data Types - *loaddata*
Definition of tool load	Data Types - *tooldata*

IDelete Cancels an interrupt

IDelete (Interrupt Delete) is used to cancel (delete) an interrupt.

If the interrupt is to be only temporarily disabled, the instruction *ISleep* or *IDisable* should be used.

Example

IDelete feeder_low;

The interrupt *feeder_low* is cancelled.

Arguments

IDelete Interrupt

Interrupt Data type: *intnum*

The interrupt identity.

Program execution

The definition of the interrupt is completely erased. To define it again, it must first be re-connected to the trap routine.

The instruction should be preceded by a stop point. Otherwise the interrupt will be deactivated before the end point is reached.

Interrupts do not have to be erased; this is done automatically when

- a new program is loaded
- the program is restarted from the beginning
- the program pointer is moved to the start of a routine

Syntax

IDelete
[Interrupt ':='] < variable (**VAR**) of *intnum* > ';'

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Temporarily disabling an interrupt	Instructions - *ISleep*
Temporarily disabling all interrupts	Instructions - *IDisable*

IDisable Disables interrupts

IDisable (Interrupt Disable) is used to disable all interrupts temporarily. It may, for example, be used in a particularly sensitive part of the program where no interrupts may be permitted to take place in case they disturb normal program execution.

Example

```
IDisable;
FOR i FROM 1 TO 100 DO
   character[i]:=ReadBin(sensor);
ENDFOR
IEnable;
```

No interrupts are permitted as long as the serial channel is reading.

Program execution

Interrupts which occur during the time in which an *IDisable* instruction is in effect are placed in a queue. When interrupts are permitted once more, the interrupt(s) of the program then immediately start generating, executed in “first in - first out” order in the queue.

Syntax

IDisable‘;’

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupt*
Permitting interrupts	Instructions - *IEnable*

IEnable Enables interrupts

IEnable (Interrupt Enable) is used to enable interrupts during program execution.

Example

```
IDisable;
FOR i FROM 1 TO 100 DO
   character[i]:=ReadBin(sensor);
ENDFOR
IEnable;
```

No interrupts are permitted as long as the serial channel is reading. When it has finished reading, interrupts are once more permitted.

Program execution

Interrupts which occur during the time in which an *IDisable* instruction is in effect, are placed in a queue. When interrupts are permitted once more (*IEnable*), the interrupt(s) of the program then immediately start generating, executed in “first in - first out” order in the queue.Program execution then continues in the ordinary program and interrupts which occur after this are dealt with as soon as they occur.

Interrupts are always permitted when a program is started from the beginning,. Interrupts disabled by the *ISleep* instruction are not affected by the *IEnable* instruction.

Syntax

IEnable‘;’

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Permitting no interrupts	Instructions - *IDisable*

Compact IF If a condition is met, then... (one instruction)

Compact IF is used when a single instruction is only to be executed if a given condition is met.

If different instructions are to be executed, depending on whether the specified condition is met or not, the *IF* instruction is used.

Examples

IF reg1 > 5 GOTO next;

If *reg1 is greater than 5,* program execution continues at the *next* label.

IF counter > 10 Set do1;

The *do1* signal is set if *counter > 10.*

Arguments

IF Condition ...

Condition Data type: *bool*

The condition that must be satisfied for the instruction to be executed.

Syntax

(EBNF)
IF <conditional expression> (<instruction> | **<SMT>**) ';'

Related information

	Described in:
Conditions (logical expressions)	Basic Characteristics - *Expressions*
IF with several instructions	Instructions - *IF*

IF If a condition is met, then ...; otherwise ...

IF is used when different instructions are to be executed depending on whether a condition is met or not.

Examples

```
IF reg1 > 5 THEN
   Set do1;
   Set do2;
ENDIF
```

The *do1* and *do2* signals are set only if *reg1 is greater than 5*.

```
IF reg1 > 5 THEN
   Set do1;
   Set do2;
ELSE
   Reset do1;
   Reset do2;
ENDIF
```

The *do1* and *do2* signals are set or reset depending on whether *reg1 is greater than 5* or not.

Arguments

IF Condition THEN ...
{ELSEIF Condition THEN ...}
[ELSE ...]
ENDIF

Condition Data type: *bool*

The condition that must be satisfied for the instructions between THEN and ELSE/ELSEIF to be executed.

Example

```
IF counter > 100 THEN
   counter := 100;
ELSEIF counter < 0 THEN
   counter := 0;
ELSE
   counter := counter + 1;
ENDIF
```

Counter is incremented by 1. However, if the value of *counter* is outside the limit *0-100, counter* is assigned the corresponding limit value.

Program execution

The conditions are tested in sequential order, until one of them is satisfied. Program execution continues with the instructions associated with that condition. If none of the conditions are satisfied, program execution continues with the instructions following ELSE. If more than one condition is met, only the instructions associated with the first of those conditions are executed.

Syntax

(EBNF)
IF <conditional expression> **THEN**
   <instruction list>
{**ELSEIF** <conditional expression> **THEN** <instruction list> | **<EIF>**}
[**ELSE**
   <instruction list>]
ENDIF

Related information

	Described in:
Conditions (logical expressions)	Basic Characteristics - *Expressions*

Incr Increments by 1

Incr is used to add 1 to a numeric variable or persistent.

Example

Incr reg1;

1 is added to *reg1*, i.e. reg1:=reg1+1.

Arguments

Incr Name

Name Data type: *num*

The name of the variable or persistent to be changed.

Example

```
WHILE stop_production=0 DO
   produce_part;
   Incr no_of_parts;
   TPWrite "No of produced parts= "\Num:=no_of_parts;
ENDWHILE
```

The number of parts produced is updated on the teach pendant each cycle. Production continues to run as long as the signal *stop_production* is not set.

Syntax

Incr
[Name ':='] < var or pers (**INOUT**) of *num* > ';'

Related information

	Described in:
Decrementing a variable by 1	Instructions - *Decr*
Adding any value to a variable	Instructions - *Add*
Changing data using an arbitrary expression, e.g. multiplication	Instructions - *:=*

InvertDO Inverts the value of a digital output signal

InvertDO (Invert Digital Output) inverts the value of a digital output signal (0 -> 1 and 1 -> 0).

Example

InvertDO do15;

The current value of the signal *do15* is inverted.

Arguments

InvertDO Signal

Signal Data type: *signaldo*

The name of the signal to be inverted.

Program execution

The current value of the signal is inverted (see Figure 19).



Figure 19 Inversion of a digital output signal.

Syntax

InvertDO
[Signal ':='] < variable (**VAR**) of *signaldo* > ';'

Related information

	Described in:
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	System Parameters

IODisable Disable I/O unit

IODisable is used to disable an I/O unit during program execution (only in the S4C system).

I/O units are automatically enabled after start-up if they are defined in the system parameters. When required for some reason, I/O units can be disabled or enabled during program execution.

Examples

IODisable “cell1”, 5;

Disable I/O unit with name *cell1*.Wait max. *5* s.

Arguments

IODisable UnitName MaxTime

UnitName Data type: *string*

The name of the I/O unit to be disabled (with same name as configured).

MaxTime Data type: *num*

The maximum period of waiting time permitted, expressed in seconds. If this time runs out before the I/O unit has finished the disable steps, the error handler will be called, if there is one, with the error code ERR_IODISABLE. If there is no error handler, the execution will be stopped.

To disable an I/O unit takes about 2-5 s.

Program execution

The specified I/O unit starts the disable steps. The instruction is ready when the disable steps are finished. If the *MaxTime* runs out before the I/O unit has finished the disable steps, a recoverable error will be generated.

After disabling an I/O unit, any setting of outputs in this unit will result in an error.

Example

```
PROC go_home()
    VAR num recover_flag :=0;
    ...
    ! Start to disable I/O unit cell1
    recover_flag := 1;
    IODisable “cell1”, 0;
    ! Move to home position
    MoveJ home, v1000,fine,tool1;
    ! Wait until disable of I/O unit cell1 is ready
    recover_flag := 2;
    IODisable “cell1”, 5;
    ...
    ERROR
        IF ERRNO = ERR_IODISABLE THEN
            IF recover_flag = 1 THEN
                TRYNEXT;
            ELSEIF recover_flag = 2 THEN
                RETRY;
            ENDIF
        ELSEIF ERRNO = ERR_EXCRTYMAX THEN
            ErrWrite “IODisable error”, “Not possible to disable I/O unit cell1”;
            Stop;
        ENDIF
ENDPROC
```

To save cycle time, the I/O unit *cell1* is disabled during robot movement to the *home* position. With the robot at the *home* position, a test is done to establish whether or not the I/O unit *cell1* is fully disabled. After the max. number of retries (5 with a waiting time of *5* s), the robot execution will stop with an error message.

The same principle can be used with *IOEnable* (this will save more cycle time compared with *IODisable*).

Syntax

```
IODisable
    [ UnitName ’:=’ ] < expression (IN) of string> ’,’
    [ MaxTime ’:=’ ] < expression (IN) of num > ’;’
```

Related information

	Described in:
Enabling an I/O unit	Instructions - *IOEnable*
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User’s Guide - *System Parameters*

IOEnable Enable I/O unit

IOEnable is used to enable an I/O unit during program execution (only in the S4C system).

I/O units are automatically enabled after start-up if they are defined in the system parameters. When required for some reason, I/O units can be disabled or enabled during program execution.

Examples

IOEnable “cell1”, 5;

Enable I/O unit with name *cell1*. Wait max. *5* s.

Arguments

IOEnable UnitName MaxTime

UnitName Data type: *string*

The name of the I/O unit to be enabled (with same name as configured).

MaxTime Data type: *num*

The maximum period of waiting time permitted, expressed in seconds. If this time runs out before the I/O unit has finished the enable steps, the error handler will be called, if there is one, with the error code ERR_IOENABLE. If there is no error handler, the execution will be stopped.

To enable an I/O unit takes about 2-5 s.

Program execution

The specified I/O unit starts the enable steps. The instruction is ready when the enable steps are finished. If the *MaxTime* runs out before the I/O unit has finished the enable steps, a recoverable error will be generated.

After a sequence of *IODisable* - *IOEnable*, all outputs for the current I/O unit will be set to the old values (before *IODisable*).

Example

IOEnable can also be used to check whether some I/O unit is disconnected for some reason.

```
VAR num max_retry:=0;
...
IOEnable “cell1”, 0;
SetDO cell1_sig3, 1;
...
ERROR
    IF ERRNO = ERR_IOENABLE THEN
        IF max_retry < 5 THEN
            WaitTime 1;
            max_retry := max_retry + 1;
            RETRY;
        ELSE
            RAISE;
        ENDIF
    ENDIF
```

Before using signals on the I/O unit *cell1*, a test is done by trying to enable the I/O unit with timeout after *0* sec. If the test fails, a jump is made to the error handler. In the error handler, the program execution waits for *1* sec. and a new retry is made. After *5* retry attempts the error ERR_IOENABLE is propagated to the caller of this routine.

Syntax

```
IOEnable
    [ UnitName ’:=’ ] < expression (IN) of string> ’,’
    [ MaxTime ’:=’ ] < expression (IN) of num > ’;’
```

Related information

	Described in:
More examples	Instructions - *IODisable*
Disabling an I/O unit	Instructions - *IODisable*
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User’s Guide - *System Parameters*

ISignalAI Interrupts from analog input signal

ISignalAI (Interrupt Signal Analog Input) is used to order and enable interrupts from an analog input signal.

Example

```
VAR intnum sig1int;
CONNECT sig1int WITH iroutine1;
ISignalAI \Single, ai1, AIO_BETWEEN, 1.5, 0.5, 0, sig1int;
```

Orders an interrupt which is to occur the first time the logical value of the analog input signal *ai1* is between *0.5* and *1.5*. A call is then made to the *iroutine1* trap routine.

```
ISignalAI ai1, AIO_BETWEEN, 1.5, 0.5, 0.1, sig1int;
```

Orders an interrupt which is to occur each time the logical value of the analog input signal *ai1* is between *0.5* and *1.5*, and the absolute signal difference compared to the stored reference value is bigger than *0.1*.

```
ISignalAI ai1, AIO_OUTSIDE, 1.5, 0.5, 0.1, sig1int;
```

Orders an interrupt which is to occur each time the logical value of the analog input signal *ai1* is lower than *0.5* or higher than *1.5*, and the absolute signal difference compared to the stored reference value is bigger than *0.1*.

Arguments

ISignalAI [\Single] Signal Condition HighValue LowValue DeltaValue [\DPos] | [\DNeg] Interrupt

[\Single] Data type: *switch*

Specifies whether the interrupt is to occur once or cyclically.

If the argument *Single* is set, the interrupt occurs once at the most. If the argument is omitted, an interrupt will occur each time its condition is satisfied.

Signal Data type: *signalai*

The name of the signal that is to generate interrupts.

Condition Data type: *aiotrigg*

Specifies how *HighValue* and *LowValue* define the condition to be satisfied:

- AIO_ABOVE_HIGH: logical value of the signal is above HighValue
- AIO_BELOW_HIGH: logical value of the signal is below HighValue
- AIO_ABOVE_LOW: logical value of the signal is above LowValue
- AIO_BELOW_LOW: logical value of the signal is below LowValue
- AIO_BETWEEN: logical value of the signal is between LowValue and HighValue
- AIO_OUTSIDE: logical value of the signal is above HighValue or below LowValue
- AIO_ALWAYS: independently of HighValue and LowValue

HighValue Data type: *num*

High logical value to define the condition.

LowValue Data type: *num*

Low logical value to define the condition.

DeltaValue Data type: *num*

Defines the minimum logical signal difference before generation of a new interrupt. The actual signal value compared to the stored reference value must be greater than the specified *DeltaValue* before generation of a new interrupt.

[\DPos] Data type: *switch*

Specifies that only positive logical signal differences will give new interrupts.

[\DNeg] Data type: *switch*

Specifies that only negative logical signal differences will give new interrupts.

If none of *\DPos* and *\DNeg* argument is used, both positive and negative differences will generate new interrupts.

Interrupt Data type: *intnum*

The interrupt identity. This interrupt should have previously been connected to a trap routine by means of the instruction CONNECT.

Program execution

When the signal fulfils the specified conditions (both *Condition* **and** *DeltaValue*), a call is made to the corresponding trap routine. When this has been executed, program execution continues from where the interrupt occurred.

Conditions for interrupt generation

Before the interrupt subscription is ordered, each time the signal is sampled, the value of the signal is read, saved, and later used as a reference value for the *DeltaValue* condition.

At the interrupt subscription time, if specified *DeltaValue* = 0 **and** after the interrupt subscription time always at each time the signal is sampled, its value is then compared to *HighValue* and *LowValue* according to *Condition* and with consideration to *DeltaValue*, to generate or not generate an interrupt. If the new read value satisfies the specified *HighValue* and *LowValue Condition*, but its difference compared to the last stored reference value is less or equal to the *DeltaValue* argument, no interrupt occurs.
If the signal difference is not in the specified direction, no interrupts will occur. (argument *DPos* or *DNeg*).

The stored reference value for the *DeltaValue* condition is updated with a newly read value for later use at any sample, if the following conditions are satisfied:

- Argument *Condition* with specified *HighValue* and *LowValue* (within limits)
- Argument *DeltaValue* (sufficient signal change in any direction, independently of specified switch *DPos* or *DNeg*)

The reference value is only updated at the sample time, not at the interrupt subscription time.

An interrupt is also generated at the sample for update of the reference value, if the direction of the signal difference is in accordance with the specified argument (any direction, *DPos* or *DNeg*).

When the *Single* switch is used, only one interrupt at the most will be generated.
If the switch *Single* (cyclic interrupt) is not used, at every sample of the signal value a new test is done of the specified conditions (both *Condition* **and** *DeltaValue*) compared to the actual signal value and the last stored reference value, to generate or not generate an interrupt.

Condition for interrupt generation at interrupt subscription time



Condition for interrupt generation at each sample after interrupt subscription



Example 1 of interrupt generation



Assuming the interrupt is ordered between sample 0 and 1, the following instruction will give the following results:

ISignalAI ai1, AIO_BETWEEN, 6.1, 2,2, 1.0, sig1int;

sample 1 will generate an interrupt, because the signal value is between *High-Value* and *LowValue* and the signal difference compared to sample 0 is more than *DeltaValue*.

sample 2 will generate an interrupt, because the signal value is between *High-Value* and *LowValue* and the signal difference compared to sample 1 is more than *DeltaValue*.

samples 3, 4, 5 will not generate any interrupt, because the signal difference is less than *DeltaValue*.

sample 6 will generate an interrupt.

samples 7 to 10 will not generate any interrupt, because the signal is above *HighValue*

sample 11 will not generate any interrupt, because the signal difference compared to sample 6 is equal to *DeltaValue*.

sample 12 will not generate any interrupt, because the signal difference compared to sample 6 is less than *DeltaValue*.

Example 2 of interrupt generation



Assuming the interrupt is ordered between sample 0 and 1, the following instruction will give the following results:

ISignalAI ai1, AIO_BETWEEN, 6.1, 2,2, 1.0 \DPos, sig1int;

A new reference value is stored at sample 1 and 2, because the signal is within limits and the absolute signal difference between the actual value and the last stored reference value is greater than 1.0.
No interrupt will be generated because the signal changes are in the negative direction.

sample 6 will generate an interrupt, because the signal value is between *High-Value* and *LowValue* and the signal difference in the positive direction compared to sample 2 is more than *DeltaValue*.

Example 3 of interrupt generation



Assuming the interrupt is ordered between sample 0 and 1, the following instruction will give the following results:

ISignalAI \Single, ai1, AIO_OUTSIDE, 6.1, 2,2, 1.0 \DPos, sig1int;

A new reference value is stored at sample 7, because the signal is within limits and the absolute signal difference between the actual value and the last stored reference value is greater than 1.0

sample 8 will generate an interrupt, because the signal value is above *HighValue* and the signal difference in the positive direction compared to sample 7 is more than *DeltaValue*.

Example 4 of interrupt generation



Assuming the interrupt is ordered between sample 0 and 1, the following instruction will give the following results:

ISignalAI ai1, AIO_ALWAYS, 6.1, 2,2, 1.0 \DPos, sig1int;

A new reference value is stored at sample 1 and 2, because the signal is within limits and the absolute signal difference between the actual value and the last stored reference value is greater than 1.0

sample 6 will generate an interrupt, because the signal difference in the positive direction compared to sample 2 is more than *DeltaValue*.

sample 7 and 8 will generate an interrupt, because the signal difference in the positive direction compared to previous sample is more than *DeltaValue*.

A new reference value is stored at sample 11 and 12, because the signal is within limits and the absolute signal difference between the actual value and the last stored reference value is greater than 1.0

Error handling

If there is a subscription of interrupt on an analog input signal, an interrupt will be given for every change in the analog value that satisfies the condition specified when ordering the interrupt subscription. If the analog value is noisy, many interrupts can be generated, even if only one or two bits in the analog value are changed.

To avoid generating interrupts for small changes of the analog input value, set the *DeltaValue* to a level greater than 0. Then no interrupts will be generated until a change of the analog value is greater than the specified *DeltaValue*.

Limitations

The *HighValue* and *LowValue* arguments should be in the range: logical maximum value, logical minimum value defined for the signal.

HighValue must be above *LowValue*.

DeltaValue must be 0 or positive.

The limitations for the interrupt identity are the same as for *ISignalDI*.

Syntax

```
ISignalAI
   [ '\'Single',']
   [ Signal':=' ]<variable (VAR) of signalai>','
   [ Condition':=' ]<expression (IN) of aiotrigg>','
   [ HighValue':=' ]<expression (IN) of num>','
   [ LowValue':=' ]<expression (IN) of num>','
   [ DeltaValue':=' ]<expression (IN) of num>
   [ '\'DPos] | [ '\'DNeg] ','
   [ Interrupt':=' ]<variable (VAR) of intnum>';'
```

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Definition of constants	Data Types - *aiotrigg*
Interrupt from analog output signal	Instructions - *ISignalAO*
Interrupt from digital input signal	Instructions - *ISignalDI*
Interrupt from digital output signal	Instructions - *ISignalDO*
More information on interrupt management	Basic Characteristics - *Interrupts*
More examples	Data Types - *intnum*
Related system parameters (filter)	System Parameters - *IO Signals*

ISignalAO Interrupts from analog output signal

ISignalAO (Interrupt Signal Analog Output) is used to order and enable interrupts from an analog output signal.

Example

```
VAR intnum sig1int;
CONNECT sig1int WITH iroutine1;
ISignalAO \Single, ao1, AIO_BETWEEN, 1.5, 0.5, 0, sig1int;
```

Orders an interrupt which is to occur the first time the logical value of the analog output signal *ao1* is between *0.5* and *1.5*. A call is then made to the *iroutine1* trap routine.

```
ISignalAO ao1, AIO_BETWEEN, 1.5, 0.5, 0.1, sig1int;
```

Orders an interrupt which is to occur each time the logical value of the analog output signal *ao1* is between *0.5* and *1.5*, and the absolute signal difference compared to the previous stored reference value is bigger than 0.1.

```
ISignalAO ao1, AIO_OUTSIDE, 1.5, 0.5, 0.1, sig1int;
```

Orders an interrupt which is to occur each time the logical value of the analog output signal *ao1* is lower than *0.5* or higher than *1.5*, and the absolute signal difference compared to the previous stored reference value is bigger than 0.1.

Arguments

ISignalAO [\Single] Signal Condition HighValue LowValue DeltaValue [\DPos] | [\DNeg] Interrupt

[\Single] Data type: *switch*

Specifies whether the interrupt is to occur once or cyclically.

If the argument *Single* is set, the interrupt occurs once at the most. If the argument is omitted, an interrupt will occur each time its condition is satisfied.

Signal Data type: *signalao*

The name of the signal that is to generate interrupts.

Condition Data type: *aiotrigg*

Specifies how *HighValue* and *LowValu*e define the condition to be satisfied:

- AIO_ABOVE_HIGH: logical value of the signal is above HighValue
- AIO_BELOW_HIGH: logical value of the signal is below HighValue
- AIO_ABOVE_LOW: logical value of the signal is above LowValue
- AIO_BELOW_LOW: logical value of the signal is below LowValue
- AIO_BETWEEN: logical value of the signal is between LowValue and HighValue
- AIO_OUTSIDE: logical value of the signal is above HighValue or below LowValue
- AIO_ALWAYS: independently of HighValue and LowValue

HighValue Data type: *num*

High logical value to define the condition.

LowValue Data type: *num*

Low logical value to define the condition.

DeltaValue Data type: *num*

Defines the minimum logical signal difference before generation of a new interrupt. The actual signal value compared to the previous stored reference value must be greater than the specified *DeltaValue* before generation of a new interrupt.

[\DPos] Data type: *switch*

Specifies that only positive logical signal differences will give new interrupts.

[\DNeg] Data type: *switch*

Specifies that only negative logical signal differences will give new interrupts.

If neither of the *\DPos* and *\DNeg* arguments are used, both positive and negative differences will generate new interrupts.

Interrupt Data type: *intnum*

The interrupt identity. This interrupt should have previously been connected to a trap routine by means of the instruction CONNECT.

Program execution

See instruction *ISignalAI* for information about:

- Program execution
- Condition for interrupt generation
- More examples

Same principles are valid for *ISignalAO* as for *ISignalAI*.

Limitations

The *HighValue* and *LowValue* arguments should be in the range: logical maximum value, logical minimum value, defined for the signal.

HighValue must be above *LowValue*.

DeltaValue must be 0 or positive.

The limitations for the interrupt identity are the same as for *ISignalDO*.

Syntax

```
ISignalAO
   [ '\'Single',']
   [ Signal':=' ]<variable (VAR) of signalao>','
   [ Condition':=' ]<expression (IN) of aiotrigg>','
   [ HighValue':=' ]<expression (IN) of num>','
   [ LowValue':=' ]<expression (IN) of num>','
   [ DeltaValue':=' ]<expression (IN) of num>
   [ '\'DPos] | [ '\'DNeg] ','
   [ Interrupt':=' ]<variable (VAR) of intnum>';'
```

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Definition of constants	Data Types - *aiotrigg*
Interrupt from analog input signal	Instructions - *ISignalAI*
Interrupt from digital input signal	Instructions - *ISignalDI*
Interrupt from digital output signal	Instructions - *ISignalDO*
More information on interrupt management	Basic Characteristics - *Interrupts*
More examples	Data Types - *intnum*
Related system parameters (filter)	System Parameters - *IO Signals*

ISignalDI Orders interrupts from a digital input signal

ISignalDI (Interrupt Signal Digital In) is used to order and enable interrupts from a digital input signal.

System signals can also generate interrupts.

Examples

```
VAR intnum sig1int;
CONNECT sig1int WITH iroutine1;
ISignalDI di1,1,sig1int;
```

Orders an interrupt which is to occur each time the digital input signal *di1* is set to *1*. A call is then made to the *iroutine1* trap routine.

```
ISignalDI di1,0,sig1int;
```

Orders an interrupt which is to occur each time the digital input signal *di1* is set to *0*.

```
ISignalDI \Single, di1,1,sig1int;
```

Orders an interrupt which is to occur only the first time the digital input signal *di1* is set to *1*.

Arguments

ISignalDI [\Single] Signal TriggValue Interrupt

[\Single] Data type: *switch*

Specifies whether the interrupt is to occur once or cyclically.

If the argument *Single* is set, the interrupt occurs once at the most. If the argument is omitted, an interrupt will occur each time its condition is satisfied.

Signal Data type: *signaldi*

The name of the signal that is to generate interrupts.

TriggValue Data type: *dionum*

The value to which the signal must change for an interrupt to occur.

The value is specified as 0 or 1 or as a symbolic value (e.g. *high/low*). The signal is edge-triggered upon changeover to 0 or 1.

TriggValue 2 or symbolic value *edge* can be used for generation of interrupts on both positive flank (0 -> 1) and negative flank (1 -> 0).

Interrupt Data type: *intnum*

The interrupt identity. This should have previously been connected to a trap routine by means of the instruction *CONNECT.*

Program execution

When the signal assumes the specified value, a call is made to the corresponding trap routine. When this has been executed, program execution continues from where the interrupt occurred.

If the signal changes to the specified value before the interrupt is ordered, no interrupt occurs (see Figure 20).



Figure 20 Interrupts from a digital input signal at signal level 1.

Limitations

The same variable for interrupt identity cannot be used more than once, without first deleting it. Interrupts should therefore be handled as shown in one of the alternatives below.

```
PROC main ( )
   VAR intnum sig1int;
   CONNECT sig1int WITH iroutine1;
   ISignalDI di1, 1, sig1int;
   WHILE TRUE DO
   :
   :
   ENDWHILE
ENDPROC
```

All activation of interrupts is done at the beginning of the program. These instructions are then kept outside the main flow of the program.

```
PROC main ( )
   VAR intnum sig1int;
   CONNECT sig1int WITH iroutine1;
   ISignalDI di1, 1, sig1int;
   :
   :
   IDelete sig1int;
ENDPROC
```

The interrupt is deleted at the end of the program, and is then reactivated. It should be noted, in this case, that the interrupt is inactive for a short period.

Syntax

```
ISignalDI
   [ '\' Single',']
   [ Signal ':=' ] < variable (VAR) of signaldi > ','
   [ TriggValue ':=' ] < expression (IN) of dionum >','
   [ Interrupt ':=' ] < variable (VAR) of intnum > ';'
```

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Interrupt from an output signal	Instructions - *ISignalDO*
More information on interrupt management	Basic Characteristics - *Interrupts*
More examples	Data Types - *intnum*

ISignalDO Interrupts from a digital output signal

ISignalDO (Interrupt Signal Digital Out) is used to order and enable interrupts from a digital output signal.

System signals can also generate interrupts.

Examples

```
VAR intnum sig1int;
CONNECT sig1int WITH iroutine1;
ISignalDO do1,1,sig1int;
```

Orders an interrupt which is to occur each time the digital output signal *do1* is set to *1*. A call is then made to the *iroutine1* trap routine.

```
ISignalDO do1,0,sig1int;
```

Orders an interrupt which is to occur each time the digital output signal *do1* is set to *0*.

```
ISignalDO\Single, do1,1,sig1int;
```

Orders an interrupt which is to occur only the first time the digital output signal *do1* is set to *1*.

Arguments

ISignalDO [\Single] Signal TriggValue Interrupt

[\Single] Data type: *switch*

Specifies whether the interrupt is to occur once or cyclically.

If the argument *Single* is set, the interrupt occurs once at the most. If the argument is omitted, an interrupt will occur each time its condition is satisfied.

Signal Data type: *signaldo*

The name of the signal that is to generate interrupts.

TriggValue Data type: *dionum*

The value to which the signal must change for an interrupt to occur.

The value is specified as 0 or 1 or as a symbolic value (e.g. *high/low*). The signal is edge-triggered upon changeover to 0 or 1.

TriggValue 2 or symbolic value *edge* can be used for generation of interrupts on both positive flank (0 -> 1) and negative flank (1 -> 0).

Interrupt Data type: *intnum*

The interrupt identity. This should have previously been connected to a trap routine by means of the instruction *CONNECT.*

Program execution

When the signal assumes the specified value 0 or 1, a call is made to the corresponding trap routine. When this has been executed, program execution continues from where the interrupt occurred.

If the signal changes to the specified value before the interrupt is ordered, no interrupt occurs (see Figure 21).



Figure 21 Interrupts from a digital output signal at signal level 1.

Limitations

The same variable for interrupt identity cannot be used more than once, without first deleting it. Interrupts should therefore be handled as shown in one of the alternatives below.

```
PROC main ( )
   VAR intnum sig1int;
   CONNECT sig1int WITH iroutine1;
   ISignalDO do1, 1, sig1int;
   WHILE TRUE DO
   :
   :
   ENDWHILE
ENDPROC
```

All activation of interrupts is done at the beginning of the program. These instructions are then kept outside the main flow of the program.

```
PROC main ( )
   VAR intnum sig1int;
   CONNECT sig1int WITH iroutine1;
   ISignalDO do1, 1, sig1int;
   :
   :
   IDelete sig1int;
ENDPROC
```

The interrupt is deleted at the end of the program, and is then reactivated. It should be noted, in this case, that the interrupt is inactive for a short period.

Syntax

```
ISignalDO
   [ '\' Single',']
   [ Signal ':=' ] < variable (VAR) of signaldo > ','
   [ TriggValue ':=' ] < expression (IN) of dionum >','
   [ Interrupt ':=' ] < variable (VAR) of intnum > ';'
```

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Interrupt from an input signal	Instructions - *ISignalDI*
More information on interrupt management	Basic Characteristics- *Interrupts*
More examples	Data Types - *intnum*

ISleep Deactivates an interrupt

ISleep (Interrupt Sleep) is used to deactivate an individual interrupt temporarily.

During the deactivation time, any generated interrupts of the specified type are discarded without any trap execution.

Example

```
ISleep sig1int;
```

The interrupt *sig1int* is deactivated.

Arguments

ISleep Interrupt

Interrupt Data type: *intnum*

The variable (interrupt identity) of the interrupt.

Program execution

Any generated interrupts of the specified type are discarded without any trap execution, until the interrupt has been re-activated by means of the instruction *IWatch*. Interrupts which are generated while *ISleep* is in effect are ignored.

Example

```
VAR intnum timeint;
CONNECT timeint WITH check_serialch;
ITimer 60, timeint;
.
ISleep timeint;
WriteBin ch1, buffer, 30;
IWatch timeint;
.
TRAP check_serialch
    WriteBin ch1, buffer, 1;
    IF ReadBin(ch1\Time:=5) < 0 THEN
        TPWrite “The serial communication is broken”;
        EXIT;
    ENDIF
```

```
ENDTRAP
```

Communication across the ch1 serial channel is monitored by means of interrupts which are generated every *60* seconds. The trap routine checks whether the communication is working. When, however, communication is in progress, these interrupts are not permitted.

Error handling

Interrupts which have neither been ordered nor enabled are not permitted. If the interrupt number is unknown, the system variable ERRNO will be set to ERR_UNKINO (see “Data types - errnum”). The error can be handled in the error handler.

Syntax

ISleep
[Interrupt ‘:=’] < variable (**VAR**) of *intnum* > ‘;’

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Enabling an interrupt	Instructions - *IWatch*
Disabling all interrupts	Instructions - *IDisable*
Cancelling an interrupt	Instructions - *IDelete*

ITimer Orders a timed interrupt

ITimer (Interrupt Timer) is used to order and enable a timed interrupt.

This instruction can be used, for example, to check the status of peripheral equipment once every minute.

Examples

```
VAR intnum timeint;
CONNECT timeint WITH iroutine1;
ITimer 60, timeint;
```

Orders an interrupt that is to occur cyclically every *60* seconds. A call is then made to the trap routine *iroutine1*.

```
ITimer \Single, 60, timeint;
```

Orders an interrupt that is to occur once, after *60* seconds.

Arguments

ITimer [\Single] Time Interrupt

[\Single] Data type: *switch*

Specifies whether the interrupt is to occur once or cyclically.

If the argument *Single* is set, the interrupt occurs only once. If the argument is omitted, an interrupt will occur each time at the specified time.

Time Data type: *num*

The amount of time that must lapse before the interrupt occurs.

The value is specified in second if *Single* is set, this time may not be less than 0.05 seconds. The corresponding time for cyclical interrupts is 0.25 seconds.

Interrupt Data type: *intnum*

The variable (interrupt identity) of the interrupt. This should have previously been connected to a trap routine by means of the instruction *CONNECT*.

Program execution

The corresponding trap routine is automatically called at a given time following the interrupt order. When this has been executed, program execution continues from where the interrupt occurred.

If the interrupt occurs cyclically, a new computation of time is started from when the interrupt occurs.

Example

```
VAR intnum timeint;
CONNECT timeint WITH check_serialch;
ITimer 60, timeint;
.
TRAP check_serialch
   WriteBin ch1, buffer, 1;
   IF ReadBin(ch1\Time:=5) < 0 THEN
      TPWrite “The serial communication is broken”;
      EXIT;
   ENDIF
ENDTRAP
```

Communication across the ch1 serial channel is monitored by means of interrupts which are generated every *60* seconds. The trap routine checks whether the communication is working. If it is not, program execution is interrupted and an error message appears.

Limitations

The same variable for interrupt identity cannot be used more than once, without being first deleted. See Instructions - *ISignalDI*.

Syntax

```
ITimer
   [ ’\’Single ’,’]
   [ Time ’:=’ ] < expression (IN) of num >’,’
   [ Interrupt ’:=’ ] < variable (VAR) of intnum > ’;’
```

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
More information on interrupt management	Basic Characteristics- *Interrupts*

IVarValue Orders a variable value interrupt

IVarVal(Interrupt Variable Value) is used to order and enable an interrupt when the value of a variable accessed via the serial sensor interface has been changed.

This instruction can be used, for example, to get seam volume or gap values from a seam tracker.

Examples

```
LOCAL PERS num adtVlt{25}:=[1,1.2,1.4,1.6,1.8,2,2.16667,2.33333,2.5,...];
LOCAL PERS num adptWfd{25}:=[2,2.2,2.4,2.6,2.8,3,3.16667,3.33333,3.5,...];
LOCAL PERS num adptSpd{25}:=10,12,14,16,18,20,21.6667,23.3333,25[,...];
LOCAL CONST num GAP_VARIABLE_NO:=11;
PERS num gap_value;
VAR intnum IntAdap;

PROC main()
! Setup the interrupt. The trap routine AdapTrp will be called
! when the gap variable with number 'GAP_VARIABLE_NO' in
! the sensor interface has been changed. The new value will be available
! in the PERS gp_value variable.
   CONNECT IntAdap WITH AdapTrp;
   IVarValue GAP_VARIABLE_NO, gap_value, IntAdap;

   ! Start welding
   ArcL\On,*,v100,adaptSm,adaptWd,adaptWv,z10,tool\j\Track:=track;
   ArcL\On,*,v100,adaptSm,adaptWd,adaptWv,z10,tool\j\Track:=track;
ENDPROC

TRAP AdapTrap
   VAR num ArrInd;

   !Scale the raw gap value received
   ArrInd:=ArrIndx(gap_value);

   ! Update active welddata PERS variable 'adaptWd' with
   ! new data from the arrays of predefined parameter arrays.
   ! The scaled gap value is used as index in the voltage, wirefeed and speed arrays.
   adaptWd.weld_voltage:=adptVlt{ArrInd};
   adaptWd.weld_wirefeed:=adptWfd{ArrInd};
   adaptWd.weld_speed:=adptSpd{ArrInd};

   !Request a refresh of AW parameters using the new data i adaptWd
   ArcRefresh;
ENDTRAP
```

Arguments

IVarValue VarNo Value, Interrupt

VarNo Data type: *num*

The number of the variable to be supervised.

Value Data type: *num*

A PERS variable which will hold the new value of Varno.

Interrupt Data type: *intnum*

The variable (interrupt identity) of the interrupt. This should have previously been connected to a trap routine by means of the instruction *CONNECT.*

Program execution

The corresponding trap routine is automatically called at a given time following the interrupt order. When this has been executed, program execution continues from where the interrupt occurred.

Limitations

The same variable for interrupt identity cannot be used more than five times, without first being deleted.

Syntax

IVarValue
[VarNo ':='] < expression (**IN**) of *num* >','
[Value ':='] < persistent(**PERS**) of *num* >','
[Interrupt ':='] < variable (**VAR**) of *intnum* > ';'

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
More information on interrupt management	Basic Characteristics- *Interrupts*

IWatch Activates an interrupt

IWatch (Interrupt Watch) is used to activate an interrupt which was previously ordered but was deactivated with *ISleep*.

Example

IWatch sig1int;

The interrupt *sig1int* that was previously deactivated is activated.

Arguments

IWatch Interrupt

Interrupt Data type: *intnum*

Variable (interrupt identity) of the interrupt.

Program execution

Re-activates interrupts of the specified type once again. However, interrupts generated during the time the *ISleep* instruction was in effect, are ignored.

Example

```
VAR intnum sig1int;
CONNECT sig1int WITH iroutine1;
ISignalDI di1,1,sig1int;
.
ISleep sig1int;
weldpart1;
IWatch sig1int;
```

During execution of the *weldpart1* routine, no interrupts are permitted from the signal *di1*.

Error handling

Interrupts which have not been ordered are not permitted. If the interrupt number is unknown, the system variable ERRNO is set to ERR_UNKINO (see “Date types - errnum”). The error can be handled in the error handler.

Syntax

IWatch
[Interrupt ‘:=’] < variable (**VAR**) of *intnum* > ‘;’

Related information

	Described in:
Summary of interrupts	RAPID Summary - *Interrupts*
Deactivating an interrupt	Instructions - *ISleep*

label Line name

Label is used to name a line in the program. Using the *GOTO* instruction, this name can then be used to move program execution.

Example

```
GOTO next;
   .
next:
```

Program execution continues with the instruction following *next*.

Arguments

Label:

Label Identifier

The name you wish to give the line.

Program execution

Nothing happens when you execute this instruction.

Limitations

The label must not be the same as

- any other label within the same routine,
- any data name within the same routine.

A label hides global data and routines with the same name within the routine it is located in.

Syntax

(EBNF)
<identifier>':'

Related information

	Described in:
Identifiers	Basic Characteristics- *Basic Elements*
Moving program execution to a label	Instructions - *GOTO*

Load Load a program module during execution

Load is used to load a program module into the program memory during execution.

The loaded program module will be added to the already existing modules in the program memory.

A program or system module can be loaded in static (default) or dynamic mode:

Static mode

Table 1 How different operations affects static loaded program or system modules

	Set PP to main from TP	Open new RAPID program
Program Module	Not affected	Unloaded
System Module	Not affected	Not affected

Dynamic mode

Table 2 How different operations affects dynamic loaded program or system modules

	Set PP to main from TP	Open new RAPID program
Program Module	Unloaded	Unloaded
System Module	Unloaded	Unloaded

Both static and dynamic loaded modules can be unloaded by the instruction *UnLoad*.

Example

Load \Dynamic, ram1disk \File:="PART_A.MOD";

Load the program module PART_A.MOD from the *ram1disk* into the program memory. (*ram1disk* is a predefined string constant "*ram1disk*:"). Load program module in dynamic mode.

Arguments

Load [\Dynamic] FilePath [\File]

[\Dynamic] Data type: *switch*

The switch enables load of a program module in dynamic mode. Otherwise the load is in static mode.

FilePath Data type: *string*

The file path and the file name to the file that will be loaded into the program memory. The file name shall be excluded when the argument *File* is used.

[\File] Data type: *string*

When the file name is excluded in the argument *FilePath* then it must be defined with this argument.

Program execution

Program execution waits for the program module to finish loading before proceeding with the next instruction.

To obtain a good program structure, that is easy to understand and maintain, all loading and unloading of program modules should be done from the main module which is always present in the program memory during execution.

After the program module is loaded it will be linked and initialised. The initialisation of the loaded module sets all variables at module level to their init values. Unresolved references will be accepted if the system parameter for *Tasks* is set (BindRef = NO). However, when the program is started or the teach pendant function Program/File/ Check is used, no check for unresolved references will be done if the parameter Bind-Ref = NO. There will be a run time error on execution of an unresolved reference.

Examples

```
Load \Dynamic, "ram1disk:DOORDIR/DOOR1.MOD";
```

Loads the program module DOOR1.MOD from the *ram1disk* at the directory DOORDIR into the program memory. The program module is loaded in dynamic mode.

```
Load \Dynamic, "ram1disk:"  \File:="DOORDIR/DOOR1.MOD";
```

Same as above but another syntax.

```
Load "ram1disk:"  \File:="DOORDIR/DOOR1.MOD";
```

Same as the two examples above but the module is loaded in static mode.

Limitations

Loading a program module that contains a main routine is not allowed.

Avoid ongoing robot movements during the loading.

Avoid using the floppy disk for loading since reading from the floppy drive is very time consuming.

Error handling

If the file in the *Load* instructions cannot be found, then the system variable ERRNO is set to ERR_FILNOTFND. If the module already is loaded into the program memory then the system variable ERRNO is set to ERR_LOADED (see "Data types - errnum"). The errors above can be handled in an error handler.

Syntax

Load
[‘\’Dynamic ‘,’]
[FilePath’:=’]<expression (**IN**) of *string*>
[’\’File’:=’ <expression (**IN**) of *string*>]’;’

Related information

	Described in:
Unload a program module	Instructions - *UnLoad*
Load a program module in parallel with another program execution	Instructions - *StartLoad-WaitLoad*
Accept unresolved references	System Parameters - *Controller / Tasks / BindRef*

MechUnitLoad Defines a payload for a mechanical unit

MechUnitLoad is used to define a payload for an external mechanical unit.
(The payload for the robot is defined with instruction *GripLoad*)

This instruction should be used for all mechanical units with dynamic model in servo to achieve the best motion performance.

The *MechUnitLoad* instruction should always be executed after execution of the instruction *ActUnit*.

Example



Figure 22 A mechanical unit named IRBP_L of type IRBP L.

```
ActUnit IRBP_L;
MechUnitLoad IRBP_L, 1, load0;
```

Activate mechanical unit *IRBP_L* and define the payload *load0* corresponding to no load (at all) mounted on axis *1*.

```
ActUnit IRBP_L;
MechUnitLoad IRBP_L, 1, fixture1;
```

Activate mechanical unit *IRBP_L* and define the payload *fixture1* corresponding to fixture *fixture1* mounted on axis *1*.

```
ActUnit IRBP_L;
MechUnitLoad IRBP_L, 1, workpiece1;
```

Activate mechanical unit *IRBP_L* and define the payload *workpiece1* corresponding to fixture and work piece named *workpiece1* mounted on axis *1*.

Arguments

MechUnitLoad MechUnit AxisNo Load

MechUnit *(Mechanical Unit)* Data type: *mecunit*

The name of the mechanical unit.

AxisNo *(Axis Number)* Data type: *num*

The axis number, within the mechanical unit, that holds the load.

Load Data type: *loaddata*

The load data that describes the current payload to be defined.

Program execution

After execution of *MechUnitLoad*, when the robot and external axes have come to a standstill, the specified load is defined for the specified mechanical unit and axis. This means that the payload is controlled and monitored by the control system.

The default payload at cold start-up, for a certain mechanical unit type, is the predefined maximal payload for this mechanical unit type.

When some other payload is used, the actual payload for the mechanical unit and axis should be redefined with this instruction. This should always be done after activation of the mechanical unit.

The defined payload will survive a power failure restart.
The defined payload will also survive a restart of the program after manual activation of some other mechanical units from the jogging window.



Figure 23 Payload mounted on the end-effector of a mechanical unit.

Example



Figure 24 A mechanical unit named IRBP_K of type IRBP K with three axes.

MoveL homeside1, v1000, fine, gun1;
...
ActUnit IRBP_K;

The whole mechanical unit *IRBP_K* is activated.

MechUnitLoad IRBP_K, 2, workpiece1;

Defines payload *workpiece1* on the mechanical unit *IRBP_K* axis *2*.

MechUnitLoad IRBP_K, 3, workpiece2;

Defines payload *workpiece2* on the mechanical unit *IRBP_K* axis *3*.

MoveL homeside2, v1000, fine, gun1

The axes of the mechanical unit *IRBP_K* move to the switch position *homeside2* with mounted payload on both axes *2* and *3*.

Limitations

The movement instruction previous to this instruction should be terminated with a stop point in order to make a restart in this instruction possible following a power failure.

Syntax

MechUnitLoad
[MechUnit ':='] < variable (**VAR**) of *mecunit*> ','
[AxisNo ':='] <expression (**IN**) of *num* ','
[Load ':='] < persistent (**PERS**) of *loaddata* > ';'

Related information

	Described in:
Identification of payload for external mechanical units	LoadID&CollDetect - Program *muloadid.prg*
Mechanical units	Data Types - *mecunit*
Definition of load data	Data Types - *loaddata*
Definition of payload for the robot	Instructions - *GripLoad* Data Types - *tooldata*

MoveCSync Moves the robot circularly and executes a RAPID procedure

MoveCSync (*Move Circular Synchronously*) is used to move the tool centre point (TCP) circularly to a given destination. The specified RAPID procedure is executed at the middle of the corner path in the destination point. During the movement, the orientation normally remains unchanged relative to the circle.

Examples

MoveCSync p1, p2, v500, z30, tool2, “proc1”;

The TCP of the tool, *tool2*, is moved circularly to the position *p2*, with speed data *v500* and zone data *z30*. The circle is defined from the start position, the circle point *p1* and the destination point *p2*. Procedure *proc1* is executed in the middle of the corner path at *p2*.

Arguments

MoveCSync CirPoint ToPoint Speed [\T] Zone Tool [\WObj] ProcName

CirPoint Data type: *robtarget*

The circle point of the robot. The circle point is a position on the circle between the start point and the destination point. To obtain the best accuracy, it should be placed about halfway between the start and destination points. If it is placed too close to the start or destination point, the robot may give a warning. The circle point is defined as a named position or stored directly in the instruction (marked with an * in the instruction). The position of the external axes are not used.

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the TCP, the tool reorientation and external axes.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot and external axes move. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point that is moved to the specified destination point.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (object coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified.

ProcName *(Procedure Name)* Data type: *string*

Name of the RAPID procedure to be executed at the middle of the corner path in the destination point.

Program execution

See the instruction *MoveC* for more information about circular movements.

The specified RAPID procedure is executed when the TCP reaches the middle of the corner path in the destination point of the *MoveCSync* instruction, as shown in Figure 25:



Figure 25 Execution of user-defined RAPID procedure at the middle of the corner path.

For stop points, we recommend the use of “normal” programming sequence with

MoveC + other RAPID instructions in sequence.

Execution of the specified RAPID procedure in different execution modes:

Execution mode:	Execution of RAPID procedure:
Continuously or Cycle	According to this description
Forward step	In the stop point
Backward step	Not at all

Limitation

General limitations according to instruction *MoveC.*

Switching execution mode after program stop from continuously or cycle to stepwise forward or backward results in an error. This error tells the user that the mode switch can result in missed execution of a RAPID procedure in the queue for execution on the path. This error can be avoided if the program is stopped with StopInstr before the mode switch.

Instruction *MoveCSync* cannot be used on TRAP level.
The specified RAPID procedure cannot be tested with stepwise execution.

Syntax

MoveCSync
[CirPoint ':='] < expression (**IN**) of *robtarget* > ','
[ToPoint ':='] < expression (**IN**) of *robtarget* > ','
[Speed ':='] < expression (**IN**) of *speeddata* >
['\' T ':=' < expression (**IN**) of *num* >] ','
[Zone ':='] < expression (**IN**) of *zonedata* > ','
[Tool ':='] < persistent (**PERS**) of *tooldata* >
['\' WObj ':=' < persistent (**PERS**) of *wobjdata* >] ','
[ProcName ':='] < expression (**IN**) of *string* >] ';'

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*

MoveAbsJ Moves the robot to an absolute joint position

MoveAbsJ (*Move Absolute Joint*) is used to move the robot to an absolute position, defined in axes positions.

This instruction need only be used when:

- the end point is a singular point
- for ambiguous positions on the IRB 6400C, e.g. for movements with the tool over the robot.

The final position of the robot, during a movement with *MoveAbsJ*, is neither affected by the given tool and work object, nor by active program displacement. However, the robot uses these data to calculating the load, TCP velocity, and the corner path. The same tools can be used as in adjacent movement instructions.

The robot and external axes move to the destination position along a non-linear path. All axes reach the destination position at the same time.

Examples

MoveAbsJ p50, v1000, z50, tool2;

The robot with the tool *tool2* is moved along a non-linear path to the absolute axis position, *p50*, with velocity data *v1000* and zone data *z50*.

MoveAbsJ *, v1000\T:=5, fine, grip3;

The robot with the tool *grip3*, is moved along a non-linear path to a stop point which is stored as an absolute axis position in the instruction (marked with an *). The entire movement takes 5 s.

Arguments

MoveAbsJ [\Conc] ToJointPos [\NoEOffs] Speed [\V] | [\T] Zone [\Z] Tool [\WObj]

[\Conc] *(Concurrent)* Data type: *switch*

Subsequent instructions are executed while the robot is moving. The argument is used to shorten the cycle time when, for example, communicating with external equipment, if synchronisation is not required.

Using the argument \Conc, the number of movement instructions in succession is limited to 5. In a program section that includes StorePath-RestoPath, movement instructions with the argument \Conc are not permitted.

If this argument is omitted and the ToPoint is not a stop point, the subsequent instruction is executed some time before the robot has reached the programmed zone.

ToJointPos (*To Joint Position*) Data type: *jointtarget*

The destination absolute joint position of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

[**NoEOffs**] (*No External Offsets)* Data type: *switch*

If the argument *NoEOffs* is set, then the movement with *MoveAbsJ* is not affected by active offsets for external axes.

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the tool reorientation and external axes.

[\V] *(Velocity)* Data type: *num*

This argument is used to specify the velocity of the TCP in mm/s directly in the instruction. It is then substituted for the corresponding velocity specified in the speed data.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

[\Z] *(Zone)* Data type: *num*

This argument is used to specify the position accuracy of the robot TCP directly in the instruction. The length of the corner path is given in mm, which is substituted for the corresponding zone specified in the zone data.

Tool Data type: *tooldata*

The tool in use during the movement.

The position of the TCP and the load on the tool are defined in the tool data. The TCP position is used to decide the velocity and the corner path for the movement.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object used during the movement.

This argument can be omitted if the tool is held by the robot. However, if the robot holds the work object, i.e. the tool is stationary, or with coordinated external axes, then the argument must be specified.

In the case of a stationary tool or coordinated external axes, the data used by the system to decide the velocity and the corner path for the movement, is defined in the work object.

Program execution

A movement with *MoveAbsJ* is not affected by active program displacement and if executed with switch *\NoEOffs*, there will be no offset for external axes.
Without switch *\NoEOffs*, the external axes in the destination target are affected by active offset for external axes.

The tool is moved to the destination absolute joint position with interpolation of the axis angles. This means that each axis is moved with constant axis velocity and that all axes reach the destination joint position at the same time, which results in a non-linear path.

Generally speaking, the TCP is moved at approximate programmed velocity. The tool is reoriented and the external axes are moved at the same time as the TCP moves. If the programmed velocity for reorientation, or for the external axes, cannot be attained, the velocity of the TCP will be reduced.

A corner path is usually generated when movement is transferred to the next section of the path. If a stop point is specified in the zone data, program execution only continues when the robot and external axes have reached the appropriate joint position.

Examples

MoveAbsJ *, v2000\V:=2200, z40 \Z:=45, grip3;

The tool, *grip3*, is moved along a non-linear path to a absolute joint position stored in the instruction. The movement is carried out with data set to *v2000* and *z40*, the velocity and zone size of the TCP are *2200* mm/s and *45* mm respectively.

MoveAbsJ \Conc, *, v2000, z40, grip3;

The tool, *grip3*, is moved along a non-linear path to a absolute joint position stored in the instruction. Subsequent logical instructions are executed while the robot moves.

```
MoveAbsJ \Conc, * \NoEOffs, v2000, z40, grip3;
```

Same movement as above but the movement is not affected by active offsets for external axes.

```
GripLoad obj_mass;
MoveAbsJ start, v2000, z40, grip3 \WObj:= obj;
```

The robot moves the work object *obj* in relation to the fixed tool *grip3* along a non-linear path to an absolute axis position *start*.

Error handling

When running the program, a check is made that the arguments Tool and \WObj do not contain contradictory data with regard to a movable or a stationary tool respectively.

Limitations

In order to be able to run backwards with the instruction *MoveAbsJ* involved, and avoiding problems with singular points or ambiguous areas, it is essential that the subsequent instructions fulfil certain requirements, as follows (see Figure 1).



Figure 1 Limitation for backward execution with MoveAbsJ.

Syntax

```
MoveAbsJ
   [ '\' Conc ',' ]
   [ ToJointPos ':=' ] < expression (IN) of jointtarget >
   [ '\' NoEoffs ] ','
   [ Speed ':=' ] < expression (IN) of speeddata >
                  [ '\' V ':=' < expression (IN) of num > ]
                  | [ '\' T ':=' < expression (IN) of num > ] ','
   [Zone ':=' ] < expression (IN) of zonedata >
                  [ '\' Z '=' < expression (IN) of num > ] ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ] ';'
```

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of jointtarget	Data Types - *jointtarget*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion and I/O Principles
Concurrent program execution	Motion and I/O Principles - *Synchronisation Using Logical Instructions*

MoveC Moves the robot circularly

MoveC is used to move the tool centre point (TCP) circularly to a given destination. During the movement, the orientation normally remains unchanged relative to the circle.

Examples

MoveC p1, p2, v500, z30, tool2;

The TCP of the tool, *tool*2, is moved circularly to the position *p2*, with speed data *v500* and zone data *z30*. The circle is defined from the start position, the circle point *p1* and the destination point *p2*.

MoveC *, *, v500 \T:=5, fine, grip3;

The TCP of the tool, *grip3*, is moved circularly to a fine point stored in the instruction (marked by the second *). The circle point is also stored in the instruction (marked by the first *). The complete movement takes *5* seconds.

MoveL p1, v500, fine, tool1;
MoveC p2, p3, v500, z20, tool1;
MoveC p4, p1, v500, fine, tool1;

A complete circle is performed if the positions are the same as those shown in Figure 2.



Figure 2 A complete circle is performed by two MoveC instructions.

Arguments

MoveC [\Conc] CirPoint ToPoint Speed [\V] | [\T] Zone [\Z] Tool [\WObj] [\Corr]

[\Conc] *(Concurrent)* Data type: *switch*

Subsequent instructions are executed at once. This argument is used to shorten the cycle time when, for example, communicating with external equipment, if synchronisation is not required.

Using the argument \Conc, the number of movement instructions in succession is limited to 5. In a program section that includes StorePath-RestoPath, movement instructions with the argument \Conc are not permitted.

If this argument is omitted, and the ToPoint is not a Stop point the subsequent instruction is executed some time before the robot has reached the programmed zone.

CirPoint Data type: *robtarget*

The circle point of the robot. The circle point is a position on the circle between the start point and the destination point. To obtain the best accuracy, it should be placed about halfway between the start and destination points. If it is placed too close to the start or destination point, the robot may give a warning. The circle point is defined as a named position or stored directly in the instruction (marked with an * in the instruction). The position of the external axes are not used.

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the TCP, the tool reorientation and external axes.

[\V] *(Velocity)* Data type: *num*

This argument is used to specify the velocity of the TCP in mm/s directly in the instruction. It is then substituted for the corresponding velocity specified in the speed data.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot and external axes move. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

[\Z] *(Zone)* Data type: *num*

This argument is used to specify the position accuracy of the robot TCP directly in the instruction. The length of the corner path is given in mm, which is substituted for the corresponding zone specified in the zone data.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point that is moved to the specified destination point.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (object coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified in order for a circle relative to the work object to be executed.

[\Corr] *(Correction)* Data type: *switch*

Correction data written to a corrections entry by the instruction *CorrWrite* will be added to the path and destination position, if this argument is present.

Program execution

The robot and external units are moved to the destination point as follows:

- The TCP of the tool is moved circularly at constant programmed velocity.
- The tool is reoriented at a constant velocity, from the orientation at the start position to the orientation at the destination point.
- The reorientation is performed relative to the circular path. Thus, if the orientation relative to the path is the same at the start and the destination points, the relative orientation remains unchanged during the movement (see Figure 3).



Figure 3 Tool orientation during circular movement.

- The orientation at the circle point is not critical; it is only used to distinguish between two possible directions of reorientation. The accuracy of the reorientation along the path depends only on the orientation at the start and destination points.
- Uncoordinated external axes are executed at constant velocity in order for them to arrive at the destination point at the same time as the robot axes. The position in the circle position is not used.

If it is not possible to attain the programmed velocity for the reorientation or for the external axes, the velocity of the TCP will be reduced.

A corner path is usually generated when movement is transferred to the next section of a path. If a stop point is specified in the zone data, program execution only continues when the robot and external axes have reached the appropriate position.

Examples

MoveC *, *, v500 \V:=550, z40 \Z:=45, grip3;

The TCP of the tool, *grip3*, is moved circularly to a position stored in the instruction. The movement is carried out with data set to *v500* and *z40*; the velocity and zone size of the TCP are *550* mm/s and *45* mm respectively.

MoveC \Conc, *, *, v500, z40, grip3;

The TCP of the tool, *grip3*, is moved circularly to a position stored in the instruction. The circle point is also stored in the instruction. Subsequent logical instructions are executed while the robot moves.

MoveC cir1, p15, v500, z40, grip3 \WObj:=fixture;

The TCP of the tool, *grip3*, is moved circularly to a position, *p15*, via the circle point *cir1*. These positions are specified in the object coordinate system for *fixture*.

Limitations

A change of execution mode from forward to backward or vice versa, while the robot is stopped on a circular path, is not permitted and will result in an error message.

The instruction *MoveC* (or any other instruction including circular movement) should never be started from the beginning, with TCP between the circle point and the end point. Otherwise the robot will not take the programmed path (positioning around the circular path in another direction compared with that programmed).

Make sure that the robot can reach the circle point during program execution and divide the circle segment if necessary.

Syntax

```
MoveC
   [ '\' Conc ',' ]
   [ CirPoint ':=' ] < expression (IN) of robtarget > ','
   [ ToPoint ':=' ] < expression (IN) of robtarget > ','
   [ Speed ':=' ] < expression (IN) of speeddata >
                  [ '\' V ':=' < expression (IN) of num > ]
                  | [ '\' T ':=' < expression (IN) of num > ] ','
   [Zone ':=' ] < expression (IN) of zonedata >
                  [ '\' Z ':=' < expression (IN) of num > ] ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ]
   [ '\' Corr ]';'
```

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Writes to a corrections entry	Instructions - *CorrWrite*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Concurrent program execution	Motion and I/O Principles - *Synchronisation Using Logical Instructions*

MoveCDO Moves the robot circularly and sets digital output in the corner

MoveCDO (*Move Circular Digital Output*) is used to move the tool centre point (TCP) circularly to a given destination. The specified digital output is set/reset in the middle of the corner path at the destination point. During the movement, the orientation normally remains unchanged relative to the circle.

Examples

MoveCDO p1, p2, v500, z30, tool2, do1,1;

The TCP of the tool, *tool2*, is moved circularly to the position *p2*, with speed data *v500* and zone data *z30*. The circle is defined from the start position, the circle point *p1* and the destination point *p2*. Output *do1* is set in the middle of the corner path at *p2*.

Arguments

MoveCDO CirPoint ToPoint Speed [\T] Zone Tool [\WObj] Signal Value

CirPoint Data type: *robtarget*

The circle point of the robot. The circle point is a position on the circle between the start point and the destination point. To obtain the best accuracy, it should be placed about halfway between the start and destination points. If it is placed too close to the start or destination point, the robot may give a warning. The circle point is defined as a named position or stored directly in the instruction (marked with an * in the instruction). The position of the external axes are not used.

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the TCP, the tool reorientation and external axes.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot and external axes move. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point that is moved to the specified destination point.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (object coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified in order for a circle relative to the work object to be executed.

Signal Data type: *signaldo*

The name of the digital output signal to be changed.

Value Data type: *dionum*

The desired value of signal (0 or 1).

Program execution

See the instruction *MoveC* for more information about circular movement.

The digital output signal is set/reset in the middle of the corner path for flying points, as shown in Figure 4.



Figure 4 Set/Reset of digital output signal in the corner path with MoveCDO.

For stop points, we recommend the use of “normal” programming sequence with *MoveC* + *SetDO*. But when using stop point in instruction *MoveCDO*, the digital output signal is set/reset when the robot reaches the stop point.

The specified I/O signal is set/reset in execution mode continuously and stepwise forward but not in stepwise backward.

Limitations

General limitations according to instruction *MoveC*.

Syntax

```
MoveCDO
   [ CirPoint ':=' ] < expression (IN) of robtarget > ','
   [ ToPoint ':=' ] < expression (IN) of robtarget > ','
   [ Speed ':=' ] < expression (IN) of speeddata >
                  [ '\' T ':=' < expression (IN) of num > ] ','
   [ Zone ':=' ] < expression (IN) of zonedata > ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ] ','
   [ Signal ':=' ] < variable (VAR) of signaldo>] ','
   [ Value ':=' ] < expression (IN) of dionum > ] ';'
```

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Movements with I/O settings	Motion and I/O Principles - *Synchronisation Using Logical Instructions*

MoveJDO Moves the robot by joint movement and sets digital output in the corner

MoveJDO (*Move Joint Digital Output*) is used to move the robot quickly from one point to another when that movement does not have to be in a straight line. The specified digital output signal is set/reset at the middle of the corner path.

The robot and external axes move to the destination position along a non-linear path. All axes reach the destination position at the same time.

Examples

MoveJDO p1, vmax, z30, tool2, do1, 1;

The tool centre point (TCP) of the tool, *tool2*, is moved along a non-linear path to the position, *p1*, with speed data *vmax* and zone data *z30*. Output *do1* is set in the middle of the corner path at *p1*.

Arguments

MoveJDO ToPoint Speed [\T] Zone Tool [\WObj] Signal Value

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the tool reorientation and external axes.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point moved to the specified destination point.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified.

Signal Data type: *signaldo*

The name of the digital output signal to be changed.

Value Data type: *dionum*

The desired value of signal (0 or 1).

Program execution

See the instruction *MoveJ* for more information about joint movement.

The digital output signal is set/reset in the middle of the corner path for flying points, as shown in Figure 5.



Figure 5 Set/Reset of digital output signal in the corner path with MoveJDO.

For stop points, we recommend the use of “normal” programming sequence with *MoveJ* + *SetDO*. But when using stop point in instruction *MoveJDO*, the digital output signal is set/reset when the robot reaches the stop point.

The specified I/O signal is set/reset in execution mode continuously and stepwise forward but not in stepwise backward.

Syntax

```
MoveJDO
   [ ToPoint ':=' ] < expression (IN) of robtarget > ','
   [ Speed ':=' ] < expression (IN) of speeddata >
                  [ '\' T ':=' < expression (IN) of num > ] ','
   [ Zone ':=' ] < expression (IN) of zonedata > ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ] ','
   [ Signal ':=' ] < variable (VAR) of signaldo>] ','
   [ Value ':=' ] < expression (IN) of dionum > ] ';'
```

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Movements with I/O settings	Motion and I/O Principles - *Synchronisation Using Logical Instructions*

MoveJ Moves the robot by joint movement

MoveJ is used to move the robot quickly from one point to another when that movement does not have to be in a straight line.

The robot and external axes move to the destination position along a non-linear path. All axes reach the destination position at the same time.

Examples

MoveJ p1, vmax, z30, tool2;

The tool centre point (TCP) of the tool, *tool*2, is moved along a non-linear path to the position, *p1*, with speed data *vmax* and zone data *z30*.

MoveJ *, vmax \T:=5, fine, grip3;

The TCP of the tool, *grip3*, is moved along a non-linear path to a stop point stored in the instruction (marked with an *). The entire movement takes *5* seconds.

Arguments

MoveJ [\Conc] ToPoint Speed [\V] | [\T] Zone [\Z] Tool [\WObj]

[\Conc] *(Concurrent)* Data type: *switch*

Subsequent instructions are executed while the robot is moving. The argument is used to shorten the cycle time when, for example, communicating with external equipment, if synchronisation is not required.

Using the argument \Conc, the number of movement instructions in succession is limited to 5. In a program section that includes StorePath-RestoPath, movement instructions with the argument \Conc are not permitted.

If this argument is omitted and the ToPoint is not a stop point, the subsequent instruction is executed some time before the robot has reached the programmed zone.

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the tool reorientation and external axes.

[\V] *(Velocity)* Data type: *num*

This argument is used to specify the velocity of the TCP in mm/s directly in the instruction. It is then substituted for the corresponding velocity specified in the speed data.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

[\Z] *(Zone)* Data type: *num*

This argument is used to specify the position accuracy of the robot TCP directly in the instruction. The length of the corner path is given in mm, which is substituted for the corresponding zone specified in the zone data.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point moved to the specified destination point.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified.

Program execution

The tool centre point is moved to the destination point with interpolation of the axis angles. This means that each axis is moved with constant axis velocity and that all axes reach the destination point at the same time, which results in a non-linear path.

Generally speaking, the TCP is moved at the approximate programmed velocity (regardless of whether or not the external axes are coordinated). The tool is reoriented and the external axes are moved at the same time as the TCP moves. If the programmed velocity for reorientation, or for the external axes, cannot be attained, the velocity of the TCP will be reduced.

A corner path is usually generated when movement is transferred to the next section of the path. If a stop point is specified in the zone data, program execution only continues when the robot and external axes have reached the appropriate position.

Examples

MoveJ *, v2000\V:=2200, z40 \Z:=45, grip3;

The TCP of the tool, *grip3*, is moved along a non-linear path to a position stored in the instruction. The movement is carried out with data set to *v2000* and *z40*; the velocity and zone size of the TCP are *2200* mm/s and *45* mm respectively.

MoveJ \Conc, *, v2000, z40, grip3;

The TCP of the tool, *grip3*, is moved along a non-linear path to a position stored in the instruction. Subsequent logical instructions are executed while the robot moves.

MoveJ start, v2000, z40, grip3 \WObj:=fixture;

The TCP of the tool, *grip3*, is moved along a non-linear path to a position, *start*. This position is specified in the object coordinate system for *fixture*.

Syntax

MoveJ
[’\’ Conc ’,’]
[ToPoint ’:=’] < expression (**IN**) of *robtarget* > ’,’
[Speed ’:=’] < expression (**IN**) of *speeddata* >
[’\’ V ’:=’ < expression (**IN**) of *num* >]
| [’\’ T ’:=’ < expression (**IN**) of *num* >] ’,’
[Zone ’:=’] < expression (**IN**) of *zonedata* >
[’\’ Z ‘:=’ < expression (**IN**) of *num* >] ’,’
[Tool ’:=’] < persistent (**PERS**) of *tooldata* >
[’\’ WObj ’:=’ < persistent (**PERS**) of *wobjdata* >] ’;’

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Concurrent program execution	Motion and I/O Principles - *Synchronisation Using Logical Instructions*

MoveLDO Moves the robot linearly and sets digital output in the corner

MoveLDO (*Move Linearly Digital Output*) is used to move the tool centre point (TCP) linearly to a given destination. The specified digital output signal is set/reset at the middle of the corner path.

When the TCP is to remain stationary, this instruction can also be used to reorient the tool.

Example

MoveLDO p1, v1000, z30, tool2, do1,1;

The TCP of the tool, *tool2*, is moved linearly to the position *p1*, with speed data *v1000* and zone data *z30*. Output *do1* is set in the middle of the corner path at *p1*.

Arguments

MoveLDO ToPoint Speed [\T] Zone Tool [\WObj] Signal Value

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity for the tool centre point, the tool reorientation and external axes.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point moved to the specified destination position.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified.

Signal Data type: *signaldo*

The name of the digital output signal to be changed.

Value Data type: *dionum*

The desired value of signal (0 or 1).

Program execution

See the instruction *MoveL* for more information about linear movements.

The digital output signal is set/reset in the middle of the corner path for flying points, as shown in Figure 6.



Figure 6 Set/Reset of digital output signal in the corner path with MoveLDO.

For stop points, we recommend the use of “normal” programming sequence with *MoveL* + *SetDO*. But when using stop point in instruction *MoveLDO*, the digital output signal is set/reset when the robot reaches the stop point.

The specified I/O signal is set/reset in execution mode continuously and stepwise forward but not in stepwise backward.

Syntax

```
MoveLDO
   [ ToPoint ':=' ] < expression (IN) of robtarget > ','
   [ Speed ':=' ] < expression (IN) of speeddata >
                  [ '\' T ':=' < expression (IN) of num > ] ','
   [ Zone ':=' ] < expression (IN) of zonedata > ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ] ','
   [ Signal ':=' ] < variable (VAR) of signaldo>] ','
   [ Value ':=' ] < expression (IN) of dionum > ] ';'
```

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Movements with I/O settings	Motion and I/O Principles - *Synchronisation Using Logical Instructions*

MoveL Moves the robot linearly

MoveL is used to move the tool centre point (TCP) linearly to a given destination. When the TCP is to remain stationary, this instruction can also be used to reorientate the tool.

Example

MoveL p1, v1000, z30, tool2;

The TCP of the tool, *tool*2, is moved linearly to the position *p1*, with speed data *v1000* and zone data *z30*.

MoveL *, v1000\T:=5, fine, grip3;

The TCP of the tool, *grip3*, is moved linearly to a fine point stored in the instruction (marked with an *). The complete movement takes *5* seconds.

Arguments

MoveL [\Conc] ToPoint Speed [\V] | [\T] Zone [\Z] Tool [\WObj] [\Corr]

[\Conc] *(Concurrent)* Data type: *switch*

Subsequent instructions are executed at once. This argument is used to shorten the cycle time when, for example, communicating with external equipment, if synchronisation is not required.

Using the argument \Conc, the number of movement instructions in succession is limited to 5. In a program section that includes StorePath-RestoPath, movement instructions with the argument \Conc are not permitted.

If this argument is omitted and the ToPoint is not a stop point, the subsequent instruction is executed some time before the robot has reached the programmed zone.

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity for the tool centre point, the tool reorientation and external axes.

[\V] *(Velocity)* Data type: *num*

This argument is used to specify the velocity of the TCP in mm/s directly in the instruction. It is then substituted for the corresponding velocity specified in the speed data.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

[\Z] *(Zone)* Data type: *num*

This argument is used to specify the position accuracy of the robot TCP directly in the instruction. The length of the corner path is given in mm, which is substituted for the corresponding zone specified in the zone data.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point moved to the specified destination position.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary tool or coordinated external axes are used, this argument must be specified in order to perform a linear movement relative to the work object.

[\Corr] *(Correction)* Data type: *switch*

Correction data written to a corrections entry by the instruction *CorrWrite* will be added to the path and destination position, if this argument is present.

Program execution

The robot and external units are moved to the destination position as follows:

- The TCP of the tool is moved linearly at constant programmed velocity.
- The tool is reoriented at equal intervals along the path.
- Uncoordinated external axes are executed at a constant velocity in order for them to arrive at the destination point at the same time as the robot axes.

If it is not possible to attain the programmed velocity for the reorientation or for the external axes, the velocity of the TCP will be reduced.

A corner path is usually generated when movement is transferred to the next section of a path. If a stop point is specified in the zone data, program execution only continues when the robot and external axes have reached the appropriate position.

Examples

MoveL *, v2000 \V:=2200, z40 \Z:=45, grip3;

The TCP of the tool, *grip3*, is moved linearly to a position stored in the instruction. The movement is carried out with data set to *v2000* and *z40*; the velocity and zone size of the TCP are *2200* mm/s and *45* mm respectively.

MoveL \Conc, *, v2000, z40, grip3;

The TCP of the tool, *grip3*, is moved linearly to a position stored in the instruction. Subsequent logical instructions are executed while the robot moves.

MoveL start, v2000, z40, grip3 \WObj:=fixture;

The TCP of the tool, *grip3*, is moved linearly to a position, *start*. This position is specified in the object coordinate system for *fixture*.

Syntax

```
MoveL
   [ '\' Conc ',' ]
   [ ToPoint ':=' ] < expression (IN) of robtarget > ','
   [ Speed ':=' ] < expression (IN) of speeddata >
                  [ '\' V ':=' < expression (IN) of num > ]
                  | [ '\' T ':=' < expression (IN) of num > ] ','
   [Zone ':=' ] < expression (IN) of zonedata >
                  [ '\' Z ':=' < expression (IN) of num > ] ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ]
   [ '\' Corr ]';'
```

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Writes to a corrections entry	Instructions - *CorrWrite*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Concurrent program execution	Motion and I/O Principles - *Synchronisation Using Logical Instructions*

MoveJSync Moves the robot by joint movement and executes a RAPID procedure

MoveJSync (*Move Joint Synchronously)* is used to move the robot quickly from one point to another when that movement does not have to be in a straight line. The specified RAPID procedure is executed at the middle of the corner path in the destination point.

The robot and external axes move to the destination position along a non-linear path. All axes reach the destination position at the same time.

Examples

MoveJSync p1, vmax, z30, tool2, “proc1”;

The tool centre point (TCP) of the tool, *tool2*, is moved along a non-linear path to the position, *p1*, with speed data *vmax* and zone data *z30*. Procedure *proc1* is executed in the middle of the corner path at *p1*.

Arguments

MoveJSync ToPoint Speed [\T] Zone Tool [\WObj] ProcName

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the tool reorientation and external axes.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point moved to the specified destination point.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified.

ProcName *(Procedure Name)* Data type: *string*

Name of the RAPID procedure to be executed at the middle of the corner path in the destination point.

Program execution

See the instruction *MoveJ* for more information about joint movements.

The specified RAPID procedure is executed when the TCP reaches the middle of the corner path in the destination point of the *MoveJSync* instruction, as shown in Figure 7:



Figure 7 Execution of user-defined RAPID procedure in the middle of the corner path.

For stop points, we recommend the use of “normal” programming sequence with *MoveJ* + other RAPID instructions in sequence.

Execution of the specified RAPID procedure in different execution modes:

Execution mode:	Execution of RAPID procedure:
Continuously or Cycle	According to this description
Forward step	In the stop point
Backward step	Not at all

Limitation

Switching execution mode after program stop from continuously or cycle to stepwise forward or backward results in an error. This error tells the user that the mode switch can result in missed execution of a RAPID procedure in the queue for execution on the path. This error can be avoided if the program is stopped with StopInstr before the mode switch.

Instruction *MoveJSync* cannot be used on TRAP level.
The specified RAPID procedure cannot be tested with stepwise execution.

Syntax

```
MoveJSync
   [ ToPoint ':=' ] < expression (IN) of robtarget > ','
   [ Speed ':=' ] < expression (IN) of speeddata >
                  [ '\' T ':=' < expression (IN) of num > ] ','
   [ Zone ':=' ] < expression (IN) of zonedata >
                  [ '\' Z ':=' < expression (IN) of num > ] ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ] ','
   [ ProcName':=' ] < expression (IN) of string > ] ';'
```

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*

MoveL Sync Moves the robot linearly and executes a RAPID procedure

MoveLSync (*Move Linearly Synchronously*) is used to move the tool centre point (TCP) linearly to a given destination.The specified RAPID procedure is executed at the middle of the corner path in the destination point.

When the TCP is to remain stationary, this instruction can also be used to reorient the tool.

Example

MoveLSync p1, v1000, z30, tool2, “proc1”;

The TCP of the tool, *tool2*, is moved linearly to the position *p1*, with speed data *v1000* and zone data *z30*. Procedure *proc1* is executed in the middle of the corner path at *p1*.

Arguments

MoveLSync ToPoint Speed [\T] Zone Tool [\WObj] ProcName

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity for the tool centre point, the tool reorientation and external axes.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point moved to the specified destination position.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified.

ProcName *(Procedure Name)* Data type: *string*

Name of the RAPID procedure to be executed at the middle of the corner path in the destination point.

Program execution

See the instruction *MoveL* for more information about linear movements.

The specified RAPID procedure is executed when the TCP reaches the middle of the corner path in the destination point of the *MoveLSync* instruction, as shown in Figure 8:



Figure 8 Execution of user-defined RAPID procedure in the middle of the corner path.

For stop points, we recommend the use of “normal” programming sequence with *MoveL* + other RAPID instructions in sequence.

Execution of the specified RAPID procedure in different execution modes:

Execution mode:	Execution of RAPID procedure:
Continuously or Cycle	According to this description
Forward step	In the stop point
Backward step	Not at all

Limitation

Switching execution mode after program stop from continuously or cycle to stepwise forward or backward results in an error. This error tells the user that the mode switch can result in missed execution of a RAPID procedure in the queue for execution on the path. This error can be avoided if the program is stopped with StopInstr before the mode switch.

Instruction *MoveLSync* cannot be used on TRAP level.
The specified RAPID procedure cannot be tested with stepwise execution.

Syntax

MoveLSync
[ToPoint ':='] < expression (**IN**) of *robtarget* > ','
[Speed ':='] < expression (**IN**) of *speeddata* >
['\' T ':=' < expression (**IN**) of *num* >] ','
[Zone ':='] < expression (**IN**) of *zonedata* > ','
[Tool ':='] < persistent (**PERS**) of *tooldata* >
['\' WObj ':=' < persistent (**PERS**) of *wobjdata* >] ','
[ProcName':='] < expression (**IN**) of *string* >] ';'

Related information

	Described in:
Other positioning instructions	RAPID Summary - *Motion*
Definition of velocity	Data Types - *speeddata*
Definition of zone data	Data Types - *zonedata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion and I/O Principles
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*

Open Opens a file or serial channel

Open is used to open a file or serial channel for reading or writing.

Example

```
VAR iodev logfile;
...
Open "flp1:" \File:= "LOGDIR/LOGFILE1.DOC", logfile;
```

The file *LOGFILE1.DOC* in unit *flp1:* (diskette), directory *LOGDIR*, is opened for writing. The reference name *logfile* is used later in the program when writing to the file.

Arguments

Open Object [\File] IODevice [\Read] | [\Write] | [\Append] [\Bin]

Object Data type: *string*

The I/O object that is to be opened, e.g. "flp1:", "ram1disk:" or "sio1:"

[\File] Data type: *string*

The name of the file to be opened, e.g. "LOGDIR/LOGFILE1.DOC"

The complete path can also be specified in the argument *Object*, e.g. "flp1:LOGDIR/LOGFILE.DOC".

IODevice Data type: *iodev*

A reference to the file or serial channel to open. This reference is then used for reading from and writing to the file or serial channel.

[\Read] Data type: *switch*

Opens a file or serial channel for reading. When reading from a file, the reading is started from the beginning of the file.

[\Write] Data type: *switch*

Opens a file or serial channel for writing. If the selected file already exists, its contents are deleted. Anything subsequently written is written at the start of the file.

[\Append] Data type: *switch*

Opens a file or serial channel for writing. If the selected file already exists, anything subsequently written is written at the end of the file.

Open a file or serial channel with *Append* and without the *Bin* arguments. The instruction opens a character-based file or serial channel for writing.

Open a file or serial channel with *Append* and *Bin* arguments. The instruction opens a binary file or serial channel for both reading and writing.

The arguments *Read*, *Write*, *Append* are mutually exclusive. If none of these are specified, the instruction acts in the same way as the *Write* argument for character-based files or a serial channel (instruction without *Bin* argument) and in the same way as the *Append* argument for binary files or a serial channel (instruction with *Bin* argument).

[\Bin] Data type: *switch*

The file or serial channel is opened in a binary mode.
If none of the arguments *Read*, *Write* or *Append* are specified, the instruction opens a binary file or serial channel for both reading and writing, with the file pointer at the end of the file

The set of instructions to access a binary file or serial channel is different from the set of instructions to access a character-based file.

Example

```
VAR iodev printer;
...
Open "sio1:", printer \Bin;
WriteStrBin printer, "This is a message to the printer\0D";
Close printer;
```

The serial channel *sio1:* is opened for binary reading and writing.
The reference name *printer* is used later when writing to and closing the serial channel.

Program execution

The specified file or serial channel is opened so that it is possible to read from or write to it.

It is possible to open the same physical file several times at the same time, but each invocation of the *Open* instruction will return a different reference to the file (data type *iodev*). E.g. it is possible to have one write pointer and one different read pointer to the same file at the same time.

The *iodev* variable used when opening a file or serial channel must be free from use. If it has been used previously to open a file, this file must be closed prior to issuing a new *Open* instruction with the same *iodev* variable.

Error handling

If a file cannot be opened, the system variable ERRNO is set to ERR_FILEOPEN. This error can then be handled in the error handler.

Syntax

Open
[Object ':='] <expression (**IN**) of *string*>
['\'File':=' <expression (**IN**) of *string*>] ','
[IODevice ':='] <variable (**VAR**) of *iodev*>
['\'Read] | ['\'Write] | ['\'Append]
['\'Bin] ';'

Related information

	Described in:
Writing to and reading from files or serial channels	RAPID Summary - *Communication*

PathResol Override path resolution

PathResol (Path Resolution) is used to override the configured geometric path sample time defined in the system parameters for the manipulator.

Description

The path resolution affects the accuracy of the interpolated path and the program cycle time. The path accuracy is improved and the cycle time is often reduced when the parameter *PathSampleTime* is decreased. A value for parameter *PathSampleTime* which is too low, may however cause CPU load problems in some demanding applications. However, use of the standard configured path resolution (*PathSampleTime* 100%) will avoid CPU load problems and provide sufficient path accuracy in most situations.

Example of *PathResol* usage:

Dynamically critical movements (max payload, high speed, combined joint motions close to the border of the work area) may cause CPU load problems. Increase the parameter *PathSampleTime*.

Low performance external axes may cause CPU load problems during coordination. Increase the parameter *PathSampleTime*.

Arc-welding with high frequency weaving may require high resolution of the interpolated path. Decrease the parameter *PathSampleTime*.

Small circles or combined small movements with direction changes can decrease the path performance quality and increase the cycle time. Decrease the parameter *PathSampleTime*.

Gluing with large reorientations and small corner zones can cause speed variations. Decrease the parameter *PathSampleTime*.

Example

```
MoveJ p1,v1000,fine,tool1;
PathResol 150;
```

With the robot at a stop point, the path sample time is increased to *150*% of the configured.

Arguments

PathResol PathSampleTime

PathSampleTime Data type: *num*

Override as a percent of the configured path sample time.
100% corresponds to the configured path sample time.
Within the range 25-400%.

A lower value of the parameter *PathSampleTime* improves the path resolution (path accuracy).

Program execution

The path resolutions of all subsequent positioning instructions are affected until a new *PathResol* instruction is executed. This will affect the path resolution during all program execution of movements (default path level and path level after *StorePath*) and also during jogging.

The default value for override of path sample time is 100%. This value is automatically set

- at a cold start-up
- when a new program is loaded
- when starting program execution from the beginning.

The current override of path sample time can be read from the variable *C_MOTSET* (data type *motsetdata*) in the component *pathresol*.

Limitations

The robot must be standing still at a stop point before overriding the path sample time. When there is a corner path in the program, the system will instead create a stop point (warning 50146) and it is not possible to restart in this instruction following a power failure.

Syntax

PathResol
[PathSampleTime ':='] < expression (**IN**) of *num*> ';'

Related information

	Described in:
Positioning instructions	Motion and I/O Principles- *Movements*
Motion settings	RAPID Summary - *Motion Settings*
Configuration of path resolution	System Parameters - *CPU Optimization*

PDispOff Deactivates program displacement

PDispOff (Program Displacement Off) is used to deactivate a program displacement.

Program displacement is activated by the instruction *PDispSet* or *PDispOn* and applies to all movements until some other program displacement is activated or until program displacement is deactivated.

Examples

```
PDispOff;
```

Deactivation of a program displacement.

```
MoveL p10, v500, z10, tool1;
PDispOn \ExeP:=p10, p11, tool1;
MoveL p20, v500, z10, tool1;
MoveL p30, v500, z10, tool1;
PDispOff;
MoveL p40, v500, z10, tool1;
```

A program displacement is defined as the difference between the positions *p10* and *p11*. This displacement affects the movement to *p20* and *p30*, but not to *p40*.

Program execution

Active program displacement is reset. This means that the program displacement coordinate system is the same as the object coordinate system, and thus all programmed positions will be related to the latter.

Syntax

PDispOff ‘;’

Related information

	Described in:
Definition of program displacement using two positions	Instructions - *PDispOn*
Definition of program displacement using values	Instructions - *PDispSet*

PDispOn Activates program displacement

PDispOn (Program Displacement On) is used to define and activate a program displacement using two robot positions.

Program displacement is used, for example, after a search has been carried out, or when similar motion patterns are repeated at several different places in the program.

Examples

```
MoveL p10, v500, z10, tool1;
PDispOn \ExeP:=p10, p20, tool1;
```

Activation of a program displacement (parallel movement). This is calculated based on the difference between positions *p10* and *p20*.

```
MoveL p10, v500, fine, tool1;
PDispOn *, tool1;
```

Activation of a program displacement (parallel movement). Since a stop point has been used in the previous instruction, the argument \ExeP does not have to be used. The displacement is calculated on the basis of the difference between the robot's actual position and the programmed point (*) stored in the instruction.

```
PDispOn \Rot \ExeP:=p10, p20, tool1;
```

Activation of a program displacement including a rotation. This is calculated based on the difference between positions *p10* and *p20*.

Arguments

PDispOn [\Rot] [\ExeP] ProgPoint Tool [\WObj]

[\Rot] *(Rotation)* Data type: *switch*

The difference in the tool orientation is taken into consideration and this involves a rotation of the program.

[\ExeP] *(Executed Point)* Data type: *robtarget*

The robot's new position at the time of the program execution.
If this argument is omitted, the robot's current position at the time of the program execution is used.

ProgPoint *(Programmed Point)* Data type: *robtarget*

The robot's original position at the time of programming.

Tool Data type: *tooldata*

The tool used during programming, i.e. the TCP to which the *ProgPoint* position is related.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the *ProgPoint* position is related.

This argument can be omitted and, if it is, the position is related to the world coordinate system. However, if a stationary TCP or coordinated external axes are used, this argument must be specified.

The arguments *Tool* and *\WObj* are used both to calculate the *ProgPoint* during programming and to calculate the current position during program execution if no ExeP argument is programmed.

Program execution

Program displacement means that the ProgDisp coordinate system is translated in relation to the object coordinate system. Since all positions are related to the ProgDisp coordinate system, all programmed positions will also be displaced. See Figure 9.



Figure 9 Displacement of a programmed position using program displacement.

Program displacement is activated when the instruction *PDispOn* is executed and remains active until some other program displacement is activated (the instruction *PDispSet* or *PDispOn*) or until program displacement is deactivated (the instruction *PDispOff*).

Only one program displacement can be active at any one time. Several *PDispOn* instructions, on the other hand, can be programmed one after the other and, in this case, the different program displacements will be added.

Program displacement is calculated as the difference between *ExeP* and *ProgPoint*. If *ExeP* has not been specified, the current position of the robot at the time of the program execution is used instead. Since it is the actual position of the robot that is used, the robot should not move when *PDispOn* is executed.

If the argument *\Rot* is used, the rotation is also calculated based on the tool orientation at the two positions. The displacement will be calculated in such a way that the new position (*ExeP)* will have the same position and orientation in relation to the displaced coordinate system, ProgDisp, as the old position (*ProgPoint*) had in relation to the original coordinate system (see Figure 10).



Figure 10 Translation and rotation of a programmed position.

The program displacement is automatically reset

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Example

```
PROC draw_square()
   PDispOn *, tool1;
   MoveL *, v500, z10, tool1;
   MoveL *, v500, z10, tool1;
   MoveL *, v500, z10, tool1;
   MoveL *, v500, z10, tool1;
   PDispOff;
ENDPROC
.
MoveL p10, v500, fine, tool1;
draw_square;
MoveL p20, v500, fine, tool1;
draw_square;
MoveL p30, v500, fine, tool1;
draw_square;
```

The routine *draw_square* is used to execute the same motion pattern at three different positions, based on the positions *p10*, *p20* and *p30*. See Figure 11.



Figure 11 Using program displacement, motion patterns can be reused.

```
SearchL sen1, psearch, p10, v100, tool1\WObj:=fixture1;
PDispOn \ExeP:=psearch, *, tool1 \WObj:=fixture1;
```

A search is carried out in which the robot's searched position is stored in the position *psearch*. Any movement carried out after this starts from this position using a program displacement (parallel movement). The latter is calculated based on the difference between the searched position and the programmed point (*) stored in the instruction. All positions are based on the *fixture1* object coordinate system.

Syntax

PDispOn
[['\' Rot]
['\' ExeP ':=' < expression (**IN**) of *robtarget* >] ',']
[ProgPoint ':='] < expression (**IN**) of *robtarget* > ','
[Tool ':='] < persistent (**PERS**) of *tooldata*>
['\'WObj ':=' < persistent (**PERS**) of *wobjdata*>] ';'

Related information

	Described in:
Deactivation of program displacement	Instructions - *PDispOff*
Definition of program displacement using values	Instructions - *PDispSet*
Coordinate systems	Motion Principles - *Coordinate Systems*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
More examples	Instructions - *PDispOff*

PDispSet Activates program displacement using a value

PDispSet (Program Displacement Set) is used to define and activate a program displacement using values.

Program displacement is used, for example, when similar motion patterns are repeated at several different places in the program.

Example

```
VAR pose xp100 :=  [ [100, 0, 0], [1, 0, 0, 0] ];
.
PDispSet xp100;
```

Activation of the *xp100* program displacement, meaning that:

- The ProgDisp coordinate system is displaced 100 mm from the object coordinate system, in the direction of the positive x-axis (see Figure 12).
- As long as this program displacement is active, all positions will be displaced 100 mm in the direction of the x-axis.



Figure 12 A 100 mm-program displacement along the x-axis.

Arguments

PDispSet DispFrame

DispFrame *(Displacement Frame)* Datatyp: *pose*

The program displacement is defined as data of the type *pose*.

Program execution

Program displacement involves translating and/or rotating the ProgDisp coordinate system relative to the object coordinate system. Since all positions are related to the ProgDisp coordinate system, all programmed positions will also be displaced.
See Figure 13.



Figure 13 Translation and rotation of a programmed position.

Program displacement is activated when the instruction *PDispSet* is executed and remains active until some other program displacement is activated (the instruction *PDispSet* or *PDispOn*) or until program displacement is deactivated (the instruction *PDispOff*).

Only one program displacement can be active at any one time. Program displacements cannot be added to one another using *PDispSet*.

The program displacement is automatically reset

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Syntax

PDispSet
[DispFrame ':='] < expression (**IN**) of *pose*> ';'

Related information

	Described in:
Deactivation of program displacement	Instructions - *PDispOff*
Definition of program displacement using two positions	Instructions - *PDispOn*
Definition of data of the type *pose*	Data Types - *pose*
Coordinate systems	Motion Principles- *Coordinate Systems*
Examples of how program displacement can be used	Instructions - *PDispOn*

PulseDO Generates a pulse on a digital output signal

PulseDO is used to generate a pulse on a digital output signal.

Examples

PulseDO do15;

A pulse with a pulse length of 0.2 s is generated on the output signal *do15*.

PulseDO \PLength:=1.0, ignition;

A pulse of length *1.0 s* is generated on the signal *ignition*.

Arguments

PulseDO [\PLength] Signal

[\PLength] *(Pulse Length)* Data type: *num*

The length of the pulse in seconds (0.1 - 32s).
If the argument is omitted, a 0.2 second pulse is generated.

Signal Data type: *signaldo*

The name of the signal on which a pulse is to be generated.

Program execution

A pulse is generated with a specified pulse length (see Figure 14).



Figure 14 Generation of a pulse on a digital output signal.

The next instruction is executed directly after the pulse starts. The pulse can then be set/reset without affecting the rest of the program execution.

Limitations

The length of the pulse has a resolution of 0.01 seconds. Programmed values that differ from this are rounded off.

Syntax

PulseDO
[’\’ PLength ’:=’ < expression (**IN**) of *num* > ’,’]
[Signal ’:=’] < variable (**VAR**) of *signaldo* > ’;’

Related information

	Described in:
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User’s Guide - *System Parameters*

RAISE Calls an error handler

RAISE is used to create an error in the program and then to call the error handler of the routine. *RAISE* can also be used in the error handler to propagate the current error to the error handler of the calling routine.

This instruction can, for example, be used to jump back to a higher level in the structure of the program, e.g. to the error handler in the main routine, if an error occurs at a lower level.

Example

```
IF ...
   IF ...
      IF ...
         RAISE escape1;
   .
ERROR
   IF ERRNO=escape1 RAISE;
```

The routine is interrupted to enable it to remove itself from a low level in the program. A jump occurs to the error handler of the called routine.

Arguments

RAISE [Error no.]

Error no. Data type: *errnum*

Error number: Any number between 1 and 90 which the error handler can use to locate the error that has occurred (the *ERRNO* system variable).

It is also possible to book an error number outside the range 1-90 with the instruction *BookErrNo*.

The error number must be specified outside the error handler in a RAISE instruction in order to be able to transfer execution to the error handler of that routine.

If the instruction is present in a routine's error handler, the error number may not be specified. In this case, the error is propagated to the error handler of the calling routine.

Program execution

Program execution continues in the routine's error handler. After the error handler has been executed, program execution can continue with:

- the routine that called the routine in question (RETURN),
- the error handler of the routine that called the routine in question (RAISE).

If the RAISE instruction is present in a routine's error handler, program execution continues in the error handler of the routine that called the routine in question. The same error number remains active.

If the RAISE instruction is present in a trap routine, the error is dealt with by the system's error handler.

Error handling

If the error number is out of range, the system variable ERRNO is set to ERR_ILLRAISE (see "Data types - errnum"). This error can be handled in the error handler.

Syntax

(EBNF)
RAISE [<error number>] ';'

<error number> ::= <expression>

Related information

	Described in:
Error handling	Basic Characteristics - *Error Recovery*
Booking error numbers	Instructions - *BookErrNo*

ReadAnyBin Read data from a binary serial channel or file

ReadAnyBin (Read Any Binary) is used to read any type of data from a binary serial channel or file.

Example

```
VAR iodev channel2;
VAR robtarget next_target;
...
Open "sio1:", channel2 \Bin;
ReadAnyBin channel2, next_target;
```

The next robot target to be executed, *next_target*, is read from the channel referred to by *channel2*.

Arguments

ReadAnyBin IODevice Data [\Time])

IODevice Data type: *iodev*

The name (reference) of the binary serial channel or file to be read.

Data Data type: *ANYTYPE*

The VAR or PERS to which the read data will be stored.

[\Time] Data type: *num*

The max. time for the reading operation (timeout) in seconds. If this argument is not specified, the max. time is set to 60 seconds.

If this time runs out before the read operation is finished, the error handler will be called with the error code ERR_DEV_MAXTIME. If there is no error handler, the execution will be stopped.

The timeout function is in use also during program stop and will be noticed in the RAPID program at program start.

Program execution

As many bytes as required for the specified data are read from the specified binary serial channel or file.

Limitations

This instruction can only be used for serial channels or files that have been opened for binary reading.

The data to be read by this instruction must have a *value* data type of *atomic*, *string*, or *record* data type. *Semi-value* and *non-value* data types cannot be used.

Array data cannot be used.

Note that the VAR or PERS variable, for storage of the data read, can be updated in several steps. Therefore, always wait until the whole data structure is updated before using read data from a TRAP or another program task.

Error handling

If an error occurs during reading, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

If the end of the file is detected before all the bytes are read, the system variable ERRNO is set to ERR_RANYBIN_EOF. This error can then be handled in the error handler.

Example

```
CONST num NEW_ROBT:=12;
CONST num NEW_WOBJ:=20;
VAR iodev channel;
VAR num input;
VAR robtarget cur_robt;
VAR wobjdata cur_wobj;

Open "sio1:", channel\Bin;

! Wait for the opcode character
input := ReadBin (channel \Time:= 0.1);
TEST input
CASE NEW_ROBT:
   ReadAnyBin channel, cur_robt;
CASE NEW_WOBJ:
   ReadAnyBin channel, cur_wobj;
ENDTEST

Close channel;
```

As a first step, the opcode of the message is read from the serial channel. According to this opcode a robtarget or a wobjdata is read from the serial channel.

Syntax

```
ReadAnyBin
   [IODevice':='] <variable (VAR) of iodev>','
   [Data':='] <var or pers (INOUT) of ANYTYPE>
   ['\'Time':=' <expression (IN) of num>]';'
```

Related information

	Described in:
Opening (etc.) of serial channels or files	RAPID Summary - *Communication*
Write data to a binary serial channel or file	Instructions - *WriteAnyBin*

Reset Resets a digital output signal

Reset is used to reset the value of a digital output signal to zero.

Examples

Reset do15;

The signal *do15* is set to 0.

Reset weld;

The signal *weld* is set to 0.

Arguments

Reset Signal

Signal Data type: *signaldo*

The name of the signal to be reset to zero.

Program execution

The true value depends on the configuration of the signal. If the signal is inverted in the system parameters, this instruction causes the physical channel to be set to 1.

Syntax

Reset
[Signal ':='] < variable (**VAR**) of *signaldo* > ';'

Related information

	Described in:
Setting a digital output signal	Instructions - *Set*
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	System Parameters

RestoPath Restores the path after an interrupt

RestoPath is used to restore a path that was stored at a previous stage using the instruction *StorePath*.

Example

RestoPath;

Restores the path that was stored earlier using *StorePath*.

Program execution

The current movement path of the robot and the external axes is deleted and the path stored earlier using *StorePath* is restored. Nothing moves, however, until the instruction *StartMove* is executed or a return is made using *RETRY* from an error handler.

Example

```
ArcL p100, v100, seam1, weld5, weave1, z10, gun1;
...
ERROR
   IF ERRNO=AW_WELD_ERR THEN
      gun_cleaning;
      RETRY;
   ENDIF
...
PROC gun_cleaning()
   VAR robtarget p1;
   StorePath;
   p1 := CRobT();
   MoveL pclean, v100, fine, gun1;
   ...
   MoveL p1, v100, fine, gun1;
   RestoPath;
ENDPROC
```

In the event of a welding error, program execution continues in the error handler of the routine, which, in turn, calls *gun_cleaning*. The movement path being executed at the time is then stored and the robot moves to the position *pclean* where the error is rectified. When this has been done, the robot returns to the position where the error occurred, *p1*, and stores the original movement once again. The weld then automatically restarts, meaning that the robot is first reversed along the path before welding starts and ordinary program execution can continue.

Limitations

Only the movement path data is stored with the instruction *StorePath*.
If the user wants to order movements on the new path level, the actual stop position must be stored directly after *StorePath* and before *RestoPath* make a movement to the stored stop position on the path.

The movement instruction which precedes this instruction should be terminated with a stop point.

Syntax

RestoPath‘;’

Related information

	Described in:
Storing paths	Instructions - *StorePath*
More examples	Instructions - *StorePath*

RETRY Restarts following an error

RETRY is used to restart program execution after an error has occurred.

Example

```
reg2 := reg3/reg4;
   .
ERROR
   IF ERRNO = ERR_DIVZERO THEN
      reg4 := 1;
      RETRY;
   ENDIF
```

An attempt is made to divide *reg3* by *reg4*. If reg4 is equal to 0 (division by zero), a jump is made to the error handler, which initialises reg4. The *RETRY* instruction is then used to jump from the error handler and another attempt is made to complete the division.

Program execution

Program execution continues with (re-executes) the instruction that caused the error.

Error handling

If the maximum number of retries (4 retries) is exceeded, the program execution stops with an error message and the system variable ERRNO is set to ERR_EXCRTYMAX (see "Data types - errnum").

Limitations

The instruction can only exist in a routine's error handler. If the error was created using a *RAISE* instruction, program execution cannot be restarted with a *RETRY* instruction, then the instruction *TRYNEXT* should be used.

Syntax

RETRY ';'

Related information

	Described in:
Error handlers	Basic Characteristics- *Error Recovery*
Continue with the next instruction	Instructions - *TRYNEXT*

RETURN Finishes execution of a routine

RETURN is used to finish the execution of a routine. If the routine is a function, the function value is also returned.

Examples

```
errormessage;
Set do1;
   .

PROC errormessage()
   TPWrite "ERROR";
   RETURN;
ENDPROC
```

The *errormessage* procedure is called. When the procedure arrives at the RETURN instruction, program execution returns to the instruction following the procedure call, *Set do1*.

```
FUNC num abs_value(num value)
   IF value<0 THEN
      RETURN -value;
   ELSE
      RETURN value;
   ENDIF
ENDFUNC
```

The function returns the absolute value of a number.

Arguments

RETURN [Return value]

Return value Data type: According to the function declaration

The return value of a function.

The return value must be specified in a RETURN instruction present in a function.

If the instruction is present in a procedure or trap routine, a return value may not be specified.

Program execution

The result of the *RETURN* instruction may vary, depending on the type of routine it is used in:

- Main routine: If a program stop has been ordered at the end of the cycle, the program stops. Otherwise, program execution continues with the first instruction of the main routine.
- Procedure: Program execution continues with the instruction following the procedure call.
- Function: Returns the value of the function.
- Trap routine: Program execution continues from where the interrupt occurred.
- Error handler: In a procedure:
  Program execution continues with the routine that called the routine with the error handler (with the instruction following the procedure call).

  In a function:
  The function value is returned.

Syntax

(EBNF)
RETURN [<expression>]';'

Related information

	Described in:
Functions and Procedures	Basic Characteristics - *Routines*
Trap routines	Basic Characteristics - *Interrupts*
Error handlers	Basic Characteristics - *Error Recovery*

Rewind Rewind file position

Rewind sets the file position to the beginning of the file.

Example

Rewind iodev1;

The file referred to by *iodev1* will have the file position set to the beginning of the file.

Arguments

Rewind IODevice

IODevice Data type: *iodev*

Name (reference) of the file to be rewound.

Program execution

The specified file is rewound to the beginning.

Example

```
! IO device and numeric variable for use together with a binary file
VAR iodev dev;
VAR num bindata;

! Open the binary file with \Write switch to erase old contents
Open "flp1:"\File := "bin_file",dev \Write;
Close dev;

! Open the binary file with \Bin switch for binary read and write access
Open "flp1:"\File := "bin_file",dev \Bin;
WriteStrBin dev,"Hello world";

! Rewind the file pointer to the beginning of the binary file
! Read contents of the file and write the binary result on TP
! (gives 72 101 108 108 111 32 119 111 114 108 100 )
Rewind dev;
bindata := ReadBin(dev);
WHILE bindata <> EOF_BIN DO
    TPWrite " " \Num:=bindata;
    bindata := ReadBin(dev);
ENDWHILE

! Close the binary file
Close dev;
```

The instruction *Rewind* is used to rewind a binary file to the beginning so that the contents of the file can be read back with *ReadBin.*

Syntax

Rewind
[IODevice ':='] <variable **(VAR)** of *iodev*>';'

Related information

	Described in:
Opening (etc.) of files	RAPID Summary - *Communication*

Save Save a program module

Save is used to save a program module.

The specified program module in the program memory will be saved with the original (specified in *Load* or *StartLoad*) or specified file path.

It is also possible to save a system module at the specified file path.

Example

```
Load "ram1disk:PART_B.MOD";
...
Save "PART_B";
```

Load the program module with the file name PART_B.MOD from the *ram1disk* into the program memory.

Save the program module PART_B with the original file path *ram1disk* with the original file name PART_B.MOD.

Arguments

Save [\Task] ModuleName [\FilePath] [\File]

[\Task] Data type: *taskid*

The program task in which the program module should be saved.

If this argument is omitted, the specified program module in the current (executing) program task will be saved.

For all program tasks in the system, predefined variables of the data type *taskid* will be available. The variable identity will be "taskname"+"Id", e.g. for the MAIN task the variable identity will be MAINId, TSK1 - TSK1Id etc.

ModuleName Data type: *string*

The program module to save.

[\FilePath] Data type: *string*

The file path and the file name to the place where the program module is to be saved. The file name shall be excluded when the argument *\File* is used.

[\File] Data type: *string*

When the file name is excluded in the argument *FilePath*, it must be specified with this argument.

The argument *FilePath* can only be omitted for program modules loaded with *Load* or *StartLoad-WaitLoad* and the program module will be stored at the same destination as specified in these instructions. To store the program module at another destination it is also possible to use the argument *FilePath*.

To be able to save a program module that previously was loaded from the teach pendant, external computer, or system configuration, the argument *FilePath* must be used.

Program execution

Program execution waits for the program module to finish saving before proceeding with the next instruction.

Example

```
Save  "PART_A" \FilePath:="ram1disk:DOORDIR/PART_A.MOD";
```

Save the program module PART_A to the *ram1disk* in the file PART_A.MOD and in the directory DOORDIR.

```
Save  "PART_A" \FilePath:="ram1disk:" \File:="DOORDIR/PART_A.MOD";
```

Same as above but another syntax.

```
Save \Task:=TSK1Id, "PART_A" \FilePath:="ram1disk:DOORDIR/PART_A.MOD";
```

Save program module PART_A in program task TSK1 to the specified destination. This is an example where the instruction *Save* is executing in one program task and the saving is done in another program task.

Limitations

TRAP routines, system I/O events and other program tasks cannot execute during the saving operation. Therefore, any such operations will be delayed.

The save operation can interrupt update of PERS data done step by step from other program tasks. This will result in inconsistent whole PERS data.

A program stop during execution of the *Save* instruction can result in a guard stop with motors off and the error message "20025 Stop order timeout" will be displayed on the Teach Pendant.

Avoid ongoing robot movements during the saving.

Error handling

If the program module cannot be saved because there is no module name, unknown, or ambiguous module name, the system variable ERRNO is set to ERR_MODULE.

If the save file cannot be opened because of permission denied, no such directory, or no space left on device, the system variable ERRNO is set to ERR_IOERROR.

If argument *\FilePath* is not specified for program modules loaded from the Teach Pendant, System Parameters, or an external computer, the system variable ERRNO is set to ERR_PATH.

The errors above can be handled in the error handler.

Syntax

```
Save
   [ '\' Task ':=' <variable (VAR) of taskid> ',' ]
   [ ModuleName ':=' ] <expression (IN) of string>
   [ '\' FilePath ':='<expression (IN) of string> ]
   [ '\' File ':=' <expression (IN) of string>] ';'
```

Related information

	Described in:
Program tasks	Data Types - *taskid*

SearchC Searches circularly using the robot

SearchC (Search Circular) is used to search for a position when moving the tool centre point (TCP) circularly.

During the movement, the robot supervises a digital input signal. When the value of the signal changes to the requested one, the robot immediately reads the current position.

This instruction can typically be used when the tool held by the robot is a probe for surface detection. Using the *SearchC* instruction, the outline coordinates of a work object can be obtained.

Examples

SearchC sen1, sp, cirpoint, p10, v100, probe;

The TCP of the *probe* is moved circularly towards the position *p10* at a speed of *v100*. When the value of the signal *do1* changes to active, the position is stored in *sp*.

SearchC \Stop, sen1, sp, cirpoint, p10, v100, probe;

The TCP of the *probe* is moved circularly towards the position *p10*. When the value of the signal *do1* changes to active, the position is stored in *sp* and the robot stops immediately.

Arguments

SearchC [\Stop] | [\PStop] | [\SStop] | [\Sup] Signal [\Flanks] SearcPoint CirPoint ToPoint Speed [\V] | [\T] Tool [\WObj] [\Corr]

[\Stop] *(Stiff Stop)* Data type: *switch*

The robot movement is stopped, as quickly as possible, without keeping the TCP on the path (hard stop), when the value of the search signal changes to active. However, the robot is moved a small distance before it stops and is not moved back to the searched position, i.e. to the position where the signal changed.

[\PStop] *(Path Stop)* Data type: *switch*

The robot movement is stopped as quickly as possible, while keeping the TCP on the path (soft stop), when the value of the search signal changes to active. However, the robot is moved a distance before it stops and is not moved back to the searched position, i.e. to the position where the signal changed.

[\SStop] *(Smooth Stop)* Data type: *switch*

The robot movement is stopped as quickly as possible, while keeping the TCP close to or on the path (smooth stop), when the value of the search signal changes to active. However, the robot is moved only a small distance before it stops and is not moved back to the searched position, i.e. to the position where the signal changed. *SStop* is faster then *PStop*. But when the robot is running faster than 100 mm/s, it stops in the direction of the tangent of the movement which causes it to marginally slide of the path.

[\Sup] *(Supervision)* Data type: *switch*

The search instruction is sensitive to signal activation during the complete movement (flying search), i.e. even after the first signal change has been reported. If more than one match occurs during a search, program execution stops.

If the argument *\Stop*, *\PStop*, *\SStop or \Sup* is omitted, the movement continues (flying search) to the position specified in the *ToPoint* argument (same as with argument *\Sup*),

Signal Data type: *signaldi*

The name of the signal to supervise.

[\Flanks] Data type: *switch*

The positive and the negative edge of the signal is valid for a search hit.

If the argument *\Flanks* is omitted, only the positive edge of the signal is valid for a search hit and a signal supervision will be activated at the beginning of a search process. This means that if the signal has a positive value already at the beginning of a search process, the robot movement is stopped as quickly as possible, while keeping the TCP on the path (soft stop). However, the robot is moved a small distance before it stops and is not moved back to the start position. A user recovery error (ERR_SIGSUPSEARCH) will be generated and can be dealt with by the error handler.

SearchPoint Data type: *robtarget*

The position of the TCP and external axes when the search signal has been triggered. The position is specified in the outermost coordinate system, taking the specified tool, work object and active ProgDisp/ExtOffs coordinate system into consideration.

CirPoint Data type: *robtarget*

The circle point of the robot. See the instruction MoveC for a more detailed description of circular movement. The circle point is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction). *SearchC* always uses a stop point as zone data for the destination.

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the external axes and of the tool reorientation.

[\V] *(Velocity)* Data type: *num*

This argument is used to specify the velocity of the TCP in mm/s directly in the instruction. It is then substituted for the corresponding velocity specified in the speed data.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point that is moved to the specified destination position.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot positions in the instruction are related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified for a linear movement relative to the work object to be performed.

[\Corr] *(Correction)* Data type: *switch*

Correction data written to a corrections entry by the instruction *CorrWrite* will be added to the path and destination position, when this argument is present.

Program execution

See the instruction *MoveC* for information about circular movement.

The movement is always ended with a stop point, i.e. the robot is stopped at the destination point.

When a flying search is used, i.e. the *\Sup* argument is specified, the robot movement always continues to the programmed destination point. When a search is made using the switch *\Stop, \PStop* or *\SStop*, the robot movement stops when the first signal is detected.

The *SearchC* instruction returns the position of the TCP when the value of the digital signal changes to the requested one, as illustrated in Figure 15.



Figure 15 Flank-triggered signal detection (the position is stored when the signal is changed the first time only).

Example

SearchC \Sup, sen1\Flanks, sp, cirpoint, p10, v100, probe;

The TCP of the *probe* is moved circularly towards the position *p10*. When the value of the signal *do1* changes to active or passive, the position is stored in *sp*. If the value of the signal changes twice, program execution stops.

Limitations

Zone data for the positioning instruction that precedes *SearchC* must be used carefully. The start of the search, i.e. when the I/O signal is ready to react, is not, in this case, the programmed destination point of the previous positioning instruction, but a point along the real robot path. Figure 16 illustrates an example of something that may go wrong when zone data other than *fine* is used.

The instruction *SearchC* should never be restarted after the circle point has been passed. Otherwise the robot will not take the programmed path (positioning around the circular path in another direction compared with that programmed).



Figure 16 A match is made on the wrong side of the object because the wrong zone data was used.

Typical stop distance using a search velocity of 50 mm/s:

- without TCP on path (switch *Stop*) 1-3 mm
- with TCP on path (switch *PStop*) 12-16 mm
- with TCP near path (switch *SStop*) 7-10 mm

Error handling

An error is reported during a search when:

- no signal detection occurred - this generates the error ERR_WHLSEARCH.
- more than one signal detection occurred – this generates the error ERR_WHLSEARCH only if the *Sup* argument is used.
- the signal has already a positive value at the beginning of the search process - this generates the error ERR_SIGSUPSEARCH only if the *Flanks* argument is omitted.

Errors can be handled in different ways depending on the selected running mode:

Continuous forward / ERR_WHLSEARCH
No position is returned and the movement always continues to the programmed destination point. The system variable ERRNO is set to ERR_WHLSEARCH and the error can be handled in the error handler of the routine.

Continuous forward / **Instruction forward** / ERR_SIGSUPSEARCH
No position is returned and the movement always stops as quickly as possible at the beginning of the search path. The system variable ERRNO is set to ERR_SIGSUPSEARCH and the error can be handled in the error handler of the routine.

Instruction forward / ERR_WHLSEARCH
No position is returned and the movement always continues to the programmed destination point. Program execution stops with an error message.

Instruction backward
During backward execution, the instruction just carries out the movement without any signal supervision.

Syntax

```
SearchC
   [ '\' Stop',' ] | [ '\' PStop ',' ] | [ '\' SStop ',' ] | [ '\' Sup ',' ]
   [ Signal ':=' ] < variable (VAR) of signaldi >
                  ['\' Flanks]','
   [ SearchPoint ':=' ] < var or pers (INOUT) of robtarget > ','
   [ CirPoint ':=' ] < expression (IN) of robtarget > ','
   [ ToPoint ':=' ] < expression (IN) of robtarget > ','
   [ Speed ':=' ] < expression (IN) of speeddata >
                  [ '\' V ':=' < expression (IN) of num > ]
                  | [ '\' T ':=' < expression (IN) of num > ] ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ]
   [ '\' Corr ]';'
```

Related information

	Described in:
Linear searches	Instructions - *SearchL*
Writes to a corrections entry	Instructions - *CorrWrite*
Circular movement	Motion and I/O Principles - *Positioning during Program Execution*
Definition of velocity	Data Types - *speeddata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Using error handlers	RAPID Summary - *Error Recovery*
Motion in general	Motion and I/O Principles
More searching examples	Instructions - *SearchL*

SearchL Searches linearly using the robot

SearchL (Search Linear) is used to search for a position when moving the tool centre point (TCP) linearly.

During the movement, the robot supervises a digital input signal. When the value of the signal changes to the requested one, the robot immediately reads the current position.

This instruction can typically be used when the tool held by the robot is a probe for surface detection. Using the *SearchL* instruction, the outline coordinates of a work object can be obtained.

Examples

SearchL do1, sp, p10, v100, probe;

The TCP of the *probe* is moved linearly towards the position *p10* at a speed of *v100*. When the value of the signal *do1* changes to active, the position is stored in *sp*.

SearchL \Stop, sen1, sp, p10, v100, probe;

The TCP of the *probe* is moved linearly towards the position *p10*. When the value of the signal *do1*changes to active, the position is stored in *sp* and the robot stops immediately.

Arguments

SearchL [\Stop] | [\PStop] | [\SStop] | [\Sup] Signal [\Flanks] SearchPoint ToPoint Speed [\V] | [\T] Tool [\WObj] [\Corr]

[\Stop] *(Stiff Stop)* Data type: *switch*

The robot movement is stopped as quickly as possible, without keeping the TCP on the path (hard stop), when the value of the search signal changes to active. However, the robot is moved a small distance before it stops and is not moved back to the searched position, i.e. to the position where the signal changed.

[\PStop] *(Path Stop)* Data type: *switch*

The robot movement is stopped as quickly as possible, while keeping the TCP on the path (soft stop), when the value of the search signal changes to active. However, the robot is moved a distance before it stops and is not moved back to the searched position, i.e. to the position where the signal changed.

[\SStop] *(Smooth Stop)* Data type: *switch*

The robot movement is stopped as quickly as possible, while keeping the TCP close to or on the path (smooth stop), when the value of the search signal changes to active. However, the robot is moved only a small distance before it stops and is not moved back to the searched position, i.e. to the position where the signal changed. *SStop* is faster then *PStop*. But when the robot is running faster than 100 mm/s it stops in the direction of the tangent of the movement which causes it to marginally slide off the path.

[\Sup] *(Supervision)* Data type: *switch*

The search instruction is sensitive to signal activation during the complete movement (flying search), i.e. even after the first signal change has been reported. If more than one match occurs during a search, program execution stops.

If the argument *Stop*, *PStop*, *SStop* or *Sup* is omitted, the movement continues (flying search) to the position specified in the *ToPoint* argument (same as with argument *Sup*).

Signal Data type: *signaldi*

The name of the signal to supervise.

[\Flanks] Data type: *switch*

The positive and the negative edge of the signal is valid for a search hit.

If the argument *Flanks* is omitted, only the positive edge of the signal is valid for a search hit and a signal supervision will be activated at the beginning of a search process. This means that if the signal has the positive value already at the beginning of a search process, the robot movement is stopped as quickly as possible, while keeping the TCP on the path (soft stop). A user recovery error (ERR_SIGSUPSEARCH) will be generated and can be handled in the error handler.

SearchPoint Data type: *robtarget*

The position of the TCP and external axes when the search signal has been triggered. The position is specified in the outermost coordinate system, taking the specified tool, work object and active ProgDisp/ExtOffs coordinate system into consideration.

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction). *SearchL* always uses a stop point as zone data for the destination.

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the external axes and of the tool reorientation.

[\V] *(Velocity)* Data type: *num*

This argument is used to specify the velocity of the TCP in mm/s directly in the instruction. It is then substituted for the corresponding velocity specified in the speed data.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point that is moved to the specified destination position.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified for a linear movement relative to the work object to be performed.

[\Corr] *(Correction)* Data type: *switch*

Correction data written to a corrections entry by the instruction *CorrWrite* will be added to the path and destination position, if this argument is present.

Program execution

See the instruction *MoveL* for information about linear movement.

The movement always ends with a stop point, i.e. the robot stops at the destination point.

If a flying search is used, i.e. the *\Sup* argument is specified, the robot movement always continues to the programmed destination point. If a search is made using the switch *\Stop, \PStop or \SStop*, the robot movement stops when the first signal is detected.

The *SearchL* instruction stores the position of the TCP when the value of the digital signal changes to the requested one, as illustrated in Figure 17.



Figure 17 Flank-triggered signal detection (the position is stored when the signal is changed the first time only).

Examples

SearchL \Sup, sen1\Flanks, sp, p10, v100, probe;

The TCP of the *probe* is moved linearly towards the position *p10*. When the value of the signal *do1* changes to active or passive, the position is stored in *sp*. If the value of the signal changes twice, program execution stops after the search process is finished.

```
SearchL \Stop, sen1, sp, p10, v100, tool1;
MoveL sp, v100, fine, tool1;
PDispOn *, tool1;
MoveL p100, v100, z10, tool1;
MoveL p110, v100, z10, tool1;
MoveL p120, v100, z10, tool1;
PDispOff;
```

At the beginning of the search process, a check on the signal *do1* will be done and if the signal already has a positive value, the program execution stops.
Otherwise the TCP of *tool1* is moved linearly towards the position *p10*. When the value of the signal *do1* changes to active, the position is stored in *sp* and the robot is moved back to this point. Using program displacement, the robot then moves relative to the searched position, *sp*.

Limitations

Zone data for the positioning instruction that precedes *SearchL* must be used carefully. The start of the search, i.e. when the I/O signal is ready to react, is not, in this case, the programmed destination point of the previous positioning instruction, but a point along the real robot path. Figure 18 to Figure 20 illustrate examples of things that may go wrong when zone data other than *fine* is used.



Figure 18 A match is made on the wrong side of the object because the wrong zone data was used.



Figure 19 No match detected because the wrong zone data was used.



Figure 20 No match detected because the wrong zone data was used.

Typical stop distance using a search velocity of 50 mm/s:

- without TCP on path (switch *Stop*) 1-3 mm
- with TCP on path (switch *PStop*) 12-16 mm
- with TCP near path (switch *SStop*) 7-10 mm

Error handling

An error is reported during a search when:

- no signal detection occurred - this generates the error ERR_WHLSEARCH.
- more than one signal detection occurred – this generates the error ERR_WHLSEARCH only if the *Sup* argument is used.
- the signal already has a positive value at the beginning of the search process - this generates the error ERR_SIGSUPSEARCH only if the *Flanks* argument is omitted.

Errors can be handled in different ways depending on the selected running mode:

Continuous forward / ERR_WHLSEARCH
No position is returned and the movement always continues to the programmed destination point. The system variable ERRNO is set to ERR_WHLSEARCH and the error can be handled in the error handler of the routine.

Continuous forward / Instruction forward / ERR_SIGSUPSEARCH
No position is returned and the movement always stops as quickly as possible at the beginning of the search path.The system variable ERRNO is set to ERR_SIGSUPSEARCH and the error can be handled in the error handler of the routine.

Instruction forward / ERR_WHLSEARCH
No position is returned and the movement continues to the programmed destination point. Program execution stops with an error message.

Instruction backward
During backward execution, the instruction just carries out the movement without any signal supervision.

Example

```
VAR num fk;
.
MoveL p10, v100, fine, tool1;
SearchL \Stop, sen1, sp, p20, v100, tool1;
.
ERROR
   IF ERRNO=ERR_WHLSEARCH THEN
      MoveL p10, v100, fine, tool1;
      RETRY;
   ELSEIF ERRNO=ERR_SIGSUPSEARCH THEN
      TPWrite “The signal of the SearchL instruction is already high!”;
      TPReadFK fk,”Try again after manual reset of signal ?”,”YES”,””,””,””,”NO”;
      IF fk = 1 THEN
         MoveL p10, v100, fine, tool1;
         RETRY;
      ELSE
         Stop;
      ENDIF
   ENDIF
```

If the signal is already active at the beginning of the search process, a user dialog will be activated (TPReadFK ...;). Reset the signal and push YES on the user dialog and the robot moves back to p10 and tries once more. Otherwise program execution will stop.

If the signal is passive at the beginning of the search process, the robot searches from position *p10* to *p20*. If no signal detection occurs, the robot moves back to *p10* and tries once more.

Syntax

```
SearchL
   [ ’\’ Stop ’,’ ] | [ ’\’ PStop ’,’] | [ ’\’ SStop ’,’] | [ ’\’ Sup ’,’ ]
   [ Signal ’:=’ ] < variable (VAR) of signaldi >
                  [‘\’ Flanks] ’,’
   [ SearchPoint ’:=’ ] < var or pers (INOUT) of robtarget > ’,’
   [ ToPoint ’:=’ ] < expression (IN) of robtarget > ’,’
   [ Speed ’:=’ ] < expression (IN) of speeddata >
                  [ ’\’ V ’:=’ < expression (IN) of num > ]
                  | [ ’\’ T ’:=’ < expression (IN) of num > ] ’,’
   [ Tool ’:=’ ] < persistent (PERS) of tooldata >
   [ ’\’ WObj ’:=’ < persistent (PERS) of wobjdata > ]
   [ ’\’ Corr ]’;’
```

Related information

	Described in:
Circular searches	Instructions - *SearchC*
Writes to a corrections entry	Instructions - *CorrWrite*
Linear movement	Motion and I/O Principles - *Positioning during Program Execution*
Definition of velocity	Data Types - *speeddata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Using error handlers	RAPID Summary - *Error Recovery*
Motion in general	Motion and I/O Principles

Set Sets a digital output signal

Set is used to set the value of a digital output signal to one.

Examples

Set do15;

The signal *do15* is set to 1.

Set weldon;

The signal *weldon* is set to 1.

Arguments

Set Signal

Signal Data type: *signaldo*

The name of the signal to be set to one.

Program execution

The true value depends on the configuration of the signal. If the signal is inverted in the system parameters, this instruction causes the physical channel to be set to zero.

Syntax

Set
[Signal ':='] < variable (**VAR**) of *signaldo* > ';'

Related information

	Described in:
Setting a digital output signal to zero	Instructions - *Reset*
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	System Parameters

SetAO Changes the value of an analog output signal

SetAO is used to change the value of an analog output signal.

Example

SetAO ao2, 5.5;

The signal *ao2* is set to *5.5*.

Arguments

SetAO Signal Value

Signal Data type: *signalao*

The name of the analog output signal to be changed.

Value Data type: *num*

The desired value of the signal.

Program execution

The programmed value is scaled (in accordance with the system parameters) before it is sent on the physical channel. See Figure 21.



Figure 21 Diagram of how analog signal values are scaled.

Example

SetAO weldcurr, curr_outp;

The signal *weldcurr* is set to the same value as the current value of the variable *curr_outp*.

Syntax

SetAO
[Signal ':='] < variable (**VAR**) of *signalao* > ','
[Value ':='] < expression (**IN**) of *num* > ';'

Related information

	Described in:
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	System Parameters

SetDO Changes the value of a digital output signal

SetDO is used to change the value of a digital output signal, with or without a time-delay.

Examples

SetDO do15, 1;

The signal *do15* is set to *1*.

SetDO weld, off;

The signal *weld* is set to *off*.

SetDO \SDelay := 0.2, weld, high;

The signal *weld* is set to *high* with a delay of *0.2* s. Program execution, however, continues with the next instruction.

Arguments

SetDO [\SDelay] Signal Value

[\SDelay] *(Signal Delay)* Data type: *num*

Delays the change for the amount of time given in seconds (0.1 - 32s). Program execution continues directly with the next instruction. After the given time-delay, the signal is changed without the rest of the program execution being affected.

If the argument is omitted, the value of the signal is changed directly.

Signal Data type: *signaldo*

The name of the signal to be changed.

Value Data type: *dionum*

The desired value of the signal.

The value is specified as 0 or 1.

Program execution

The true value depends on the configuration of the signal. If the signal is inverted in the system parameters, the value of the physical channel is the opposite.

Syntax

SetDO
[’\’ SDelay ’:=’ < expression (**IN**) of *num* > ’,’]
[Signal ’:=’] < variable (**VAR**) of *signaldo* > ’,’
[Value ’:=’] < expression (**IN**) of *dionum* > ’;’

Related information

	Described in:
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User’s Guide - *System Parameters*

SetGO Changes the value of a group of digital output signals

SetGO is used to change the value of a group of digital output signals, with or without a time delay.

Example

SetGO go2, 12;

The signal *go2* is set to *12*. If *go2* comprises 4 signals, e.g. outputs 6-9, outputs 6 and 7 are set to zero, while outputs 8 and 9 are set to one.

SetGO \SDelay := 0.4, go2, 10;

The signal *go2* is set to *10*. If *go2* comprises 4 signals, e.g. outputs 6-9, outputs 6 and 8 are set to zero, while outputs 7 and 9 are set to one, with a delay of *0.4* s. Program execution, however, continues with the next instruction.

Arguments

SetGO [\SDelay] Signal Value

[\SDelay] *(Signal Delay)* Data type: *num*

Delays the change for the period of time stated in seconds (0.1 - 32s). Program execution continues directly with the next instruction. After the specified time delay, the value of the signals is changed without the rest of the program execution being affected.

If the argument is omitted, the value is changed directly.

Signal Data type: *signalgo*

The name of the signal group to be changed.

Value Data type: *num*

The desired value of the signal group (a positive integer).

The permitted value is dependent on the number of signals in the group:

No. of signals	Permitted value	No. of signals	Permitted value
1	0 - 1	9	0 - 511
2	0 - 3	10	0 - 1023
3	0 - 7	11	0 - 2047
4	0 - 15	12	0 - 4095
5	0 - 31	13	0 - 8191
6	0 - 63	14	0 - 16383
7	0 - 127	15	0 - 32767
8	0 - 255	16	0 - 65535

Program execution

The programmed value is converted to an unsigned binary number. This binary number is sent on the signal group, with the result that individual signals in the group are set to 0 or 1. Due to internal delays, the value of the signal may be undefined for a short period of time.

Syntax

SetDO
[’\’ SDelay ’:=’ < expression (**IN**) of *num* > ’,’]
[Signal ’:=’] < variable (**VAR**) of *signalgo* > ’,’
[Value ’:=’] < expression (**IN**) of *num* > ’;’

Related information

	Described in:
Other input/output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O (system parameters)	System Parameters

SingArea Defines interpolation around singular points

SingArea is used to define how the robot is to move in the proximity of singular points.

SingArea is also used to define linear and circular interpolation for robots with less than six axes.

Examples

SingArea \Wrist;

The orientation of the tool may be changed slightly in order to pass a singular point (axes 4 and 6 in line).

Robots with less than six axes may not be able to reach an interpolated tool orientation. By using SingArea \Wrist, the robot can achieve the movement but the orientation of the tool will be slightly changed.

SingArea \Off;

The tool orientation is not allowed to differ from the programmed orientation. If a singular point is passed, one or more axes may perform a sweeping movement, resulting in a reduction in velocity.

Robots with less than six axes may not be able to reach a programmed tool orientation. As a result the robot will stop.

Arguments

SingArea [\Wrist] | [\Off]

[\Wrist] Data type: *switch*

The tool orientation is allowed to differ somewhat in order to avoid wrist singularity. Used when axes 4 and 6 are parallel (axis 5 at 0 degrees). Also used for linear and circular interpolation of robots with less than six axes where the tool orientation is allowed to differ.

[\Off] Data type: *switch*

The tool orientation is not allowed to differ. Used when no singular points are passed, or when the orientation is not permitted to be changed.

If none of the arguments are specified, program execution automatically uses the robot's default argument. For robots with six axes the default argument is *\Off*.

Program execution

If the arguments *\Wrist* is specified, the orientation is joint-interpolated to avoid singular points. In this way, the TCP follows the correct path, but the orientation of the tool deviates somewhat. This will also happen when a singular point is not passed.

The specified interpolation applies to all subsequent movements until a new *SingArea* instruction is executed.

The movement is only affected on execution of linear or circular interpolation.

By default, program execution automatically uses the */Off* argument for robots with six axes. Robots with less than six axes may use either the */Off* argument (IRB640) or the */Wrist* argument by default. This is automatically set in event routine SYS_RESET.

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Syntax

```
SingArea
   [ '\' Wrist ] | [ '\' Off ] ';'
```

Related information

	Described in:
Singularity	Motion Principles- *Singularity*
Interpolation	Motion Principles - *Positioning during Program Execution*

SoftAct Activating the soft servo

SoftAct (Soft Servo Activate) is used to activate the so called “soft” servo on any axis of the robot or external mechanical unit.

Example

SoftAct 3, 20;

Activation of soft servo on robot axis *3*, with softness value *20*%.

SoftAct 1, 90 \Ramp:=150;

Activation of the soft servo on robot axis *1*, with softness value *90*% and ramp factor *150*%.

SoftAct \MechUnit:=orbit1, 1, 40 \Ramp:=120;

Activation of soft servo on axis *1* for the mechanical unit *orbit1*, with softness value *40*% and ramp factor *120*%.

Arguments

SoftAct [\MechUnit] Axis Softness [\Ramp]

[\MechUnit] *(Mechanical Unit* Data type: *mecunit*

The name of the mechanical unit. If this argument is omitted, it means activation of the soft servo for specified robot axis.

Axis Data type: *num*

Number of the robot or external axis to work with soft servo.

Softness Data type: *num*

Softness value in percent (0 - 100%). 0% denotes min. softness (max. stiffness), and 100% denotes max. softness.

Ramp Data type: *num*

Ramp factor in percent (>= 100%). The ramp factor is used to control the engagement of the soft servo. A factor 100% denotes the normal value; with greater values the soft servo is engaged more slowly (longer ramp). The default value for ramp factor is 100 %.

Program execution

Softness is activated at the value specified for the current axis. The softness value is valid for all movements, until a new softness value is programmed for the current axis, or until the soft servo is deactivated by an instruction.

Limitations

Soft servo for any robot or external axis is always deactivated when there is a power failure. This limitation can be handled in the user program when restarting after a power failure.

The same axis must not be activated twice, unless there is a moving instruction in between. Thus, the following program sequence should be avoided, otherwise there will be a jerk in the robot movement:

```
SoftAct n , x ;
SoftAct n , y ;
```

(n = robot axis n, x and y softness values)

Syntax

SoftAct
[’\’MechUnit ’:=’ < variable (**VAR**) of *mecunit*> ’,’]
[Axis ’:=’] < expression (**IN**) of *num*> ’,’
[Softness ’:=’] < expression (**IN**) of *num*>
[’\’Ramp ’:=’ < expression (**IN**) of *num*>]’;’

Related information

	Described in:
Behaviour with the soft servo engaged	Motion and I/O Principles- *Positioning during program execution*

SoftDeact Deactivating the soft servo

SoftDeact (Soft Servo Deactivate) is used to deactivate the so called “soft” servo on all robot and external axes.

Example

SoftDeact;

Deactivating the soft servo on all axes.

SoftDeact \Ramp:=150;

Deactivating the soft servo on all axes, with ramp factor 150%.

Arguments

SoftDeact [\Ramp]

Ramp Data type: *num*

Ramp factor in percent (>= 100%). The ramp factor is used to control the deactivating of the soft servo. A factor 100% denotes the normal value; with greater values the soft servo is deactivated more slowly (longer ramp). The default value for ramp factor is 100 %.

Program execution

The soft servo is deactivated for all robot and external axes.

Syntax

SoftDeact
[’\’Ramp ’:=’ < expression (**IN**) of *num*>]’;’

Related information

	Described in:
Activating the soft servo	Instructions - *SoftAct*

SpyStart Start recording of execution time data

SpyStart is used to start the recording of instruction and time data during execution.

The execution data will be stored in a file for later analysis.

The stored data is intended for debugging RAPID programs, specifically for multi-tasking systems (only necessary to have *SpyStart* - *SpyStop* in one program task).

Example

SpyStart "ram1disk:spy.log";

Starts recording the execution time data in the file *spy.log* on the ramdisk.

Arguments

SpyStart File

File Data type: *string*

The file path and the file name to the file that will contain the execution data.

Program execution

The specified file is opened for writing and the execution time data begins to be recorded in the file.

Recording of execution time data is active until:

- execution of instruction SpyStop
- starting program execution from the beginning
- loading a new program
- next warm start-up

Limitations

Avoid using the floppy disk for recording since writing to the floppy is very time consuming.

Never use the spy function in production programs because the function increases the cycle time and consumes memory on the mass memory device (ramdisk) in use.

Error handling

If the file in the *SpyStart* instruction can't be opened then the system variable ERRNO is set to ERR_FILEOPEN (see "Data types - errnum"). This error can then be handled in the error handler.

File format

```
TASK            INSTR                                   IN       CODE  OUT

MAIN ConfJ\Off;                                         691310:READY :691310
----- SYSTEM TRAP-----
MAIN ConfL\Off;                                         691450:READY :691450
MAIN def_wz;                                            691450:READY :691450
MAIN WZSphDef\Inside,volume,[p1.trans.x+xtrans          691450:READY :691450
MAIN WZDOSet\Temp,wz1\Inside,volume,do1,1;               691460:READY :691460
----- SYSTEM TRAP-----
MAIN WZSphDef\Inside,volume,[p2.trans.x+xtrans,          691460:READY :691460
MAIN WZDOSet\Temp,wz2\Inside,volume,do2,1;               691860:READY :691860
...
MAIN MoveL home,s,z,toolx\WObj:=wobjx;                  693910:WAIT :694010
----- SYSTEM TRAP -----
MAIN MoveL home,s,z,toolx\WObj:=wobjx;                  726820:WAIT :726820
----- SYSTEM TRAP -----
MAIN MoveL home,s,z,toolx\WObj:=wobjx;                  740300:READY :740300
MAIN writepos;                                          740300:READY :740300
...
MAIN SpyStop;                                           827610:
```

TASK column shows executed program task
INSTR column shows executed instruction in specified program task
IN column shows the time in ms at enter of the executed instruction
CODE column shows if the instruction is READY or
if the instruction WAIT for completion at **OU**T time
OUT column shows the time in ms at leave of the executed instruction

All time is given in ms (relative values) with resolution 10 ms.

----- SYSTEM TRAP----- means that the system is doing something else than execution of RAPID instructions.

If procedure call to some NOSTEPIN procedure (module) the output list shows only the name of the called procedure. This is repeated for every executed instruction in the NOSTEPIN routine.

Syntax

SpyStart
[File':=']<expression (**IN**) of *string*>';'

Related information

	Described in:
Stop recording of execution data	Instructions - *SpyStop*

SpyStop Stop recording of time execution data

SpyStop is used to stop the recording of time data during execution.

The data, which can be useful for optimising the execution cycle time, is stored in a file for later analysis.

Example

SpyStop;

Stops recording the execution time data in the file specified by the previous *SpyStart* instruction.

Program execution

The execution data recording is stopped and the file specified by the previous *SpyStart* instruction is closed.
If no *SpyStart* instruction has been executed before, the *SpyStop* instruction is ignored.

Examples

```
IF debug = TRUE SpyStart "ram1disk:spy.log";
produce_sheets;
IF debug = TRUE SpyStop;
```

If the debug flag is true, start recording execution data in the file *spy.log* on the ramdisk; perform actual production; stop recording and close the file *spy.log*.

Limitations

Avoid using the floppy disk for recording since writing to the floppy is very time consuming.

Never use the spy function in production programs because the function increases the cycle time and consumes memory on the mass memory device (ramdisk) in use.

Syntax

SpyStop’;’

Related information

	Described in:
Start recording of execution data	Instructions - *SpyStart*

StartLoad Load a program module during execution

The loaded program module will be added to the modules already existing in the program memory.

A program or system module can be loaded in static (default) or dynamic mode:

Static mode

Table 3 How different operations affect static loaded program or system modules

	Set PP to main from TP	Open new RAPID program
Program Module	Not affected	Unloaded
System Module	Not affected	Not affected

Dynamic mode

Table 4 How different operations affect dynamic loaded program or system modules

	Set PP to main from TP	Open new RAPID program
Program Module	Unloaded	Unloaded
System Module	Unloaded	Unloaded

Both static and dynamic loaded modules can be unloaded by the instruction *UnLoad*.

Example

```
VAR loadsession load1;

! Start loading of new program module PART_B containing routine routine_b
! in dynamic mode
StartLoad \Dynamic, ram1disk \File:="PART_B.MOD", load1;

! Executing in parallel in old module PART_A containing routine_a
%"routine_a"%;

! Unload of old program module PART_A
UnLoad ram1disk \File:="PART_A.MOD";

! Wait until loading and linking of new program module PART_B is ready
WaitLoad load1;


! Execution in new program module PART_B
%"routine_b"%;
```

Starts the loading of program module *PART_B.MOD* from *ram1disk* into the program memory with instruction *StartLoad*. In parallel with the loading, the program executes *routine_a* in module PART_A.MOD. Then instruction *WaitLoad* waits until the loading and linking is finished. The module is loaded in dynamic mode.

Variable *load1* holds the identity of the load session, updated by *StartLoad* and referenced by *WaitLoad*.

To save linking time, the instruction *UnLoad* and *WaitLoad* can be combined in the instruction *WaitLoad* by using the option argument *\UnLoadPath*.

Arguments

StartLoad [\Dynamic] FilePath [\File] LoadNo

[\Dynamic] Data type: *switch*

The switch enables load of a program module in dynamic mode. Otherwise the loading is in static mode.

FilePath Data type: *string*

The file path and the file name to the file that will be loaded into the program memory. The file name shall be excluded when the argument *\File* is used.

[\File] Data type: *string*

When the file name is excluded in the argument *FilePath*, then it must be defined with this argument.

LoadNo Data type: *loadsession*

This is a reference to the load session that should be used in the instruction *WaitLoad* to connect the loaded program module to the program task.

Program execution

Execution of *StartLoad* will only order the loading and then proceed directly with the next instruction, without waiting for the loading to be completed.

The instruction *WaitLoad* will then wait at first for the loading to be completed, if it is not already finished, and then it will be linked and initialised. The initialisation of the loaded module sets all variables at module level to their init values.

Unsolved references will be accepted if the system parameter for *Tasks/BindRef* is set to NO. However, when the program is started or the teach pendant function *Program Window/File/Check Program* is used, no check for unsolved references will be made if *BindRef* = NO. There will be a run time error on execution of an unsolved reference.

Another way to use references to instructions, that are not in the task from the beginning, is to use *Late Binding*. This makes it possible to specify the routine to call with a string expression, quoted between two %%. In this case the *BindRef* parameter could be set to YES (default behaviour). The *Late Binding* way is preferable.

To obtain a good program structure, that is easy to understand and maintain, all loading and unloading of program modules should be done from the main module, which is always present in the program memory during execution.

Examples

```
StartLoad \Dynamic, "ram1disk:DOORDIR/DOOR1.MOD", load1;
```

Loads the program module *DOOR1.MOD* from the *ram1disk* at the directory *DOORDIR* into the program memory. The program module is loaded in dynamic mode.

```
StartLoad \Dynamic, "ram1disk:" \File:="DOORDIR/DOOR1.MOD", load1;
```

Same as above but with another syntax.

```
StartLoad "ram1disk:" \File:="DOORDIR/DOOR1.MOD", load1;
```

Same as the two examples above but the module is loaded in static mode.

```
StartLoad \Dynamic, "ram1disk:" \File:="DOORDIR/DOOR1.MOD", load1;
...
WaitLoad load1;
```

is the same as

```
Load \Dynamic, "ram1disk:" \File:="DOORDIR/DOOR1.MOD";
```

Limitations

It is not allowed to load a system module or a program module that contains a main routine.

Syntax

```
StartLoad
   ['\'Dynamic ',']
   [FilePath ':='] <expression (IN) of string>
   ['\'File ':=' <expression (IN) of string> ] ','
   [LoadNo ':='] <variable (VAR) of loadsession> ';'
```

Related information

	Described in:
Connect the loaded module to the task	Instructions - *WaitLoad*
Load session	Data Types - *loadsession*
Load a program module	Instructions - *Load*
Unload a program module	Instructions - *UnLoad*
Accept unsolved references	System Parameters - *Controller/Task/ BindRef*

StartMove Restarts robot motion

StartMove is used to resume robot and external axes motion when this has been stopped by the instruction *StopMove*.

Example

```
StopMove;
WaitDI ready_input, 1;
StartMove;
```

The robot starts to move again when the input *ready_input* is set.

Program execution

Any processes associated with the stopped movement are restarted at the same time as motion resumes.

Error handling

If the robot is too far from the path (more than 10 mm or 20 degrees) to perform a start of the interrupted movement, the system variable *ERRNO* is set to ERR_PATHDIST. This error can then be handled in the error handler.

Syntax

StartMove’;’

Related information

	Described in:
Stopping movements	Instructions - *StopMove*
More examples	Instructions - *StorePath*

Stop Stops program execution

Stop is used to temporarily stop program execution.

Program execution can also be stopped using the instruction *EXIT*. This, however, should only be done if a task is complete, or if a fatal error occurs, since program execution cannot be restarted with *EXIT*.

Example

```
TPWrite "The line to the host computer is broken";
Stop;
```

Program execution stops after a message has been written on the teach pendant.

Arguments

Stop [\NoRegain]

[\NoRegain] Data type: *switch*

Specifies for the next program start in manual mode, whether or not the robot and external axes should regain to the stop position. In automatic mode the robot and external axes always regain to the stop position.

If the argument *NoRegain* is set, the robot and external axes will not regain to the stop position (if they have been jogged away from it).

If the argument is omitted and if the robot or external axes have been jogged away from the stop position, the robot displays a question on the teach pendant. The user can then answer, whether or not the robot should regain to the stop position.

Program execution

The instruction stops program execution as soon as the robot and external axes reach the programmed destination point for the movement it is performing at the time. Program execution can then be restarted from the next instruction.

If there is a *Stop* instruction in some event routine, the routine will be executed from the beginning in the next event.

Example

```
MoveL p1, v500, fine, tool1;
TPWrite “Jog the robot to the position for pallet corner 1”;
Stop \NoRegain;
p1_read := CRobT();
MoveL p2, v500, z50, tool1;
```

Program execution stops with the robot at *p1*. The operator jogs the robot to *p1_read*. For the next program start, the robot does not regain to *p1*, so the position *p1_read* can be stored in the program.

Limitations

The movement instruction which precedes this instruction should be terminated with a stop point, in order to be able to restart in this instruction following a power failure.

Syntax

```
Stop
[ ’\’ NoRegain ]’;’
```

Related information

	Described in:
Stopping after a fatal error	Instructions - *EXIT*
Terminating program execution	Instructions - *EXIT*
Only stopping robot movements	Instructions - *StopMove*

StopMove Stops robot motion

StopMove is used to stop robot and external axes movements temporarily. If the instruction *StartMove* is given, movement resumes.

This instruction can, for example, be used in a trap routine to stop the robot temporarily when an interrupt occurs.

Example

```
StopMove;
WaitDI ready_input, 1;
StartMove;
```

The robot movement is stopped until the input, *ready_input*, is set.

Program execution

The movements of the robot and external axes stop without the brakes being engaged. Any processes associated with the movement in progress are stopped at the same time as the movement is stopped.

Program execution continues without waiting for the robot and external axes to stop (standing still).

Examples

```
VAR intnum intno1;
...
CONNECT intno1 WITH go_to_home_pos;
ISignalDI di1,1,intno1;

TRAP go_to_home_pos
   VAR robtarget p10;

   StopMove;
   StorePath;
   p10:=CRobT();
   MoveL home,v500,fine,tool1;
   WaitDI di1,0;
   Move L p10,v500,fine,tool1;
   RestoPath;
   StartMove;
ENDTRAP
```

When the input *di1* is set to 1, an interrupt is activated which in turn activates the interrupt routine *go_to_home_pos*. The current movement is stopped immediately and the robot moves instead to the *home* position. When *di1* is set to 0, the robot returns to the position at which the interrupt occurred and continues to move along the programmed path.

```
VAR intnum intno1;
...
CONNECT intno1 WITH go_to_home_pos;
ISignalDI di1,1,intno1;

TRAP go_to_home_pos ()
   VAR robtarget p10;

   StorePath;
   p10:=CRobT();
   MoveL home,v500,fine,tool1;
   WaitDI di1,0;
   Move L p10,v500,fine,tool1;
   RestoPath;
   StartMove;
ENDTRAP
```

Similar to the previous example, but the robot does not move to the *home* position until the current movement instruction is finished.

Syntax

StopMove';'

Related information

	Described in:
Continuing a movement	Instructions - *StartMove*
Interrupts	RAPID Summary - *Interrupts* Basic Characteristics- *Interrupts*

StorePath Stores the path when an interrupt occurs

StorePath is used to store the movement path being executed when an error or interrupt occurs. The error handler or trap routine can then start a new movement and, following this, restart the movement that was stored earlier.

This instruction can be used to go to a service position or to clean the gun, for example, when an error occurs.

Example

```
StorePath;
```

The current movement path is stored for later use.

Program execution

The current movement path of the robot and external axes is saved. After this, another movement can be started in a trap routine or an error handler. When the reason for the error or interrupt has been rectified, the saved movement path can be restarted.

Example

```
TRAP machine_ready
   VAR robtarget p1;
   StorePath;
   p1 := CRobT();
   MoveL p100, v100, fine, tool1;
   ...
   MoveL p1, v100, fine, tool1;
   RestoPath;
   StartMove;
ENDTRAP
```

When an interrupt occurs that activates the trap routine *machine_ready*, the movement path which the robot is executing at the time is stopped at the end of the instruction (ToPoint) and stored. After this, the robot remedies the interrupt by, for example, replacing a part in the machine and the normal movement is restarted.

Limitations

Only the movement path data is stored with the instruction *StorePath*.
If the user wants to order movements on the new path level, the actual stop position must be stored directly after *StorePath* and before *RestoPath* make a movement to the stored stop position on the path.

Only one movement path can be stored at a time.

Syntax

StorePath‘;’

Related information

	Described in:
Restoring a path	Instructions - *RestoPath*
More examples	Instructions - *RestoPath*

TEST Depending on the value of an expression ...

TEST is used when different instructions are to be executed depending on the value of an expression or data.

If there are not too many alternatives, the *IF..ELSE* instruction can also be used.

Example

```
TEST reg1
CASE 1,2,3 :
   routine1;
CASE 4 :
   routine2;
DEFAULT :
   TPWrite "Illegal choice";
   Stop;
ENDTEST
```

Different instructions are executed depending on the value of *reg1*. If the value is 1-3 *routine1* is executed. If the value is 4, *routine2* is executed. Otherwise, an error message is printed and execution stops.

Arguments

TEST Test data {CASE Test value {, Test value} : ...}
[DEFAULT: ...] ENDTEST

Test data Data type: All

The data or expression with which the test value will be compared.

Test value Data type: Same as test data

The value which the test data must have for the associated instructions to be executed.

Program execution

The test data is compared with the test values in the first CASE condition. If the comparison is true, the associated instructions are executed. After that, program execution continues with the instruction following ENDTEST.

If the first CASE condition is not satisfied, other CASE conditions are tested, and so on. If none of the conditions are satisfied, the instructions associated with DEFAULT are executed (if this is present).

Syntax

(EBNF)
TEST <expression>
{(**CASE** <test value> { ’,’ <test value> } ’:’
<instruction list>) | **<CSE>** }
[**DEFAULT** ’:’ <instruction list>]
ENDTEST

<test value> ::= <expression>

Related information

	Described in:
Expressions	Basic Characteristics - *Expressions*

TestSign Output of test signals

TestSign is used when output of test signals from the robot system is needed. A test signals mirror the resolver angle for an axis, for example. The test signal is an analogue output to one of the two test output connection points.

The test signals for the master robot cannot be reached.

Example

TestSign 1, resolver_angle, Orbit, 2, 4, 0;

Test signal connection point 1 will give the value of the resolver_angle for Orbit axis 2, scaled 4 times the nominal value and sampled at maximum rate (indicated by 0).

Arguments

TestSign Output SignalId MechUnit Axis Scale Stime

Output Data type: *num*

This argument specifies which of the two test signal output connector points that the test signal will be available at. Possible values are 1 or 2.

SignalId Data type: *testsignal*

The name of the test signal to send to the output.

MechUnit *(Mechanical Unit)* Data type: *mecunit*

Argument that holds the name of the mechanical unit to retrieve test signal from.

Axis Data type: *num*

Argument that holds the axis number of the mechanical unit to retrieve test signal from.

Scale Data type: *num*

The scaling factor for output. Possible values are 1, 2, 4, 8, 16, and so on.

Stime *(Sample Time)* Data type: *num*

Sample time in seconds. The output signal is updated with a new value at every sample. A value of 1 updates the output every second and a value of 0 performs an update as often as possible. The value 0.01 corresponds to a sampling of 100 times per second.

Program execution

The TestSign instruction mirrors a signal in the robot system to an output. The mirroring of a specified test signal is active until a new TestSign instruction for the output is performed. A warm start of the robot system removes currently activated test signals.

Error handling

If there is an error in the parameter MechUnit, the system parameter ERRNO is set to ERR_UNIT_PAR. If there is an error in the parameter Axis, ERRNO is set to ERR_AXIS_PAR.

Syntax

TestSign
[Output ':='] < expression (**IN**) of *num*> ','
[SignalId ':='] < expression (**IN**) of *testsignal*>
[MechUnit ':='] < variable (**VAR**) of mecunit> ','
[Axis ':='] < expression (**IN**) of *num*> ','
[Scale ':='] < expression (**IN**) of *num* > ','
[Stime ':='] < expression (**IN**) of *num*> ';'

TPErase Erases text printed on the teach pendant

TPErase (Teach Pendant Erase) is used to clear the display of the teach pendant.

Example

```
TPErase;
TPWrite "Execution started";
```

The teach pendant display is cleared before *Execution started* is written.

Program execution

The teach pendant display is completely cleared of all text. The next time text is written, it will be entered on the uppermost line of the display.

Syntax

```
TPErase;
```

Related information

	Described in:
Writing on the teach pendant	RAPID Summary - *Communication*

TPReadFK Reads function keys

TPReadFK (Teach Pendant Read Function Key) is used to write text above the functions keys and to find out which key is depressed.

Example

TPReadFK reg1, “More ?”, stEmpty, stEmpty, stEmpty, “Yes”, “No”;

The text *More ?* is written on the teach pendant display and the function keys 4 and 5 are activated by means of the text strings *Yes* and *No* respectively (see Figure 22). Program execution waits until one of the function keys 4 or 5 is pressed. In other words, *reg1* will be assigned 4 or 5 depending on which of the keys is depressed.



Figure 22 The operator can input information via the function keys.

Arguments

TPReadFK Answer Text FK1 FK2 FK3 FK4 FK5 [\MaxTime] [\DIBreak] [\BreakFlag]

Answer Data type: *num*

The variable for which, depending on which key is pressed, the numeric value 1..5 is returned. If the function key 1 is pressed, 1 is returned, and so on.

Text Data type: *string*

The information text to be written on the display (a maximum of 80 characters).

FKx *(Function key text)* Data type: *string*

The text to be written as a prompt for the appropriate function key (a maximum of 7 characters). FK1 is the left-most key.

Function keys without prompts are specified by the predefined string constant *stEmpty* with value empty string (“”).

[\MaxTime] Data type: *num*

The maximum amount of time [s] that program execution waits. If no function key is depressed within this time, the program continues to execute in the error handler unless the BreakFlag is used (see below). The constant ERR_TP_MAXTIME can be used to test whether or not the maximum time has elapsed.

[\DIBreak] *(Digital Input Break)* Data type: *signaldi*

The digital signal that may interrupt the operator dialog. If no function key is depressed when the signal is set to 1 (or is already 1), the program continues to execute in the error handler, unless the BreakFlag is used (see below). The constant ERR_TP_DIBREAK can be used to test whether or not this has occurred.

[\BreakFlag] Data type: *errnum*

A variable that will hold the error code if maxtime or dibreak is used. If this optional variable is omitted, the error handler will be executed. The constants ERR_TP_MAXTIME and ERR_TP_ DIBREAK can be used to select the reason.

Program execution

The information text is always written on a new line. If the display is full of text, this body of text is moved up one line first. Strings longer than the width of the teach pendant (40 characters) are split into two lines.

Prompts are written above the appropriate function keys. Keys without prompts are deactivated.

Program execution waits until one of the activated function keys is depressed.

Description of concurrent *TPReadFK* or *TPReadNum* request on Teach Pendant (TP request) from same or other program tasks:

- New TP request from other program task will not take focus (new put in queue)
- New TP request from TRAP in the same program task will take focus (old put in queue)
- Program stop take focus (old put in queue)
- New TP request in program stop state takes focus (old put in queue)

Example

```
VAR errnum errvar;
...
TPReadFK reg1, “Go to service position?”, stEmpty, stEmpty, stEmpty, “Yes”, “No”
\MaxTime:= 600
   \DIBreak:= di5\BreakFlag:= errvar;
IF reg1 = 4 or OR errvar = ERR_TP_DIBREAK THEN
   MoveL service, v500, fine, tool1;
   Stop;
ENDIF
IF errvar = ERR_TP_MAXTIME EXIT;
```

The robot is moved to the service position if the forth function key (“Yes”) is pressed, or if the input 5 is activated. If no answer is given within 10 minutes, the execution is terminated.

Predefined data

```
CONST string stEmpty := “”;
```

The predefined constant *stEmpty* should be used for Function Keys without prompts. Using *stEmpty* instead of “”saves about 80 bytes for every Function Key without prompts.

Syntax

```
TPReadFK
   [Answer’:=’] <var or pers (INOUT) of num>’,’
   [Text’:=’] <expression (IN) of string>’,’
   [FK1 ’:=’] <expression (IN) of string>’,’
   [FK2 ’:=’] <expression (IN) of string>’,’
   [FK3 ’:=’] <expression (IN) of string>’,’
   [FK4 ’:=’] <expression (IN) of string>’,’
   [FK5 ’:=’] <expression (IN) of string>
   [’\’MaxTime ’:=’ <expression (IN) of num>]
   [’\’DIBreak ’:=’ <variable (VAR) of signaldi>]
   [’\’BreakFlag ’:=’ <var or pers (INOUT) of errnum>]’;’
```

Related information

	Described in:
Writing to and reading from the teach pendant	RAPID Summary - *Communication*
Replying via the teach pendant	Running Production

TPReadNum Reads a number from the teach pendant

TPReadNum (Teach Pendant Read Numerical) is used to read a number from the teach pendant.

Example

TPReadNum reg1, “How many units should be produced?“;

The text *How many units should be produced?* is written on the teach pendant display. Program execution waits until a number has been input from the numeric keyboard on the teach pendant. That number is stored in *reg1*.

Arguments

TPReadNum Answer String [\MaxTime] [\DIBreak] [\BreakFlag]

Answer Data type: *num*

The variable for which the number input via the teach pendant is returned.

String Data type: *string*

The information text to be written on the teach pendant (a maximum of 80 characters).

[\MaxTime] Data type: *num*

The maximum amount of time that program execution waits. If no number is input within this time, the program continues to execute in the error handler unless the BreakFlag is used (see below). The constant ERR_TP_MAXTIME can be used to test whether or not the maximum time has elapsed.

[\DIBreak] *(Digital Input Break)* Data type: *signaldi*

The digital signal that may interrupt the operator dialog. If no number is input when the signal is set to 1 (or is already 1), the program continues to execute in the error handler unless the BreakFlag is used (see below). The constant ERR_TP_DIBREAK can be used to test whether or not this has occurred.

[\BreakFlag] Data type: *errnum*

A variable that will hold the error code if maxtime or dibreak is used. If this optional variable is omitted, the error handler will be executed.The constants ERR_TP_MAXTIME and ERR_TP_ DIBREAK can be used to select the reason.

Program execution

The information text is always written on a new line. If the display is full of text, this body of text is moved up one line first. Strings longer than the width of the teach pendant (40 characters) are split into two lines.

Program execution waits until a number is typed on the numeric keyboard (followed by Enter or ***OK***).

Reference to *TPReadFK* about description of concurrent *TPReadFK* or *TPReadNum* request on Teach Pendant from same or other program tasks.

Example

```
TPReadNum reg1, “How many units should be produced?“;
FOR i FROM 1 TO reg1 DO
   produce_part;
ENDFOR
```

The text *How many units should be produced?* is written on the teach pendant display. The routine *produce_part* is then repeated the number of times that is input via the teach pendant.

Syntax

```
TPReadNum
   [Answer’:=’] <var or pers (INOUT) of num>’,’
   [String’:=’] <expression (IN) of string>
   [’\’MaxTime ’:=’ <expression (IN) of num>]
   [’\’DIBreak ’:=’ <variable (VAR) of signaldi>]
   [’\’BreakFlag ’:=’ <var or pers (INOUT) of errnum>] ’;’
```

Related information

	Described in:
Writing to and reading from the teach pendant	RAPID Summary - *Communication*
Entering a number on the teach pendant	Production Running
Examples of how to use the arguments MaxTime, DIBreak and BreakFlag	Instructions - *TPReadFK*

TPShow Switch window on the teach pendant

TPShow (Teach Pendant Show) is used to select Teach Pendant Window from RAPID.

Examples

TPShow TP_PROGRAM;

The *Production Window* will be active if the system is in *AUTO* mode and the *Program Window* will be active if the system is in *MAN* mode after execution of this instruction.

TPShow TP_LATEST;

The latest used Teach Pendant Window before the current Teach Pendant Window will be active after execution of this instruction.

Arguments

TPShow Window

Window Data type: *tpnum*

The window to show:

TP_PROGRAM	=	*Production Window* if in *AUTO* mode. *Program Window* if in *MAN* mode.
TP_LATEST	=	Latest used Teach Pendant Window before current Teach Pendant Window.
TP_SCREENVIEWER	=	*Screen Viewer Window*, if the Screen Viewer option is active.

Predefined data

```
CONST tpnum TP_PROGRAM := 1;
CONST tpnum TP_LATEST := 2;
CONST tpnum TP_SCREENVIEWER := 3;
```

Program execution

The selected Teach Pendant Window will be activated.

Syntax

TPShow
[Window':='] <expression **(IN)** of *tpnum*> ';'

Related information

	Described in:
Communicating using the teach pendant	RAPID Summary - *Communication*
Teach Pendant Window number	Data Types - *tpnum*

TPWrite Writes on the teach pendant

TPWrite (Teach Pendant Write) is used to write text on the teach pendant. The value of certain data can be written as well as text.

Examples

TPWrite "Execution started";

The text *Execution started* is written on the teach pendant.

TPWrite "No of produced parts="\Num:=reg1;

If, for example, the answer to *No of produced parts=5*, enter 5 instead of *reg1* on the teach pendant.

Arguments

TPWrite String [\Num] | [\Bool] | [\Pos] | [\Orient]

String Data type: *string*

The text string to be written (a maximum of 80 characters).

[\Num] *(Numeric)* Data type: *num*

The data whose numeric value is to be written after the text string.

[\Bool] *(Boolean)* Data type: *bool*

The data whose logical value is to be written after the text string.

[\Pos] *(Position)* Data type: *pos*

The data whose position is to be written after the text string.

[\Orient] *(Orientation)* Data type: *orient*

The data whose orientation is to be written after the text string.

Program execution

Text written on the teach pendant always begins on a new line. When the display is full of text, this text is moved up one line first. Strings that are longer than the width of the teach pendant (40 characters) are divided up into two lines.

If one of the arguments *\Num*, *\Bool*, *\Pos* or *\Orient* is used, its value is first converted to a text string before it is added to the first string. The conversion from value to text string takes place as follows:

Argument	Value	Text string
\Num	23	"23"
\Num	1.141367	"1.14137"
\Bool	TRUE	"TRUE"
\Pos	[1817.3,905.17,879.11]	"[1817.3,905.17,879.11]"
\Orient	[0.96593,0,0.25882,0]	"[0.96593,0,0.25882,0]"

The value is converted to a string with standard RAPID format. This means in principle 6 significant digits. If the decimal part is less than 0.000005 or greater than 0.999995, the number is rounded to an integer.

Limitations

The arguments *\Num*, *\Bool*, *\Pos* and *\Orient* are mutually exclusive and thus cannot be used simultaneously in the same instruction.

Syntax

TPWrite
[String':='] <expression **(IN)** of *string*>
['\'Num':=' <expression **(IN)** of *num*>]
| ['\'Bool':=' <expression **(IN)** of *bool*>]
| ['\'Pos':=' <expression **(IN)** of *pos*>]
| ['\'Orient':=' <expression **(IN)** of *orient*>]';'

Related information

	Described in:
Clearing and reading the teach pendant	RAPID Summary - *Communication*

TriggC Circular robot movement with events

TriggC (Trigg Circular) is used to set output signals and/or run interrupt routines at fixed positions, at the same time as the robot is moving on a circular path.

One or more (max. 4) events can be defined using the instructions *TriggIO*, *TriggEquip* or *TriggInt* and afterwards these definitions are referred to in the instruction *TriggC*.

Examples

```
VAR triggdata gunon;

TriggIO gunon, 0 \Start \DOp:=gun, on;

MoveL p1, v500, z50, gun1;
TriggC p2, p3, v500, gunon, fine, gun1;
```

The digital output signal *gun* is set when the robot's TCP passes the midpoint of the corner path of the point *p1*.



Figure 23 Example of fixed-position IO event.

Arguments

TriggC [\Conc] CirPoint ToPoint Speed [\T]
Trigg_1 [\T2] [\T3] [\T4] Zone Tool [\WObj] [\Corr]

[\Conc] *(Concurrent)* Data type: *switch*

Subsequent instructions are executed at once. This argument is used to shorten the cycle time when, for example, communicating with external equipment, and synchronisation is not required. It can also be used
to tune the execution of the robot path, to avoid warning 50024 Corner path failure, or error 40082 Deceleration limit.

When using the argument \Conc, the number of movement instructions in succession is limited to 5. In a program section that includes StorePath-RestoPath, movement instructions with the argument \Conc are not permitted.

If this argument is omitted and the ToPoint is not a stop point, the subsequent instruction is executed some time before the robot has reached the programmed zone.

CirPoint Data type: *robtarget*

The circle point of the robot. See the instruction *MoveC* for a more detailed description of circular movement. The circle point is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the external axes and of the tool reorientation.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Trigg_1 Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO, TriggEquip* or *TriggInt.*

[\T2] (*Trigg 2*) Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO, TriggEquip* or *TriggInt.*

[\T3] (*Trigg 3*) Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO, TriggEquip* or *TriggInt.*

[\T4] (*Trigg 4)* Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO, TriggEquip* or *TriggInt.*

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point that is moved to the specified destination position.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified for a linear movement relative to the work object to be performed.

[\Corr] *(Correction)* Data type: *switch*

Correction data written to a corrections entry by the instruction *CorrWrite* will be added to the path and destination position, if this argument is present.

Program execution

See the instruction *MoveC* for information about circular movement.

As the trigger conditions are fulfilled when the robot is positioned closer and closer to the end point, the defined trigger activities are carried out. The trigger conditions are fulfilled either at a certain distance before the end point of the instruction, or at a certain distance after the start point of the instruction, or at a certain point in time (limited to a short time) before the end point of the instruction.

During stepping execution forwards, the I/O activities are carried out but the interrupt routines are not run. During stepping execution backwards, no trigger activities at all are carried out.

Examples

```
VAR intnum intno1;
VAR triggdata trigg1;
...
CONNECT intno1 WITH trap1;
TriggInt trigg1, 0.1 \Time, intno1;
...
TriggC p1, p2, v500, trigg1, fine, gun1;
TriggC p3, p4, v500, trigg1, fine, gun1;
...
IDelete intno1;
```

The interrupt routine *trap1* is run when the work point is at a position *0.1* s before the point *p2* or *p4* respectively.

Limitations

If the current start point deviates from the usual, so that the total positioning length of the instruction *TriggC* is shorter than usual, it may happen that several or all of the trigger conditions are fulfilled immediately and at the same position. In such cases, the sequence in which the trigger activities are carried out will be undefined. The program logic in the user program may not be based on a normal sequence of trigger activities for an “incomplete movement”.

The instruction *TriggC* should never be started from the beginning with the robot in position after the circle point. Otherwise the robot will not take the programmed path (positioning around the circular path in another direction compared with that programmed).

Syntax

TriggC
[’\’ Conc ’,’]
[CirPoint ’:=’] < expression (**IN**) of *robtarget* > ’,’
[ToPoint ’:=’] < expression (**IN**) of *robtarget* > ’,’
[Speed ’:=’] < expression (**IN**) of *speeddata* >
[’\’ T ’:=’ < expression (**IN**) of *num* >] ’,’
[Trigg_1 ’:=’] < variable (**VAR**) of *triggdata* >
[’\’ T2 ’:=’ < variable (**VAR**) of *triggdata* >]
[’\’ T3 ’:=’ < variable (**VAR**) of *triggdata* >]
[’\’ T4 ’:=’ < variable (**VAR**) of *triggdata* >] ‘,’
[Zone ’:=’] < expression (**IN**) of *zonedata* > ‘,’
[Tool ’:=’] < persistent (**PERS**) of *tooldata* >
[’\’ WObj ’:=’ < persistent (**PERS**) of *wobjdata* >]
[’\’ Corr]’;’

Related information

	Described in:
Linear movement with triggers	Instructions - *TriggL*
Joint movement with triggers	Instructions - *TriggJ*
Definition of triggers	Instructions - *TriggIO, TriggEquip TriggInt*
Writes to a corrections entry	Instructions - *CorrWrite*
Circular movement	Motion Principles - *Positioning during Program Execution*
Definition of velocity	Data Types - *speeddata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion Principles

TriggEquip Defines a fixed position-time I/O event

TriggEquip (Trigg Equipment) is used to define conditions and actions for setting a digital, a group of digital, or an analog output signal at a fixed position along the robot's movement path with possibility to do time compensation for the lag in the external equipment.

The data defined is used for implementation in one or more subsequent *TriggL, TriggC* or *TriggJ* instructions.

Examples

```
VAR triggdata gunon;

TriggEquip gunon, 10, 0.1 \DOp:=gun, 1;

TriggL p1, v500, gunon, z50, gun1;
```

The tool *gun1* opens in point p2, when the TCP is *10* mm before the point *p1*. To reach this, the digital output signal *gun* is set to the value *1*, when TCP is *0.1* s before the point p2. The gun is full open when TCP reach point p2.



Figure 24 Example of fixed position-time I/O event.

Arguments

TriggEquip TriggData Distance [\Start] EquipLag [\DOp] | [\GOp] | [\AOp] | [\ProcID] SetValue [\Inhib]

TriggData Data type: *triggdata*

Variable for storing the *triggdata* returned from this instruction. These *triggdata* are then used in the subsequent *TriggL*, *TriggC* or *TriggJ* instructions.

Distance Data type: *num*

Defines the position on the path where the I/O equipment event shall occur.

Specified as the distance in mm (positive value) from the end point of the movement path (applicable if the argument \ *Start* is not set).

See the section entitled Program execution for further details.

[\Start] Data type: *switch*

Used when the distance for the argument *Distance* starts at the movement start point instead of the end point.

EquipLag (*Equipment Lag*) Data type: *num*

Specify the lag for the external equipment in s.

For compensation of external equipment lag, use positive argument value. Positive argument value means that the I/O signal is set by the robot system at specified time before the TCP physical reach the specified distance in relation to the movement start or end point.

Negative argument value means that the I/O signal is set by the robot system at specified time after that the TCP physical has passed the specified distance in relation to the movement start or end point.



Figure 25 Use of argument EquipLag.

[\DOp] (*Digital OutPut*) Data type: *signaldo*

The name of the signal, when a digital output signal shall be changed.

[\GOp] (*Group OutPut*) Data type: *signalgo*

The name of the signal, when a group of digital output signals shall be changed.

[\AOp] (*Analog Output*) Data type: *signalao*

The name of the signal, when a analog output signal shall be changed.

[\ProcID] (*Process Identity*) Data type: *num*

Not implemented for customer use.

(The identity of the IPM process to receive the event. The selector is specified in the argument *SetValue*.)

SetValue Data type: *num*

Desired value of output signal (within the allowed range for the current signal).

[\Inhib] (*Inhibit*) Data type: *bool*

The name of a persistent variable flag for inhibit the setting of the signal at runtime.

If this optional argument is used and the actual value of the specified flag is TRUE at the position-time for setting of the signal then the specified signal (*DOp*, *GOp* or *AOp*) will be set to 0 in stead of specified value.

Program execution

When running the instruction *TriggEquip*, the trigger condition is stored in the specified variable for the argument *TriggData*.

Afterwards, when one of the instructions *TriggL*, *TriggC* or *TriggJ* is executed, the following are applicable, with regard to the definitions in *TriggEquip*:

The distance specified in the argument *Distance*:

Linear movement	The straight line distance
Circular movement	The circle arc length
Non-linear movement	The approximate arc length along the path (to obtain adequate accuracy, the distance should not exceed one half of the arc length).



Figure 26 Fixed position-time I/O on a corner path.

The position-time related event will be generated when the start point (end point) is passed, if the specified distance from the end point (start point) is not within the length of movement of the current instruction (*Trigg...*). With use of argument *EquipLag* with negative time (delay), the I/O signal can be set after the end point.

Examples

```
VAR triggdata glueflow;

TriggEquip glueflow, 1 \Start, 0.05 \AOp:=glue, 5.3;

MoveJ p1, v1000, z50, tool1;
TriggL p2, v500, glueflow, z50, tool1;
```

The analog output signal *glue* is set to the value *5.3* when the TCP passes a point located *1* mm after the start point *p1* with compensation for equipment lag *0.05* s.

```
...
TriggL p3, v500, glueflow, z50, tool1;
```

The analog output signal *glue* is set once more to the value *5.3* when the TCP passes a point located *1* mm after the start point *p2*.

Limitations

I/O events with distance (without the argument *Time*) is intended for flying points (corner path). I/O events with distance, using stop points, results in worse accuracy than specified below.

Regarding the accuracy for I/O events with distance and using flying points, the following is applicable when setting a digital output at a specified distance from the start point or end point in the instruction *TriggL* or *TriggC*:

- Accuracy specified below is valid for positive *EquipLag* parameter < 60 ms, equivalent to the lag in the robot servo (without changing the system parameter *Event Preset Time*).
- Accuracy specified below is valid for positive *EquipLag* parameter < configured *Event Preset Time* (system parameter).
- Accuracy specified below is not valid for positive *EquipLag* parameter > configured *Event Preset Time* (system parameter). In this case, an approximate method is used in which the dynamic limitations of the robot are not taken into consideration. *SingArea \Wrist* must be used in order to achieve an acceptable accuracy.
- Accuracy specified below is valid for negative *EquipLag*.

I/O events with time (with the argument *Time*) is intended for stop points. I/O events with time, using flying points, results in worse accuracy than specified below.
I/O events with time can only be specified from the end point of the movement. This time cannot exceed the current braking time of the robot, which is max. approx. 0.5 s (typical values at speed 500 mm/s for IRB2400 150 ms and for IRB6400 250 ms). If the specified time is greater that the current braking time, the event will be generated anyhow, but not until braking is started (later than specified). However, the whole of the movement time for the current movement can be utilised during small and fast movements.

Typical absolute accuracy values for set of digital outputs +/- 5 ms.
Typical repeat accuracy values for set of digital outputs +/- 2 ms.

Syntax

TriggEquip
[TriggData ':='] < variable (**VAR**) of *triggdata*> ','
[Distance ':='] < expression (**IN**) of *num*>
['\' Start] ','
[EquipLag ':='] < expression (**IN**) of *num*>
['\' DOp ':=' < variable (**VAR**) of *signaldo*>]
| ['\' GOp ':=' < variable (**VAR**) of *signalgo*>]
| ['\' AOp ':=' < variable (**VAR**) of *signalao*>]
| ['\' ProcID ':=' < expression (**IN**) of *num*>] ','
[SetValue ':='] < expression (**IN**) of *num*>
['\' Inhibit ':=' < persistent (**PERS**) of *bool*>] ','

Related information

	Described in:
Use of triggers	Instructions - *TriggL, TriggC, TriggJ*
Definition of other triggs	Instruction - *TriggIO, TriggInt*
More examples	Data Types - *triggdata*
Set of I/O	Instructions - *SetDO, SetGO, SetAO*
Configuration of Event preset time	User's guide System Parameters - *Manipulator*

TriggInt Defines a position related interrupt

TriggInt is used to define conditions and actions for running an interrupt routine at a position on the robot's movement path.

The data defined is used for implementation in one or more subsequent *TriggL, TriggC* or *TriggJ* instructions.

Examples

```
VAR intnum intno1;
VAR triggdata trigg1;
...
CONNECT intno1 WITH trap1;
TriggInt trigg1, 5, intno1;
...
TriggL p1, v500, trigg1, z50, gun1;
TriggL p2, v500, trigg1, z50, gun1;
...
IDelete intno1;
```

The interrupt routine *trap1* is run when the TCP is at a position *5* mm before the point *p1* or *p2* respectively.



Figure 27 Example position related interrupt.

Arguments

TriggInt TriggData Distance [\Start] | [\Time] Interrupt

TriggData Data type: *triggdata*

Variable for storing the *triggdata* returned from this instruction. These *triggdata* are then used in the subsequent *TriggL, TriggC* or *TriggJ* instructions.

Distance Data type: *num*

Defines the position on the path where the interrupt shall be generated.

Specified as the distance in mm (positive value) from the end point of the movement path (applicable if the argument \ *Start* or *Time* is not set).

See the section entitled Program execution for further details.

[\Start] Data type: *switch*

Used when the distance for the argument *Distance* starts at the movement start point instead of the end point.

[\Time] Data type: *switch*

Used when the value specified for the argument *Distance* is in fact a time in seconds (positive value) instead of a distance.

Position related interrupts in time can only be used for short times (< 0.5 s) before the robot reaches the end point of the instruction. See the section entitled Limitations for more details.

Interrupt Data type: *intnum*

Variable used to identify an interrupt.

Program execution

When running the instruction *TriggInt*, data is stored in a specified variable for the argument *TriggData* and the interrupt that is specified in the variable for the argument *Interrupt* is activated.

Afterwards, when one of the instructions *TriggL*, *TriggC* or *TriggJ* is executed, the following are applicable, with regard to the definitions in *TriggInt*:

The distance specified in the argument *Distance*:

Linear movement	The straight line distance
Circular movement	The circle arc length
Non-linear movement	The approximate arc length along the path (to obtain adequate accuracy, the distance should not exceed one half of the arc length).



Figure 28 Position related interrupt on a corner path.

The position related interrupt will be generated when the start point (end point) is passed, if the specified distance from the end point (start point) is not within the length of movement of the current instruction (*Trigg...*).

Examples

This example describes programming of the instructions that interact to generate position related interrupts:

```
VAR intnum intno2;
VAR triggdata trigg2;
```

- Declaration of the variables *intno2* and *trigg2 (*shall not be initiated).

```
CONNECT intno2 WITH trap2;
```

- Allocation of interrupt numbers that are stored in the variable *intno2*
- The interrupt number is coupled to the interrupt routine *trap2*

```
TriggInt trigg2, 0, intno2;
```

- The interrupt number in the variable *intno2* is flagged as used
- The interrupt is activated
- Defined trigger conditions and interrupt number are stored in the variable *trigg2*

```
TriggL p1, v500, trigg2, z50, gun1;
```

- The robot is moved to the point *p1*.
- When the TCP reaches the point *p1*, an interrupt is generated and the interrupt routine *trap2* is run.

TriggL p2, v500, trigg2, z50, gun1;

- The robot is moved to the point *p2*
- When the TCP reaches the point *p2*, an interrupt is generated and the interrupt routine *trap2* is run once more.

IDelete intno2;

- The interrupt number in the variable *intno2 is* de-allocated.

Limitations

Interrupt events with distance (without the argument *Time*) is intended for flying points (corner path). Interrupt events with distance, using stop points, results in worse accuracy than specified below.

Interrupt events with time (with the argument *Time*) is intended for stop points. Interrupt events with time, using flying points, results in worse accuracy than specified below.
I/O events with time can only be specified from the end point of the movement. This time cannot exceed the current braking time of the robot, which is max. approx. 0.5 s (typical values at speed 500 mm/s for IRB2400 150 ms and for IRB6400 250 ms). If the specified time is greater that the current braking time, the event will be generated anyhow, but not until braking is started (later than specified). However, the whole of the movement time for the current movement can be utilised during small and fast movements.

Typical absolute accuracy values for generation of interrupts +/- 5 ms.
Typical repeat accuracy values for generation of interrupts +/- 2 ms.

Normally there is a delay of 5 to 120 ms between interrupt generation and response, depending on the type of movement being performed at the time of the interrupt. (Ref. to Basic Characteristics RAPID - *Interrupts*).

To obtain the best accuracy when setting an output at a fixed position along the robot's path, use the instructions *TriggIO* or *TriggEquip* in preference to the instructions *TriggInt* with *SetDO/SetGO/SetAO* in an interrupt routine.

Syntax

TriggInt
[TriggData ':='] < variable (**VAR**) of *triggdata*> ','
[Distance ':='] < expression (**IN**) of *num*>
['\' Start] | ['\' Time] ','
[Interrupt ':='] < variable (**VAR**) of *intnum*> ';'

Related information

	Described in:
Use of triggers	Instructions - *TriggL, TriggC, TriggJ*
Definition of position fix I/O	Instruction - *TriggIO, TriggEquip*
More examples	Data Types - *triggdata*
Interrupts	Basic Characteristics - *Interrupts*

TriggIO Defines a fixed position I/O event

TriggIO is used to define conditions and actions for setting a digital, a group of digital, or an analog output signal at a fixed position along the robot's movement path.

To obtain a fixed position I/O event, *TriggIO* compensates for the lag in the control system (lag between robot and servo) but not for any lag in the external equipment. For compensation of both lags use *TriggEquip*.

The data defined is used for implementation in one or more subsequent *TriggL*, *TriggC* or *TriggJ* instructions.

Examples

```
VAR triggdata gunon;

TriggIO gunon, 10 \DOp:=gun, 1;

TriggL p1, v500, gunon, z50, gun1;
```

The digital output signal *gun* is set to the value *1* when the TCP is *10* mm before the point *p1*.



Figure 29 Example of fixed-position IO event.

Arguments

TriggIO TriggData Distance [\Start] | [\Time] [\DOp] | [\GOp] | [\AOp] | [\ProcID] SetValue [\DODelay]

TriggData Data type: *triggdata*

Variable for storing the *triggdata* returned from this instruction. These *triggdata* are then used in the subsequent *TriggL*, *TriggC* or *TriggJ* instructions.

Distance Data type: *num*

Defines the position on the path where the I/O event shall occur.

Specified as the distance in mm (positive value) from the end point of the movement path (applicable if the argument \ *Start* or *Time* is not set).

See the section entitled Program execution for further details.

[\Start] Data type: *switch*

Used when the distance for the argument *Distance* starts at the movement start point instead of the end point.

[\Time] Data type: *switch*

Used when the value specified for the argument *Distance* is in fact a time in seconds (positive value) instead of a distance.

Fixed position I/O in time can only be used for short times (< 0.5 s) before the robot reaches the end point of the instruction. See the section entitled Limitations for more details.

[\DOp] (*Digital OutPut*) Data type: *signaldo*

The name of the signal, when a digital output signal shall be changed.

[\GOp] (*Group OutPut*) Data type: *signalgo*

The name of the signal, when a group of digital output signals shall be changed.

[\AOp] (*Analog Output*) Data type: *signalao*

The name of the signal, when a analog output signal shall be changed.

[\ProcID] (*Process Identity*) Data type: *num*

Not implemented for customer use.

(The identity of the IPM process to receive the event. The selector is specified in the argument *SetValue*.)

SetValue Data type: *num*

Desired value of output signal (within the allowed range for the current signal).

[\DODelay] (*Digital Output Delay)* Data type: *num*

Time delay in seconds (positive value) for a digital output signal or group of digital output signals.

Only used to delay setting digital output signals, after the robot has reached the specified position. There will be no delay if the argument is omitted.

The delay is not synchronised with the movement.

Program execution

When running the instruction *TriggIO*, the trigger condition is stored in a specified variable for the argument *TriggData*.

Afterwards, when one of the instructions *TriggL*, *TriggC* or *TriggJ* is executed, the following are applicable, with regard to the definitions in *TriggIO*:

The distance specified in the argument *Distance*:

Linear movement	The straight line distance
Circular movement	The circle arc length
Non-linear movement	The approximate arc length along the path (to obtain adequate accuracy, the distance should not exceed one half of the arc length).



Figure 30 Fixed position I/O on a corner path.

The fixed position I/O will be generated when the start point (end point) is passed, if the specified distance from the end point (start point) is not within the length of movement of the current instruction (*Trigg...*).

Examples

```
VAR triggdata glueflow;

TriggIO glueflow, 1 \Start \AOp:=glue, 5.3;

MoveJ p1, v1000, z50, tool1;
TriggL p2, v500, glueflow, z50, tool1;
```

The analog output signal *glue* is set to the value *5.3* when the work point passes a point located *1* mm after the start point *p1*.

```
...
TriggL p3, v500, glueflow, z50, tool1;
```

The analog output signal *glue* is set once more to the value *5.3* when the work point passes a point located *1* mm after the start point *p2*.

Limitations

I/O events with distance (without the argument *Time*) is intended for flying points (corner path). I/O events with distance, using stop points, results in worse accuracy than specified below.

I/O events with time (with the argument *Time*) is intended for stop points. I/O events with time, using flying points, results in worse accuracy than specified below.
I/O events with time can only be specified from the end point of the movement. This time cannot exceed the current braking time of the robot, which is max. approx. 0.5 s (typical values at speed 500 mm/s for IRB2400 150 ms and for IRB6400 250 ms). If the specified time is greater that the current braking time, the event will be generated anyhow, but not until braking is started (later than specified). However, the whole of the movement time for the current movement can be utilised during small and fast movements.

Typical absolute accuracy values for set of digital outputs +/- 5 ms.
Typical repeat accuracy values for set of digital outputs +/- 2 ms.

Syntax

TriggIO
[TriggData ':='] < variable (**VAR**) of *triggdata*> ','
[Distance ':='] < expression (**IN**) of *num*>
['\' Start] | ['\' Time]
['\' DOp ':=' < variable (**VAR**) of *signaldo*>]
| ['\' GOp ':=' < variable (**VAR**) of *signalgo*>]
| ['\' AOp ':=' < variable (**VAR**) of *signalao*>]
| ['\' ProcID ':=' < expression (**IN**) of *num*>] ','
[SetValue ':='] < expression (**IN**) of *num*>
['\' DODelay ':=' < expression (**IN**) of *num*>] ';'

Related information

	Described in:
Use of triggers	Instructions - *TriggL*, *TriggC*, *TriggJ*
Definition of position-time I/O event	Instruction - *TriggEquip*
Definition of position related interrupts	Instruction - *TriggInt*
More examples	Data Types - *triggdata*
Set of I/O	Instructions - *SetDO*, *SetGO*, *SetAO*

TriggJ Axis-wise robot movements with events

TriggJ (Trigg Joint) is used to set output signals and/or run interrupt routines at fixed positions, at the same time as the robot is moving on a circular path.

One or more (max. 4) events can be defined using the instructions *TriggIO, TriggEquip* or *TriggInt* and afterwards these definitions are referred to in the instruction *TriggJ*.

Examples

```
VAR triggdata gunon;

TriggIO gunon, 0 \Start \DOp:=gun, on;

MoveL p1, v500, z50, gun1;
TriggJ p2, v500, gunon, fine, gun1;
```

The digital output signal *gun* is set when the robot's TCP passes the midpoint of the corner path of the point *p1*.



Figure 31 Example of fixed-position IO event.

Arguments

TriggJ [\Conc] ToPoint Speed [\T] Trigg_1 [\T2] [\T3] [\T4] Zone Tool [\WObj]

[\Conc] *(Concurrent)* Data type: *switch*

Subsequent instructions are executed at once. This argument is used to shorten the cycle time when, for example, communicating with external equipment, if synchronisation is not required. It can also be used
to tune the execution of the robot path, to avoid warning 50024 Corner path failure or error 40082 Deceleration limit.

Using the argument \Conc, the number of movement instructions in succession is limited to 5. In a program section that includes StorePath-RestoPath, movement instructions with the argument \Conc are not permitted.

If this argument is omitted and the ToPoint is not a stop point, the subsequent instruction is executed some time before the robot has reached the programmed zone .

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the external axes and of the tool reorientation.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Trigg_1 Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO*, *TriggEquip* or *TriggInt*.

[\T2] (*Trigg 2*) Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO*, *TriggEquip* or *TriggInt*.

[\T3] (*Trigg 3*) Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO*, *TriggEquip* or *TriggInt*.

[\T4] (*Trigg 4)* Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO*, *TriggEquip* or *TriggInt*.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point that is moved to the specified destination position.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified for a linear movement relative to the work object to be performed.

Program execution

See the instruction *MoveJ* for information about joint movement.

As the trigger conditions are fulfilled when the robot is positioned closer and closer to the end point, the defined trigger activities are carried out. The trigger conditions are fulfilled either at a certain distance before the end point of the instruction, or at a certain distance after the start point of the instruction, or at a certain point in time (limited to a short time) before the end point of the instruction.

During stepping execution forwards, the I/O activities are carried out but the interrupt routines are not run. During stepping execution backwards, no trigger activities at all are carried out.

Examples

```
VAR intnum intno1;
VAR triggdata trigg1;
...
CONNECT intno1 WITH trap1;
TriggInt trigg1, 0.1 \Time , intno1;
...
TriggJ p1, v500, trigg1, fine, gun1;
TriggJ p2, v500, trigg1, fine, gun1;
...
IDelete intno1;
```

The interrupt routine *trap1* is run when the work point is at a position *0.1* s before the point *p1* or *p2* respectively.

Limitations

If the current start point deviates from the usual, so that the total positioning length of the instruction *TriggJ* is shorter than usual (e.g. at the start of *TriggJ* with the robot position at the end point), it may happen that several or all of the trigger conditions are fulfilled immediately and at the same position. In such cases, the sequence in which the trigger activities are carried will be undefined. The program logic in the user program may not be based on a normal sequences of trigger activities for an ”incomplete movement”.

Syntax

```
TriggJ
   ['\' Conc ',']
   [ ToPoint ':=' ] < expression (IN) of robtarget > ','
   [ Speed ':=' ] < expression (IN) of speeddata >
      [ '\' T ':=' < expression (IN) of num > ] ','
   [Trigg_1 ':=' ] < variable (VAR) of triggdata >
   [ '\' T2 ':=' < variable (VAR) of triggdata > ]
   [ '\' T3 ':=' < variable (VAR) of triggdata > ]
   [ '\' T4 ':=' < variable (VAR) of triggdata > ] ','
   [Zone ':=' ] < expression (IN) of zonedata > ','
   [ Tool ':=' ] < persistent (PERS) of tooldata >
   [ '\' WObj ':=' < persistent (PERS) of wobjdata > ] ';'
```

Related information

	Described in:
Linear movement with triggs	Instructions - *TriggL*
Circular movement with triggers	Instructions - *TriggC*
Definition of triggers	Instructions - *TriggIO*, *TriggEquip* or *TriggInt*
Joint movement	Motion Principles - *Positioning during Program Execution*
Definition of velocity	Data Types - *speeddata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion Principles

TriggL Linear robot movements with events

TriggL (Trigg Linear) is used to set output signals and/or run interrupt routines at fixed positions, at the same time as the robot is making a linear movement.

One or more (max. 4) events can be defined using the instructions *TriggIO*, *TriggEquip* or *TriggInt* and afterwards these definitions are referred to in the instruction *TriggL*.

Examples

```
VAR triggdata gunon;

TriggIO gunon, 0 \Start \DOp:=gun, on;

MoveJ p1, v500, z50, gun1;
TriggL p2, v500, gunon, fine, gun1;
```

The digital output signal *gun* is set when the robot's TCP passes the midpoint of the corner path of the point *p1*.



Figure 32 Example of fixed-position IO event.

Arguments

TriggL [\Conc] ToPoint Speed [\T] Trigg_1 [\T2] [\T3] [\T4] Zone Tool [\WObj] [\Corr]

[\Conc] *(Concurrent)* Data type: *switch*

Subsequent instructions are executed at once. This argument is used to shorten the cycle time when, for example, communicating with external equipment, if synchronisation is not required. It can also be used
to tune the execution of the robot path, to avoid warning 50024 Corner path failure or error 40082 Deceleration limit.

Using the argument \Conc, the number of movement instructions in succession is limited to 5. In a program section that includes StorePath-RestoPath, movement instructions with the argument \Conc are not permitted.

If this argument is omitted and the ToPoint is not a stop point, the subsequent instruction is executed some time before the robot has reached the programmed zone.

ToPoint Data type: *robtarget*

The destination point of the robot and external axes. It is defined as a named position or stored directly in the instruction (marked with an * in the instruction).

Speed Data type: *speeddata*

The speed data that applies to movements. Speed data defines the velocity of the tool centre point, the external axes and of the tool reorientation.

[\T] *(Time)* Data type: *num*

This argument is used to specify the total time in seconds during which the robot moves. It is then substituted for the corresponding speed data.

Trigg_1 Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO*, *TriggEquip* or *TriggInt*.

[\T2] (*Trigg 2*) Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO*, *TriggEquip* or *TriggInt*.

[\T3] (*Trigg 3*) Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO*, *TriggEquip* or *TriggInt*.

[\T4] (*Trigg 4)* Data type: *triggdata*

Variable that refers to trigger conditions and trigger activity, defined earlier in the program using the instructions *TriggIO*, *TriggEquip* or *TriggInt*.

Zone Data type: *zonedata*

Zone data for the movement. Zone data describes the size of the generated corner path.

Tool Data type: *tooldata*

The tool in use when the robot moves. The tool centre point is the point that is moved to the specified destination position.

[\WObj] *(Work Object)* Data type: *wobjdata*

The work object (coordinate system) to which the robot position in the instruction is related.

This argument can be omitted, and if it is, the position is related to the world coordinate system. If, on the other hand, a stationary TCP or coordinated external axes are used, this argument must be specified for a linear movement relative to the work object to be performed.

[\Corr] *(Correction)* Data type: *switch*

Correction data written to a corrections entry by the instruction *CorrWrite* will be added to the path and destination position, if this argument is present.

Program execution

See the instruction *MoveL* for information about linear movement.

As the trigger conditions are fulfilled when the robot is positioned closer and closer to the end point, the defined trigger activities are carried out. The trigger conditions are fulfilled either at a certain distance before the end point of the instruction, or at a certain distance after the start point of the instruction, or at a certain point in time (limited to a short time) before the end point of the instruction.

During stepping execution forwards, the I/O activities are carried out but the interrupt routines are not run. During stepping execution backwards, no trigger activities at all are carried out.

Examples

```
VAR intnum intno1;
VAR triggdata trigg1;
...
CONNECT intno1 WITH trap1;
TriggInt trigg1, 0.1 \Time, intno1;
...
TriggL p1, v500, trigg1, fine, gun1;
TriggL p2, v500, trigg1, fine, gun1;
...
IDelete intno1;
```

The interrupt routine *trap1* is run when the work point is at a position *0.1* s before the point *p1* or *p2* respectively.

Limitations

If the current start point deviates from the usual, so that the total positioning length of the instruction *TriggL* is shorter than usual (e.g. at the start of *TriggL* with the robot position at the end point), it may happen that several or all of the trigger conditions are fulfilled immediately and at the same position. In such cases, the sequence in which the trigger activities are carried out will be undefined. The program logic in the user program may not be based on a normal sequence of trigger activities for an “incomplete movement”.

Syntax

```
TriggL
   [’\’ Conc ’,’]
   [ ToPoint ’:=’ ] < expression (IN) of robtarget > ’,’
   [ Speed ’:=’ ] < expression (IN) of speeddata >
      [ ’\’ T ’:=’ < expression (IN) of num > ] ’,’
   [Trigg_1 ’:=’ ] < variable (VAR) of triggdata >
   [ ’\’ T2 ’:=’ < variable (VAR) of triggdata > ]
   [ ’\’ T3 ’:=’ < variable (VAR) of triggdata > ]
   [ ’\’ T4 ’:=’ < variable (VAR) of triggdata > ] ‘,’
   [Zone ’:=’ ] < expression (IN) of zonedata > ‘,’
   [ Tool ’:=’ ] < persistent (PERS) of tooldata >
   [ ’\’ WObj ’:=’ < persistent (PERS) of wobjdata > ]
   [ ’\’ Corr ]’;’
```

Related information

	Described in:
Circular movement with triggers	Instructions - *TriggC*
Joint movement with triggers	Instructions - *TriggJ*
Definition of triggers	Instructions - *TriggIO*, *TriggEquip* or *TriggInt*
Writes to a corrections entry	Instructions - *CorrWrite*
Linear movement	Motion Principles - *Positioning during Program Execution*
Definition of velocity	Data Types - *speeddata*
Definition of tools	Data Types - *tooldata*
Definition of work objects	Data Types - *wobjdata*
Motion in general	Motion Principles

TRYNEXT Jumps over an instruction which has caused an error

TRYNEXT is used to jump over an instruction which has caused an error. Instead, the next instruction is run.

Example

```
reg2 := reg3/reg4;
   .
ERROR
   IF ERRNO = ERR_DIVZERO THEN
      reg2:=0;
      TRYNEXT;
   ENDIF
```

An attempt is made to divide *reg3* by *reg4*. If reg4 is equal to 0 (division by zero), a jump is made to the error handler, where *reg2* is set to 0. The *TRYNEXT* instruction is then used to continue with the next instruction.

Program execution

Program execution continues with the instruction subsequent to the instruction that caused the error.

Limitations

The instruction can only exist in a routine's error handler.

Syntax

TRYNEXT';'

Related information

	Described in:
Error handlers	Basic Characteristics- *Error Recovery*

TuneReset Resetting servo tuning

TuneReset is used to reset the dynamic behaviour of all robot axes and external mechanical units to their normal values.

Example

TuneReset;

Resetting tuning values for all axes to 100%.

Program execution

The tuning values for all axes are reset to 100%.

The default servo tuning values for all axes are automatically set by executing instruction *TuneReset*

- at a cold start-up
- when a new program is loaded
- when starting program execution from the beginning.

Syntax

TuneReset ';'

Related information

	Described in:
Tuning servos	Instructions - *TuneServo*

TuneServo Tuning servos

TuneServo is used to tune the dynamic behaviour of separate axes on the robot. It is not necessary to use *TuneServo* under normal circumstances, but sometimes tuning can be optimised depending on the robot configuration and the load characteristics. For external axes *TuneServo* can be used for load adaptation.

Incorrect use of the *TuneServo* can cause oscillating movements or torques that can damage the robot. You must bear this in mind and be careful when using the *TuneServo*.

Avoid doing TuneServo commands at the same time as the robot is moving. It can result in momentary high CPU loads causing error indication and stops.

Note. To obtain optimal tuning it is essential that the correct load data is used. Check on this before using *TuneServo*.

Generally, optimal tuning values often differ between different robots. Optimal tuning may also change with time.

Improving path accuracy

For robots running at lower speeds, TuneServo can be used to improve the path accuracy by:

- Tuning tune_kv and tune_ti (see the tune types description below).
- Tuning friction compensation parameters (see below).

These two methods can be combined.

Other possibilities to improve the path accuracy:

- Decreasing path resolution can improve the path. Note: a value of path resolution which is too low will cause CPU load problems.
- The accuracy of straight lines can be improved by decreasing acceleration using AccSet. Example: AccSet 20, 10.

Description

Tune_df

Tune_df is used for reducing overshoots or oscillations along the path.

There is always an optimum tuning value that can vary depending on position and movement length. This optimum value can be found by changing the tuning in small steps (1 - 2%) on the axes that are involved in this unwanted behaviour. Normally the optimal tuning will be found in the range 70% - 130%. Too low or too high tuning values have a negative effect and will impair movements considerably.

When the tuning value at the start point of a long movement differs considerably from the tuning value at the end point, it can be advantageous in some cases to use an intermediate point with a corner zone to define where the tuning value will change.

Some examples of the use of *TuneServo* to optimise tuning follow below:

IRB 6400, in a press service application (extended and flexible load), axes 4 - 6: Reduce the tuning value for the current wrist axis until the movement is acceptable. A change in the movement will not be noticeable until the optimum value is approached. A low value will impair the movement considerably. Typical tuning value 25%.

IRB 6400, upper parts of working area. Axis 1 can often be optimised with a tuning value of 85% - 95%.

IRB 6400, short movement (< 80 mm). Axis 1 can often be optimised with a tuning value of 94% - 98%.

IRB 2400, with track motion. In some cases axes 2 - 3 can be optimised with a tuning value of 110% - 130%. The movement along the track can require a different tuning value compared with movement at right angles to the track.

Overshoots and oscillations can be reduced by decreasing the acceleration or the acceleration ramp (*AccSet*), which will however increase the cycle time. This is an alternative method to the use of *TuneServo*.

Tune_dg

Tune_dg can reduce overshoots on rare occasions. Normally it should not be used.

Tune_df should always be tried first in cases of overshoots.

Tuning of tune_dg can be performed with large steps in tune value (eg. 50%, 100%, 200%, 400%).

Never use tune_dg when the robot is moving.

Tune_dh

Tune_dh can be used for reducing vibrations and overshoots (eg. large flexible load).

Tune value must always be lower than 100. Tune_dh increases path deviation and normally also increases cycle time.

Example:

IRB6400 with large flexible load which vibrates when the robot has stopped. Use tune_dh with tune value 15.

Tune_dh should only be executed for one axis. All axes in the same mechanical unit automatically get the same tune_value.

Never use tune_dh when the robot is moving.

Tune_di

Tune_di can be used for reducing path deviation at high speeds.

A tune value in the range 50 - 80 is recommended for reducing path deviation. Overshoots can increase (lower tune value means larger overshoot).

A higher tune value than 100 can reduce overshoot (but increases path deviation at high speed).

Tune_di should only be executed for one axis. All axes in the same mechanical unit automatically get the same tune_value.

Tune_dk

For future use.

Tune_kp, tune_kv, tune_ti external axes

These tune types affect position control gain (kp), speed control gain (kv) and speed control integration time (ti) for external axes. These are used for adapting external axes to different load inertias. Basic tuning of external axes can also be simplified by using these tune types.

Tune_kp, tune_kv, tune_ti robot axes

For robot axes, these tune types have another significance and can be used for reducing path errors at low speeds (< 500 mm/s).

Recommended values: tune_kv 100 - 180%, tune_ti 50 - 100%. Tune_kp should not be used for robot axes. Values of tune_kv/tune_ti which are too high or too low will cause vibrations or oscillations. Be careful if trying to exceed these recommended values. Make changes in small steps and avoid oscillating motors.

Always tune one axis at a time. Change the tuning values in small steps. Try to improve the path where this specific axis changes its direction of movement or where it accelerates or deccelerates.

Never use these tune types at high speeds or when the required path accuracy is fulfilled.

Friction compensation: tune_fric_lev and tune_fric_ramp

These tune types can be used to reduce robot path errors caused by friction and backlash at low speeds (10 - 200 mm/s). These path errors appear when a robot axis changes direction of movement. Activate friction compensation for an axis by setting the system parameter *Friction ffw on* to TRUE (topic: Manipulator, type: Control parameters).

The friction model is a constant level with opposite sign of the axis speed direction. *Friction ffw level (Nm)* is the absolute friction level at (low) speeds and is greater than *Friction ffw ramp (rad/s) (see figure)*.



Figure 33 Friction model

Tune_fric_lev overrides the value of the system parameter *Friction ffw level*.

Tuning *Friction ffw level* (using tune_fric_lev) for each robot axis can improve the robots path accuracy considerably in the speed range 20 - 100 mm/s. For larger robots (especially the IRB6400 family) the effect will however be minimal as other sources of tracking errors dominate these robots.

Tune_fric_ramp overrides the value of the system parameter *Friction ffw ramp*. In **most** cases there is no need to tune the *Friction ffw ramp*. The default setting will be appropriate.

Tune one axis at a time. Change the tuning value in small steps and find the level that minimises the robot path error at positions on the path where this specific axis changes direction of movement. Repeat the same procedure for the next axis etc.

The final tuning values can be transferred to the system parameters. Example:

Friction ffw level = 1. Final tune value (tune_fric_lev) = 150%.

> Set Friction ffw level = 1.5 and tune value = 100% (default value) which is equivalent.

Arguments

TuneServo MecUnit Axis TuneValue [\Type]

MecUnit *(Mechanical Unit)* Data type: *mecunit*

The name of the mechanical unit.

Axis Data type: *num*

The number of the current axis for the mechanical unit (1 - 6).

TuneValue Data type: *num*

Tuning value in percent (1 - 500). 100% is the normal value.

[\Type] Data type: *tunetype*

Type of servo tuning. Available types are *TUNE_DF*, *TUNE_KP*, *TUNE_KV*, *TUNE_TI*, *TUNE_FRIC_LEV*, *TUNE_FRIC_RAMP*, *TUNE_DG*, *TUNE_DH*, *TUNE_DI* and *TUNE_DK*. These types are predefined in the system with constants.

This argument can be omitted when using tuning type *TUNE_DF*.

Example

```
TuneServo MHA160R1, 1, 110 \Type:= TUNE_KP;
```

Activating of tuning type *TUNE_KP* with the tuning value *110*% on axis *1* in the mechanical unit *MHA160R1*.

Program execution

The specified tuning type and tuning value are activated for the specified axis. This value is applicable for all movements until a new value is programmed for the current axis, or until the tuning types and values for all axes are reset using the instruction *TuneReset*.

The default servo tuning values for all axes are automatically set by executing instruction *TuneReset*

- at a cold start-up
- when a new program is loaded
- when starting program execution from the beginning.

Limitations

Any active servo tuning are always set to default values at power fail.
This limitation can be handled in the user program at restart after power failure.

Syntax

```
TuneServo
   [MecUnit ':=' ] < variable (VAR) of mecunit> ','
   [Axis ':=' ] < expression (IN) of num> ','
   [TuneValue ':=' ] < expression (IN) of num>
   ['\' Type ':=' <expression (IN) of tunetype>]';'
```

Related information

	Described in:
Other motion settings	Summary Rapid - *Motion Settings*
Types of servo tuning	Data Types - *tunetype*
Reset of all servo tunings	Instructions - *TuneReset*
Tuning of external axes	System parameters - *Manipulator*
Friction compensation	System parameters - *Manipulator*

UnLoad UnLoad a program module during execution

UnLoad is used to unload a program module from the program memory during execution.

The program module must previously have been loaded into the program memory using the instruction *Load* or *StartLoad - WaitLoad*.

Example

UnLoad ram1disk \File:="PART_A.MOD";

UnLoad the program module PART_A.MOD from the program memory, that previously was loaded into the program memory with *Load*. (See instructions *Load*). (*ram1disk* is a predefined string constant "ram1disk:").

Arguments

UnLoad [\Save] FilePath [\File]

[\Save] Data type: *switch*

If this argument is used, the program module is saved before the unloading starts. The program module will be saved at the original place specified in the *Load* or *StartLoad* instruction.

FilePath Data type: *string*

The file path and the file name to the file that will be unloaded from the program memory. The file path and the file name must be the same as in the previously executed *Load* or *StartLoad* instruction. The file name shall be excluded when the argument *\File* is used.

[\File] Data type: *string*

When the file name is excluded in the argument *FilePath*, then it must be defined with this argument. The file name must be the same as in the previously executed *Load* or *StartLoad* instruction.

Program execution

To be able to execute an *UnLoad* instruction in the program, a *Load* or *StartLoad - WaitLoad* instruction with the same file path and name must have been executed earlier in the program.

The program execution waits for the program module to finish unloading before the execution proceeds with the next instruction.

After that the program module is unloaded and the rest of the program modules will be linked.

For more information see the instructions *Load* or *StartLoad-Waitload*.

Examples

UnLoad "ram1disk:DOORDIR/DOOR1.MOD";

UnLoad the program module DOOR1.MOD from the program memory, that previously was loaded into the program memory with *Load*. (See instructions *Load*).

UnLoad "ram1disk:" \File:="DOORDIR/DOOR1.MOD";

Same as above but another syntax.

Unload \Save, "ram1disk:" \File:="DOORDIR/DOOR1.MOD";

Same as above but save the program module before unloading.

Limitations

It is not allowed to unload a program module that is executing.

TRAP routines, system I/O events and other program tasks cannot execute during the unloading.

Avoid ongoing robot movements during the unloading.

Program stop during execution of *UnLoad* instruction results in guard stop with motors off and error message "20025 Stop order timeout" on the Teach Pendant.

Error handling

If the file in the *UnLoad* instruction cannot be unloaded because of ongoing execution within the module or wrong path (module not loaded with *Load* or *StartLoad*), the system variable ERRNO is set to ERR_UNLOAD. This error can then be handled in the error handler.

Syntax

```
UnLoad
   ['\'Save ',']
   [FilePath':=']<expression (IN) of string>
   ['\'File':=' <expression (IN) of string>]';'
```

Related information

	Described in:
Load a program module	Instructions - *Load* Instructions - *StartLoad-WaitLoad*
Accept unresolved references	System Parameters - *Controller* System Parameters - *Tasks* System Parameters - *BindRef*

WaitDI Waits until a digital input signal is set

WaitDI (Wait Digital Input) is used to wait until a digital input is set.

Example

WaitDI di4, 1;

Program execution continues only after the *di4* input has been set.

WaitDI grip_status, 0;

Program execution continues only after the *grip_status* input has been reset.

Arguments

WaitDI Signal Value [\MaxTime] [\TimeFlag]

Signal Data type: *signaldi*

The name of the signal.

Value Data type: *dionum*

The desired value of the signal.

[\MaxTime] *(Maximum Time)* Data type: *num*

The maximum period of waiting time permitted, expressed in seconds. If this time runs out before the condition is met, the error handler will be called, if there is one, with the error code ERR_WAIT_MAXTIME. If there is no error handler, the execution will be stopped.

[\TimeFlag] *(Timeout Flag)* Data type: *bool*

The output parameter that contains the value TRUE if the maximum permitted waiting time runs out before the condition is met. If this parameter is included in the instruction, it is not considered to be an error if the max. time runs out. This argument is ignored if the *MaxTime* argument is not included in the instruction.

Program Running

If the value of the signal is correct, when the instruction is executed, the program simply continues with the following instruction.

If the signal value is not correct, the robot enters a waiting state and when the signal changes to the correct value, the program continues. The change is detected with an interrupt, which gives a fast response (not polled).

When the robot is waiting, the time is supervised, and if it exceeds the max time value, the program will continue if a Time Flag is specified, or raise an error if it's not. If a Time Flag is specified, this will be set to true if the time is exceeded, otherwise it will be set to false.

Syntax

WaitDI
[Signal ':='] < variable (**VAR**) of *signaldi* > ','
[Value ':='] < expression (**IN**) of *dionum* >
['\'MaxTime ':='<expression (**IN**) of *num*>]
['\'TimeFlag':='<variable (**VAR**) of *bool*>] ';'

Related information

	Described in:
Waiting until a condition is satisfied	Instructions - *WaitUntil*
Waiting for a specified period of time	Instructions - *WaitTime*

WaitDO Waits until a digital output signal is set

WaitDO (Wait Digital Output) is used to wait until a digital output is set.

Example

WaitDO do4, 1;

Program execution continues only after the *do4* output has been set.

WaitDO grip_status, 0;

Program execution continues only after the *grip_status* output has been reset.

Arguments

WaitDO Signal Value [\MaxTime] [\TimeFlag]

Signal Data type: *signaldo*

The name of the signal.

Value Data type: *dionum*

The desired value of the signal.

[\MaxTime] *(Maximum Time)* Data type: *num*

The maximum period of waiting time permitted, expressed in seconds. If this time runs out before the condition is met, the error handler will be called, if there is one, with the error code ERR_WAIT_MAXTIME. If there is no error handler, the execution will be stopped.

[\TimeFlag] *(Timeout Flag)* Data type: *bool*

The output parameter that contains the value TRUE if the maximum permitted waiting time runs out before the condition is met. If this parameter is included in the instruction, it is not considered to be an error if the max. time runs out. This argument is ignored if the *MaxTime* argument is not included in the instruction.

Program Running

If the value of the signal is correct, when the instruction is executed, the program simply continues with the following instruction.

If the signal value is not correct, the robot enters a waiting state and when the signal changes to the correct value, the program continues. The change is detected with an interrupt, which gives a fast response (not polled).

When the robot is waiting, the time is supervised, and if it exceeds the max time value, the program will continue if a Time Flag is specified, or raise an error if its not. If a Time Flag is specified, this will be set to true if the time is exceeded, otherwise it will be set to false.

Syntax

WaitDO
[Signal ':='] < variable (**VAR**) of *signaldo* > ','
[Value ':='] < expression (**IN**) of *dionum* >
['\'MaxTime ':='<expression (**IN**) of *num*>]
['\'TimeFlag':='<variable (**VAR**) of *bool*>] ';'

Related information

	Described in:
Waiting until a condition is satisfied	Instructions - *WaitUntil*
Waiting for a specified period of time	Instructions - *WaitTime*

WaitLoad Connect the loaded module to the task

WaitLoad is used to connect the module, if loaded with *StartLoad*, to the program task.

The loaded module must be connected to the program task with the instruction *WaitLoad* before any of its symbol/routines can be used.

The loaded program module will be added to the modules already existing in the program memory.

This instruction can also be combined with the function to unload some other program module, in order to minimise the number of links (1 instead of 2).

Example

```
VAR loadsession load1;
...
StartLoad "ram1disk:PART_A.MOD", load1;
MoveL p10, v1000, z50, tool1 \WObj:=wobj1;
MoveL p20, v1000, z50, tool1 \WObj:=wobj1;
MoveL p30, v1000, z50, tool1 \WObj:=wobj1;
MoveL p40, v1000, z50, tool1 \WObj:=wobj1;
WaitLoad load1;
%"routine_x"%;
UnLoad "ram1disk:PART_A.MOD";
```

Load the program module PART_A.MOD from the *ram1disk* into the program memory. In parallel, move the robot. Then connect the new program module to the program task and call the routine *routine_x* in the module PART_A.

Arguments

WaitLoad [\UnloadPath] [\UnloadFile] LoadNo

[\UnloadPath] Data type: *string*

The file path and the file name to the file that will be unloaded from the program memory. The file name should be excluded when the argument *\UnloadFile* is used.

[\UnloadFile] Data type: *string*

When the file name is excluded in the argument *\UnloadPath,* then it must be defined with this argument.

LoadNo Data type: *loadsession*

This is a reference to the load session, fetched by the instruction *StartLoad,* to connect the loaded program module to the program task.

Program execution

The instruction *WaitLoad* will first wait for the loading to be completed, if it is not already done, and then it will be linked and initialised. The initialisation of the loaded module sets all variables at module level to their init values.

Unsolved references will be accepted, if the system parameter for *Tasks/BindRef* is set to NO. However, when the program is started or the teach pendant function *Program Window/File/Check Program* is used, no check for unsolved references will be done if *BindRef* = NO. There will be a run time error on execution of an unsolved reference.

Another way to use references to instructions, that are not in the task from the beginning, is to use *Late Binding.* This makes it possible to specify the routine to call with a string expression, quoted between two %%. In this case the *BindRef* parameter could be set to YES (default behaviour). The *Late Binding* way is preferable.

To obtain a good program structure, that is easy to understand and maintain, all loading and unloading of program modules should be done from the main module, which is always present in the program memory during execution.

Examples

```
StartLoad  "ram1disk:DOORDIR/DOOR2.MOD",  load1;
...
WaitLoad  \UnloadPath:="ram1disk:DOORDIR/DOOR1.MOD", load1;
```

Load the program module *DOOR2.MOD* from the *ram1disk* at the directory *DOORDIR* into the program memory and connect the new module to the task. The program module DOOR1.MOD will be unloaded from the program memory.

```
StartLoad  "ram1disk:" \File:="DOORDIR/DOOR2.MOD",  load1;
! The robot can do some other work
WaitLoad  \UnloadPath:="ram1disk:" \File:= "DOORDIR/DOOR1.MOD",  load1;
```

is the same as the instructions below but the robot can do some other work during the loading time and also do it faster (only one link).

```
Load  "ram1disk:"  \File:="DOORDIR/DOOR2.MOD";
UnLoad "ram1disk:" \File:="DOORDIR/DOOR1.MOD";
```

Error handling

If the file specified in the *StartLoad* instruction cannot be found, the system variable ERRNO is set to ERR_FILNOTFND at execution of *WaitLoad*.

If argument *LoadNo* refers to an unknown load session, the system variable ERRNO is set to ERR_UNKPROC.

If the module is already loaded into the program memory, the system variable ERRNO is set to ERR_LOADED.

The following errors can only occur when the argument *UnloadPath* is used in the instruction *WaitLoad*:

- If the program module specified in the argument *UnloadPath* cannot be unloaded because of ongoing execution within the module, the system variable ERRNO is set to ERR_UNLOAD.
- If the program module specified in the argument *UnloadPath* cannot be unloaded because the program module is not loaded with *Load* or *StartLoad-WaitLoad* from the RAPID program, the system variable ERRNO is also set to ERR_UNLOAD.

These errors can then be handled in the error handler.

Syntax

WaitLoad
[['\' UnloadPath ':=' <expression (**IN**) of *string*>]
['\' UnloadFile ':=' <expression (**IN**) of *string*>] ',']
[LoadNo ':='] <variable (**VAR**) of *loadsession*> ';'

Related information

Load a program module during execution	Instructions - *StartLoad*
Load session	Data Types - *loadsession*
Load a program module	Instructions - *Load*
Unload a program module	Instructions - *UnLoad*
Accept unsolved references	System Parameters - *Controller/Task/ BindRef*

VelSet Changes the programmed velocity

VelSet is used to increase or decrease the programmed velocity of all subsequent positioning instructions. This instruction is also used to maximize the velocity.

Example

VelSet 50, 800;

All the programmed velocities are decreased to 50% of the value in the instruction. The TCP velocity is not, however, permitted to exceed 800 mm/s.

Arguments

VelSet Override Max

Override Data type: *num*

Desired velocity as a percentage of programmed velocity. 100% corresponds to the programmed velocity.

Max Data type: *num*

Maximum TCP velocity in mm/s.

Program execution

The programmed velocity of all subsequent positioning instructions is affected until a new *VelSet* instruction is executed.

The argument *Override* affects:

- All velocity components (TCP, orientation, rotating and linear external axes) in *speeddata*.
- The programmed velocity override in the positioning instruction (the argument *\V*).
- Timed movements.

The argument *Override* does not affect:

- The welding speed in *welddata*.
- The heating and filling speed in *seamdata*.

The argument *Max* only affects the velocity of the TCP.

The default values for *Override* and *Max* are 100% and 5000 mm/s respectively. These values are automatically set

- at a cold start-up
- when a new program is loaded
- when starting program executing from the beginning.

Example

```
VelSet 50, 800;
MoveL p1, v1000, z10, tool1;
MoveL p2, v2000, z10, tool1;
MoveL p3, v1000\T:=5, z10, tool1;
```

The speed is *500* mm/s to point *p1* and *800* mm/s to *p2*. It takes *10* seconds to move from p2 to *p3*.

Limitations

The maximum speed is not taken into consideration when the time is specified in the positioning instruction.

Syntax

```
VelSet
   [ Override ':=' ] < expression (IN) of num > ','
   [ Max ':=' ] < expression (IN) of num > ';'
```

Related information

	Described in:
Definition of velocity	Data Types - *speeddata*
Positioning instructions	RAPID Summary - *Motion*

WHILE Repeats as long as ...

WHILE is used when a number of instructions are to be repeated as long as a given condition is met.

If it is possible to determine the number of repetitions in advance, the *FOR* instruction can be used.

Example

```
WHILE reg1 < reg2 DO
   ...
   reg1 := reg1 +1;
ENDWHILE
```

Repeats the instructions in the WHILE loop as long as *reg1 < reg2*.

Arguments

WHILE Condition DO ... ENDWHILE

Condition Data type: *bool*

The condition that must be met for the instructions in the WHILE loop to be executed.

Program execution

7. The condition is calculated. If the condition is not met, the WHILE loop terminates and program execution continues with the instruction following ENDWHILE.
8. The instructions in the WHILE loop are executed.
9. The WHILE loop is repeated, starting from point 1.

Syntax

(EBNF)
WHILE <conditional expression> **DO**
   <instruction list>
ENDWHILE

Related information

	Described in:
Expressions	Basic Characteristics - *Expressions*

Write Writes to a character-based file or serial channel

Write is used to write to a character-based file or serial channel. The value of certain data can be written as well as text.

Examples

Write logfile, "Execution started";

The text *Execution started* is written to the file with reference name *logfile*.

Write logfile, "No of produced parts="\Num:=reg1;

The text *No of produced parts=5*, for example, is written to the file with the reference name *logfile* (assuming that the contents of *reg1* is 5).

Arguments

Write IODevice String [\Num] | [\Bool] | [\Pos] | [\Orient] [\NoNewLine]

IODevice Data type: *iodev*

The name (reference) of the current file or serial channel.

String Data type: *string*

The text to be written.

[\Num] *(Numeric)* Data type: *num*

The data whose numeric values are to be written after the text string.

[\Bool] *(Boolean)* Data type: *bool*

The data whose logical values are to be written after the text string.

[\Pos] *(Position)* Data type: *pos*

The data whose position is to be written after the text string.

[\Orient] *(Orientation)* Data type: *orient*

The data whose orientation is to be written after the text string.

[\NoNewLine] Data type: *switch*

Omits the line-feed character that normally indicates the end of the text.

Program execution

The text string is written to a specified file or serial channel. If the argument *\NoNewLine* is not used, a line-feed character (LF) is also written.

If one of the arguments *\Num*, *\Bool*, *\Pos* or *\Orient* is used, its value is first converted to a text string before being added to the first string. The conversion from value to text string takes place as follows:

Argument	Value	Text string
\Num	23	"23"
\Num	1.141367	"1.14137"
\Bool	TRUE	"TRUE"
\Pos	[1817.3,905.17,879.11]	"[1817.3,905.17,879.11]"
\Orient	[0.96593,0,0.25882,0]	"[0.96593,0,0.25882,0]"

The value is converted to a string with standard RAPID format. This means in principle 6 significant digits. If the decimal part is less than 0.000005 or greater than 0.999995, the number is rounded to an integer.

Example

```
VAR iodev printer;
.
Open "sio1:", printer\Write;
WHILE DInput(stopprod)=0 DO
    produce_part;
    Write printer, "Produced part="\Num:=reg1\NoNewLine;
    Write printer, "          "\NoNewLine;
    Write printer, CTime();
ENDWHILE
Close printer;
```

A line, including the number of the produced part and the time, is output to a printer each cycle. The printer is connected to serial channel *sio1:*. The printed message could look like this:

Produced part=473 09:47:15

Limitations

The arguments *\Num*, *\Bool*, *\Pos* and *\Orient* are mutually exclusive and thus cannot be used simultaneously in the same instruction.

This instruction can only be used for files or serial channels that have been opened for writing.

Error handling

If an error occurs during writing, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

Syntax

```
Write
   [IODevice':='] <variable (VAR) of iodev>','
   [String':='] <expression (IN) of string>
   ['\'Num':=' <expression (IN) of num> ]
   | ['\'Bool':=' <expression (IN) of bool> ]
   | ['\'Pos':=' <expression (IN) of pos> ]
   | ['\'Orient':=' <expression (IN) of orient> ]
   ['\'NoNewLine]';'
```

Related information

	Described in:
Opening a file or serial channel	RAPID Summary - *Communication*

WriteBin Writes to a binary serial channel

WriteBin is used to write a number of bytes to a binary serial channel.

Example

WriteBin channel2, text_buffer, 10;

10 characters from the *text_buffer* list are written to the channel referred to by *channel2*.

Arguments

WriteBin IODevice Buffer NChar

IODevice Data type: *iodev*

Name (reference) of the current serial channel.

Buffer Data type: *array of num*

The list (array) containing the numbers (characters) to be written.

NChar *(Number of Characters)* Data type*: num*

The number of characters to be written from the *Buffer.*

Program execution

The specified number of numbers (characters) in the list is written to the serial channel.

Limitations

This instruction can only be used for serial channels that have been opened for binary reading and writing.

Error handling

If an error occurs during writing, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

Example

```
VAR iodev channel;
VAR num out_buffer{20};
VAR num input;
VAR num nchar;
Open "sio1:", channel\Bin;

out_buffer{1} := 5;                          ( enq )
WriteBin channel, out_buffer, 1;
input := ReadBin (channel \Time:= 0.1);

IF input = 6 THEN                            ( ack )
   out_buffer{1} := 2;                       ( stx )
   out_buffer{2} := 72;                      ( 'H' )
   out_buffer{3} := 101;                     ( 'e' )
   out_buffer{4} := 108;                     ( 'l' )
   out_buffer{5} := 108;                     ( 'l' )
   out_buffer{6} := 111;                     ( 'o' )
   out_buffer{7} := 32;                      ( ' ' )
   out_buffer{8} := StrToByte("w"\Char);     ( 'w' )
   out_buffer{9} := StrToByte("o"\Char);     ( 'o' )
   out_buffer{10} := StrToByte("r"\Char);    ( 'r' )
   out_buffer{11} := StrToByte("l"\Char);    ( 'l' )
   out_buffer{12} := StrToByte("d"\Char);    ( 'd' )
   out_buffer{13} := 3;                      ( etx )
   WriteBin channel, out_buffer, 13;
ENDIF
```

The text string *Hello world* (with associated control characters) is written to a serial channel. The function *StrToByte* is used in the same cases to convert a string into a *byte* (*num*) data.

Syntax

WriteBin
[IODevice':='] <variable (**VAR**) of *iodev*>','
[Buffer':='] <array {*} (**IN**) of *num*>','
[NChar':='] <expression (**IN**) of *num*>';'

Related information

	Described in:
Opening (etc.) of serial channels	RAPID Summary - *Communication*
Convert a string to a byte data	Functions - *StrToByte*
Byte data	Data Types - *byte*

WriteAnyBin Writes data to a binary serial channel or file

WriteAnyBin (Write Any Binary) is used to write any type of data to a binary serial channel or file.

Example

```
VAR iodev channel2;
VAR orient quat1 := [1, 0, 0, 0];
...
Open "sio1:", channel2 \Bin;
WriteAnyBin channel2, quat1;
```

The *orient* data *quat1* is written to the channel referred to by *channel2*.

Arguments

WriteAnyBin IODevice Data

IODevice Data type: *iodev*

The name (reference) of the binary serial channel or file for the writing operation.

Data Data type: *ANYTYPE*

The VAR or PERS containing the data to be written.

Program execution

As many bytes as required for the specified data are written to the specified binary serial channel or file.

Limitations

This instruction can only be used for serial channels or files that have been opened for binary writing.

The data to be written by this instruction must have a *value* data type of *atomic*, *string*, or *record* data type. *Semi-value* and *non-value* data types cannot be used.

Array data cannot be used.

Error handling

If an error occurs during writing, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

Example

```
VAR iodev channel;
VAR num input;
VAR robtarget cur_robt;

Open "sio1:", channel\Bin;

! Send the control character enq
WriteStrBin channel, "\05";
! Wait for the control character ack
input := ReadBin (channel \Time:= 0.1);
IF input = 6 THEN
   ! Send current robot position
   cur_robt := CRobT(\Tool:= tool1\WObj:= wobj1);
   WriteAnyBin channel, cur_robt;
ENDIF

Close channel;
```

The current position of the robot is written to a binary serial channel.

Syntax

```
WriteAnyBin
   [IODevice':='] <variable (VAR) of iodev>','
   [Data':='] <var or pers (INOUT) of ANYTYPE>';'
```

Related information

	Described in:
Opening (etc.) of serial channels or files	RAPID Summary - *Communication*
Read data from a binary serial channel or file	Functions - *ReadAnyBin*

WriteStrBin Writes a string to a binary serial channel

WriteStrBin (Write String Binary) is used to write a string to a binary serial channel or binary file.

Example

```
WriteStrBin channel2, "Hello World\0A";
```

The string *"Hello World\0A"* is written to the channel referred to by *channel2*. The string is in this case ended with new line \0A. All characters and hexadecimal values written with *WriteStrBin* will be unchanged by the system.

Arguments

WriteStrBin IODevice Str

IODevice Data type: *iodev*

Name (reference) of the current serial channel.

Str (*String*) Data type: *string*

The text to be written.

Program execution

The text string is written to the specified serial channel or file.

Limitations

This instruction can only be used for serial channels or files that have been opened for binary reading and writing.

Error handling

If an error occurs during writing, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

Example

```
VAR iodev channel;
VAR num input;
Open "sio1:", channel\Bin;

! Send the control character enq
WriteStrBin channel, "\05";
! Wait for the control character ack
input := ReadBin (channel \Time:= 0.1);
IF input = 6 THEN
   ! Send a text starting with control character stx and ending with etx
   WriteStrBin channel, "\02Hello world\03";
ENDIF

Close channel;
```

The text string *Hello world* (with associated control characters in hexadecimal) is written to a binary serial channel.

Syntax

```
WriteStrBin
   [IODevice':='] <variable (VAR) of iodev>','
   [Str':='] <expression (IN) of string>';'
```

Related information

	Described in:
Opening (etc.) of serial channels	RAPID Summary - *Communication*

WaitTime Waits a given amount of time

WaitTime is used to wait a given amount of time. This instruction can also be used to wait until the robot and external axes have come to a standstill.

Example

WaitTime 0.5;

Program execution waits 0.5 seconds.

Arguments

WaitTime [\InPos] Time

[\InPos] Data type: *switch*

If this argument is used, the robot and external axes must have come to a standstill before the waiting time starts to be counted.

Time Data type: *num*

The time, expressed in seconds, that program execution is to wait.

Program execution

Program execution temporarily stops for the given amount of time. Interrupt handling and other similar functions, nevertheless, are still active.

Example

WaitTime \InPos,0;

Program execution waits until the robot and the external axes have come to a standstill.

Limitations

If the argument *Inpos* is used, the movement instruction which precedes this instruction should be terminated with a stop point, in order to be able to restart in this instruction following a power failure.

Argument *Inpos* cannot be used together with SoftServo.

Syntax

```
WaitTime
   ['\'InPos',']
   [Time ':='] <expression (IN) of num>';'
```

Related information

	Described in:
Waiting until a condition is met	Instructions - *WaitUntil*
Waiting until an I/O is set/reset	Instruction - *WaitDI*

WaitUntil Waits until a condition is met

WaitUntil is used to wait until a logical condition is met; for example, it can wait until one or several inputs have been set.

Example

WaitUntil di4 = 1;

Program execution continues only after the *di4* input has been set.

Arguments

WaitUntil [\InPos] Cond [\MaxTime] [\TimeFlag]

[\InPos] Data type: *switch*

If this argument is used, the robot and external axes must have stopped moving before the condition starts being evaluated.

Cond Data type: *bool*

The logical expression that is to be waited for.

[\MaxTime] Data type: *num*

The maximum period of waiting time permitted, expressed in seconds. If this time runs out before the condition is set, the error handler will be called, if there is one, with the error code ERR_WAIT_MAXTIME. If there is no error handler, the execution will be stopped.

[\TimeFlag] *(Timeout Flag)* Data type: *bool*

The output parameter that contains the value TRUE if the maximum permitted waiting time runs out before the condition is met. If this parameter is included in the instruction, it is not considered to be an error if the max. time runs out. This argument is ignored if the *MaxTime* argument is not included in the instruction.

Program execution

If the programmed condition is not met on execution of a *WaitUntil* instruction, the condition is checked again every 100 ms.

When the robot is waiting, the time is supervised, and if it exceeds the max time value, the program will continue if a *TimeFlag* is specified, or raise an error if it’s not. If a *TimeFlag* is specified, this will be set to TRUE if the time is exceeded, otherwise it will be set to false.

Examples

```
VAR bool timeout;
WaitUntil start_input = 1 AND grip_status = 1\MaxTime := 60
          \TimeFlag := timeout;
IF timeout THEN
   TPWrite "No start order received within expected time";
ELSE
   start_next_cycle;
ENDIF
```

If the two input conditions are not met within *60* seconds, an error message will be written on the display of the teach pendant.

```
WaitUntil \Inpos, di4 = 1;
```

Program execution waits until the robot has come to a standstill and the *di4* input has been set.

Limitation

If the argument *\Inpos is used*, the movement instruction which precedes this instruction should be terminated with a stop point, in order to be able to restart in this instruction following a power failure.

Syntax

```
WaitUntil
   ['\'InPos',']
   [Cond ':='] <expression (IN) of bool>
   ['\'MaxTime ':='<expression (IN) of num>]
   ['\'TimeFlag':='<variable (VAR) of bool>] ';'
```

Related information

	Described in:
Waiting until an input is set/reset	Instructions - *WaitDI*
Waiting a given amount of time	Instructions - *WaitTime*
Expressions	Basic Characteristics - *Expressions*

WZBoxDef Define a box-shaped world zone

WZBoxDef (World Zone Box Definition) is used to define a world zone that has the shape of a straight box with all its sides parallel to the axes of the World Coordinate System.

Example




```
VAR shapedata volume;
CONST pos corner1:=[200,100,100];
CONST pos corner2:=[600,400,400];
...
WZBoxDef \Inside, volume, corner1, corner2;
```

Define a straight box with coordinates parallel to the axes of the world coordinate system and defined by the opposite corners *corner1* and *corner2*.

Arguments

WZBoxDef [\Inside] | [\Outside] Shape LowPoint HighPoint

\Inside Data type: *switch*

Define the volume inside the box.

\Outside Data type: *switch*

Define the volume outside the box (inverse volume).

One of the arguments *\Inside* or *\Outside* must be specified.

Shape Data type: *shapedata*

Variable for storage of the defined volume (private data for the system).

LowPoint Data type: *pos*

Position (x,y,x) in mm defining one lower corner of the box.

HighPoint Data type: *pos*

Position (x,y,z) in mm defining the corner diagonally opposite to the previous one.

Program execution

The definition of the box is stored in the variable of type *shapedata* (argument *Shape*), for future use in *WZLimSup* or *WZDOSet* instructions.

Limitations

The *LowPoint* and *HighPoint* positions must be valid for opposite corners (with different x, y and z coordinate values).

If the robot is used to point out the *LowPoint* or *HighPoint*, work object *wobj0* must be active (use of component *trans* in *robtarget* e.g. p1.trans as argument).

Syntax

WZBoxDef
[’\’Inside] | [’\’Outside] ’,’
[Shape’:=’]<variable (**VAR**) of *shapedata*>’,’
[LowPoint’:=’]<expression (**IN**) of *pos*>’,’
[HighPoint’:=’]<expression (**IN**) of *pos*>’;’

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Define sphere-shaped world zone	Instructions - *WZSphDef*
Define cylinder-shaped world zone	Instructions - *WZCylDef*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone digital output set	Instructions - *WZDOSet*

WZCylDef Define a cylinder-shaped world zone

WZCylDef (World Zone Cylinder Definition) is used to define a world zone that has the shape of a cylinder with the cylinder axis parallel to the z-axis of the World Coordinate System.

Example




```
VAR shapedata volume;
CONST pos C2:=[300,200,200];
CONST num R2:=100;
CONST num H2:=200;
...
WZCylDef \Inside, volume, C2, R2, H2;
```

Define a cylinder with the centre of the bottom circle in *C2*, radius *R2* and height *H2*.

Arguments

WZCylDef [\Inside] | [\Outside] Shape CentrePoint Radius Height

\Inside Data type: *switch*

Define the volume inside the cylinder.

\Outside Data type: *switch*

Define the volume outside the cylinder (inverse volume).

One of the arguments *\Inside* or *\Outside* must be specified.

Shape Data type: *shapedata*

Variable for storage of the defined volume (private data for the system).

CentrePoint Data type: *pos*

Position (x,y,z) in mm defining the centre of one circular end of the cylinder.

Radius Data type: *num*

The radius of the cylinder in mm.

Height Data type: *num*

The height of the cylinder in mm.
If it is positive (+z direction), the *CentrePoint* argument is the centre of the lower end of the cylinder (as in the above example).
If it is negative (-z direction), the *CentrePoint* argument is the centre of the upper end of the cylinder.

Program execution

The definition of the cylinder is stored in the variable of type *shapedata* (argument *Shape*), for future use in *WZLimSup* or *WZDOSet* instructions.

Limitations

If the robot is used to point out the *CentrePoint*, work object *wobj0* must be active (use of component *trans* in *robtarget* e.g. p1.trans as argument).

Syntax

WZCylDef
[’\’Inside] | [’\’Outside]’,’
[Shape’:=’]<variable (**VAR**) of *shapedata*>’,’
[CentrePoint’:=’]<expression (**IN**) of *pos*>’,’
[Radius’:=’]<expression (**IN**) of *num*>’,’
[Height’:=’]<expression (**IN**) of *num*>’;’

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Define box-shaped world zone	Instructions - *WZBoxDef*
Define sphere-shaped world zone	Instructions - *WZSphDef*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone digital output set	Instructions - *WZDOSet*

WZDisable Deactivate temporary world zone supervision

WZDisable (World Zone Disable) is used to deactivate the supervision of a temporary world zone, previously defined either to stop the movement or to set an output.

Example

```
VAR wztemporary wzone;
...
PROC ...
   WZLimSup \Temp, wzone, volume;
   MoveL p_pick, v500, z40, tool1;
   WZDisable wzone;
   MoveL p_place, v200, z30, tool1;
ENDPROC
```

When moving to *p_pick*, the position of the robot's TCP is checked so that it will not go inside the specified volume *wzone*. This supervision is not performed when going to *p_place*.

Arguments

WZDisable WorldZone

WorldZone

Data type: *wztemporary*

Variable or persistent variable of type *wztemporary*, which contains the identity of the world zone to be deactivated.

Program execution

The temporary world zone is deactivated. This means that the supervision of the robot's TCP, relative to the corresponding volume, is temporarily stopped. It can be re-activated via the *WZEnable* instruction.

Limitations

Only a temporary world zone can be deactivated. A stationary world zone is always active.

Syntax

WZDisable
[WorldZone':=']<variable or persistent (**INOUT**) of *wztemporary*>';'

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Temporary world zone data	Data Types - *wztemporary*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone set digital output	Instructions - *WZDOSet*
Activate world zone	Instructions - *WZEnable*
Erase world zone	Instructions - *WZFree*

WZDOSet Activate world zone to set digital output

WZDOSet (World Zone Digital Output Set) is used to define the action and to activate a world zone for supervision of the robot movements.

After this instruction is executed, when the robot's TCP is inside the defined world zone or is approaching close to it, a digital output signal is set to the specified value.

Example

```
VAR wztemporary service;

PROC zone_output()
   VAR shapedata volume;
   CONST pos p_service:=[500,500,700];
   ...
   WZSphDef \Inside, volume, p_service, 50;
   WZDOSet \Temp, service \Inside, volume, do_service, 1;
ENDPROC
```

Definition of temporary world zone *service* in the application program, that sets the signal *do_service*, when the robot's TCP is inside the defined sphere during program execution or when jogging.

Arguments

WZDOSet [\Temp] | [\Stat] WorldZone [\Inside] | [\Before] Shape Signal SetValue

\Temp (*Temporary*) Data type: *switch*

The world zone to define is a temporary world zone.

\Stat (*Stationary*) Data type: *switch*

The world zone to define is a stationary world zone.

One of the arguments *\Temp* or *\Stat* must be specified.

WorldZone Data type: *wztemporary*

Variable or persistent variable, that will be updated with the identity (numeric value) of the world zone.

If use of switch *\Temp*, the data type must be *wztemporary.*
If use of switch *\Stat*, the data type must be *wzstationary.*

\Inside Data type: *switch*

The digital output signal will be set when the robot's TCP is inside the defined volume.

\Before Data type: *switch*

The digital output signal will be set before the robot's TCP reaches the defined volume (as soon as possible before the volume).

One of the arguments *Inside* or *Outside* must be specified.

Shape Data type: *shapedata*

The variable that defines the volume of the world zone.

Signal Data type: *signaldo*

The name of the digital output signal that will be changed.

If a stationary worldzone is used, the signal must be write protected for access from the user (RAPID, TP). Set Access = System for the signal in System Parameters.

SetValue Data type: *dionum*

Desired value of the signal (0 or 1) when the robot's TCP is inside the volume or just before it enters the volume.

When outside or just outside the volume, the signal is set to the opposite value.

Program execution

The defined world zone is activated. From this moment, the robot's TCP position is supervised and the output will be set, when the robot's TCP position is inside the volume (*Inside*) or comes close to the border of the volume (*Before*).

Example

```
VAR wztemporary home;
VAR wztemporary service;
PERS wztemporary equip1:=[0];

PROC main()
   ...
   ! Definition of all temporary world zones
   zone_output;
   ...
   ! equip1 in robot work area
   WZEnable equip1;
   ...
   ! equip1 out of robot work area
   WZDisable equip1;
   ...
   ! No use for equip1 any more
   WZFree equip1;
   ...
ENDPROC

PROC zone_output()
   VAR shapedata volume;
   CONST pos p_home:=[800,0,800];
   CONST pos p_service:=[800,800,800];
   CONST pos p_equip1:=[-800,-800,0];
   ...
   WZSphDef \Inside, volume, p_home, 50;
   WZDOSet \Temp, home \Inside, volume, do_home, 1;
   WZSphDef \Inside, volume, p_service, 50;
   WZDOSet \Temp, service \Inside, volume, do_service, 1;
   WZCylDef \Inside, volume, p_equip1, 300, 1000;
   WZLimSup \Temp, equip1, volume;
   ! equip1 not in robot work area
   WZDisable equip1;
ENDPROC
```

Definition of temporary world zones *home* and *service* in the application program, that sets the signals *do_home* and *do_service*, when the robot is inside the sphere *home* or *service* respectively during program execution or when jogging.

Also, definition of a temporary world zone *equip1*, which is active only in the part of the robot program when *equip1* is inside the working area for the robot. At that time the robot stops before entering the *equip1* volume, both during program execution and manual jogging. *equip1* can be disabled or enabled from other program tasks by using the persistent variable *equip1* value.

Limitations

A world zone cannot be redefined by using the same variable in the argument *WorldZone*.

A stationary world zone cannot be deactivated, activated again or erased in the RAPID program.

A temporary world zone can be deactivated (*WZDisable*), activated again (*WZEnable*) or erased (*WZFree)* in the RAPID program.

Syntax

WZDOSet
(’\’Temp) | (’\’Stat) ’,’
[WorldZone’:=’]<variable or persistent (**INOUT**) of *wztemporary*>
(’\’Inside) | (’\’Before) ’,’
[Shape’:=’]<variable (**VAR**) of *shapedata*>’,’
[Signal’:=’]<variable (**VAR**) of *signaldo*>’,’
[SetValue’:=’]<expression (**IN**) of dio*num*>’;’

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Temporary world zone	Data Types - *wztemporary*
Stationary world zone	Data Types - *wzstationary*
Define straight box-shaped world zone	Instructions - *WZBoxDef*
Define sphere-shaped world zone	Instructions - *WZSphDef*
Define cylinder-shaped world zone	Instructions - *WZCylDef*
Activate world zone limit supervision	Instructions - *WZLimSup*
Signal access mode	User’s Guide - System Parameters I/O Signals

WZEnable Activate temporary world zone supervision

WZEnable (World Zone Enable) is used to re-activate the supervision of a temporary world zone, previously defined either to stop the movement or to set an output.

Example

```
VAR wztemporary wzone;
...
PROC ...
   WZLimSup \Temp, wzone, volume;
   MoveL p_pick, v500, z40, tool1;
   WZDisable wzone;
   MoveL p_place, v200, z30, tool1;
   WZEnable wzone;
   MoveL p_home, v200, z30, tool1;
ENDPROC
```

When moving to *p_pick*, the position of the robot's TCP is checked so that it will not go inside the specified volume *wzone*. This supervision is not performed when going to *p_place*, but is reactivated before going to *p_home*

Arguments

WZEnable WorldZone

WorldZone

Data type: *wztemporary*

Variable or persistent variable of the type *wztemporary*, which contains the identity of the world zone to be activated.

Program execution

The temporary world zone is re-activated.
Please note that a world zone is automatically activated when it is created. It need only be re-activated when it has previously been deactivated by *WZDisable*.

Limitations

Only a temporary world zone can be deactivated and reactivated. A stationary world zone is always active.

Syntax

```
WZEnable
   [WorldZone':=']<variable or persistent (INOUT) of wztemporary>';'
```

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Temporary world zone data	Data Types - *wztemporary*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone set digital output	Instructions - *WZDOSet*
Deactivate world zone	Instructions - *WZDisable*
Erase world zone	Instructions - *WZFree*

WZFree Erase temporary world zone supervision

WZFree (World Zone Free) is used to erase the definition of a temporary world zone, previously defined either to stop the movement or to set an output.

Example

```
VAR wztemporary wzone;
...
PROC ...
   WZLimSup \Temp, wzone, volume;
   MoveL p_pick, v500, z40, tool1;
   WZDisable wzone;
   MoveL p_place, v200, z30, tool1;
   WZEnable wzone;
   MoveL p_home, v200, z30, tool1;
   WZFree wzone;
ENDPROC
```

When moving to *p_pick*, the position of the robot's TCP is checked so that it will not go inside a specified volume *wzone*. This supervision is not performed when going to *p_place*, but is reactivated before going to *p_home*. When this position is reached, the world zone definition is erased.

Arguments

WZFree WorldZone

WorldZone Data type: *wztemporary*

Variable or persistent variable of the type *wztemporary*, which contains the identity of the world zone to be erased.

Program execution

The temporary world zone is first deactivated and then its definition is erased.

Once erased, a temporary world zone cannot be either re-activated nor deactivated.

Limitations

Only a temporary world zone can be deactivated, reactivated or erased. A stationary world zone is always active.

Syntax

WZFree
[WorldZone':=']<variable or persistent (**INOUT**) of *wztemporary*>';'

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Temporary world zone data	Data Types - *wztemporary*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone set digital output	Instructions - *WZDOSet*
Deactivate world zone	Instructions - *WZDisable*
Activate world zone	Instructions - *WZEnable*

WZLimSup Activate world zone limit supervision

WZLimSup (World Zone Limit Supervision) is used to define the action and to activate a world zone for supervision of the working area of the robot.

After this instruction is executed, when the robot's TCP reaches the defined world zone, the movement is stopped both during program execution and when jogging.

Example

```
VAR wzstationary max_workarea;
...
PROC POWER_ON()
   VAR shapedata volume;
   ...
   WZBoxDef \Outside, volume, corner1, corner2;
   WZLimSup \Stat, max_workarea, volume;
ENDPROC
```

Definition and activation of stationary world zone *max_workarea*, with the shape of the area outside a box (temporarily stored in *volume*) and the action work-area supervision. The robot stops with an error message before entering the area outside the box.

Arguments

WZLimSup [\Temp] | [\Stat] WorldZone Shape

\Temp (*Temporary*) Data type: *switch*

The world zone to define is a temporary world zone.

\Stat (*Stationary*) Data type: *switch*

The world zone to define is a stationary world zone.

One of the arguments *\Temp* or *\Stat* must be specified.

WorldZone Data type: *wztemporary*

Variable or persistent variable that will be updated with the identity (numeric value) of the world zone.

If use of switch *\Temp*, the data type must be *wztemporary.*
If use of switch *\Stat*, the data type must be *wzstationary.*

Shape

Data type: *shapedata*

The variable that defines the volume of the world zone.

Program execution

The defined world zone is activated. From this moment the robot's TCP position is supervised. If it reaches the defined area the movement is stopped.

Example

```
VAR wzstationary box1_invers;
VAR wzstationary box2;

PROC wzone_power_on()
   VAR shapedata volume;
   CONST pos box1_c1:=[500,-500,0];
   CONST pos box1_c2:=[-500,500,500];
   CONST pos box2_c1:=[500,-500,0];
   CONST pos box2_c2:=[200,-200,300];
   ...
   WZBoxDef \Outside, volume, box1_c1, box1_c2;
   WZLimSup \Stat, box1_invers, volume;
   WZBoxDef \Inside, volume, box2_c1, box2_c2;
   WZLimSup \Stat, box2, volume;
ENDPROC
```

Limitation of work area for the robot with the following stationary world zones:

- Outside working area when outside box1_invers
- Outside working area when inside box2

If this routine is connected to the system event POWER ON, these world zones will always be active in the system, both for program movements and manual jogging.

Limitations

A world zone cannot be redefined using the same variable in argument *WorldZone*.

A stationary world zone cannot be deactivated, activated again or erased in the RAPID program.

A temporary world zone can be deactivated (*WZDisable*), activated again (*WZEnable*) or erased (*WZFree)* in the RAPID program.

Syntax

```
WZLimSup
    ['\'Temp] | ['\Stat]','
    [WorldZone':=']<variable or persistent (INOUT) of wztemporary>','
    [Shape':='] <variable (VAR) of shapedata>';'
```

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Temporary world zone	Data Types - *wztemporary*
Stationary world zone	Data Types - *wzstationary*
Define straight box-shaped world zone	Instructions - *WZBoxDef*
Define sphere-shaped world zone	Instructions - *WZSphDef*
Define cylinder-shaped world zone	Instructions - *WZCylDef*
Activate world zone digital output set	Instructions - *WZDOSet*

WZSphDef Define a sphere-shaped world zone

WZSphDef (World Zone Sphere Definition) is used to define a world zone that has the shape of a sphere.

Example




```
VAR shapedata volume;
CONST pos C1:=[300,300,200];
CONST num R1:=200;
...
WZSphDef \Inside, volume, C1, R1;
```

Define a sphere named *volume* by its centre *C1* and its radius *R1*.

Arguments

WZSphDef [\Inside] | [\Outside] Shape CentrePoint Radius

\Inside Data type: *switch*

Define the volume inside the sphere.

\Outside Data type: *switch*

Define the volume outside the sphere (inverse volume).

One of the arguments *Inside* or *Outside* must be specified.

Shape Data type: *shapedata*

Variable for storage of the defined volume (private data for the system).

CentrePoint

Data type: *pos*

Position (x,y,z) in mm defining the centre of the sphere.

Radius

Data type: *num*

The radius of the sphere in mm.

Program execution

The definition of the sphere is stored in the variable of type *shapedata* (argument *Shape*), for future use in *WZLimSup* or *WZDOSet* instructions.

Limitations

If the robot is used to point out the *CentrePoint*, work object *wobj0* must be active (use of component *trans* in *robtarget* e.g. p1.trans as argument).

Syntax

WZSphDef
[’\’Inside] | [’\’Outside]’,’
[Shape’:=’]<variable (**VAR**) of *shapedata*>’,’
[CentrePoint’:=’]<expression (**IN**) of *pos*>’,’
[Radius’:=’]<expression (**IN**) of *num*>’;’

Related information

	Described in:
World Zones	Motion and I/O Principles - *World Zones*
World zone shape	Data Types - *shapedata*
Define box-shaped world zone	Instructions - *WZBoxDef*
Define cylinder-shaped world zone	Instructions - *WZCylDef*
Activate world zone limit supervision	Instructions - *WZLimSup*
Activate world zone digital output set	Instructions - *WZDOSet*

CONTENTS

Offs	Displaces a robot position
OpMode	Read the operating mode
OrientZYX	Builds an orient from Euler angles
ORobT	Removes a program displacement from a position
PoseInv	Inverts the pose
PoseMult	Multiplies pose data
PoseVect	Applies a transformation to a vector
Pow	Calculates the power of a value
Present	Tests if an optional parameter is used
ReadBin	Reads a byte from a file or serial channel
ReadMotor	Reads the current motor angles
ReadNum	Reads a number from a file or serial channel
ReadStr	Reads a string from a file or serial channel
ReadStrBin	Reads a string from a binary serial channel or file
RelTool	Make a displacement relative to the tool
Round	Round is a numeric value
RunMode	Read the running mode
Sin	Calculates the sine value
Sqrt	Calculates the square root value
StrFind	Searches for a character in a string
StrLen	Gets the string length
StrMap	Maps a string
StrMatch	Search for pattern in string
StrMemb	Checks if a character belongs to a set
StrOrder	Checks if strings are ordered
StrPart	Finds a part of a string
StrToByte	Converts a string to a byte data
StrToVal	Converts a string to a value
Tan	Calculates the tangent value
TestAndSet	Test variable and set if unset
TestDI	Tests if a digital input is set
Trunc	Truncates a numeric value
ValToStr	Converts a value to a string
VectMagn	Magnitude of a pos vector

Abs Gets the absolute value

Abs is used to get the absolute value, i.e. a positive value of numeric data.

Example

```
reg1 := Abs(reg2);
```

Reg1 is assigned the absolute value of *reg2*.

Return value

Data type: *num*

The absolute value, i.e. a positive numeric value.

e.g.	Input value	Returned value
	3	3
	-3	3
	-2.53	2.53

Arguments

Abs (Input)

Input

Data type: *num*

The input value.

Example

```
TPReadNum no_of_parts, "How many parts should be produced? ";
no_of_parts := Abs(no_of_parts);
```

The operator is asked to input the number of parts to be produced. To ensure that the value is greater than zero, the value given by the operator is made positive.

Syntax

```
Abs '('
   [ Input ':=' ] < expression (IN) of num > ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

ACos Calculates the arc cosine value

ACos (Arc Cosine) is used to calculate the arc cosine value.

Example

```
VAR num angle;
VAR num value;
.
.
angle := ACos(value);
```

Return value

Data type: *num*

The arc cosine value, expressed in degrees, range [0, 180].

Arguments

ACos (Value)

Value Data type: *num*

The argument value, range [-1, 1].

Syntax

```
Acos'('
   [Value ':='] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

AOutput Reads the value of an analog output signal

AOutput is used to read the current value of an analog output signal.

Example

IF AOutput(ao4) > 5 THEN ...

If the current value of the signal *ao4* is greater than *5*, then ...

Return value

Data type: *num*

The current value of the signal.

The current value is scaled (in accordance with the system parameters) before it is read by the RAPID program. See Figure 34.



Figure 34 Diagram of how analog signal values are scaled.

Arguments

AOutput (Signal)

Signal Data type: *signalao*

The name of the analog output to be read.

Syntax

AOutput ’(’
[Signal ’:=’] < variable (**VAR**) of *signalao* > ’)’

A function with a return value of data type *num*.

Related information

	Described in:
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User’s Guide - *System Parameters*

ASin Calculates the arc sine value

ASin (Arc Sine) is used to calculate the arc sine value.

Example

```
VAR num angle;
VAR num value;
.
.
angle := ASin(value);
```

Return value

Data type: *num*

The arc sine value, expressed in degrees, range [-90, 90].

Arguments

ASin (Value)

Value Data type: *num*

The argument value, range [-1, 1].

Syntax

```
ASin'('
   [Value ':='] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

ATan Calculates the arc tangent value

ATan (Arc Tangent) is used to calculate the arc tangent value.

Example

```
VAR num angle;
VAR num value;
.
.
angle := ATan(value);
```

Return value

Data type: *num*

The arc tangent value, expressed in degrees, range [-90, 90].

Arguments

ATan (Value)

Value Data type: *num*

The argument value.

Syntax

```
ATan'('
   [Value ':='] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*
Arc tangent with a return value in the range [-180, 180]	Functions - *ATan2*

ATan2 Calculates the arc tangent2 value

ATan2 (Arc Tangent2) is used to calculate the arc tangent2 value.

Example

```
VAR num angle;
VAR num x_value;
VAR num y_value;
.
.
angle := ATan2(y_value, x_value);
```

Return value

Data type: *num*

The arc tangent value, expressed in degrees, range [-180, 180].

The value will be equal to ATan(y/x), but in the range [-180, 180], since the function uses the sign of both arguments to determine the quadrant of the return value.

Arguments

ATan2 (Y X)

Y Data type: *num*

The numerator argument value.

X Data type: *num*

The denominator argument value.

Syntax

ATan2’(’
[Y ’:=’] <expression (**IN**) of *num*> ’,’
[X ’:=’] <expression (**IN**) of *num*>
’)’

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*
Arc tangent with only one argument	Functions - *ATan*

ByteToStr Converts a byte to a string data

ByteToStr (Byte To String) is used to convert a *byte* into a *string* data with a defined byte data format.

Example

```
VAR string con_data_buffer{5};
VAR byte data1 := 122;

con_data_buffer{1} := ByteToStr(data1);
```

The content of the array component *con_data_buffer{1}* will be "122" after the *ByteToStr* ... function.

```
con_data_buffer{2} := ByteToStr(data1\Hex);
```

The content of the array component *con_data_buffer{2}* will be "7A" after the *ByteToStr* ... function.

```
con_data_buffer{3} := ByteToStr(data1\Okt);
```

The content of the array component *con_data_buffer{3}* will be "172" after the *ByteToStr* ... function.

```
con_data_buffer{4} := ByteToStr(data1\Bin);
```

The content of the array component *con_data_buffer{4}* will be "01111010"after the *ByteToStr* ... function.

```
con_data_buffer{5} := ByteToStr(data1\Char);
```

The content of the array component *con_data_buffer{5}* will be "z" after the *ByteToStr* ... function.

Return value

Data type: *string*

The result of the conversion operation with the following format:

Format:	Characters:	String length:	Range:
Dec .....:	’0’ - ’9’	1-3	"0" - "255"
Hex .....:	’0’ - ’9’, ’A’ -’F’	2	"00" - "FF"
Okt .....:	’0’ - ’7’	3	"000" - "377"
Bin .....:	’0’ - ’1’	8	"00000000" - "11111111"
Char ....:	Any ASCII char (*)	1	One ASCII char

(*) If non-writable ASCII char, the return format will be RAPID character code format (e.g. “\07” for BEL control character).

Arguments

ByteToStr (BitData [\Hex] | [\Oct] | [\Bin] | [\Char])

BitData Data type: *byte*

The bit data to be converted.

If the optional switch argument is omitted, the data will be converted in *decimal* (Dec) format.

[\Hex] *(Hexadecimal)* Data type: *switch*

The data will be converted in *hexadecimal* format.

[\Okt] *(Octal)* Data type: *switch*

The data will be converted in *octal* format.

[\Bin] *(Binary)* Data type: *switch*

The data will be converted in *binary* format.

[\Char] *(Character)* Data type: *switch*

The data will be converted in *ASCII character* format.

Limitations

The range for a data type *byte* is 0 to 255 decimal.

Syntax

```
ByteToStr'('
   [BitData ':='] <expression (IN) of byte>
   ['\' Hex ] | ['\' Okt] | ['\' Bin] | ['\' Char]
   ')' ';'
```

A function with a return value of the data type *string*.

Related information

	Described in:
Convert a string to a byte data	Instructions - *StrToByte*
Other bit (byte) functions	RAPID Summary - Bit Functions
Other string functions	RAPID Summary - String Functions

CalcJointT Calculates joint angles from robtarget

CalcJointT (Calculate Joint Target) is used to calculate joint angles of the robot axes and external axes from a specified *robtarget* data.

The input *robtarget* data should be specified in the same coordinate system as specified in argument for *Tool*, *WObj* and at execution time active program displacement (*ProgDisp*) and external axes offset (*EOffs)*.
The returned *jointtarget* data is expressed in the calibration coordinate system.

Example

```
VAR jointtarget jointpos1;
CONST robtarget p1 := [...];

jointpos1 := CalcJointT(p1, tool1 \WObj:=wobj1);
```

The jointtarget value corresponding to the robtarget value *p1* is stored in *jointpos1*. The tool *tool1* and work object *wobj1* are used for calculating the joint angles *jointpos1*.

Return value

Data type: *jointtarget*

The angles in degrees for the axes of the robot on the arm side.

The values for the external axes, in mm for linear axes, in degrees for rotational axes.

The returned values are always related to the calibration position.

Arguments

CalcJointT (Rob_target Tool [\WObj])

Rob_target

Data type: *robtarget*

The position of the robot and external axes in the outermost coordinate system, related to the specified tool and work object and at execution time active program displacement (*ProgDisp*) and/or external axes offset (*EOffs*).

Tool

Data type: *tooldata*

The tool used for calculation of the robot joint angles.

[\WObj] **(***Work Object***)** Data type: *wobjdata*

The work object (coordinate system) to which the robot position is related.

If this argument is omitted the work object *wobj0* is used.
This argument must be specified when using stationary tool, coordinated external axes, or conveyor

Program execution

The returned *jointtarget* is calculated from the input *robtarget*.
To calculate the robot joint angles, the specified *Tool*, *WObj* (including coordinated user frame) and the *ProgDisp* active at execution time, are taken into consideration.
To calculate the external axis position at the execution time, active *EOffs* is taken into consideration.

The calculation always selects the robot configuration according to the specified configuration data in the input *robtarget* data. Instructions *ConfL* and *ConfJ* do not affect this calculation principle. When wrist singularity is used, robot axis 4 will be set to 0 degrees.

If there is any active program displacement (*ProgDisp*) and/or external axis offset (*EOffs*) at the time *the robtarget* is stored, then the same program displacement and/or external axis offset must be active when *CalcJointT* is executed.

Syntax

CalcJointT'('
[Rob_target ':='] <expression (**IN**) of *robtarget*> ','
[Tool ':='] <persistent (**PERS**) of *tooldata*>
['\'WObj ':=' <persistent (**PERS**) of *wobjdata*>] ')'

A function with a return value of the data type *jointtarget*.

Related information

	Described in:
Calculate robtarget from jointtarget	Functions - *CalcRobT*
Definition of position	Data Types - *robtarget*
Definition of joint position	Data Types - *jointtarget*
Definition of tools	Data Types- *tooldata*
Definition of work objects	Data Types - *wobjdata*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Program displacement coordinate system	Instructions - *PDispOn*
External axis offset coordinate system	Instructions - *EOffsOn*

CalcRobT Calculates robtarget from jointtarget

CalcRobT (Calculate Robot Target) is used to calculate a robtarget data from a given jointtarget data.

This function returns a *robtarget* value with position (x, y, z), orientation (q1 ... q4), robot axes configuration and external axes position.

The input *jointtarget* data should be specified in the calibration coordinate system. The returned *robtarget* data is expressed in the outermost coordinate system, taking the specified tool, work object and at execution time active program displacement (*ProgDisp*) and external axis offset (*EOffs*) into consideration.

Example

```
VAR robtarget p1;
CONST jointtarget jointpos1 := [...];

p1 := CalcRobT(jointpos1, tool1 \WObj:=wobj1);
```

The robtarget value corresponding to the jointtarget value *jointpos1* is stored in *p1*. The tool *tool1* and work object *wobj1* are used for calculating of the position *p1*.

Return value

Data type: *robtarget*

The robot and external axis position is returned in data type *robtarget* and expressed in the outermost coordinate system, taking the specified tool, work object and at execution time active program displacement (*ProgDisp*) and external axes offset (*EOffs*) into consideration.

If there is no active *ProgDisp,* the robot position is expressed in the object coordinate system.
If there are no active *EOffs,* the external axis position is expressed in the calibration coordinate system.

Arguments

CalcRobT (Joint_target Tool [\WObj])

Joint_target Data type: *jointtarget*

The joint position for the robot axes and external axes related to the calibration coordinate system.

Tool Data type: *tooldata*

The tool used for calculation of the robot position.

[\WObj] **(***Work Object***)** Data type: *wobjdata*

The work object (coordinate system) to which the robot position returned by the function is related.

If this argument is omitted the work object *wobj0* is used.
This argument must be specified when using stationary tool, coordinated external axes, or conveyor.

Program execution

The returned *robtarget* is calculated from the input *jointtarget*.
To calculate the cartesian robot position, the specified *Tool*, *WObj* (including coordinated user frame) and at the execution time active *ProgDisp* are taken into consideration.
To calculate the external axes position, the *EOffs* active at execution time is taken into consideration.

Syntax

CalcRobT'('
[Joint_target ':='] <expression (**IN**) of *jointtarget*> ','
[Tool ':='] <persistent (**PERS**) of *tooldata*>
['\'WObj ':=' <persistent (**PERS**) of *wobjdata*>] ')'

A function with a return value of the data type *robtarget*.

Related information

	Described in:
Calculate jointtarget from robtarget	Functions - *CalcJointT*
Definition of position	Data Types - *robtarget*
Definition of joint position	Data Types - *jointtarget*
Definition of tools	Data Types- *tooldata*
Definition of work objects	Data Types - *wobjdata*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Program displacement coordinate system	Instructions - *PDispOn*
External axes offset coordinate system	Instructions - *EOffsOn*

CDate Reads the current date as a string

CDate (Current Date) is used to read the current system date.

This function can be used to present the current date to the operator on the teach pendant display or to paste the current date into a text file that the program writes to.

Example

```
VAR string date;

date := CDate();
```

The current date is stored in the variable *date*.

Return value

Data type: *string*

The current date in a string.

The standard date format is “year-month-day”, e.g. ”1998-01-29”.

Example

```
date := CDate();
TPWrite “The current date is: “+date;
Write logfile, date;
```

The current date is written to the teach pendant display and into a text file.

Syntax

```
CDate ’(’ ’)’
```

A function with a return value of the type *string*.

Related Information

	Described in:
Time instructions	RAPID Summary - *System & Time*
Setting the system clock	User’s Guide - *Service*

CJointT Reads the current joint angles

CJointT (Current Joint Target) is used to read the current angles of the robot axes and external axes.

Example

VAR jointtarget joints;

joints := CJointT();

The current angles of the axes for the robot and external axes are stored in *joints*.

Return value

Data type: *jointtarget*

The current angles in degrees for the axes of the robot on the arm side.

The current values for the external axes, in mm for linear axes, in degrees for rotational axes.

The returned values are related to the calibration position.

Syntax

CJointT’(’’)’

A function with a return value of the data type *jointtarget.*

Related information

	Described in:
Definition of joint	Data Types - *jointtarget*
Reading the current motor angle	Functions - *ReadMotor*

ClkRead Reads a clock used for timing

ClkRead is used to read a clock that functions as a stop-watch used for timing.

Example

```
reg1:=ClkRead(clock1);
```

The clock *clock1* is read and the time in seconds is stored in the variable *reg1*.

Return value

Data type: *num*

The time in seconds stored in the clock. Resolution 0.010 seconds.

Argument

ClkRead (Clock)

Clock Data type: *clock*

The name of the clock to read.

Program execution

A clock can be read when it is stopped or running.

Once a clock is read it can be read again, started again, stopped or reset.

If the clock has overflowed, program execution is stopped with an error message.

Syntax

```
ClkRead '('
   [ Clock ':=' ] < variable (VAR) of clock > ')'
```

A function with a return value of the type *num*.

Related Information

	Described in:
Clock instructions	RAPID Summary - *System & Time*
Clock overflow	Data Types - *clock*
More examples	Instructions - *ClkStart*

Cos Calculates the cosine value

Cos (Cosine) is used to calculate the cosine value from an angle value.

Example

```
VAR num angle;
VAR num value;
.
.
value := Cos(angle);
```

Return value

Data type: *num*

The cosine value, range = [-1, 1] .

Arguments

Cos (Angle)

Angle Data type: *num*

The angle value, expressed in degrees.

Syntax

```
Cos'('
   [Angle ':='] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

CPos Reads the current position (pos) data

CPos (Current Position) is used to read the current position of the robot.

This function returns the x, y, and z values of the robot TCP as data of type *pos*. If the complete robot position (*robtarget*) is to be read, use the function *CRobT* instead.

Example

VAR pos pos1;

pos1 := CPos(\Tool:=tool1 \WObj:=wobj0);

The current position of the robot TCP is stored in variable *pos1*. The tool *tool1* and work object *wobj0* are used for calculating the position.

Return value

Data type: *pos*

The current position (pos) of the robot with x, y, and z in the outermost coordinate system, taking the specified tool, work object and active ProgDisp coordinate system into consideration.

Arguments

CPos ([\Tool] [\WObj])

[\Tool] Data type: *tooldata*

The tool used for calculation of the current robot position.

If this argument is omitted the current active tool is used.

[\WObj] **(***Work Object***)** Data type: *wobjdata*

The work object (coordinate system) to which the current robot position returned by the function is related.

If this argument is omitted the current active work object is used.

When programming, it is very sensible to always specify arguments \Tool and \WObj. The function will always then return the wanted position, although some other tool or work object has been activated manually.

Program execution

The coordinates returned represent the TCP position in the ProgDisp coordinate system.

Example

```
VAR pos pos2;
VAR pos pos3;
VAR pos pos4;

pos2 := CPos(\Tool:=grip3 \WObj:=fixture);
.
.
pos3 := CPos(\Tool:=grip3 \WObj:=fixture);
pos4 := pos3-pos2;
```

The x, y, and z position of the robot is captured at two places within the program using the *CPos* function. The tool *grip3* and work object *fixture* are used for calculating the position. The x, y and z distances travelled between these positions are then calculated and stored in the *pos* variable *pos4*.

Syntax

CPos ’(’
[’\’Tool ’:=’ <persistent (**PERS**) of *tooldata*>]
[’\’WObj ’:=’ <persistent (**PERS**) of *wobjdata*>] ’)’

A function with a return value of the data type *pos*.

Related information

	Described in:
Definition of position	Data Types - *pos*
Definition of tools	Data Types- *tooldata*
Definition of work objects	Data Types - *wobjdata*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
Reading the current *robtarget*	Functions - *CRobT*

CRobT Reads the current position (robtarget) data

CRobT (Current Robot Target) is used to read the current position of the robot and external axes.

This function returns a *robtarget* value with position (x, y, z), orientation (q1 ... q4), robot axes configuration and external axes position. If only the x, y, and z values of the robot TCP (*pos*) are to be read, use the function *CPos* instead.

Example

```
VAR robtarget p1;

p1 := CRobT(\Tool:=tool1 \WObj:=wobj0);
```

The current position of the robot and external axes is stored in *p1*. The tool *tool1* and work object *wobj0* are used for calculating the position.

Return value

Data type: *robtarget*

The current position of the robot and external axes in the outermost coordinate system, taking the specified tool, work object and active ProgDisp/ExtOffs coordinate system into consideration.

Arguments

CRobT ([\Tool] [\WObj])

[\Tool] Data type: *tooldata*

The tool used for calculation of the current robot position.

If this argument is omitted the current active tool is used.

[\WObj] **(***Work Object***)** Data type: *wobjdata*

The work object (coordinate system) to which the current robot position returned by the function is related.

If this argument is omitted the current active work object is used.

When programming, it is very sensible to always specify arguments \Tool and \WObj. The function will always then return the wanted position, although some other tool or work object has been activated manually.

Program execution

The coordinates returned represent the TCP position in the ProgDisp coordinate system. External axes are represented in the ExtOffs coordinate system.

Example

```
VAR robtarget p2;

p2 := ORobT( RobT(\Tool:=grip3 \WObj:=fixture) );
```

The current position in the object coordinate system (without any ProgDisp or ExtOffs) of the robot and external axes is stored in *p2*. The tool *grip3* and work object *fixture* are used for calculating the position.

Syntax

CRobT'('
['\'Tool ':=' <persistent (**PERS**) of *tooldata*>]
['\'WObj ':=' <persistent (**PERS**) of *wobjdata*>] ')'

A function with a return value of the data type *robtarget*.

Related information

	Described in:
Definition of position	Data Types - *robtarget*
Definition of tools	Data Types- *tooldata*
Definition of work objects	Data Types - *wobjdata*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*
ExtOffs coordinate system	Instructions - *EOffsOn*
Reading the current *pos* (x, y, z only)	Functions - *CPos*

CTime Reads the current time as a string

CTime is used to read the current system time.

This function can be used to present the current time to the operator on the teach pendant display or to paste the current time into a text file that the program writes to.

Example

```
VAR string time;

time := CTime();
```

The current time is stored in the variable *time*.

Return value

Data type: *string*

The current time in a string.

The standard time format is "hours:minutes:seconds", e.g. "18:20:46".

Example

```
time := CTime();
TPWrite “The current time is: “+time;
Write logfile, time;
```

The current time is written to the teach pendant display and written into a text file.

Syntax

```
CTime ’(’ ’)’
```

A function with a return value of the type *string*.

Related Information

	Described in:
Time and date instructions	RAPID Summary - *System & Time*
Setting the system clock	User’s Guide - *System Parameters*

CTool Reads the current tool data

CTool (Current Tool) is used to read the data of the current tool.

Example

```
PERS tooldata temp_tool:= [ TRUE, [ [0, 0, 0], [1, 0, 0 ,0] ],
                              [0.001, [0, 0, 0.001], [1, 0, 0, 0], 0, 0, 0] ];

temp_tool := CTool();
```

The value of the current tool is stored in the variable *temp_tool*.

Return value

Data type: *tooldata*

This function returns a *tooldata* value holding the value of the current tool, i.e. the tool last used in a movement instruction.

The value returned represents the TCP position and orientation in the wrist centre coordinate system, see *tooldata*.

Syntax

CTool’(’’)’

A function with a return value of the data type *tooldata*.

Related information

	Described in:
Definition of tools	Data Types- *tooldata*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*

CWObj Reads the current work object data

CWObj (Current Work Object) is used to read the data of the current work object.

Example

```
PERS wobjdata temp_wobj;

temp_wobj := CWObj();
```

The value of the current work object is stored in the variable *temp_wobj*.

Return value

Data type: *wobjdata*

This function returns a *wobjdata* value holding the value of the current work object, i.e. the work object last used in a movement instruction.

The value returned represents the work object position and orientation in the world coordinate system, see *wobjdata*.

Syntax

```
CWObj'('')'
```

A function with a return value of the data type *wobjdata*.

Related information

	Described in:
Definition of work objects	Data Types- *wobjdata*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*

DefDFrame Define a displacement frame

DefDFrame (Define Displacement Frame) is used to calculate a displacement frame from three original positions and three displaced positions.

Example



Three positions, *p1- p3,* related to an object in an original position, have been stored. After a displacement of the object the same positions are searched for and stored as *p4-p6*. From these six positions the displacement frame is calculated. Then the calculated frame is used to displace all the stored positions in the program.

```
CONST robtarget p1 := [...];
CONST robtarget p2 := [...];
CONST robtarget p3 := [...];
VAR robtarget p4;
VAR robtarget p5;
VAR robtarget p6;
VAR pose frame1;
.
!Search for the new positions
SearchL sen1, p4, *, v50, tool1;
.
SearchL sen1, p5, *, v50, tool1;
.
SearchL sen1, p6, *, v50, tool1;
frame1 := DefDframe (p1, p2, p3, p4, p5, p6);
.
!activation of the displacement defined by frame1
PDispSet frame1;
```

Return value

Data type: *pose*

The displacement frame.

Arguments

DefDFrame (OldP1 OldP2 OldP3 NewP1 NewP2 NewP3)

OldP1 Data type: *robtarget*

The first original position.

OldP2 Data type: *robtarget*

The second original position.

OldP3 Data type: *robtarget*

The third original position.

NewP1 Data type: *robtarget*

The first displaced position. This position must be measured and determined with great accuracy.

NewP2 Data type: *robtarget*

The second displaced position. It should be noted that this position can be measured and determined with less accuracy in one direction, e.g. this position must be placed on a line describing the new direction of *p1* to *p2*.

NewP3 Data type: *robtarget*

The third displaced position. This position can be measured and determined with less accuracy in two directions, e.g. it has to be placed in a plane describing the new plane of *p1*, *p2* and *p3*.

Error handling

If it is not possible to calculate the frame because of bad accuracy in the positions, the system variable ERRNO is set to ERR_FRAME. This error can then be handled in the error handler.

Syntax

DefDFrame'('
[OldP1 ':='] <expression (**IN**) of *robtarget*> ','
[OldP2 ':='] <expression (**IN**) of *robtarget*> ','
[OldP3 ':='] <expression (**IN**) of *robtarget*> ','
[NewP1 ':='] <expression (**IN**) of *robtarget*> ','
[NewP2 ':='] <expression (**IN**) of *robtarget*> ','
[NewP3 ':='] <expression (**IN**) of *robtarget*> ')'

A function with a return value of the data type *pose*.

Related information

	Described in:
Activation of displacement frame	Instructions - *PDispSet*
Manual definition of displacement frame	User’s Guide - *Calibration*

DefFrame Define a frame

DefFrame (Define Frame) is used to calculate a frame, from three positions defining the frame.

Example



Three positions, *p1- p3*, related to the object coordinate system, are used to define the new coordinate system, *frame1*. The first position, *p1*, is defining the origin of *frame1*, the second position, *p2*, is defining the direction of the x-axis and the third position, *p3*, is defining the location of the xy-plane. The defined *frame1* may be used as a displacement frame, as shown in the example below:

```
CONST robtarget p1 := [...];
CONST robtarget p2 := [...];
CONST robtarget p3 := [...];
VAR pose frame1;
.
.
frame1 := DefFrame (p1, p2, p3);
.
.
!activation of the displacement defined by frame1
PDispSet frame1;
```

Return value

Data type: *pose*

The calculated frame.

The calculation is related to the active object coordinate system.

Arguments

DefFrame (NewP1 NewP2 NewP3 [\Origin])

NewP1 Data type: *robtarget*

The first position, which will define the origin of the new frame.

NewP2 Data type: *robtarget*

The second position, which will define the direction of the x-axis of the new frame.

NewP3 Data type: *robtarget*

The third position, which will define the xy-plane of the new frame. The position of point 3 will be on the positive y side, see the figure above.

[**Origin**] Data type: *num*

Optional argument, which will define how the origin of the frame will be placed. Origin = 1, means that the origin is placed in NewP1, i.e. the same as if this argument is omitted. Origin = 2 means that the origin is placed in NewP2, see the figure below.



Origin = 3 means that the origin is placed on the line going through NewP1 and NewP2 and so that NewP3 will be placed on the y axis, see the figure below.



Other values, or if Origin is omitted, will place the origin in NewP1.

Limitations

The three positions *p1 - p3*, defining the frame, must define a well shaped triangle. The most well shaped triangle is the one with all sides of equal length.



The triangle is not considered to be well shaped if the angle θ a is too small. The angle θ is too small if:

$$|\cos\Theta| < 1 - 10^{-4}$$

The triangle *p1, p2, p3* must not be too small, i.e. the positions cannot be too close. The distances between the positions *p1 - p2* and *p1 - p3* must not be less than 0.1 mm.

Error handling

If the frame cannot be calculated because of the above limitations, the system variable ERRNO is set to ERR_FRAME. This error can then be handled in the error handler.

Syntax

DefFrame’(’
[NewP1 ’:=’] <expression (**IN**) of *robtarget*> ’,’
[NewP2 ’:=’] <expression (**IN**) of *robtarget*> ’,’
[NewP3 ’:=’] <expression (**IN**) of *robtarget*>
[’\’Origin ’:=’ <expression (**IN**) of *num*>]’)’

A function with a return value of the data type *pose*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*
Activation of displacement frame	Instructions - *PDispSet*

Dim Obtains the size of an array

Dim (Dimension) is used to obtain the number of elements in an array.

Example

```
PROC arrmul(VAR num array{*}, num factor)

  FOR index FROM 1 TO Dim(array, 1) DO
    array{index} := array{index} * factor;
  ENDFOR

ENDPROC
```

All elements of a num array are multiplied by a factor.
This procedure can take any one-dimensional array of data type *num* as an input.

Return value

Data type: *num*

The number of array elements of the specified dimension.

Arguments

Dim (ArrPar DimNo)

ArrPar *(Array Parameter)* Data type: Any type

The name of the array.

DimNo *(Dimension Number)* Data type: *num*

The desired array dimension: 1 = first dimension
2 = second dimension
3 = third dimension

Example

```
PROC add_matrix(VAR num array1{*,*,*}, num array2{*,*,*})

   IF Dim(array1,1) <> Dim(array2,1) OR Dim(array1,2) <> Dim(array2,2) OR
      Dim(array1,3) <> Dim(array2,3) THEN
      TPWrite "The size of the matrices are not the same";
      Stop;
   ELSE
      FOR i1 FROM 1 TO Dim(array1, 1) DO
         FOR i2 FROM 1 TO Dim(array1, 2) DO
            FOR i3 FROM 1 TO Dim(array1, 3) DO
               array1{i1,i2,i3} := array1{i1,i2,i3} + array2{i1,i2,i3};
            ENDFOR
         ENDFOR
      ENDFOR
   ENDIF
   RETURN;

ENDPROC
```

Two matrices are added. If the size of the matrices differs, the program stops and an error message appears.
This procedure can take any three-dimensional arrays of data type *num* as an input.

Syntax

```
Dim '('
   [ArrPar':='] <reference (REF) of any type> ','
   [DimNo':='] <expression (IN) of num> ')'
```

A REF parameter requires that the corresponding argument be either a constant, a variable or an entire persistent. The argument could also be an IN parameter, a VAR parameter or an entire PERS parameter.

A function with a return value of the data type *num*.

Related information

	Described in:
Array parameters	Basic Characteristics - *Routines*
Array declaration	Basic Characteristics - *Data*

Distance Distance between two points

Distance is used to calculate the distance between two points in the space.

Example




```
VAR num dist;
CONST pos p1 := [4,0,4];
CONST pos p2 := [-4,4,4];
...
dist := Distance(p1, p2);
```

The distance in space between the points *p1* and *p2* is calculated and stored in the variable *dist*.

Return value

Data type: *num*

The distance (always positive) between the points.

Arguments

Distance (Point1 Point2)

Point1 Data type: *pos*

The first point described by the *pos* data type.

Point2 Data type: *pos*

The second point described by the *pos* data type.

Program execution

Calculation of the distance between the two points:



$$\text{distance} = \sqrt{\left((x_1 - x_2)^2 + (y_1 - y_2)^2 + (z_1 - z_2)^2\right)}$$

Syntax

Distance’(’
[Point1 ’:=’] <expression (**IN**) of *pos*> ’,’
[Point2 ’:=’] <expression (**IN**) of *pos*>
’)’

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*
Definition of pos	Data Type - *pos*

DotProd Dot product of two pos vectors

DotProd (Dot Product) is used to calculate the dot (or scalar) product of two pos vectors. The typical use is to calculate the projection of one vector upon the other or to calculate the angle between the two vectors.

Example



The dot or scalar product of two vectors **A** and **B** is a scalar, which equals the products of the magnitudes of **A** and **B** and the cosine of the angle between them.

$$A \cdot B = |A||B|\cos\theta_{AB}$$

The dot product:

- is less than or equal to the product of their magnitudes.
- can be either a positive or a negative quantity, depending whether the angle between them is smaller or larger then 90 degrees.
- is equal to the product of the magnitude of one vector and the projection of the other vector upon the first one.
- is zero when the vectors are perpendicular to each other.

The vectors are described by the data type *pos* and the dot product by the data type *num*:

```
VAR num dotprod;
VAR pos vector1;
VAR pos vector2;
.
.
vector1 := [1,1,1];
vector2 := [1,2,3];
dotprod := DotProd(vector1, vector2);
```

Return value

Data type: *num*

The value of the dot product of the two vectors.

Arguments

DotProd (Vector1 Vector2)

Vector1 Data type: *pos*

The first vector described by the *pos* data type.

Vector2 Data type: *pos*

The second vector described by the *pos* data type.

Syntax

```
DotProd'('
    [Vector1 ':='] <expression (IN) of pos> ','
    [Vector2 ':='] <expression (IN) of pos>
    ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

DOutput Reads the value of a digital output signal

DOutput is used to read the current value of a digital output signal.

Example

IF DOutput(do2) = 1 THEN . . .

If the current value of the signal *do2* is equal to *1*, then . . .

Return value

Data type: *dionum*

The current value of the signal (0 or 1).

Arguments

DOutput (Signal)

Signal Data type: *signaldo*

The name of the signal to be read.

Program execution

The value read depends on the configuration of the signal. If the signal is inverted in the system parameters, the value returned by this function is the opposite of the true value of the physical channel.

Example

IF DOutput(auto_on) <> active THEN . . .

If the current value of the system signal *auto_on* is *not active*, then ..., i.e. if the robot is in the manual operating mode, then ... Note that the signal must first be defined as a system output in the system parameters.

Syntax

DOutput ’(’
[Signal ’:=’] < variable (**VAR**) of *signaldo* > ’)’

A function with a return value of the data type *dionum*.

Related information

	Described in:
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User’s Guide - *System Parameters*

EulerZYX Gets Euler angles from orient

EulerZYX (Euler ZYX rotations) is used to get an Euler angle component from an orient type variable.

Example

```
VAR num anglex;
VAR num angley;
VAR num anglez;
VAR pose object;
.
.
anglex := GetEuler(\X, object.rot);
angley := GetEuler(\Y, object.rot);
anglez := GetEuler(\Z, object.rot);
```

Return value

Data type: *num*

The corresponding Euler angle, expressed in degrees, range [-180, 180].

Arguments

EulerZYX ([\X] | [\Y] | [\Z] Rotation)

The arguments \X, \Y and \Z are mutually exclusive. If none of these are specified, a run-time error is generated.

[\X] Data type: *switch*

Gets the rotation around the X axis.

[**Y**] Data type: *switch*

Gets the rotation around the Y axis.

[**Z**] Data type: *switch*

Gets the rotation around the Z axis.

Rotation Data type: *orient*

The rotation in its quaternion representation.

Syntax

```
EulerZYX’(’
    [’\’X ’,’] | [’\’Y ’,’] | [’\’Z ’,’]
    [Rotation ’:=’] <expression (IN) of orient>
    ’)’
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

Exp Calculates the exponential value

Exp (Exponential) is used to calculate the exponential value, e^x.

Example

```
VAR num x;
VAR num value;
.
.
value:= Exp( x);
```

Return value

Data type: *num*

The exponential value e^x .

Arguments

Exp (Exponent)

Exponent — Data type: *num*

The exponent argument value.

Syntax

```
Exp'('
   [Exponent ':='] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

FileTime Retrieve time information about a file

FileTime is used to retrieve the last time for modification, access or file status change of a file. The time is measured in secs since 00:00:00 GMT, Jan. 1 1970. The time is returned as a num.

Example

```
Load "ram1disk:notmymod.mod";
WHILE TRUE DO
   ! Call some routine in notmymod
   notmymodrout;
   IF FileTime("ram1disk:notmymod.mod" \ModifyTime)
      > ModTime("notmymod") THEN
      UnLoad "ram1disk:notmymod.mod";
      Load "ram1disk:notmymod.mod";
   ENDIF
ENDWHILE
```

This program reloads a module if there is a newer at the source. It uses the *ModTime* to retrieve the latest loading time for the specified module, and to compare it to the *FileTime\ModifyTime* at the source. Then, if the source is newer, the program unloads and loads the module again.

Return value

Data type: *num*

The time measured in secs since 00:00:00 GMT, Jan 1 1970.

Arguments

FileTime (Path [\ModifyTime] | [\AccessTime] | [\StatCTime])

Path Data type: *string*

The file specified with a full or relative path.

ModifyTime Data type: *switch*

Last modification time.

AccessTime Data type: *switch*

Time of last access (read, execute of modify).

StatCTime Data type: *switch*

Last file status (access qualification) change time.

Program execution

This function returns a numeric that specifies the time since the last:

- Modification
- Access
- File status change

of the specified file.

Example

This is a complete example that implements an alert service for maximum 10 files.

```
LOCAL RECORD falert
   string filename;
   num ftime;
ENDRECORD

LOCAL VAR falert myfiles[10];
LOCAL VAR num currentpos:=0;
LOCAL VAR intnum timeint;

LOCAL TRAP mytrap
   VAR num pos:=1;
   WHILE pos <= currentpos DO
      IF FileTime(myfiles{pos}.filename \ModifyTime) > myfiles{pos}.ftime THEN
         TPWrite "The file "+myfiles{pos}.filename+" is changed";
      ENDIF
      pos := pos+1;
   ENDWHILE
ENDTRAP

PROC alertInit(num freq)
   currentpos:=0;
   CONNECT timeint WITH mytrap;
   ITimer freq,timeint;
ENDPROC

PROC alertFree()
   IDelete timeint;
ENDPROC
```

```
PROC alertNew(string filename)
   currentpos := currentpos+1;
   IF currentpos <= 10 THEN
      myfiles{currentpos}.filename := filename;
      myfiles{currentpos}.ftime := FileTime (filename \ModifyTime);
   ENDIF
ENDPROC
```

Error handling

If the file does not exist, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

Syntax

FileTime ’(’
[Path ’:=’] < expression (**IN**) of *string*>
['\'ModifyTime] |
['\'AccessTime] |
['\'StatCTime] ’)’

A function with a return value of the data type *num*.

Related information

	Described in:
Last time a module was loaded	Functions - *ModTime*

GetTime Reads the current time as a numeric value

GetTime is used to read a specified component of the current system time as a numeric value.

GetTime can be used to :

- have the program perform an action at a certain time
- perform certain activities on a weekday
- abstain from performing certain activities on the weekend
- respond to errors differently depending on the time of day.

Example

```
hour := GetTime(\Hour);
```

The current hour is stored in the variable *hour.*

Return value

Data type: *num*

One of the four time components specified below.

Argument

GetTime ([\WDay] | [\Hour] | [\Min] | [\Sec])

[\WDay] Data type: *switch*

Return the current weekday.
Range: 1 to 7 (Monday to Sunday).

[\Hour] Data type: *switch*

Return the current hour.
Range: 0 to 23.

[\Min] Data type: *switch*

Return the current minute.
Range: 0 to 59.

[\Sec] Data type: *switch*

Return the current second.
Range: 0 to 59.

One of the arguments must be specified, otherwise program execution stops with an error message.

Example

```
weekday := GetTime(\WDay);
hour := GetTime(\Hour);
IF weekday < 6 AND hour >6 AND hour < 16 THEN
   production;
ELSE
   maintenance;
ENDIF
```

If it is a weekday and the time is between 7:00 and 15:59 the robot performs production. At all other times, the robot is in the maintenance mode.

Syntax

```
GetTime ’(’
   [’\’ WDay ]
   | [ ’\’ Hour ]
   | [ ’\’ Min ]
   | [ ’\’ Sec ] ’)’
```

A function with a return value of the type *num*.

Related Information

	Described in:
Time and date instructions	RAPID Summary - *System & Time*
Setting the system clock	User’s Guide - *System Parameters*

GOutput Reads the value of a group of digital output signals

G*Output* is used to read the current value of a group of digital output signals.

Example

IF GOutput(go2) = 5 THEN ...

If the current value of the signal *go2* is equal to *5*, then ...

Return value

Data type: *num*

The current value of the signal (a positive integer).

The values of each signal in the group are read and interpreted as an unsigned binary number. This binary number is then converted to an integer.

The value returned lies within a range that is dependent on the number of signals in the group.

No. of signals	Return value	No. of signals	Return value
1	0 - 1	9	0 - 511
2	0 - 3	10	0 - 1023
3	0 - 7	11	0 - 2047
4	0 - 15	12	0 - 4095
5	0 - 31	13	0 - 8191
6	0 - 63	14	0 - 16383
7	0 - 127	15	0 - 32767
8	0 - 255	16	0 - 65535

Arguments

GOutput (Signal)

Signal Data type: *signalgo*

The name of the signal group to be read.

Syntax

GOutput ’(’
[Signal ’:=’] < variable (**VAR**) of *signalgo* > ’)’

A function with a return value of data type *num*.

Related information

	Described in:
Input/Output instructions	RAPID Summary - *Input and Output Signals*
Input/Output functionality in general	Motion and I/O Principles - *I/O Principles*
Configuration of I/O	User’s Guide - *System Parameters*

IsPers Is Persistent

IsPers is used to test if a data object is a persistent variable or not.

Example

```
PROC procedure1 (INOUT num parameter1)
   IF IsVar(parameter1) THEN
      ! For this call reference to a variable
      ...
   ELSEIF IsPers(parameter1) THEN
      ! For this call reference to a persistent variable
      ...
   ELSE
      ! Should not happen
      EXIT;
   ENDIF
ENDPROC
```

The procedure *procedure1* will take different actions depending on whether the actual parameter *parameter1* is a variable or a persistent variable.

Return value

Data type: *bool*

TRUE if the tested actual INOUT parameter is a persistent variable.
FALSE if the tested actual INOUT parameter is not a persistent variable.

Arguments

IsPers (DatObj)

DatObj (*Data Object*) Data type: any type

The name of the formal INOUT parameter.

Syntax

```
IsPers'('
   [ DatObj ':=' ] < var or pers (INOUT) of any type > ')'
```

A function with a return value of the data type *bool*.

Related information

	Described in:
Test if variable	Function - *IsVar*
Types of parameters (access modes)	RAPID Characteristics - *Routines*

IsVar Is Variable

IsVar is used to test whether a data object is a variable or not.

Example

```
PROC procedure1 (INOUT num parameter1)
   IF IsVAR(parameter1) THEN
      ! For this call reference to a variable
      ...
   ELSEIF IsPers(parameter1) THEN
      ! For this call reference to a persistent variable
      ...
   ELSE
      ! Should not happen
      EXIT;
   ENDIF
ENDPROC
```

The procedure *procedure1* will take different actions, depending on whether the actual parameter *parameter1* is a variable or a persistent variable.

Return value

Data type: *bool*

TRUE if the tested actual INOUT parameter is a variable.
FALSE if the tested actual INOUT parameter is not a variable.

Arguments

IsVar (DatObj)

DatObj (*Data Object*) Data type: any type

The name of the formal INOUT parameter.

Syntax

IsVar’(’
[DatObj ’:=’] < var or pers (**INOUT**) of any type > ’)’

A function with a return value of the data type *bool*.

Related information

	Described in:
Test if persistent	Function - *IsPers*
Types of parameters (access modes)	RAPID Characteristics - *Routines*

MirPos Mirroring of a position

MirPos (Mirror Position) is used to mirror the translation and rotation parts of a position.

Example

```
CONST robtarget p1;
VAR robtarget p2;
PERS wobjdata mirror;
.
.
p2 := MirPos(p1, mirror);
```

p1 is a robtarget storing a position of the robot and an orientation of the tool. This position is mirrored in the xy-plane of the frame defined by *mirror*, relative to the world coordinate system. The result is new robtarget data, which is stored in *p2*.

Return value

Data type: *robtarget*

The new position which is the mirrored position of the input position.

Arguments

MirPos (Point MirPlane [\WObj] [\MirY])

Point Data type: *robtarget*

The input robot position. The orientation part of this position defines the current orientation of the tool coordinate system.

MirPlane *(Mirror Plane)* Data type: *wobjdata*

The work object data defining the mirror plane. The mirror plane is the xy-plane of the object frame defined in *MirPlane*. The location of the object frame is defined relative to the user frame, also defined in *MirPlane*, which in turn is defined relative to the world frame.

[**\WObj**] (*Work Object)* Data type: *wobjdata*

The work object data defining the object frame, and user frame, relative to which the input position, *Point*, is defined. If this argument is left out, the position is defined relative to the World coordinate system.
Note. If the position is created with a work object active, this work object must be referred to in the argument.

[**MirY*] *(Mirror Y)* Data type: *switch*

If this switch is left out, which is the default rule, the tool frame will be mirrored as regards the x-axis and the z-axis. If the switch is specified, the tool frame will be mirrored as regards the y-axis and the z-axis.

Limitations

No recalculation is done of the robot configuration part of the input robtarget data.

Syntax

MirPos'('
[Point ':='] < expression (**IN**) of *robtarget*>','
[MirPlane ':='] <expression (**IN**) of *wobjdata*> ','
['\'WObj ':=' <expression (**IN**) of *wobjdata*>]
['\'MirY]')'

A function with a return value of the data type *robtarget*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

ModTime Get time of load for a loaded module

ModTime (*Module Time*) is used to retrieve the time of loading a specified module. The module is specified by its name and must be in the task memory. The time is measured in secs since 00:00:00 GMT, Jan 1 1970. The time is returned as a num.

Example

```
MODULE mymod

  VAR num mytime;

  PROC printMyTime()
    mytime := ModTime("mymod");
    TPWrite "My time is "+NumToStr(mytime,0);
  ENDPROC
```

Return value

Data type: *num*

The time measured in secs since 00:00:00 GMT, Jan 1 1970.

Arguments

ModTime (Object)

Object Data type: *string*

The name of the module.

Program execution

This function return a numeric that specify the time when the module was loaded.

Example

This is a complete example that implements an “update if newer” service.

```
MODULE updmod
   PROC callrout()
      Load "ram1disk:mymod.mod";
      WHILE TRUE DO
         ! Call some routine in mymod
         mymodrout;
         IF FileTime("ram1disk:mymod.mod" \ModifyTime)
            > ModTime("mymod") THEN
            UnLoad "ram1disk:mymod.mod";
            Load "ram1disk:mymod.mod";
         ENDIF
      ENDWHILE
   ENDPROC
ENDMODULE
```

This program reloads a module if there is a newer one at the source. It uses the *ModTime* to retrieve the latest loading time for the specified module, and compares it to the *FileTime\ModifyTime* at the source. Then, if the source is newer, the program unloads and loads the module again.

Syntax

ModTime ’(’
   [Object ’:=’] < expression (**IN**) of *string*>’)’

A function with a return value of the data type *num*.

Related information

	Described in:
Retrieve time info. about a file	Functions - *FileTime*

NOrient Normalise Orientation

NOrient (Normalise Orientation) is used to normalise unnormalised orientation (quaternion).

Description

An orientation must be normalised, i.e. the sum of the squares must equal 1:

$$q_1^2 + q_2^2 + q_3^2 + q_4^2 = 1$$

If the orientation is slightly unnormalised, it is possible to normalise it.
The normalisation error is the absolute value of the sum of the squares of the orientation components.
The orientation is considered to be slightly unnormalised if the normalisation error is greater then 0.00001 and less then 0.1. If the normalisation error is greater then 0.1 the orient is unusable.

$$ABS(\sqrt{q_1^2 + q_2^2 + q_3^2 + q_4^2} - 1) = normerr$$

normerr > 0.1	Unusable
normerr > 0.00001 AND err <= 0.1	Slightly unnormalised
normerr <= 0.00001	Normalised

Example

We have a slightly unnormalised position (0.707170, 0, 0, 0.707170)

$$ABS(\sqrt{0{,}707170^2 + 0^2 + 0^2 + 0{,}707170^2} - 1) = 0{,}0000894$$

$$0{,}0000894 > 0{,}00001 \Rightarrow unnormalized$$

```
VAR orient unnormorient := [0.707170, 0, 0, 0.707170];
VAR orient normorient;
.
.
normorient := NOrient(unnormorient);
```

The normalisation of the orientation (0.707170, 0, 0, 0.707170) becomes (0.707107, 0, 0, 0.707107).

Return value

Data type: *orient*

The normalised orientation.

Arguments

NOrient (Rotation)

Orient Data type: *orient*

The orientation to be normalised.

Syntax

NOrient’(’
[Rotation ’:=’] <expression (**IN**) of *orient*>
’)’

A function with a return value of the data type *orient*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

NumToStr Converts numeric value to string

NumToStr (Numeric To String) is used to convert a numeric value to a string.

Example

```
VAR string str;
str := NumToStr(0.38521,3);
```

The variable *str* is given the value "0.385".

```
reg1 := 0.38521
str := NumToStr(reg1, 2\Exp);
```

The variable *str* is given the value "3.85E-01".

Return value

Data type: *string*

The numeric value converted to a string with the specified number of decimals, with exponent if so requested. The numeric value is rounded if necessary. The decimal point is suppressed if no decimals are included.

Arguments

NumToStr (Val Dec [\Exp])

Val (*Value*) Data type: *num*

The numeric value to be converted.

Dec (*Decimals*) Data type: *num*

Number of decimals. The number of decimals must not be negative or greater than the available precision for numeric values.

[\Exp] (*Exponent*) Data type: *switch*

To use exponent.

Syntax

NumToStr’(’
[Val ’:=’] <expression (**IN**) of *num*> ’,’
[Dec ’:=’] <expression (**IN**) of *num*>
[\Exp]
’)’

A function with a return value of the data type *string*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

Offs Displaces a robot position

Offs is used to add an offset to a robot position.

Examples

MoveL Offs(p2, 0, 0, 10), v1000, z50, tool1;

The robot is moved to a point *10* mm from the position *p2* (in the z-direction).

p1 := Offs (p1, 5, 10, 15);

The robot position *p1* is displaced *5* mm in the x-direction, *10* mm in the y-direction and *15* mm in the z-direction.

Return value

Data type: *robtarget*

The displaced position data.

Arguments

Offs (Point XOffset YOffset ZOffset)

Point Data type: *robtarget*

The position data to be displaced.

XOffset Data type: *num*

The displacement in the x-direction.

YOffset Data type: *num*

The displacement in the y-direction.

ZOffset Data type: *num*

The displacement in the z-direction.

Example

```
PROC pallet (num row, num column, num distance, PERS tooldata tool,
             PERS wobjdata wobj)

  VAR robtarget palletpos:=[[0, 0, 0], [1, 0, 0, 0], [0, 0, 0, 0],
                            [9E9, 9E9, 9E9, 9E9, 9E9, 9E9]];

  palettpos := Offs (palettpos, (row-1)*distance, (column-1)*distance, 0);
  MoveL palettpos, v100, fine, tool\WObj:=wobj;

ENDPROC
```

A routine for picking parts from a pallet is made. Each pallet is defined as a work object (see Figure 35). The part to be picked (row and column) and the distance between the parts are given as input parameters.
Incrementing the row and column index is performed outside the routine.



Figure 35 The position and orientation of the pallet is specified by defining a work object.

Syntax

```
Offs '('
   [Point ':='] <expression (IN) of robtarget> ','
   [XOffset ':='] <expression (IN) of num> ','
   [YOffset ':='] <expression (IN) of num> ','
   [ZOffset ':='] <expression (IN) of num> ')'
```

A function with a return value of the data type *robtarget*.

Related information

	Described in:
Position data	Data Types - *robtarget*

OpMode Read the operating mode

OpMode (Operating Mode) is used to read the current operating mode of the system.

Example

```
TEST OpMode()
CASE OP_AUTO:
      ...
CASE OP_MAN_PROG:
      ...
CASE OP_MAN_TEST:
      ...
DEFAULT:
      ...
ENDTEST
```

Different program sections are executed depending on the current operating mode.

Return value

Data type: *symnum*

The current operating mode as defined in the table below.

Return value	Symbolic constant	Comment
0	OP_UNDEF	Undefined operating mode
1	OP_AUTO	Automatic operating mode
2	OP_MAN_PROG	Manual operating mode max. 250 mm/s
3	OP_MAN_TEST	Manual operating mode full speed, 100 %

Syntax

OpMode’(’ ’)’

A function with a return value of the data type *symnum*.

Related information

	Described in:
Different operating modes	User’s Guide - *Starting up*
Reading running mode	Functions - *RunMode*

OrientZYX Builds an orient from Euler angles

OrientZYX (Orient from Euler ZYX angles) is used to build an orient type variable out of Euler angles.

Example

```
VAR num anglex;
VAR num angley;
VAR num anglez;
VAR pose object;
.
object.rot := OrientZYX(anglez, angley, anglex)
```

Return value

Data type: *orient*

The orientation made from the Euler angles.

The rotations will be performed in the following order:
-rotation around the z axis,
-rotation around the new y axis
-rotation around the new x axis.

Arguments

OrientZYX (ZAngle YAngle XAngle)

ZAngle Data type: *num*

The rotation, in degrees, around the Z axis.

YAngle Data type: *num*

The rotation, in degrees, around the Y axis.

XAngle Data type: *num*

The rotation, in degrees, around the X axis.

The rotations will be performed in the following order:
-rotation around the z axis,
-rotation around the new y axis
-rotation around the new x axis.

Syntax

OrientZYX’(’
[ZAngle ’:=’] <expression (**IN**) of *num*> ’,’
[YAngle ’:=’] <expression (**IN**) of *num*> ’,’
[XAngle ’:=’] <expression (**IN**) of *num*>
’)’

A function with a return value of the data type *orient*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

ORobT Removes a program displacement from a position

ORobT (Object Robot Target) is used to transform a robot position from the program displacement coordinate system to the object coordinate system and/or to remove an offset for the external axes.

Example

```
VAR robtarget p10;
VAR robtarget p11;

p10 := CRobT();
p11 := ORobT(p10);
```

The current positions of the robot and the external axes are stored in *p10* and *p11*. The values stored in *p10* are related to the ProgDisp/ExtOffs coordinate system. The values stored in *p11* are related to the object coordinate system without any offset on the external axes.

Return value

Data type: *robtarget*

The transformed position data.

Arguments

ORobT (OrgPoint [\InPDisp] | [\InEOffs])

OrgPoint *(Original Point)* Data type: *robtarget*

The original point to be transformed.

[\InPDisp] *(In Program Displacement)* Data type: *switch*

Returns the TCP position in the ProgDisp coordinate system, i.e. removes external axes offset only.

[\InEOffs] *(In External Offset)* Data type: *switch*

Returns the external axes in the offset coordinate system, i.e. removes program displacement for the robot only.

Examples

p10 := ORobT(p10 \InEOffs);

The ORobT function will remove any program displacement that is active, leaving the TCP position relative to the object coordinate system. The external axes will remain in the offset coordinate system.

p10 := ORobT(p10 \InPDisp);

The ORobT function will remove any offset of the external axes. The TCP position will remain in the ProgDisp coordinate system.

Syntax

ORobT ’(’
[OrgPoint ’:=’] < expression (**IN**) of *robtarget*>
[’\’InPDisp] | [’\’InEOffs]’)’

A function with a return value of the data type *robtarget*.

Related information

	Described in:
Definition of program displacement for the robot	Instructions - *PDispOn, PDispSet*
Definition of offset for external axes	Instructions - *EOffsOn, EOffsSet*
Coordinate systems	Motion and I/O Principles - *Coordinate Systems*

PoseInv Inverts the pose

PoseInv (Pose Invert) calculates the reverse transformation of a pose.

Example



Pose1 represents the coordinates of Frame1 related to Frame0.
The transformation giving the coordinates of Frame0 related to Frame1 is obtained by the reverse transformation:

```
VAR pose pose1;
VAR pose pose2;
.
.
pose2 := PoseInv(pose1);
```

Return value

Data type: *pose*

The value of the reverse pose.

Arguments

PoseInv (Pose)

Pose Data type: *pose*

The pose to invert.

Syntax

PoseInv’(’
[Pose ’:=’] <expression (**IN**) of *pose*>
’)’

A function with a return value of the data type *pose*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

PoseMult Multiplies pose data

PoseMult (Pose Multiply) is used to calculate the product of two frame transformations. A typical use is to calculate a new frame as the result of a displacement acting on an original frame.

Example



pose1 represents the coordinates of Frame1 related to Frame0.
pose2 represents the coordinates of Frame2 related to Frame1.

The transformation giving pose3, the coordinates of Frame2 related to Frame0, is obtained by the product of the two transformations:

```
VAR pose pose1;
VAR pose pose2;
VAR pose pose3;
.
.
pose3 := PoseMult(pose1, pose2);
```

Return value

Data type: *pose*

The value of the product of the two poses.

Arguments

PoseMult (Pose1 Pose2)

Pose1 Data type: *pose*

The first pose.

Pose2 Data type: *pose*

The second pose.

Syntax

PoseMult’(’
[Pose1 ’:=’] <expression (**IN**) of *pose*> ’,’
[Pose2 ’:=’] <expression (**IN**) of *pose*>
’)’

A function with a return value of the data type *pose*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

PoseVect Applies a transformation to a vector

PoseVect (Pose Vector) is used to calculate the product of a pose and a vector.
It is typically used to calculate a vector as the result of the effect of a displacement on an original vector.

Example



pose1 represents the coordinates of Frame1 related to Frame0.
pos1 is a vector related to Frame1.

The corresponding vector related to Frame0 is obtained by the product:

```
VAR pose pose1;
VAR pos pos1;
VAR pos pos2;
.
.
pos2:= PoseVect(pose1, pos1);
```

Return value

Data type: *pos*

The value of the product of the pose and the original pos.

Arguments

PoseVect (Pose Pos)

Pose Data type: *pose*

The transformation to be applied.

Pos Data type: *pos*

The pos to be transformed.

Syntax

PoseVect’(’
[Pose ’:=’] <expression (**IN**) of *pose*> ’,’
[Pos ’:=’] <expression (**IN**) of *pos*>
’)’

A function with a return value of the data type *pos*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

Pow Calculates the power of a value

Pow (Power) is used to calculate the exponential value in any base.

Example

```
VAR num x;
VAR num y
VAR num reg1;
.
reg1:= Pow(x, y);
```

reg1 is assigned the value x^y.

Return value

Data type: *num*

The value of the base x raised to the power of the exponent y (x^y).

Arguments

Pow (Base Exponent)

Base Data type: *num*

The base argument value.

Exponent Data type: *num*

The exponent argument value.

Limitations

The execution of the function x^y will give an error if:

. x < 0 and y is not an integer;
. x = 0 and y < 0.

Syntax

```
Pow'('
   [Base ':='] <expression (IN) of num> ','
   [Exponent ':='] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

Present Tests if an optional parameter is used

Present is used to test if an optional argument has been used when calling a routine.

An optional parameter may not be used if it was not specified when calling the routine. This function can be used to test if a parameter has been specified, in order to prevent errors from occurring.

Example

```
PROC feeder (\switch on | \switch off)

  IF Present (on) Set do1;
  IF Present (off) Reset do1;

ENDPROC
```

The output *do1*, which controls a feeder, is set or reset depending on the argument used when calling the routine.

Return value

Data type: *bool*

TRUE = The parameter value or a switch has been defined when calling the routine.

FALSE = The parameter value or a switch has not been defined.

Arguments

Present (OptPar)

OptPar *(Optional Parameter)* Data type: Any type

The name of the optional parameter to be tested.

Example

```
PROC glue (\switch on, num glueflow, robtarget topoint, speeddata speed,
           zonedata zone, PERS tooldata tool, \PERS wobjdata wobj)

  IF Present (on) PulseDO glue_on;
  SetAO gluesignal, glueflow;
  IF Present (wobj) THEN
    MoveL topoint, speed, zone, tool \WObj=wobj;
  ELSE
    MoveL topoint, speed, zone, tool;
  ENDIF

ENDPROC
```

A glue routine is made. If the argument *\on* is specified when calling the routine, a pulse is generated on the signal *glue_on*. The robot then sets an analog output *gluesignal*, which controls the glue gun, and moves to the end position. As the wobj parameter is optional, different MoveL instructions are used depending on whether this argument is used or not.

Syntax

```
Present '('
  [OptPar':='] <reference (REF) of any type> ')'
```

A REF parameter requires, in this case, the optional parameter name.

A function with a return value of the data type *bool*.

Related information

	Described in:
Routine parameters	Basic Characteristics - *Routines*

ReadBin Reads a byte from a file or serial channel

ReadBin (Read Binary) is used to read a byte (8 bits) from a file or serial channel.

This function works on both binary and character-based files or serial channels.

Example

```
VAR num character;
VAR iodev inchannel;
...
Open "sio1:", inchannel\Bin;
character := ReadBin(inchannel);
```

A byte is read from the binary serial channel *inchannel*.

Return value

Data type: *num*

A byte (8 bits) is read from a specified file or serial channel. This byte is converted to the corresponding positive numeric value and returned as a *num* data type.
If a file is empty (end of file), the number -1 is returned.

Arguments

ReadBin (IODevice [\Time])

IODevice Data type: *iodev*

The name (reference) of the file or serial channel to be read.

[\Time] Data type: *num*

The max. time for the reading operation (timeout) in seconds. If this argument is not specified, the max. time is set to 60 seconds.

If this time runs out before the reading operation is finished, the error handler will be called with the error code ERR_DEV_MAXTIME. If there is no error handler, the execution will be stopped.

The timeout function is in use also during program stop and will be noticed in the RAPID program at program start.

Program execution

Program execution waits until a byte (8 bits) can be read from the file or serial channel.

Example

```
VAR num bindata;
VAR iodev file;

Open “flp1:myfile.bin”, file \Read \Bin;
bindata := ReadBin(file);
WHILE bindata <> EOF_BIN DO
   TPWrite ByteToStr(bindata\Char);
   bindata := ReadBin(file);
ENDWHILE
```

Read the contents of a binary file *myfile.bin* from the beginning to the end and displays the received binary data converted to chars on the teach pendant (one char on each line).

Limitations

The function can only be used for files and serial channels that have been opened with read access (*\Read* for character based files, *\Bin* or *\Append \Bin* for binary files).

Error handling

If an error occurs during reading, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

Predefined data

The constant *EOF_BIN* can be used to stop reading at the end of the file.

```
CONST num EOF_BIN := -1;
```

Syntax

```
ReadBin’(’
   [IODevice ’:=’] <variable (VAR) of iodev>
   [’\’Time’:=’ <expression (IN) of num>]’)’
```

A function with a return value of the type *num*.

Related information

	Described in:
Opening (etc.) files or serial channels	RAPID Summary - *Communication*
Convert a byte to a string data	Functions - *ByteToStr*

ReadMotor Reads the current motor angles

ReadMotor is used to read the current angles of the different motors of the robot and external axes. The primary use of this function is in the calibration procedure of the robot.

Example

```
VAR num motor_angle2;

motor_angle2 := ReadMotor(2);
```

The current motor angle of the second axis of the robot is stored in *motor_angle2*.

Return value

Data type: *num*

The current motor angle in radians of the stated axis of the robot or external axes.

Arguments

ReadMotor [\MecUnit] Axis

MecUnit *(Mechanical Unit)* Data type: *mecunit*

The name of the mechanical unit for which an axis is to be read. If this argument is omitted, the axis for the robot is read. (Note, in this release only robot is permitted for this argument).

Axis Data type: *num*

The number of the axis to be read (1 - 6).

Program execution

The motor angle returned represents the current position in radians for the motor and independently of any calibration offset. The value is not related to a fix position of the robot, only to the resolver internal zero position, i.e. normally the resolver zero position closest to the calibration position (the difference between the resolver zero position and the calibration position is the calibration offset value). The value represents the full movement of each axis, although this may be several turns.

Example

```
VAR num motor_angle3;

motor_angle3 := ReadMotor(\MecUnit:=robot, 3);
```

The current motor angle of the third axis of the robot is stored in *motor_angle3*.

Syntax

```
ReadMotor'('
   ['\'MecUnit ':=' < variable (VAR) of mecunit>','']
   [Axis ':=' ] < expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Reading the current joint angle	Functions - *CJointT*

ReadNum Reads a number from a file or serial channel

ReadNum (Read Numeric) is used to read a number from a character-based file or serial channel.

Example

```
VAR iodev infile;
...
Open "flp1:file.doc", infile\Read;
reg1 := ReadNum(infile);
```

Reg1 is assigned a number read from the file *file.doc* on the diskette.

Return value

Data type: *num*

The numeric value read from a specified file or serial channel.
If the file is empty (end of file), the number 9.999E36 is returned.

Arguments

ReadNum (IODevice [\Delim] [\Time])

IODevice Data type: *iodev*

The name (reference) of the file or serial channel to be read.

[\Delim] (*Delimiters*) Data type: *string*

A string containing the delimiters to use when parsing a line in the file or serial channel. By default (without *Delim*), the file is read line by line and the line-feed character (\0A) is the only delimiter considered by the parsing. When the *Delim* argument is used, any character in the specified string argument will be considered to determine the significant part of the line.

When using the argument *Delim*, the control system always adds the characters carriage return (\0D) and line-feed (\0A) to the delimiters specified by the user.

To specify non-alphanumeric characters, use \xx, where xx is the hexadecimal representation of the ASCII code of the character (example: TAB is specified by \09).

[\Time] Data type: *num*

The max. time for the reading operation (timeout) in seconds. If this argument is not specified, the max. time is set to 60 seconds.

If this time runs out before the read operation is finished, the error handler will be called with the error code ERR_DEV_MAXTIME. If there is no error handler, the execution will be stopped.

The timeout function is also in use during program stop and will be noticed in the RAPID program at program start.

Program execution

Starting at the current file position, the function reads and discards any heading delimiters. A heading delimiter without the argument *\Delim* is a line-feed character. Heading delimiters with the argument *\Delim* are any characters specified in the *\Delim* argument plus carriage return and line-feed characters. It then reads everything up to and including the next delimiter character (will be discarded), but not more than 80 characters. If the significant part exceeds 80 characters, the remainder of the characters will be read on the next reading.

The string that is read is then converted to a numeric value; e.g. "234.4" is converted to the numeric value 234.4.

Example

```
reg1 := ReadNum(infile\Delim:="\09  ");
IF reg1 > EOF_NUM THEN
    TPWrite "The file is empty";
    ...
```

Reads a number in a line where numbers are separated by TAB ("\09") or SPACE (" ") characters.
Before using the number read from the file, a check is performed to make sure that the file is not empty.

Limitations

The function can only be used for character based files that have been opened for reading.

Error handling

If an access error occurs during reading, the system variable ERRNO is set to ERR_FILEACC.
If there is an attempt to read non-numeric data, the system variable ERRNO is set to ERR_RCVDATA. These errors can then be dealt with by the error handler.

Predefined data

The constant *EOF_NUM* can be used to stop reading, at the end of the file.

CONST num EOF_NUM := 9.998E36;

Syntax

ReadNum '('
[IODevice ':=']<variable (**VAR**) of *iodev*>
['\'Delim':='<expression (**IN**) of *string*>]
['\'Time':='<expression (**IN**) of *num*>]')'

A function with a return value of the type *num*.

Related information

	Described in:
Opening (etc.) files or serial channels	RAPID Summary - *Communication*

ReadStr Reads a string from a file or serial channel

ReadStr (Read String) is used to read a string from a character-based file or serial channel.

Example

```
VAR string text;
VAR iodev infile;
...
Open "flp1:file.doc", infile\Read;
text := ReadStr(infile);
```

Text is assigned a string read from the file *file.doc* on the diskette.

Return value

Data type: *string*

The string read from the specified file or serial channel.
If the file is empty (end of file), the string "EOF" is returned.

Arguments

ReadStr (IODevice [\Delim] [\RemoveCR] [\DiscardHeaders] [\Time])

IODevice Data type: *iodev*

The name (reference) of the file or serial channel to be read.

[\Delim] (*Delimiters*) Data type: *string*

A string containing the delimiters to use when parsing a line in the file or serial channel. By default the file is read line by line and the line-feed character (\0A) is the only delimiter considered by the parsing. When the *\Delim* argument is used, any character in the specified string argument plus by default line-feed character will be considered to determine the significant part of the line.

To specify non-alphanumeric characters, use \xx, where xx is the hexadecimal representation of the ASCII code of the character (example: TAB is specified by \09).

[\RemoveCR] Data type: *switch*

A switch used to remove the trailing carriage return character when reading PC files. In PC files, a new line is specified by carriage return and line feed (CRLF). When reading a line in such files, the carriage return character is by default read into the return string. When using this argument, the carriage return character will be read from the file but not included in the return string.

[\DiscardHeaders] Data type: *switch*

This argument specifies whether the heading delimiters (specified in *\Delim* plus default line-feed) are skipped or not before transferring data to the return string. By default, if the first character at the current file position is a delimiter, it is read but not transferred to the return string, the line parsing is stopped and the return will be an empty string. If this argument is used, all delimiters included in the line will be read from the file but discarded, and the return string will contain the data starting at the first non-delimiter character in the line.

[\Time] Data type: *num*

The max. time for the reading operation (timeout) in seconds. If this argument is not specified, the max. time is set to 60 seconds.

If this time runs out before the read operation is finished, the error handler will be called with the error code ERR_DEV_MAXTIME. If there is no error handler, the execution will be stopped.

The timeout function is in use also during program stop and will be noticed in the RAPID program at program start.

Program execution

Starting at the current file position, if the *\DiscardHeaders* argument is used, the function reads and discards any heading delimiters (line-feed characters and any character specified in the *\Delim* argument). In all cases, it then reads everything up to the next delimiter character, but not more than 80 characters. If the significant part exceeds 80 characters, the remainder of the characters will be read on the next reading. The delimiter that caused the parsing to stop is read from the file but not transferred to the return string. If the last character in the string is a carriage return character and the *\RemoveCR* argument is used, this character will be removed from the string.

Example

```
text := ReadStr(infile);
IF text = EOF THEN
    TPWrite "The file is empty";
    ...
```

Before using the string read from the file, a check is performed to make sure that the file is not empty.

Example

Consider a file containing:
<LF><SPACE><TAB>Hello<SPACE><SPACE>World<CR><LF>

```
text := ReadStr(infile);
```

text will be an empty string: the first character in the file is the default <LF> delimiter.

```
text := ReadStr(infile\DiscardHeaders);
```

text will contain <SPACE><TAB>Hello<SPACE><SPACE>World<CR>: the first character in the file, the default <LF> delimiter, is discarded.

```
text := ReadStr(infile\RemoveCR\DiscardHeaders);
```

text will contain <SPACE><TAB>Hello<SPACE><SPACE>World: the first character in the file, the default <LF> delimiter, is discarded; the final carriage return character is removed

```
text := ReadStr(infile\Delim:=" \09"\RemoveCR\DiscardHeaders);
```

text will contain “Hello”: the first characters in the file that match either the default <LF> delimiter or the character set defined by *\Delim* (space and tab) are discarded. Data is then transferred up to the first delimiter that is read from the file but not transferred into the string. A new invocation of the same statement will return “World”.

Example

Consider a file containing:
<CR><LF>Hello<CR><LF>

```
text := ReadStr(infile);
```

text will contain the <CR> (\0d) character: <CR> and <LF> characters are read from the file, but only <CR> is transferred to the string. A new invocation of the same statement will return “Hello\0d”.

text := ReadStr(infile\RemoveCR);

text will contain an empty string: <CR> and <LF> characters are read from the file; <CR> is transferred but removed from the string. A new invocation of the same statement will return “Hello”.

text := ReadStr(infile\Delim:=”\0d”);

text will contain an empty string: <CR> is read from the file but not transferred to the return string. A new invocation of the same instruction will return an empty string again: <LF> is read from the file but not transferred to the return string.

text := ReadStr(infile\Delim:=”\0d”\DiscardHeaders);

text will contain “Hello”. A new invocation of the same instruction will return “EOF” (end of file).

Limitations

The function can only be used for files or serial channels that have been opened for reading in a character-based mode.

Error handling

If an error occurs during reading, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

Predefined data

The constant *EOF* can be used to check if the file was empty when trying to read from the file or to stop reading at the end of the file.

CONST string EOF := "EOF";

Syntax

ReadStr ’(’
[IODevice ’:=’] <variable (**VAR**) of *iodev*>
[‘\’Delim’:=’<expression (**IN**) of *string*>]
[‘\’RemoveCR]
[‘\’DiscardHeaders]
[’\’Time’:=’ <expression (**IN**) of *num*>]’)’

A function with a return value of the type *string*.

Related information

	Described in:
Opening (etc.) files or serial channels	RAPID Summary - *Communication*

RelTool Make a displacement relative to the tool

RelTool (Relative Tool) is used to add a displacement and/or a rotation, expressed in the tool coordinate system, to a robot position.

Example

MoveL RelTool (p1, 0, 0, 100), v100, fine, tool1;

The robot is moved to a position that is 100 mm from p1 in the direction of the tool.

MoveL RelTool (p1, 0, 0, 0 \Rz:= 25), v100, fine, tool1;

The tool is rotated 25º around its z-axis.

Return value

Data type: *robtarget*

The new position with the addition of a displacement and/or a rotation, if any, relative to the active tool.

Arguments

RelTool (Point Dx Dy Dz [\Rx] [\Ry] [\Rz])

Point Data type: *robtarget*

The input robot position. The orientation part of this position defines the current orientation of the tool coordinate system.

Dx Data type: *num*

The displacement in mm in the x direction of the tool coordinate system.

Dy Data type: *num*

The displacement in mm in the y direction of the tool coordinate system.

Dz Data type: *num*

The displacement in mm in the z direction of the tool coordinate system.

[\Rx] Data type: *num*

The rotation in degrees around the x axis of the tool coordinate system.

[\Ry] Data type: *num*

The rotation in degrees around the y axis of the tool coordinate system.

[\Rz] Data type: *num*

The rotation in degrees around the z axis of the tool coordinate system.

In the event that two or three rotations are specified at the same time, these will be performed first around the x-axis, then around the new y-axis, and then around the new z-axis.

Syntax

RelTool'('
[Point ':='] < expression (**IN**) of *robtarget*>','
[Dx ':='] <expression (**IN**) of *num*> ','
[Dy ':='] <expression (**IN**) of *num*> ','
[Dz ':='] <expression (**IN**) of *num*>
['\'Rx ':=' <expression (**IN**) of *num*>]
['\'Ry ':=' <expression (**IN**) of *num*>]
['\'Rz ':=' <expression (**IN**) of *num*>]')'

A function with a return value of the data type *robtarget*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*
Positioning instructions	RAPID Summary - *Motion*

Round Round is a numeric value

Round is used to round a numeric value to a specified number of decimals or to an integer value.

Example

```
VAR num val;
val := Round(0.38521\Dec:=3);
```

The variable *val* is given the value 0.385.

```
val := Round(0.38521\Dec:=1);
```

The variable *val* is given the value 0.4.

```
val := Round(0.38521);
```

The variable *val* is given the value 0.

Return value

Data type: *num*

The numeric value rounded to the specified number of decimals.

Arguments

Round (Val [\Dec])

Val (*Value*) Data type: *num*

The numeric value to be rounded.

[\Dec] (*Decimals* Data type: *num*

Number of decimals.

If the specified number of decimals is 0 or if the argument is omitted, the value is rounded to an integer.

The number of decimals must not be negative or greater than the available precision for numeric values.

Syntax

Round’(’
[Val ’:=’] <expression (**IN**) of *num*>
[\Dec ’:=’ <expression (**IN**) of *num*>]
’)’

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*
Truncating a value	Functions - *Trunc*

ReadStrBin Reads a string from a binary serial channel or file

ReadStrBin (Read String Binary) is used to read a string from a binary serial channel or file.

Example

```
VAR iodev channel2;
VAR string text;
...
Open “sio1:”, channel2 \Bin;
text := ReadStrBin (channel2, 10);
```

Text is assigned a *10* characters text string read from the serial channel referred to by *channel2*.

Return value

Data type: *string*

The text string read from the specified serial channel or file. If the file is empty (end of file), the string "EOF" is returned.

Arguments

ReadStrBin (IODevice NoOfChars [\Time])

IODevice Data type: *iodev*

The name (reference) of the binary serial channel or file to be read.

NoOfChars Data type: *num*

The number of characters to be read from the binary serial channel or file.

[\Time] Data type: *num*

The max. time for the reading operation (timeout) in seconds. If this argument is not specified, the max. time is set to 60 seconds.

If this time runs out before the read operation is finished, the error handler will be called with the error code ERR_DEV_MAXTIME. If there is no error handler, the execution will be stopped.

The timeout function is in use also during program stop and will be noticed in the RAPID program at program start.

Program execution

The function reads the specified number of characters from the binary serial channel or file.

Example

```
text := ReadStrBin(infile,20);
IF text = EOF THEN
   TPWrite "The file is empty";
```

Before using the string read from the file, a check is performed to make sure that the file is not empty.

Limitations

The function can only be used for serial channels or files that have been opened for reading in a binary mode.

Error handling

If an error occurs during reading, the system variable ERRNO is set to ERR_FILEACC. This error can then be handled in the error handler.

Predefined data

The constant *EOF* can be used to check if the file was empty, when trying to read from the file or to stop reading at the end of the file.

```
CONST string EOF := "EOF";
```

Syntax

```
ReadStrBin '('
   [IODevice ':='] <variable (VAR) of iodev>','
   [NoOfChars ':='] <expression (IN) of num>
   ['\'Time':=' <expression (IN) of num>]')'
```

A function with a return value of the type *string*.

Related information

	Described in:
Opening (etc.) serial channels or files	RAPID Summary - *Communication*
Write binary string	Instructions - *WriteStrBin*

RunMode Read the running mode

RunMode (Running Mode) is used to read the current running mode of the program task.

Example

```
IF RunMode() = RUN_CONT_CYCLE THEN
..
ENDIF
```

The program section is executed only for continuous or cycle running.

Return value

Data type: *symnum*

The current running mode as defined in the table below.

Return value	Symbolic constant	Comment
0	RUN_UNDEF	Undefined running mode
1	RUN_CONT_CYCLE	Continuous or cycle running mode
2	RUN_INSTR_FWD	Instruction forward running mode
3	RUN_INSTR_BWD	Instruction backward running mode
4	RUN_SIM	Simulated running mode

Arguments

RunMode ([\Main])

[\Main] Data type: *switch*

Return current running mode for program task *main*.
Used in multi-tasking system to get current running mode for program task *main* from some other program task.

If this argument is omitted, the return value always mirrors the current running mode for the program task which executes the function *RunMode*.

Syntax

RunMode ’(’ [’\’Main] ’)’

A function with a return value of the data type *symnum*.

Related information

	Described in:
Reading operating mode	Functions - *OpMode*

Sin Calculates the sine value

Sin (Sine) is used to calculate the sine value from an angle value.

Example

```
VAR num angle;
VAR num value;
.
.
value := Sin(angle);
```

Return value

Data type: *num*

The sine value, range [-1, 1] .

Arguments

Sin (Angle)

Angle Data type: *num*

The angle value, expressed in degrees.

Syntax

```
Sin'('
   [Angle':='] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

Sqrt Calculates the square root value

Sqrt (Square root) is used to calculate the square root value.

Example

```
VAR num x_value;
VAR num y_value;
.
.
y_value := Sqrt( x_value);
```

Return value

Data type: *num*

The square root value.

Arguments

Sqrt (Value)

Value Data type: *num*

The argument value for square root ($\sqrt{\ }$); it has to be ≥ 0.

Syntax

Sqrt'('
[Value':='] <expression (**IN**) of *num*>
')'

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

StrFind Searches for a character in a string

StrFind (String Find) is used to search in a string, starting at a specified position, for a character that belongs to a specified set of characters.

Example

```
VAR num found;

found := StrFind("Robotics",1,"aeiou");
```

The variable *found* is given the value 2.

```
found := StrFind("Robotics",1,"aeiou"\NotInSet);
```

The variable *found* is given the value 1.

```
found := StrFind("IRB 6400",1,STR_DIGIT);
```

The variable *found* is given the value 5.

```
found := StrFind("IRB 6400",1,STR_WHITE);
```

The variable *found* is given the value 4.

Return value

Data type: *num*

The character position of the first character, at or past the specified position, that belongs to the specified set. If no such character is found, String length +1 is returned.

Arguments

StrFind (Str ChPos Set [\NotInSet])

Str (*String*) Data type: *string*

The string to search in.

ChPos (*Character Position*) Data type: *num*

Start character position. A runtime error is generated if the position is outside the string.

Set Data type: *string*

Set of characters to test against.

[\NotInSet] Data type: *switch*

Search for a character not in the set of characters.

Syntax

```
StrFind'('
    [ Str ':=' ] <expression (IN) of string> ','
    [ ChPos ':=' ] <expression (IN) of num> ','
    [ Set':=' ] <expression (IN) of string>
    ['\'NotInSet ]
    ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

StrLen Gets the string length

StrLen (String Length) is used to find the current length of a string.

Example

```
VAR num len;

len := StrLen("Robotics");
```

The variable *len* is given the value 8.

Return value

Data type: *num*

The number of characters in the string (>=0).

Arguments

StrLen (Str)

Str (*String*) Data type: *string*

The string in which the number of characters is to be counted.

Syntax

```
StrLen'('
    [ Str ':=' ] <expression (IN) of string>
    ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

StrMap Maps a string

StrMap (String Mapping) is used to create a copy of a string in which all characters are translated according to a specified mapping.

Example

VAR string str;

str := StrMap("Robotics","aeiou","AEIOU");

The variable *str* is given the value "RObOtIcs".

str := StrMap("Robotics",STR_LOWER, STR_UPPER);

The variable *str* is given the value "ROBOTICS".

Return value

Data type: *string*

The string created by translating the characters in the specified string, as specified by the "from" and "to" strings. Each character, from the specified string, that is found in the "from" string is replaced by the character at the corresponding position in the "to" string. Characters for which no mapping is defined are copied unchanged to the resulting string.

Arguments

StrMap (Str FromMap ToMap)

Str (*String*) Data type: *string*

The string to translate.

FromMap Data type: *string*

Index part of mapping.

ToMap Data type: *string*

Value part of mapping.

Syntax

```
StrMap'('
    [ Str ':=' ] <expression (IN) of string> ','
    [ FromMap':=' ] <expression (IN) of string> ','
    [ ToMap':=' ] <expression (IN) of string>
    ')'
```

A function with a return value of the data type *string*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

StrMatch Search for pattern in string

StrMatch (String Match) is used to search in a string, starting at a specified position, for a specified pattern.

Example

```
VAR num found;

found := StrMatch("Robotics",1,"bo");
```

The variable *found* is given the value 3.

Return value

Data type: *num*

The character position of the first substring, at or past the specified position, that is equal to the specified pattern string. If no such substring is found, string length +1 is returned.

Arguments

StrMatch (Str ChPos Pattern)

Str (*String*) Data type: *string*

The string to search in.

ChPos (*Character Position*) Data type: *num*

Start character position. A runtime error is generated if the position is outside the string.

Pattern Data type: *string*

Pattern string to search for.

Syntax

```
StrMatch'('
   [ Str ':=' ] <expression (IN) of string> ','
   [ ChPos ':=' ] <expression (IN) of num> ','
   [ Pattern':=' ] <expression (IN) of string>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

StrMemb Checks if a character belongs to a set

StrMemb (String Member) is used to check whether a specified character in a string belongs to a specified set of characters.

Example

VAR bool memb;

memb := StrMemb("Robotics",2,"aeiou");

The variable *memb* is given the value TRUE, as o is a member of the set "aeiou".

memb := StrMemb("Robotics",3,"aeiou");

The variable *memb* is given the value FALSE, as b is not a member of the set "aeiou".

memb := StrMemb("S-721 68 VÄSTERÅS",3,STR_DIGIT);

The variable *memb* is given the value TRUE.

Return value

Data type: *bool*

TRUE if the character at the specified position in the specified string belongs to the specified set of characters.

Arguments

StrMemb (Str ChPos Set)

Str (*String*) Data type: *string*

The string to check in.

ChPos (*Character Position*) Data type: *num*

The character position to check. A runtime error is generated if the position is outside the string.

Set Data type: *string*

Set of characters to test against.

Syntax

```
StrMemb'('
    [ Str ':=' ] <expression (IN) of string> ','
    [ ChPos ':=' ] <expression (IN) of num> ','
    [ Set':=' ] <expression (IN) of string>
    ')'
```

A function with a return value of the data type *bool*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

StrOrder Checks if strings are ordered

StrOrder (String Order) is used to check whether two strings are in order, according to a specified character ordering sequence.

Example

```
VAR bool le;

le := StrOrder("FIRST","SECOND",STR_UPPER);
```

The variable *le* is given the value TRUE, because "FIRST" comes before "SECOND" in the character ordering sequence STR_UPPER.

Return value

Data type: *bool*

TRUE if the first string comes before the second string (Str1 <= Str2) when characters are ordered as specified.

Characters that are not included in the defined ordering are all assumed to follow the present ones.

Arguments

StrOrder (Str1 Str2 Order)

Str1 (*String 1*) Data type: *string*

First string value.

Str2 (*String 2*) Data type: *string*

Second string value.

Order Data type: *string*

Sequence of characters that define the ordering.

Syntax

```
StrOrder'('
    [ Str1 ':=' ] <expression (IN) of string> ','
    [ Str2 ':=' ] <expression (IN) of string> ','
    [ Order ':=' ] <expression (IN) of string>
    ')'
```

A function with a return value of the data type *bool*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

StrPart Finds a part of a string

StrPart (String Part) is used to find a part of a string, as a new string.

Example

```
VAR string part;

part := StrPart("Robotics",1,5);
```

The variable *part* is given the value "Robot".

Return value

Data type: *string*

The substring of the specified string, which has the specified length and starts at the specified character position.

Arguments

StrPart (Str ChPos Len)

Str (*String*) Data type: *string*

The string in which a part is to be found.

ChPos (*Character Position*) Data type: *num*

Start character position. A runtime error is generated if the position is outside the string.

Len (*Length*) Data type: *num*

Length of string part. A runtime error is generated if the length is negative or greater than the length of the string, or if the substring is (partially) outside the string.

Syntax

```
StrPart'('
   [ Str ':=' ] <expression (IN) of string> ','
   [ ChPos ':=' ] <expression (IN) of num> ','
   [ Len':=' ] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *string*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

StrToByte Converts a string to a byte data

StrToByte (String To Byte) is used to convert a *string* with a defined byte data format into a *byte* data.

Example

```
VAR string con_data_buffer{5} := ["10", "AE", "176", "00001010", "A"];
VAR byte data_buffer{5};


data_buffer{1} := StrToByte(con_data_buffer{1});
```

The content of the array component *data_buffer{1}* will be 10 decimal after the *StrToByte* ... function.

```
data_buffer{2} := StrToByte(con_data_buffer{2}\Hex);
```

The content of the array component *data_buffer{2}* will be 174 decimal after the *StrToByte* ... function.

```
data_buffer{3} := StrToByte(con_data_buffer{3}\Okt);
```

The content of the array component *data_buffer{3}* will be 126 decimal after the *StrToByte* ... function.

```
data_buffer{4} := StrToByte(con_data_buffer{4}\Bin);
```

The content of the array component *data_buffer{4}* will be 10 decimal after the *StrToByte* ... function.

```
data_buffer{5} := StrToByte(con_data_buffer{5}\Char);
```

The content of the array component *data_buffer{5}* will be 65 decimal after the *StrToByte* ... function.

Return value

Data type: *byte*

The result of the conversion operation in decimal representation.

Arguments

StrToByte (ConStr [\Hex] | [\Okt] | [\Bin] | [\Char])

ConStr *(Convert String)* Data type: *string*

The string data to be converted.

If the optional switch argument is omitted, the string to be converted has *decimal* (Dec) format.

[\Hex] *(Hexadecimal)* Data type: *switch*

The string to be converted has *hexadecimal* format.

[\Okt] *(Octal)* Data type: *switch*

The string to be converted has *octal* format.

[\Bin] *(Binary)* Data type: *switch*

The string to be converted has *binary* format.

[\Char] *(Character)* Data type: *switch*

The string to be converted has *ASCII* character format.

Limitations

Depending on the format of the string to be converted, the following string data is valid:

Format:	String length:	Range:
Dec .....: ’0’ - ’9’	3	"0" - "255"
Hex .....: ’0’ - ’9’, ’a’ -’f’, ’A’ - ’F’	2	"0" - "FF"
Okt ......: ’0’ - ’7’	3	"0" - "377"
Bin ......: ’0’ - ’1’	8	"0" - "11111111"
Char ....: Any ASCII char (*)	1	One ASCII char

(*) RAPID character codes (e.g. "\07" for BEL control character) can be used as arguments in *ConStr*.

Syntax

```
StrToByte'('
    [ConStr ':='] <expression (IN) of string>
    ['\' Hex ] | ['\' Okt] | ['\' Bin] | ['\' Char]
    ')' ';'
```

A function with a return value of the data type *byte*.

Related information

	Described in:
Convert a byte to a string data	Instructions - *ByteToStr*
Other bit (byte) functions	RAPID Summary - *Bit Functions*
Other string functions	RAPID Summary - *String Functions*

StrToVal Converts a string to a value

StrToVal (String To Value) is used to convert a string to a value of any data type.

Example

```
VAR bool ok;
VAR num nval;

ok := StrToVal("3.85",nval);
```

The variable *ok* is given the value TRUE and *nval* is given the value *3.85*.

Return value

Data type: *bool*

TRUE if the requested conversion succeeded, FALSE otherwise.

Arguments

StrToVal (Str Val)

Str (*String*) Data type: *string*

A string value containing literal data with format corresponding to the data type used in argument *Val*. Valid format as for RAPID literal aggregates.

Val (*Value*) Data type: *ANYTYPE*

Name of the variable or persistent of any data type for storage of the result from the conversion. The data is unchanged if the requested conversion failed.

Example

```
VAR string 15 := “[600, 500, 225.3]”;
VAR bool ok;
VAR pos pos15;

ok := StrToVal(str15,pos15);
```

The variable *ok* is given the value TRUE and the variable *p15* is given the value that are specified in the string *str15*.

Syntax

StrToVal’(’
[Str ’:=’] <expression (**IN**) of *string*> ’,’
[Val ’:=’] <var or pers (**INOUT**) of *ANYTYPE*>
’)’

A function with a return value of the data type *bool*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

Tan Calculates the tangent value

Tan (Tangent) is used to calculate the tangent value from an angle value.

Example

```
VAR num angle;
VAR num value;
.
.
value := Tan(angle);
```

Return value

Data type: *num*

The tangent value.

Arguments

Tan (Angle)

Angle Data type: *num*

The angle value, expressed in degrees.

Syntax

```
Tan'('
   [Angle ':='] <expression (IN) of num>
   ')'
```

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*
Arc tangent with return value in the range [-180, 180]	Functions - *ATan2*

TestDI Tests if a digital input is set

TestDI is used to test whether a digital input is set.

Examples

IF TestDI (di2) THEN . . .

If the current value of the signal *di2* is equal to 1, then . . .

IF NOT TestDI (di2) THEN . . .

If the current value of the signal *di2* is equal to 0, then . . .

WaitUntil TestDI(di1) AND TestDI(di2);

Program execution continues only after both the *di1* input and the *di2* input have been set.

Return value

Data type: *bool*

TRUE = The current value of the signal is equal to 1.

FALSE = The current value of the signal is equal to 0.

Arguments

TestDI (Signal)

Signal Data type: *signaldi*

The name of the signal to be tested.

Syntax

TestDI ’(’
[Signal ’:=’] < variable (**VAR**) of *signaldi* > ’)’

A function with a return value of the data type *bool*.

Related information

	Described in:
Reading the value of a digital input signal	Functions - *DInput*
Input/Output instructions	RAPID Summary - *Input and Output Signals*

TestAndSet Test variable and set if unset

TestAndSet can be used together with a normal data object of the type *bool*, as a binary semaphore, to retrieve exclusive right to specific RAPID code areas or system resources. The function could be used both between different program tasks and different execution levels (TRAP or Event Routines) within the same program task.

Example of resources that can need protection from access at the same time:

- Use of some RAPID routines with function problems when executed in parallel.
- Use of the Teach Pendant - Operator Output & Input

Example

MAIN program task:

```
PERS bool tproutine_inuse := FALSE;
....
WaitUntil TestAndSet(tproutine_inuse);
TPWrite “First line from MAIN”;
TPWrite “Second line from MAIN”;
TPWrite “Third line from MAIN”;
tproutine_inuse := FALSE;
```

BACK1 program task:

```
PERS bool tproutine_inuse := FALSE;
....
WaitUntil TestAndSet(tproutine_inuse);
TPWrite “First line from BACK1”;
TPWrite “Second line from BACK1”;
TPWrite “Third line from BACK1”;
tproutine_inuse := FALSE;
```

To avoid mixing up the lines, one from MAIN and one from BACK1, the use of the TestAndSet function guarantees that all three lines from each task are not separated.

If program task MAIN takes the semaphore *TestAndSet(tproutine_inuse)* first, then program task BACK1 must wait until the program task MAIN has left the semaphore.

Return value

Data type: *num*

TRUE if the semaphore has been taken by me (executor of TestAndSet function), otherwise FALSE. ???

Arguments

TestAndSet Object

Object Data type: *bool*

User defined data object to be used as semaphore. The data object could be a VAR or a PERS. If TestAndSet are used between different program tasks, the object must be a PERS or an installed VAR (intertask objects).

Program execution

This function will in one indivisible step check the user defined variable and, if it is unset, will set it and return TRUE, otherwise it will return FALSE.

```
IF Object = FALSE THEN
    Object := TRUE;
    RETURN TRUE;
ELSE
    RETURN FALSE;
ENDIF
```

Example

```
LOCAL VAR bool doit_inuse := FALSE;
...
PROC doit(...)
    WaitUntil TestAndSet (doit_inuse);
    ....
    doit_inuse := FALSE;
ENDPROC
```

If a module is installed built-in and shared, it is possible to use a local module variable for protection of access from different program tasks at the same time.

Note in this case: If program execution is stopped in the routine *doit* and the program pointer is moved to *main*, the variable *doit_inuse* will not be reset. To avoid this, reset the variable *doit_inuse* to FALSE in the START event routine.

Syntax

```
TestAndSet '('
    [ Object ':=' ] < variable or persistent (INOUT) of bool> ')'
```

A function with a return value of the data type *bool*.

Related information

	Described in:
Built-in and shared module	User’s Guide - *System parameters*
Intertask objects	RAPID Developer’s Manual - RAPID Kernel Reference Manual - *Intertask objects*

Trunc Truncates a numeric value

Trunc (Truncate) is used to truncate a numeric value to a specified number of decimals or to an integer value.

Example

VAR num val;

val := Trunc(0.38521\Dec:=3);

The variable *val* is given the value 0.385.

reg1 := 0.38521

val := Trunc(reg1\Dec:=1);

The variable *val* is given the value 0.3.

val := Trunc(0.38521);

The variable *val* is given the value 0.

Return value

Data type: *num*

The numeric value truncated to the specified number of decimals.

Arguments

Trunc (Val [\Dec])

Val (*Value*) Data type: *num*

The numeric value to be truncated.

[\Dec] (*Decimals*) Data type: *num*

Number of decimals.

If the specified number of decimals is 0 or if the argument is omitted, the value is truncated to an integer.

The number of decimals must not be negative or greater than the available precision for numeric values.

Syntax

Trunc’(’
[Val ’:=’] <expression (**IN**) of *num*>
[\Dec ’:=’ <expression (**IN**) of *num*>]
’)’

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*
Rounding a value	Functions - *Round*

ValToStr Converts a value to a string

ValToStr (Value To String) is used to convert a value of any data type to a string.

Example

```
VAR string str;
VAR pos p := [100,200,300];

str := ValToStr(1.234567);
```

The variable *str* is given the value "1.23457".

```
str := ValToStr(TRUE);
```

The variable *str* is given the value "TRUE".

```
str := ValToStr(p);
```

The variable *str* is given the value "[100,200,300]".

Return value

Data type: *string*

The value is converted to a string with standard RAPID format. This means in principle 6 significant digits. If the decimal part is less than 0.000005 or greater than 0.999995, the number is rounded to an integer.

A runtime error is generated if the resulting string is too long.

Arguments

ValToStr (Val)

Val (*Value*) Data type: *ANYTYPE*

A value of any data type.

Syntax

```
ValToStr'('
   [ Val ':=' ] <expression (IN) of ANYTYPE>
   ')'
```

A function with a return value of the data type *string*.

Related information

	Described in:
String functions	RAPID Summary - *String Functions*
Definition of string	Data Types - *string*
String values	Basic Characteristics - *Basic Elements*

VectMagn Magnitude of a pos vector

VectMagn (Vector Magnitude) is used to calculate the magnitude of a *pos* vector.

Example



A vector **A** can be written as the sum of its components in the three orthogonal directions:

$$A = A_x x + A_y y + A_z z$$

The magnitude of **A** is:

$$|A| = \sqrt{A_x^2 + A_y^2 + A_z^2}$$

The vector is described by the data type *pos* and the magnitude by the data type *num*:

```
VAR num magnitude;
VAR pos vector;
.
.
vector := [1,1,1];
magnitude := VectMagn(vector);
```

Return value

Data type: *num*

The magnitude of the vector (data type *pos*).

Arguments

VectMagn (Vector)

Vector Data type: *pos*

The vector described by the data type *pos*.

Syntax

VectMagn’(’
[Vector ’:=’] <expression (**IN**) of *pos*>
’)’

A function with a return value of the data type *num*.

Related information

	Described in:
Mathematical instructions and functions	RAPID Summary - *Mathematics*

S

T

U

V

W

Z